
Abstrakt

Série učebnı́ch textů z paradigmat programovánı́ seznamuje čtenáře s vybranými styly v programovánı́.
Tento text je zaměřen na problematiku funkcionálnı́ho programovánı́. Text je koncipován jako cvičebnice,
teoretické partie jsou prezentovány v redukované podobě a důraz je kladen na prezentaci přı́kladů a pro-
tipřı́kladů, na kterých jsou dané problémy demonstrovány. V textu se nacházejı́ klasické partie z funkcio-
nálnı́ho programovánı́ (napřı́klad procedury vyššı́ch řádů, rekurze, indukce, práce s hierarchickými daty),
ale i partie, které jsou v soudobé literatuře zastoupeny minimálně nebo vůbec (napřı́klad detailnı́ popis
vyhodnocovacı́ho procesu a konstrukce interpretu funkcionálnı́ho jazyka). Text je rozdělen do dvanácti na
sebe navazujı́cı́ch lekcı́. Pro správné pochopenı́ a procvičenı́ problematiky je vhodné čı́st lekce postupně bez
přeskakovánı́ a snažit se řešit všechny úkoly. Text u čtenářů nepředpokládá žádné znalosti programovánı́.
Pro pochopenı́ většiny přı́kladů stačı́ mı́t znalost středoškolské matematiky. Přı́klady vyžadujı́cı́ znalosti
(úvodnı́ch kurzů) vysokoškolské matematiky jsou doplněny o odkazy na vhodnou literaturu.

5.1 Definice seznamu a přı́klady . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 114
5.2 Program jako data . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 117
5.3 Procedury pro manipulaci se seznamy . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 120

5.6 Správa paměti během činnosti interpretu . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 130

8.6 Reprezentace polynomů . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 228

11.2 Zjednodušovánı́ aritmetických výrazů . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 272
11.3 Symbolická derivace . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 277

Lekce 7: Akumulace

Obsah lekce: V této lekci se budeme zabývat akumulacı́, což je speciálnı́ postupná aplikace procedur.
Pomocı́ akumulace ukážeme efektivnějšı́ řešenı́ vybraných úkolů, které jsme řešili v předchozı́ch lekcı́ch,
napřı́klad mapovánı́, filtrace a nahrazovánı́ prvků seznamu. Dále ukážeme, jak lze pomocı́ akumulace
rozšı́řit některé procedury dvou argumentů tak, aby pracovaly s libovolným počtem argumentů.

Klı́čová slova: akumulace, filtrace, procedura foldl, procedura foldr.

7.1 Procedura FOLDR

V této sekci se budeme zabývat akumulacı́ prvků seznamu. Pod pojmem akumulace máme obvykle na mysli
vytvořenı́ jedné hodnoty pomocı́ vı́ce hodnot obsažených v seznamu (sečtenı́ čı́sel v seznamu, nalezenı́
maxima, vytvořenı́ seznamu pouze z některých hodnot a podobně). Akumulace má blı́zko k explicitnı́ apli-
kaci prováděné pomocı́ primitivnı́ procedury apply, o které jsme se bavili v předchozı́ lekci. Použitı́ apply
při akumulaci má nevýhodu v tom, že při neznámé délce seznamu můžeme aplikovat pouze procedury
s libovolnými argumenty. Jedině tak je totiž při použitı́ apply zaručeno, že nedojde k chybě vlivem předánı́
špatného počtu argumentů. Nynı́ budeme k problému akumulace přistupovat jinak. Seznam libovolné
délky budeme akumulovat pomocı́ procedur dvou argumentů, které budou aplikovány postupně pro jednotlivé
prvky seznamu a to bud’ zleva nebo zprava.

Nejprve si problém demonstrujeme na přı́kladu. Uvažujme seznam čı́sel

(1 2 3 4).

Z předchozı́ch lekcı́ vı́me, že tento seznam lze zkonstruovat pomocı́ konstruktoru párů cons a pomocı́
prázdného seznamu vyhodnocenı́m následujı́cı́ho výrazu:

(cons 1 (cons 2 (cons 3 (cons 4 '())))) Z=⇒ (1 2 3 4)

Ve výrazu si můžeme všimnout toho, že se na sebe postupně „nabalujı́“ výsledky aplikace procedury
cons. Hodnota vzniklá vyhodnocenı́m (cons 4 '()), což je jednoprvkový seznam obsahujı́cı́ čtyřku je
dále použita jako druhý argument při dalšı́ aplikaci cons spolu s argumentem 3, výsledek této aplikace
je opět použit při dalšı́ aplikci cons jako druhý argument, a tak dále. Analogické postupné „nabalovánı́“
výsledků aplikacı́ bychom použili, kdybychom chtěli sečı́st prvky seznamu za předpokladu, že procedura
+ by akceptovala pouze dva argumenty:

(+ 1 (+ 2 (+ 3 4)))

Aby podobnost obou výrazů vı́ce vynikla, přepı́šeme předchozı́ s využitı́m součtu s nulou:

(+ 1 (+ 2 (+ 3 (+ 4 0))))

V podobném stylu bychom mohli vyjádřit součin prvků seznamu:

(* 1 (* 2 (* 3 (* 4 1))))

Nynı́ uvedeme ještě jeden přı́klad, ve kterém pro změnu nebudeme použı́vat při postupném „nabalovánı́“
výsledných hodnot primitivnı́ procedury jako dosud (procedury cons, + a *), ale následujı́cı́ uživatelsky
definovanou proceduru:

(define y+1

(lambda (x y)

(+ y 1)))

Všimněte si, že předchozı́ procedura zcela ignoruje svůj prvnı́ argument a vracı́ hodnotu druhého ar-
gumentu zvětšenou o jedna (druhý argument tedy musı́ být čı́slo). V následujı́cı́ ukázce jsou zachyceny
výsledky aplikacı́ této procedury:

(y+1 'a 0) Z=⇒ 1

(y+1 'a (y+1 'b 0)) Z=⇒ 2

(y+1 'a (y+1 'b (y+1 'c 0))) Z=⇒ 3

(y+1 'a (y+1 'b (y+1 'c (y+1 'd 0)))) Z=⇒ 4

Jak je asi z ukázky a z definice procedury y+1 patrné, proceduru můžeme tı́mto způsobem použı́t k počı́tánı́
počtu prvků seznamu. Pro náš výchozı́ seznam bychom tedy měli:

(y+1 1 (y+1 2 (y+1 3 (y+1 4 0)))) Z=⇒ 4

Výraz je opět ve tvaru postupného „nabalovánı́“ aplikacı́. V čem se lišily všechny předchozı́ ukázky
postupně vnořených aplikacı́, které jsme uvedli? Pro zopakovánı́, jednalo se o tyto výrazy:

(cons 1 (cons 2 (cons 3 (cons 4 '()))))

(+ 1 (+ 2 (+ 3 (+ 4 0))))

(* 1 (* 2 (* 3 (* 4 1))))

(y+1 1 (y+1 2 (y+1 3 (y+1 4 0))))

Z hlediska jejich tvaru, ze lišily jen „nepatrně“, protože je lze všechny chápat jako výrazy tvaru:

(〈procedura〉 1 (〈procedura〉 2 (〈procedura〉 3 (〈procedura〉 4 〈terminátor〉)))),

kde 〈procedura〉 je procedura dvou argumentů a 〈terminátor〉 je element. Vskutku, v prvnı́m přı́padě
byla 〈procedura〉 zastoupena cons a terminátor byl (). V druhém přı́padě byla 〈procedura〉 zastoupena
+ a 〈terminátor〉 byl 0. V poslednı́m přı́padě byla 〈procedura〉 zastoupena uživatelsky definovanou procedu-
rou y+1 a 〈terminátor〉 byl 0. Z pohledu tvaru se tedy výrazy přı́liš nelišı́. Lišı́ se ale výrazně z pohledu svého
významu (výsledných hodnot): vytvořenı́ seznamu, součet hodnot, součin hodnot, výpočet délky (což je
v podstatě „počet zanořenı́ “ v daném výrazu).

Doposud jsme všechny úvahy prováděli nad seznamem pevné délky. Úvahy bychom ale mohli rozšı́řit na
seznamy libovolných délek. Nabı́zı́ se mı́t k dispozici obecnou proceduru, která pro danou proceduru dvou
argumentů, terminátor a seznam provede postupnou aplikaci dané procedury dvou argumentů přes všechny
prvky seznamu tak, jak jsme nynı́ v několika přı́padech ukázali. V jazyku Scheme budeme uvažovat
proceduru foldr (z anglického fold right, neboli „zabal směrem doprava“), která je ve své základnı́ podobě
aplikována ve tvaru:

(foldr 〈procedura〉 〈terminátor〉 〈seznam〉).

Při své aplikaci procedura foldr provede postupnou aplikaci

(〈procedura〉 〈prvek1〉 (〈procedura〉 〈prvek2〉 (〈procedura〉 · · · (〈procedura〉 〈prvekn〉 〈terminátor〉) · · ·))),

pro 〈seznam〉 ve tvaru (〈prvek1〉 〈prvek2〉 · · · 〈prvekn〉). To jest, použitı́m foldr na prázdný seznam je vrácen
element označený jako 〈terminátor〉. Použitı́m foldr na jednoprvkový, dvouprvkový, třı́prvkový, čtyřprv-
kový (a tak dále) seznam je vrácena hodnota aplikace:

(〈procedura〉 〈prvek1〉 〈terminátor〉)
(〈procedura〉 〈prvek1〉 (〈procedura〉 〈prvek2〉 〈terminátor〉))
(〈procedura〉 〈prvek1〉 (〈procedura〉 〈prvek2〉 (〈procedura〉 〈prvek3〉 〈terminátor〉)))
(〈procedura〉 〈prvek1〉 (〈procedura〉 〈prvek2〉 (〈procedura〉 〈prvek3〉 (〈procedura〉 〈prvek4〉 〈terminátor〉))))

...

Předchozı́ přı́klady bychom tedy pomocı́ foldr mohli vyřešit takto:

(define s '(1 2 3 4))

(foldr cons '() s) Z=⇒ (1 2 3 4)

(foldr + 0 s) Z=⇒ 10

(foldr * 1 s) Z=⇒ 24

(foldr y+1 0 s) Z=⇒ 4

Prvnı́ aplikacı́ foldr jsme vytvořili duplikát výchozı́ho seznamu, následuje součet a součin prvků seznamu
a poslednı́ přı́klad bychom mohli v souladu s našimi předchozı́mi pozorovánı́mi charakterizovat jako
vypočtenı́ délky seznamu. Použitı́ foldr má zjevnou výhodu v tom, že funguje pro libovolně dlouhý
seznam o jehož transformaci na sérii postupných aplikacı́ se jako programátoři nemusı́me starat. O tom se
ostatně můžete přesvědčit sami, zkuste několikrát změnit seznam nabázaný na s a proved’te výše uvedené
aplikace foldr.

Poznámka 7.1. Všimněte si, jakou roli majı́ argumenty procedury, kterou při aplikaci předáváme proceduře
foldr. Prvnı́ argument této procedury postupně nabývá hodnot vyskytujı́cı́ch se v seznamu. Na druhou stranu
druhý argument zastupuje element, který vznikl zabalenı́m hodnot v seznamu vyskytujı́cı́ch se za průběžným
prvkem. Napřı́klad při následujı́cı́ aplikaci foldr:

(foldr (lambda (x y)

(list #f x y))

'base

'(a b c d)) Z=⇒ (#f a (#f b (#f c (#f d base))))

bude procedura vzniklá vyhodnocenı́m λ-výrazu (lambda (x y) (list #f x y)) nejprve aplikována
s hodnotami d (poslednı́ prvek seznamu) a base (terminátor). Výsledkem tedy bude seznam (#f d base).
V dalšı́m kroku bude procedura aplikována s prvkem c (předposlednı́ prvek seznamu) a druhým argu-
mentem bude výsledek zabalenı́ vytvořený v předchozı́m kroku, tedy seznam (#f d base). Výsledkem
aplikace procedury proto bude seznam (#f c (#f d base)). V dalšı́m kroku bude procedura aplikována
s prvkem b a seznamem (#f c (#f d base)). Jako výsledek vznikne (#f b (#f c (#f d base))). Ko-
nečně v poslednı́m kroku bude procedura aplikována s a (prvnı́ prvek seznamu) a s poslednı́m uvedeným
seznamem. Výsledek této poslednı́ aplikace bude vrácen jako výsledek aplikace foldr.

Následujı́cı́ přı́klady ukazujı́ roli terminátoru.

(foldr cons 'base s) Z=⇒ (1 2 3 4 . base)

(foldr cons '() s) Z=⇒ (1 2 3 4)

(foldr list 'base s) Z=⇒ (1 (2 (3 (4 base))))

(foldr list '() s) Z=⇒ (1 (2 (3 (4 ()))))

(foldr list '() '()) Z=⇒ ()

(foldr list 'base '()) Z=⇒ base

(foldr list 666 '()) Z=⇒ 666

Napřı́klad na prvnı́ch dvou použitı́ch foldr je vidět, že v prvnı́m přı́padě je aplikována procedura cons

s argumenty 4 (poslednı́ prvek seznamu) a base, což vede k vytvořenı́ tečkového páru (4.base), který
se ve výsledku vyskytuje na konci vytvořené hierarchické struktury. V druhém přı́padě je aplikace cons

provedena s hodnotami 4 a (), což vede na vytvořenı́ jednoprvkového seznamu (4) – výsledná struktura
vytvořená aplikacı́ foldr je v tomto přı́padě seznam. V poslednı́ch třech přı́kladech vidı́me meznı́ přı́pad:
při pokusu o zabalenı́ prázdného seznamu je vrácen terminátor.

Pomocı́ procedury foldr lze efektivně implementovat řadu operacı́ nad seznamy. Klı́čem ke správnému
použitı́ foldr je pochopit roli procedury dvou argumentů, která je při aplikaci předávána jako prvnı́
argument, a pomocı́ niž je provedeno samotné „zabalenı́ hodnot“. Musı́me si uvědomit, že tato pro-
cedura je postupně aplikována tak, že jejı́ prvnı́ argument je průběžný prvek seznamu a druhý argument
je výsledek zabalenı́ prvků nacházejı́cı́ch se v seznamu za průběžným prvkem.

Proceduru foldr lze použı́t v obecnějšı́m tvaru. Podobně jako tomu bylo u procedury map, procedura
foldr připouštı́ při aplikaci i vı́ce seznamů než jen jeden. Proceduru foldr lze tedy aplikovat ve tvaru

(foldr 〈procedura〉 〈terminátor〉 〈seznam1〉 〈seznam2〉 · · · 〈seznamn〉),

kde 〈seznam1〉, . . . , 〈seznamn〉 jsou seznamy (n ≥ 1), a 〈procedura〉 je procedura n+1 argumentů. Při aplikaci
foldr je provedeno analogické zabalenı́ jako ve verzi s jednı́m seznamem, rozdı́l je pouze v tom, že
〈procedura〉 je aplikována s n+1 argumenty, jimiž je n průběžných prvků z předaných seznamů a poslednı́m
argumentem je výsledek zabalenı́ prvků následujı́cı́ch za průběžnými prvky.

Nejlépe činnost foldr pro vı́c argumentů vysvětlı́ následujı́cı́ přı́klady:

(foldr list 'base

'(1 2 3) '(a b c) '(i j k)) Z=⇒ (1 a i (2 b j (3 c k base)))

(foldr (lambda (x y z result)

(cons (list x y z) result))

'()

'(1 2 3) '(a b c) '(i j k)) Z=⇒ ((1 a i) (2 b j) (3 c k))

V následujı́cı́ sekci uvedeme praktické přı́klady použitı́ foldr.

7.2 Použitı́ FOLDR pro definici efektivnı́ch procedur

Nynı́ ukážeme implementaci několika procedur pro práci se seznamy, které už jsme představili v předcho-
zı́ch lekcı́ch. Uvidı́me, že vytvořenı́ těchto procedur pomocı́ foldr bude nejen kratšı́ z hlediska jejich zápisu,
ale procedury budou mnohem efektivnějšı́. I když nenı́ výpočtová efektivita hlavnı́m předmětem, na který
se v tomto kurzu soustředı́me, u každé vytvářené procedury je vždy vhodné zamýšlet se nad jejı́ efektivi-
tou. Pro programovánı́ ve funkcionálnı́ch jazycı́ch je typické vyvı́jet program v několika etapách. V prvnı́
fázi vývoj jde vlastně o obohacenı́ jazyka o nové procedury, pomocı́ nichž budeme schopni vyřešit daný
problém. Po vyřešenı́ problému se můžeme opět vrátit k naprogramovaným procedurám a pokoušet se
zvýšit jejich efektivitu tı́m, že je implementujeme znovu, samozřejmě při zachovánı́ dosavadnı́ funkčnosti.

Tento pohled uplatnı́me v malém měřı́tku i nynı́. Ukážeme si, jak lze efektivně vytvořit procedury, které
jsme již měli (méně efektivně) vytvořené. Prvnı́ z nich bude procedura length vracejı́cı́ délku seznamu.
Původnı́ kód, který jsme vytvořili pomocı́ map a apply je uveden v programu 6.1 na straně 144. Před uve-
denı́m nové verze length se nejdřı́v zamysleme nad efektivitou původnı́ verze z programu 6.1. Předně,
jak můžeme snadno kvantitativně vyjádřit efektivitu procedury pracujı́cı́ se seznamy? Seznamy jsou sek-
venčnı́ lineárnı́ dynamické datové struktury, k jejichž (dalšı́m) prvkům přistupujeme pomocı́ cdr. Tato
operace má vzhledem k dalšı́m použı́vaným operacı́m největšı́ vliv na rychlost zpracovánı́ seznamu. Proto
budeme vyjadřovat efektivitu amortizovaně vzhledem k počtu operacı́ cdr provedených nad vstupnı́mi daty.
Budeme přitom provádět běžná zjednodušenı́, která jsou studentům známá z kurzu algoritmické matematiky,
nebudeme se zabývat samotnou strukturou seznamů, ale efektivitu (časovou složitost procedur) budeme
vyjadřovat vzhledem k délce vstupnı́ch seznamů.

Procedura length v programu 6.1 počı́tá výslednou hodnotu tak, že nejprve provede mapovánı́ přes celý
vstupnı́ seznam. Pokud délku vstupnı́ho seznamu označı́me n, pak tato fáze zabere n kroků (je potřeba
n aplikacı́ cdr na průchod seznamem10). V dalšı́ fázı́ je na vzniklý seznam aplikována operace sčı́tánı́
ta potřebuje opět n kroků k tomu, aby prošla prvky seznamu (jedničky) a sečetla je. Dohromady tedy
procedura pro seznam délky n provede 2n kroků. Časovou složitost budeme dále zapisovat v běžné O-
-notaci, v přı́padě našı́ procedury tedy O(2n). Intuitivně bychom očekávali, že délku n prvkového seznamu
bychom měli stanovit právě v n krocı́ch, procedura length z programu 6.1 tedy nenı́ přı́liš efektivnı́. Na
druhou strany nenı́ těžké nahlédnout, že délku seznamu se nám nepodařı́ stanovit v „sublineárnı́m čase“,
protože pro výpočet délky seznamu musı́me skutečně každý prvek seznamu navštı́vit aspoň jednou.

Podı́vejme se nynı́ na proceduru length z programu 7.1. Jde v podstatě jen o formalizaci myšlenky z před-

Program 7.1. Výpočet délky seznamu pomocı́ foldr.

(define length

(lambda (l)

(foldr (lambda (x y)

(+ 1 y))

0 l)))

10Někdo by v tuto chvı́li mohl namı́tnout, že kroků je potřeba pouze n− 1, protože u jednoprvkového seznamu už žádný dalšı́
prvek nehledáme. Tato úvaha ale nenı́ správná, abychom zjistili, že se skutečně jedná o jednoprvkový seznam, musı́me otestovat,
jaký element se nacházı́ na druhé pozici páru, tı́m pádem cdr musı́me skutečně aplikovat n-krát. Uvědomte si, že interpret nevidı́
seznam „z vnějšku“ tak, jako jej vidı́me my.

chozı́ sekce. Jelikož je v těle procedury aplikována procedura foldr se vstupnı́m seznamem, jejı́ činnost
zabere právě n kroků, během kterých jsou navštı́veny všechny prvky seznamu. Hodnota terminátoru 0

znamená, že prázdný seznam má délku nula. Během akumulace je sečteno právě tolik jedniček, kolik je
navštı́veno prvků, výsledná hodnota je tedy čı́slo – délka seznamu. Celková časová složitost našeho řešenı́
je tedy O(n).

Z kurzu algoritmické matematiky možná vı́te, že složitost se stanovuje řádově. Z tohoto pohledu jsou
složitosti O(2n) a O(n) obě stejného řádu (lineárnı́ složitost), protože multiplikativnı́ konstanta je z hle-
diska řádové složitosti zanedbatelná11. Zde upozorněme na to, že orientovat se podle řádové složitosti,
použı́vané třeba ke klasifikaci řešitelných problémů podle jejich časové nebo prostorové složitosti, by bylo
z našeho pohledu dost ošidné. Multiplikativnı́ konstanta 2 je z pohledu procedury jako je length (která
bude v programu zřejmě často použı́vána), dost kritická a naše nová implementace majı́cı́ složitost O(n)
je výrazně lepšı́ než původnı́ se složitostı́ O(2n)12. Dalšı́ nově naprogramovanou procedurou bude spojenı́
dvou seznamů. Původnı́ kód procedury append2 je k dispozici v programu 5.2 na straně 123. Tato imple-
mentace využı́vajı́cı́ build-list je extrémně neefektivnı́. Označı́me-li délku prvnı́ho seznamu n a druhého
m, pak je nejprve spotřebováno n+m kroků na stanovenı́ délek obou seznamů. Potom je konstruován nový
seznam délky n +m. Proto, abychom zjistili složitost konstrukce, musı́me rozebrat tělo procedury volané
procedurou build-list. Při pohledu na tělo je jasné, že jsou postupně vraceny prvky z obou seznamů
pomocı́ list-ref. Samotná procedura list-ref vrátı́ prvek na k-té pozici (nejdřı́v) během k kroků. Pro
vrácenı́ všech prvků z prvnı́ho seznamu tedy potřebujeme 1 + 2 + · · · + n kroků, což je (sečtenı́m prvků
aritmetické posloupnosti) dohromady n(1+n)

2 kroků. Pro druhý seznam potřebujeme analogicky m(1+m)
2

kroků. Celkovou složitost stanovı́me součtem všech třı́ částı́ (výpočet délek a sekvenčnı́ přı́stup ke všem
prvkům obou seznamů), to jest

O
(
n+m+ n(1+n)

2 + m(1+m)
2

)
,

což je ekvivalentnı́

O
(n(n+3)+m(m+3)

2

)
.

Řádově je tedy časová složitost procedury z programu 5.2 dokonce kvadratická. To je přı́mo tristnı́, pro-
tože pro spojenı́ seznamu délky m a seznamu délky n bychom intuitivně očekávali složitost O(m + n).
Program 7.2 dokonce ukazuje, že na tom můžeme být ještě o něco lépe. Nejdřı́ve objasněme tělo nové

Program 7.2. Spojenı́ dvou seznamů pomocı́ foldr.

(define append2

(lambda (l1 l2)

(foldr cons l2 l1)))

implementace procedury append2. Jedná se o akumulaci prvků prvnı́ho seznamu pomocı́ cons, která je
terminována druhým předaným seznamem. Všechny prvky prvnı́ho seznamu jsou při této akumulaci po-
stupně navštı́veny (jeden po druhém), což zabere n kroků. Výsledná složitost je tedy O(n). Pro někoho
poněkud překvapivě, protože se do složitosti vůbec nepromı́tla délka druhého seznamu. Vskutku, druhý
seznam je použit jako terminujı́cı́ element a nenı́ tedy vůbec procházen. Činnost append2 napsané pomocı́
foldr si možná lépe uvědomı́me, když si představı́me, že foldr provede sérii aplikacı́ typu:

(cons 〈prvek1〉 (cons 〈prvek2〉 (cons · · · (cons 〈prvekn〉 〈seznam〉) · · ·))),

která skutečně vede na spojenı́ dvou seznamů: (〈prvek1〉 · · · 〈prvekn〉) a 〈seznam〉.

Pomocı́ foldr a append2 můžeme efektivně naprogramovat spojenı́ libovolného množstvı́ seznamů tak,
jak to ukazuje procedura append v programu 7.3. Proceduru jsme pochopitelně museli definovat jako

11Z prektického pohledu bychom se měli zajı́mat i o multiplikativnı́ konstanty zvlášt’v přı́padě, pokud jsou velké.
12Zde opět připomeňme, že řádově jsou obě složitostnı́ třı́dy stejné, to jest O(n) = O(2n). Z praktického hlediska je však v tomto

přı́padě konstanta 2 výraznou přı́těžı́. Pokud budeme chtı́t multiplikativnı́ konstanty zdůrazňovat, budeme je v O-notacı́ch uvádět,
i když to nenı́ běžné.

Program 7.3. Spojenı́ libovolného počtu seznamů pomocı́ foldr.

(define append

(lambda lists

(foldr append2 '() lists)))

proceduru s libovolnými argumenty, ty budou při jejı́ aplikaci navázané na symbol lists. V těle procedury
je použito jedno volánı́ foldr pomocı́ nějž provádı́me akumulaci procedury append2, kterou jsme vytvořili
v předchozı́m kroku. Tato procedura bude akumulovat prvky ze seznamu lists, což jsou seznamy předané
proceduře append při jejı́ aplikaci. Terminátorem je prázdný seznam, protože spojenı́m „žádných seznamů“
vzniká prázdný seznam. Složitost této procedury je rovna O(n), kde n je součet délek všech vstupnı́ch
seznamů. Pokud použijeme append na spojenı́ pouze dvou seznamů, bude mı́t složitost O(n +m), což je
zhoršenı́ oproti append2. Důvodem je fakt, že append terminuje spojenı́ prázdným seznamem a procházı́
prvky všech předaných seznamů (tedy i toho poslednı́ho, v našem přı́padě druhého). To je jakási daň, kterou
jsme zaplatili za obecnost řešenı́.

V programu 5.3 na straně 125 jsme ukázali implementaci procedury map1, což je varianta mappracujı́cı́ pouze
s jednı́m seznamem. Tato implementace byla opět velmi neefektivnı́ a použı́vala build-list a sekvenčnı́
vyhledávánı́ prvků pomocı́ opakované aplikaci list-ref. Časová složitost této procedury byla O

(n(n+3)
2

)
,

protože n kroků bylo spotřebováno vypočtenı́m délky seznamu a 1 + 2 + · · · + n kroků bylo potřeba na
procházenı́ jeho prvků. Složitost takto napsané map1 byla opět kvadratická. Proceduru lze ale vytvořit se
složitostı́ O(n) tak, jak je to ukázáno v programu 7.4. V nové implementaci map1 jsme prováděli akumulaci

Program 7.4. Mapovacı́ procedura pracujı́cı́ s jednı́m seznamem pomocı́ foldr.

(define map1

(lambda (f l)

(foldr (lambda (x y)

(cons (f x) y))

'()

l)))

hodnot pomocı́ uživatelsky definované procedury, která mı́sto prostého použitı́ cons tak, jak jsme jej
použili v přı́padě append2, provede napojenı́ modifikace prvnı́ho prvku s již zpracovanou částı́. Akumulace
je terminována prázdným seznamem. Pomocı́ foldr můžeme nynı́ naprogramovat i obecný map pracujı́cı́
s libovolným (ale nenulovým) počtem seznamů. Obecná verze map se nacházı́ v programu 7.5. Program 7.5
obsahuje pomocnou proceduru separate-last-argument, která má za účel pro daný neprázdný seznam
vrátit tečkový pár, jehož prvnı́m prvkem bude seznam prvků z původnı́ho seznamu kromě poslednı́ho
a druhým prvkem tečkového páru bude poslednı́ prvek seznamu. Viz následujı́cı́ přı́klady použitı́:

(separate-last-argument '()) Z=⇒ #f

(separate-last-argument '(a)) Z=⇒ (() . a)

(separate-last-argument '(a b)) Z=⇒ ((a) . b)

(separate-last-argument '(a b c)) Z=⇒ ((a b) . c)

(separate-last-argument '(a b c d)) Z=⇒ ((a b c) . d)

Procedura separate-last-argument nám tedy umožňuje přistoupit k prvkům seznamu (vyjma posled-
nı́ho) a k poslednı́mu prvku. Jedná se tedy o jakési „car a cdr naruby“. Tuto pomocnou proceduru jsme
v programu 7.5 dále použili na implementaci map. Samotný map jsme realizovali aplikacı́ foldr. Jelikož
však dopředu nevı́me, kolik seznamů bude proceduře map předáno, museli jsme proceduru předanou
foldr vytvořit jako proceduru s libovolným počtem argumentů. Pro n seznamů bude argumentů n + 1,
prvnı́ch n argumentů bude reprezentovat průběžné prvky seznamů a poslednı́ argument bude zastupovat

Program 7.5. Obecná mapovacı́ procedura pomocı́ foldr.

(define separate-last-argument

(lambda (l)

(foldr (lambda (x y)

(if (not y)

(cons '() x)

(cons (cons x (car y)) (cdr y))))

#f

l)))

(define map

(lambda (f . lists)

(apply foldr

(lambda args

(let ((separation (separate-last-argument args)))

(cons (apply f (car separation))

(cdr separation))))

'()

lists)))

akumulovanou hodnotu. Jelikož chceme k akumulované hodnotě přidat hodnotu vzniklou aplikacı́ prv-
nı́ch n argumentů, potřebujeme od sebe nutně oddělit prvnı́ch n argumentů a poslednı́ argument. K tomu
jsme použili právě pomocnou proceduru separate-last-argument, která byla rovněž vytvořená pomocı́
foldr.

Všimněte si, že v proceduře separate-last-argument je foldr terminován elementem #f. V těle pro-
cedury předané foldr je vidět, že hned při prvnı́ aplikaci, kdy je na x navázaný poslednı́ prvek seznamu,
je vytvořen pár ve tvaru ((). 〈poslednı́〉). V každém dalšı́m kroku již se druhý prvek tohoto páru neměnı́,
a do prvnı́ho prvku se přidávajı́ postupně procházené prvky. Tı́m vytvořı́me požadovaný výstup, viz výše
uvedené přı́klady.

Složitost obecného map můžeme stanovit zhruba takto. Procházı́me m seznamů délky n postupně prvek po
prvku, to zabere celkem mn kroků. K tomu musı́me připočı́st režii spojenou s násobnou aplikacı́ pomocné
procedury separate-last-argument. Tato procedura je aplikována právě tolikrát, jaká je délka seznamů,
tedy n-krát. Při každé aplikaci potřebuje m+1 kroků na separaci poslednı́ho argumentu. Celková složitost
je tedy O(mn+ (m+ 1) · n), což je ekvivalentnı́ O((2m+ 1) · n).

Program 7.6 obsahuje efektivnı́ implementaci filtračnı́ procedury filter, kterou jsme představili v pro-
gramu 6.2 na straně 146. Původnı́ filtračnı́ procedura měla časovou složitost O(2n), zdůvodněnı́ je analogické
tomu, jaké jsme provedli u původnı́ procedury length. Efektivnı́ implementace z programu 7.6 ukazuje
dalšı́ použitı́ foldr. V tomto přı́padě je terminátorem opět prázdný seznam a uživatelsky definovaná
procedura předaná foldr nejprve otestuje, zda-li průběžný prvek splňuje vlastnost danou procedurou
navázanou na symbol f. Pokud ano, je prvek přidán k seznamu v němž se akumulujı́ prvky splňujı́cı́
tyto vlastnost. V opačném přı́padě nenı́ seznam akumulovaných prvků změněn. Časová složitost nového
provedenı́ filter je O(n).

Dalšı́ ukázkou je efektivnı́ implementace predikátumember?. Tento predikát jsme představili v programu 6.3
na straně 146. Složitost původnı́ implementace byla O(2n), protože byla založena na původnı́ implementaci
filter. Kdybychom nynı́ uvažovali, že ponecháme původnı́ kód member?, ale budeme v něm použı́vat
novou implementaci filter z programu 7.6, pak bude mı́t member? časovou složitost O(n). Můžeme ale
provést úplně novou implementaci member? přı́mo použitı́m foldr bez vazby na filter. Viz program 7.7.

Poslednı́ procedurou, kterou v této sekci ukážeme je replace, která při své aplikaci vyžaduje tři argu-

Program 7.6. Filtrace prvků seznamu splňujı́cı́ch danou vlastnost pomocı́ foldr.

(define filter

(lambda (f l)

(foldr (lambda (x y)

(if (f x)

(cons x y)

y))

'()

l)))

Program 7.7. Test přı́tomnosti prvku v seznamu pomocı́ foldr.

(define member?

(lambda (elem l)

(foldr (lambda (x y)

(if (equal? x elem) #t y))

#f

l)))

menty: prvnı́m je predikát jednoho argumentu reprezentujı́cı́ vlastnost prvku seznamu (analogická role
jako u filter), druhým je procedura jednoho argumentu sloužı́cı́ k modifikaci prvků seznamu (analo-
gická role jako u map) a třetı́m argumentem je seznam. Výsledkem aplikace procedury replace je seznam
elementů vzniklý ze vstupnı́ho seznamu tak, že každý prvek seznamu splňujı́cı́ vlastnost danou prvnı́m
argumentem je modifikován pomocı́ procedury dané druhým argumentem. Viz program 7.8.

Program 7.8. Nahrazenı́ prvku dané vlastnosti modifikacı́ prvku pomocı́ foldr.

(define replace

(lambda (prop? modifier l)

(foldr (lambda (x y)

(if (prop? x)

(cons (modifier x) y)

(cons x y)))

'()

l)))

Následujı́cı́ přı́klady ukazujı́ použitı́ replace:

(define s '(1 2 3 4 5))

(replace even? (lambda (x) (- x)) s) Z=⇒ (1 -2 3 -4 5)

(replace (lambda (x) #t) (lambda (x) 1) s) Z=⇒ (1 1 1 1 1)

(replace (lambda (x) (<= x 3)) list s) Z=⇒ ((1) (2) (3) 4 5)

(replace (lambda (x) (= 1 (modulo x 3)))

(lambda (x) (+ x 10))

s) Z=⇒ (11 2 3 14 5)

7.3 Dalšı́ přı́klady akumulace

V této sekci si ukážeme dalšı́ přı́klady akumulace pomocı́ foldr. Nejprve se budeme zabývat problema-
tikou rozšı́řenı́ operace (procedury) dvou argumentů na proceduru libovolných argumentů. Konkrétně se
budeme zabývat touto problematikou u monoidálnı́ch operacı́, viz sekci 2.5. Pokud je totiž operace } na
dané množině asociativnı́ a má neutrálnı́ prvek, pak můžeme bez újmy psát

a1 } a2 } · · ·} an,

protože dı́ky asociativitě nezáležı́ na uzávorkovánı́ předchozı́ho výrazu. Dı́ky neutralitě navı́c platı́, že

a1 } a2 } · · ·} an = a1 } a2 } · · ·} an } e,

kde e je neutrálnı́ prvek vzhledem k operaci }. Z hlediska akumulace pomocı́ foldr je pro nás zajı́mavé

a1 } a2 } · · ·} an = (a1 } (a2 } · · · (an } e) · · · )).
Pravá strana předchozı́ rovnosti je ve tvaru vhodném pro akumulaci pomocı́ foldr, protože monoidálnı́
operace } zde hraje analogickou roli jako procedura předávaná foldr a terminátor je neutrálnı́ prvek e.
Kdybychom tedy ve Scheme neměli k dispozici +, * a podobné operace jako procedury libovolných ar-
gumentů, ale pouze dvou, pak bychom je pomocı́ foldr mohli snadno rozšı́řit na procedury libovolných
argumentů.

V následujı́cı́m přı́kladu máme definovány procedury, které provádějı́ součet a součin dvou prvků:

(define add2 (lambda (x y) (+ x y)))

(define mul2 (lambda (x y) (* x y)))

Pomocı́ foldr je můžeme zobecnit na operace pro libovolný počet argumentů:

(define ++

(lambda args

(foldr add2 0 args)))

(define **

(lambda args

(foldr mul2 1 args)))

Samozřejmě, že předchozı́ přı́klad byl pouze „školský“, protože v interpretu máme k dispozici + a *pracujı́cı́
s libovolnými argumenty, takže nenı́ potřeba je „redukovat na dva argumenty“ pak „opět vyrábět“. Přı́klad
měl sloužit pro demonstraci této obecné techniky. Analogicky jako v předchozı́m přı́padě bychom mohli
na libovolný počet argumentů zobecnit proceduru pro sčı́tánı́ vektorů pevné délky:

(define vec+ (lambda (v1 v2) (map + v1 v2)))

Zde je ale malý problém s neutrálnı́m prvkem. Neutrálnı́ prvek pro sčı́tánı́ vektorů je pochopitelně nulový
vektor. Pokud jsou vektory reprezentovány seznamem hodnot, pak by to měl být seznam skládajı́cı́ se
ze samých nul. Potı́ž je ale v tom, že předchozı́ procedura byla schopná sčı́tat vektory libovolné délky,
pro každou z délek máme jeden neutrálnı́ prvek. Situaci bychom mohli vyřešit tak, že bychom vytvořili
proceduru vyššı́ho řádu make-vec+, která by pro danou délku vrátila proceduru pro sčı́tánı́ libovolně
mnoha vektorů:

(define make-vec+

(lambda (n)

(let ((null-vector (build-list n (lambda (i) 0))))

(lambda vectors

(foldr vec+ null-vector vectors)))))

Použitı́ procedury by pak bylo následujı́cı́:

((make-vec+ 3)) Z=⇒ (0 0 0)

((make-vec+ 3) '(1 2 3)) Z=⇒ (1 2 3)

((make-vec+ 3) '(1 2 3) '(10 20 30)) Z=⇒ (11 22 33)

((make-vec+ 3) '(1 2 3) '(10 20 30) '(2 4 6)) Z=⇒ (13 26 39)

Proceduru make-vec+ bychom mı́sto foldr a vec+ mohli implementovat s použitı́m map pro libovolné
argumenty. Následujı́cı́ přı́klad rovněž ukazuje použitı́ apply s nepovinnými argumenty.

(define make-vec+

(lambda (n)

(let ((null-vector (build-list n (lambda (i) 0))))

(lambda vectors

(apply map + null-vector vectors)))))

Zamysleme se nynı́ nad možnostı́ rozšı́řit proceduru pro výpočet minima ze dvou prvků na proceduru
zpracovávajı́cı́ libovolné argumenty:

(define min2

(lambda (x y)

(if (<= x y) x y)))

Problémem je, že „operace minimum“ se sice chová asociativně, to jest platı́

min(a,min(b, c)) = min(min(a, b), c),

ale nemá neutrálnı́ prvek. Nynı́ máme několik možnostı́, jak postupovat. Jednou z možnostı́ je implementovat
procedury pro výpočet minima tak, jak je v současném standardu R5RS jazyka Scheme, viz [R5RS]. To jest
uvažujeme minimum z jednoho a vı́ce čı́sel:

(define min

(lambda numbers

(foldr min2 (car numbers) (cdr numbers))))

V předchozı́m kódu jsme jako terminátor zvolili prvnı́ prvek seznamu čı́sel, se kterými je procedura min

aplikována. Samotnou akumulaci pak provádı́me přes seznam předaných čı́sel bez prvnı́ho. Zde jsme
vlastně tiše využili i komutativitu sčı́tánı́ čı́sel, protože pro seznam obsahujı́cı́ hodnoty a1, . . . , an počı́táme
výsledek takto:

min(a2,min(a3, . . .min(an, a1) · · · )).
Z těla výše uvedené procedury je taky jasné, že aplikace min bez argumentu by skončila chybovým hlášenı́m
způsobeným použitı́m car a cdr na prázdný seznam.

Druhým způsobem řešenı́ problému je neutrálnı́ prvek nějak „dodat“. Napřı́klad bychom mohli uvažovat
symbol +infty, který by nám zastupoval „plus nekonečno“. Tedy jakési nestandardnı́ „největšı́ čı́slo“.
Tento krok by znamenal upravit predikát <= (a v důsledku i dalšı́ aritmetické procedury) tak, aby pracoval
i s touto novou hodnotu. Úprava by mohla vypadat takto, nejprve nadefinujeme +infty, který se bude
vyhodnocovat na sebe sama:

(define +infty '+infty)

Dále vytvořı́me novou verzi predikátu porovnávánı́ čı́sel:

(define <=

(let ((<= <=))

(lambda (x y)

(or (equal? y +infty)

(and (not (equal? x +infty))

(<= x y))))))

Nový <= se na čı́selných hodnotách chová stejně jako stará verze, pro +infty se chová tak, že +infty je
„většı́ než všechno ostatnı́“. Viz následujı́cı́ přı́klady použitı́.

(<= 2 3) Z=⇒ #t

(<= 3 2) Z=⇒ #f

(<= 2 +infty) Z=⇒ #t

(<= +infty 3) Z=⇒ #f

(<= +infty +infty) Z=⇒ #t

Nynı́ můžeme ponechat kód min2 tak, jak jej máme, a pouze definujeme novou obecnou verzi min:

(define min

(lambda numbers

(foldr min2 +infty numbers)))

Takto definovanou proceduru je možné použı́t běžným způsobem, nynı́ i bez argumentů:

(min) Z=⇒ +infty

(min 30) Z=⇒ 30

(min 30 10) Z=⇒ 10

(min 30 10 20) Z=⇒ 10

Nynı́ se vrat’me k reprezentaci množin uvedené v sekci 6.5, kde jsme implementovali konstruktor množiny
list->set. Ten ze seznamu vytvářel množinu tı́m, že podle tohoto seznamu odstranil duplicitnı́ výskyty
prvků. Pomocı́ procedury foldr, můžeme napsat elegantnějšı́ řešenı́:

(define list->set

(lambda (l)

(foldr (lambda (x y)

(if (member? x y)

y

(cons x y)))

'()

l)))

Takto nadefinovaná procedura list->set provádı́ akumulaci pomocı́ foldr přes zadaný seznam. Jako
terminátor je zvolen prázdný seznam. Vstupnı́ procedura procedury foldr pak testuje přı́tomnost prů-
běžného prvku v seznamu, který vznikl v předchozı́m kroku (použı́vá se zde predikátu member?, který
jsme napsali v programu 7.7). Pokud zjistı́, že průběžný prvek v seznamu ještě nenı́, přidá tento prvek
do seznamu. V opačném přı́padě vracı́ nezměněný seznam. Tak jsou odstraněny duplicitnı́ výskyty, viz
přı́klady použitı́.

(list->set '()) Z=⇒ ()

(list->set '(1 2 3)) Z=⇒ (1 2 3)

(list->set '(1 2 2 1 2 3 1)) Z=⇒ (2 3 1)

Ted’ se budeme zabývat možným zobecněnı́m procedury foldr. Jednı́m z argumentů procedury foldr je
〈procedura〉. Tato procedura 〈procedura〉 nese vlastně dvě informace. Řı́ká, jakým způsobem se modifikuje
průběžný prvek a jakým způsobem se tato modifikace „nabalı́“ na zabalenı́ modifikovaných hodnot za průběžným
prvkem. Vzhledem k tomu můžeme tuto proceduru rozdělit na dvě, tak aby každá z nich obsahovala jen
jednu z těchto informacı́. Napřı́klad:

• V programu 7.4 jsme definovali proceduru mapovánı́ přes jeden seznam map1. Proceduru foldr jsme
aplikovali na proceduru, která je výsledkem vyhodnocenı́ λ-výrazu

(lambda (x y) (cons (f x) y)).

Procedurou modifikujı́cı́ průběžný prvek je v tomto přı́padě vstupnı́ procedura procedury map1 navá-
zaná na symbol f. Výsledek aplikace této procedury na průběžný prvek pak kombinujeme se zbytkem
pomocı́ konstruktoru cons.

• V programu 7.1, ve kterém jsme definovali proceduru lenght, předáváme proceduře foldrproceduru,
která vznikne vyhodnocenı́m výrazu (lambda (x y) (+ 1 y)). Argument x je v nı́ ignorován a je
mı́sto něj uvažováno čı́slo 1. A toto čı́slo je přičı́táno k zabalenı́ zbytku. Rozdělit bychom ji mohli na
konstantnı́ proceduru vracejı́cı́ vždy 1 a na primitivnı́ proceduru sčı́tánı́ navázanou na symbol +.

Toto zobecněnı́ napı́šeme s pomocı́ procedury foldr. Procedura bude brát čtyři argumenty. Prvnı́m z nich
bude procedura combinator o dvou argumentech, která bude určovat způsob nabalovanı́. Smysl těchto
argumentů je v podstatě stejný jako u procedury, která je argumentem procedury foldr. Rozdı́l je jen v tom,
že jı́ jako prvnı́ argument nenı́ předáván přı́mo průběžný prvek, ale jeho modifikace. Modifikacı́ myslı́me

výsledek aplikace procedury modifier, která je druhým argumentem procedury accum, na průběžný
prvek.

(define accum

(lambda (combinator modifier nil l)

(foldr (lambda (x y)

(combinator (modifier x) y))

nil l)))

Uvádı́me několik přı́kladů volánı́ této akumulačnı́ procedury:

(accum + (lambda (x) x) 0 '(1 2 3 4)) Z=⇒ 10

(accum + (lambda (x) (* x x)) 0 '(1 2 3 4)) Z=⇒ 30

(accum + (lambda (x) 1) 0 '(1 2 3 4)) Z=⇒ 4

Už jsme naznačili, jak by se pomocı́ této obecné akumulačnı́ procedury daly napsat procedurymap1alenght.
Na závěr ještě ukážeme, jak bychom ji mohli použı́t k filtrovánı́ seznamu. Kombinačnı́ procedurou bude
spojovánı́ seznamu append, procedurou modifikujı́cı́ průběžné prvky bude procedura, která v závislosti
na platnosti vstupnı́ho predikátu vracı́ bud’to prázdný seznam (), nebo jednoprvkový seznam obsahujı́cı́
tento průběžný prvek. Jako terminátor použijeme prázdný seznam. Následuje celý kód:

(define filter

(lambda (f l)

(accum append

(lambda (x)

(if (f x)

(list x)

'()))

'()

l)))

7.4 Procedura FOLDL

V této sekci se zaměřı́me na variantu akumulačnı́ procedury foldr. Jak jsme si již mohli všimnout, foldr
pracoval tı́m způsobem, že provedl sérii aplikacı́ procedury dvou argumentů, čı́mž nám umožnil postupně
na sebe „nabalovat“ výsledky aplikacı́. Toto „nabalovánı́“ přitom postupovalo směrem doprava:

(〈procedura〉 〈prvek1〉 (〈procedura〉 〈prvek2〉 (〈procedura〉 · · ·(〈procedura〉 〈prvekn〉 〈terminátor〉) · · ·))).

Prvnı́ aplikace, která je dokončena, je aplikace provedená s poslednı́m prvkem seznamu. Následuje aplikace
provedená nad předposlednı́m prvkem seznamu a tak se postupuje až k prvnı́mu prvku. Tento proces
bychom také mohli obrátit. Mohli bychom uvažovat zabalenı́ v tomto směru:

(〈procedura〉 (〈procedura〉 · · · (〈procedura〉 (〈procedura〉 〈terminátor〉 〈prvek1〉) 〈prvek2〉) · · ·) 〈prvekn〉),

kdy je jako prvnı́ aplikována procedura na terminátor a prvnı́ prvek seznamu, výsledek je použit při aplikaci
s druhým prvkem seznamu a tak dále. Jako poslednı́ je provedena aplikace s poslednı́m prvkem seznamu.
U procedury foldr tedy probı́haly aplikace směrem zprava (odtud název fold right). U nově uvedeného
typu „zabalenı́“ probı́há aplikace směrem zleva. Nabı́zı́ se tedy uvažovat proceduru vyššı́ho řádku, která by
byla duálnı́ k foldr a prováděla zabalenı́ druhým z uvedených způsobů (zleva). Tuto proceduru nazveme
foldl (z anglického fold left).

Procedura foldl bude mı́t argumenty stejného typu a významu jako měla procedura foldr, nebudeme
je tedy opakovat. Při své aplikaci provede postupnou sérii aplikacı́ dané procedury na prvky seznamu
(směrem zleva), která je ukončena terminátorem. V literatuře [BW88] se lze setkat s různými variantami
foldl, které se lišı́ tı́m, jaký význam má prvnı́ a druhý argument procedury, která je předaná foldl jako
prvnı́ argument. Podle [BW88] je výsledkem aplikace

(foldl 〈procedura〉 〈terminátor〉 〈seznam〉)

série aplikacı́ tvaru

(〈procedura〉 (〈procedura〉 · · · (〈procedura〉 (〈procedura〉 〈terminátor〉 〈prvek1〉) 〈prvek2〉) · · ·) 〈prvekn〉),

to jest přesně tak, jak jsme naznačili v úvodu sekce. Lze také uvažovat sérii aplikacı́ vypadajı́cı́ takto:

(〈procedura〉 〈prvekn〉 (〈procedura〉 〈prvekn−1〉 (〈procedura〉 · · · (〈procedura〉 〈prvek1〉 〈terminátor〉) · · ·))).

Mezi oběma předchozı́mi sériemi aplikacı́ je zcela zřejmě jediný rozdı́l. V prvnı́m přı́padě je 〈procedura〉
aplikována tak, že jejı́m prvnı́m argumentem je výsledek předchozı́ akumulace a druhým argumentem je
průběžný prvek seznamu. V druhém přı́padě je tomu obráceně: prvnı́m argumentem je průběžný prvek
a druhým argumentem je výsledek předchozı́ akumulace. V obou přı́padech je ale akumulace zahájena od
prvnı́ho prvku seznamu.

Z toho, co jsme ted’ uvedli, by mělo být zřejmé, že procedury provádějı́cı́ výše uvedené „zabalenı́ zleva“
budeme schopni naprogramovat pomocı́ foldr a to v přı́padě obou typů sériı́ aplikacı́. Pro druhý typ
je to jednoduššı́, protože stačı́ použı́t foldr na převrácený seznam. V přı́padě prvnı́ho typu pak už jen
stačı́ obrátit argumenty při aplikaci procedury. Procedury provádějı́cı́ obě zabalenı́ jsou prezentovány
v programu 7.9. Procedura pojmenovaná genuine-foldl reprezentuje „zabalenı́ zleva“ podle [BW88],

Program 7.9. Procedury genuine-foldl a foldl vytvořené pomocı́ foldr a reverze seznamu.

(define genuine-foldl

(lambda (f term l)

(foldr (lambda (x y)

(f y x))

term

(reverse l))))

(define foldl

(lambda (f term l)

(foldr f term (reverse l))))

tedy prvnı́ z uvedených typů. Procedura foldl reprezentuje druhý z typů. Rozdı́ly mezi oběma typy
zabalenı́ a rozdı́l oproti foldr si nejlépe uvědomı́me na následujı́cı́m přı́kladu. Nejprve nadefinujeme
pomocnou proceduru:

(define proc

(lambda (x y)

(list #f x y))),

kterou pak použijeme s týmž seznamem při aplikaci foldr, foldl a genuine-foldl:

(define s '(a b c d))

(foldr proc 'base s) Z=⇒ (#f a (#f b (#f c (#f d base))))

(foldl proc 'base s) Z=⇒ (#f d (#f c (#f b (#f a base))))

(genuine-foldl proc 'base s) Z=⇒ (#f (#f (#f (#f base a) b) c) d)

Jak vidı́me, výsledky aplikaci odpovı́dajı́ oběma typům, které jsme uvedli v této sekci.

Poznámka 7.2. Jednı́m ze základnı́ch vztahů, který platı́ mezi foldr a genuine-foldl je ten, že pokud
je při akumulaci použita monoidálnı́ procedura a jako terminátor je použit jejı́ neutrálnı́ prvek, pak je výsledek
použitı́ foldr a genuine-foldl stejný. Toto pozorovánı́ lze jednoduše dokázat.

Jako přı́klad použitı́ foldl si můžeme uvést proceduru reverse provádějı́cı́ otočenı́ seznamu:

(define reverse

(lambda (l)

(foldl cons '() l)))

Tento přı́klad je poněkud „umělý“, protože v programu 7.9 jsme samotný foldl zavedli pomocı́ reverse.
Kdybychom to učinili a poté definovali reverse předchozı́m způsobem, při pokusu o jeho aplikaci bychom
se dostali do nekončı́cı́ série aplikacı́ (protože foldl aplikuje reverse a obráceně). Ukažme tedy o něco
přirozenějšı́ přı́klad. V sekci 2.5 jsme představili proceduru vyššı́ho řádu compose2 vracejı́cı́, pro dvě
vstupnı́ procedury jednoho argumentu, proceduru reprezentujı́cı́ jejich složenı́, viz přı́klad 2.6 na straně 60.
Nynı́ bychom mohli pomocı́ foldl naprogramovat složenı́ libovolného množstvı́ procedur tak, jak to
ukazuje program 7.10. Procedura compose je pomocı́ foldl vytvořena přı́močaře. Terminátorem je identita,

Program 7.10. Složenı́ libovolného množstvı́ procedur pomocı́ foldl.

(define compose

(lambda functions

(foldl (lambda (f g)

(lambda (x) (f (g x))))

(lambda (x) x)

functions)))

což je neutrálnı́ prvek vzhledem ke skládánı́. Procedura dvou argumentů, která je při aplikaci předána
foldl, provádı́ složenı́ akumulované hodnoty (procedury vzniklé předchozı́mi složenı́mi) s průběžnou
procedurou ze seznamu procedur functions. Proč jsme při skládánı́ nepoužili foldr jako u všech ostatnı́ch
procedur v této lekci? Protože jsme chtěli pro seznam procedur reprezentujı́cı́ funkce f1, . . . , fn (v tomto
pořadı́) vrátit proceduru reprezentujı́cı́ jejich kompozici f1 ◦ f2 ◦ · · · ◦ fn−1 ◦ fn, která je daná

(f1 ◦ f2 ◦ · · · ◦ fn−1 ◦ fn)(x) = (fn(fn−1(· · · (f2(f1(x))) · · · ))),
což vede k použitı́ „zabalenı́ zleva“ – skládánı́ je potřeba aplikovat směrem „zepředu“ seznamu (tedy
použı́váme „zabalenı́ zleva“). Vzhledem k tomu, že skládánı́ funkcı́ na množině je monoidálnı́ operace,
použitı́ genuine-foldl by nám nepomohlo (vedlo by to na stejný výsledek jako použitı́ foldr), protože
bychom tı́m provedli složenı́ v opačném pořadı́, což je vzhledem k nekomutativitě skládánı́ funkcı́ problém.

Následujı́cı́ přı́klady ukazujı́ použitı́ compose. Nejprve použijeme pomocné definice

(define s '(0 1 2 3 4))

(define f1 (lambda (x) (* 2 x)))

(define f2 (lambda (x) (* x x)))

(define f3 (lambda (x) (+ x 1))),

které dále použijeme při skládánı́ pomocı́ compose:

(map (compose) s) Z=⇒ (0 1 2 3 4)

(map (compose f1) s) Z=⇒ (0 2 4 6 8)

(map (compose f1 f2) s) Z=⇒ (0 4 16 36 64)

(map (compose f2 f1) s) Z=⇒ (0 2 8 18 32)

(map (compose f1 f2 f3) s) Z=⇒ (1 5 17 37 65)

(map (compose f3 f2 f1) s) Z=⇒ (2 8 18 32 50)
...

Poznamenejme, že k procedury genuine-foldl a foldl budeme rovněž chápat jako procedury pracujı́cı́
nad libovolným počtem seznamů (vždy alespoň nad jednı́m) stejně tak, jako tomu bylo i u procedur foldr,
map a podobně. V programu 7.11 je uvedeno rozšı́řenı́ těchto procedur tak, aby nepracovaly pouze s jednı́m
seznamem, ale obecně s vı́ce seznamy. V obou přı́padech jsme rozšı́řili seznam argumentů o volitelnou část
a v těle jsme provedli explicitnı́ aplikaci foldr pomocı́ apply. Následujı́cı́ přı́klady ukazujı́ činnost obou
procedur pro vı́ce seznamů:

(foldl list 'base '(a b) '(1 2) '(#f #t)) Z=⇒ (b 2 #t (a 1 #f base))

(genuine-foldl list 'base '(a b) '(1 2) '(#f #t)) Z=⇒ ((base #f 1 a) #t 2 b)

184



Program 7.11. Procedury genuine-foldl a foldl pracujı́cı́ s libovolným počtem seznamů.

(define genuine-foldl

(lambda (f term . lists)

(apply foldr

(lambda args

(apply f (reverse args)))

term

(map reverse lists))))

(define foldl

(lambda (f term . lists)

(apply foldr f term (map reverse lists))))

Poznámka 7.3. Z poslednı́ho přı́kladu a z implementace genuine-list jsme si mohli všimnout, že pro-
ceduru předávanou genuine-list jsme aplikovali s převráceným seznamem argumentů. U procedur vı́ce
než dvou argumentů ale nenı́ jasné, zda-li by se toto „převrácenı́ “ mělo týkat všech argumentů nebo jestli
bychom pouze neměli dát poslednı́ argument (zastupujı́cı́ akumulovanou hodnotu) na začátek seznamu.
Při definici genuine-foldl pracujı́cı́ pouze s jednı́m seznamem jsme tento problém nemuseli vůbec uva-
žovat. Také si všimněte, že procedury foldl se tento problém netýká, protože má argumenty pořád ve
stejném pořadı́. Z tohoto důvodu budeme dále preferovat použı́vánı́ procedury foldl nad procedurou
genuine-foldl (nemluvě o tom, že ve Scheme má foldl praktičtějšı́ uplatněnı́).

Procedury foldr a foldl nejsou přı́tomny ve standardu R5RS jazyka Scheme, ačkoliv některé interprety
jazyka Scheme jimi disponujı́. Tyto procedury jsou přı́tomny v mnoha funkcionálnı́ch programovacı́ch
jazycı́ch. Ve Scheme si foldr i foldl můžeme naprogramovat, což ukážeme v dalšı́ch lekcı́ch.

7.5 Dalšı́ přı́klady na FOLDR a FOLDL

V této sekci uvedeme praktické použitı́ akumulačnı́ procedury foldr. Jako prvnı́ se budeme zabývat pro-
cedurou, která bere jako argument libovolný seznam a vracı́ seznam všech jeho suffixů – včetně prázdného.
Použijeme proceduru foldr tı́mto způsobem: Procedura, která je jejı́m prvnı́m argumentem, vybere ze
seznamu doposud nalezených suffixů prvnı́ prvek, přidá do něj průběžný prvek seznamu a výsledný se-
znam přibalı́ k seznamu nalezených suffixů. Tento seznam obsahuje z počátku jen prázdný seznam, protože
prázdný seznam je suffixem jakéhokoli seznamu. Tı́m je dán druhý argument procedury foldr. Poslednı́m
(třetı́m) argumentem předaným foldr je samotný seznam, jehož suffixy hledáme. Implementace by pak
vypadala takto:

(define suffixes

(lambda (l)

(foldr (lambda (x y)

(cons (cons x (car y)) y))

'(())

l)))

Aplikacı́ takto nadefinované procedury dostáváme seznam suffixů seznamu seřazené od nejdelšı́ho po
nejkratšı́ (to jest po prázdný seznam):

(suffixes '()) Z=⇒ (())

(suffixes '(1)) Z=⇒ ((1) ())

(suffixes '(1 2)) Z=⇒ ((1 2) (2) ())

(suffixes '(1 2 3)) Z=⇒ ((1 2 3) (2 3) (3) ())

Pokud bychom chtěli jen neprázdné suffixy, mohli bychom to udělat mnoha způsoby s použitı́m procedury
suffixes, kterou jsme právě nadefinovali. Ze seznamu suffixů, který je výsledkem aplikace této procedury,
pak můžeme odstranit prázdný seznam vyfiltrovánı́m neprázdných seznamů, odstraněnı́m poslednı́ho
prvku, a tak dále. Též bychom mohli použı́t následujı́cı́ elegantnı́ řešenı́:

(define safe-car

(lambda (x)

(if (null? x)

'()

(car x))))

(define suffixes

(lambda (l)

(foldr (lambda (x y)

(cons (cons x (safe-car y)) y))

'()

l)))

Uvedený program obsahuje definici bezpečné verze selektoru car. Tuto bezpečnějšı́ verzi safe-car jsme
popsal i v sekci 5.4. Jinak se nová procedura suffixes lišı́ jen použitı́ procedury safe-car namı́sto car,
a v použitı́ prázdného seznamu jako terminátoru. Použitı́ safe-car je důležité při prvnı́ aplikaci procedury,
která je argumentem foldr, kdy je aplikována na prázdný seznam.

Použitı́ této procedury dostáváme podobné výsledky jako dřı́ve. Lišı́ se jen v absenci prázdného seznamu
v seznamu nalezených sufixů.

(suffixes '()) Z=⇒ ()

(suffixes '(a)) Z=⇒ ((a))

(suffixes '(a b c)) Z=⇒ ((a b c) (b c) (c))

(suffixes '(a b c d)) Z=⇒ ((a b c d) (b c d) (c d) (d))

Kdybychom namı́sto procedury foldr použili proceduru foldl, nebyl by výsledkem seznam sufixů, ale
naopak seznam prefixů. To je samozřejmě způsobeno změnou směru, kterým je akumulace prováděna.

(define prefixes

(lambda (l)

(foldl (lambda (x y)

(cons (append (safe-car y) (list x)) y))

'()

l)))

Takto nadefinovanou procedurou můžeme hledat všechny neprázdné prefixy zadaného seznamu.

(prefixes '()) Z=⇒ ()

(prefixes '(a)) Z=⇒ (a)

(prefixes '(a b c)) Z=⇒ ((a b c) (a b) (a))

(prefixes '(a b c d)) Z=⇒ ((a b c d) (a b c) (a b) (a))

7.6 Výpočet faktoriálu a Fibonacciho čı́sel

Procedury foldr a foldl lze použı́t i pro výpočet hodnot matematických funkcı́. V této sekci si ukážeme
dvě ukázky použitı́ foldr při výpočtu faktoriálu a Fibonacciho čı́sel. Hned na počátku však řekněme, že
přı́klady reprezentované v této sekci majı́ spı́š „odstrašujı́cı́ charakter“. Jejich smyslem je poukázat na fakt,
že i když se nám podařilo pomocı́ foldr vytvořit řadu užitečných a efektivnı́ch procedur (s krátkým
a přehledným tělem), ne vždy je použitı́ foldr na mı́stě.

Připomeňme, že faktoriál n! nezáporného čı́sla n je definován jako součin přirozených čı́sel od 1 do n.
Neformálně jej tedy lze chápat jako čı́slo dané

n! = 1 · 2 · · · · · n︸ ︷︷ ︸
n činitelů

.

V dalšı́ sekci ukážeme zavedenı́ faktoriálu, které je z matematického hlediska přesnějšı́. V tuto chvı́li si ale
vystačı́me s touto poněkud neformálnı́ definicı́. Pro n = 0, 1, 2, . . . nabývá faktoriál n! následujı́cı́ch hodnot:

1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880, 3628800, 39916800, 479001600, . . .

Našı́m úkolem nynı́ je naprogramovat proceduru fac, která pro daný argument jı́mž bude nezáporné
čı́slo n, vrátı́ hodnotu faktoriálu n!. Z předchozı́ho je tedy jasné, že naše procedura musı́ provést součin
1 · 2 · · · · · n. Nabı́zı́ se tedy pomocı́ build-list nejprve vytvořit seznam činitelů a potom jej vynásobit
pomocı́ apply nebo foldr. Obě verze jsou uvedeny v programu 7.12. Procedury skutečně počı́tajı́ to, co

Program 7.12. Výpočet faktoriálu pomocı́ procedur vyššı́ch řádů.

(define fac

(lambda (n)

(apply * (build-list n (lambda (x) (+ x 1))))))

(define fac

(lambda (n)

(foldr * 1 (build-list n (lambda (x) (+ x 1))))))

majı́, o čemž se můžeme přesvědčit napřı́klad vyhodnocenı́m následujı́cı́ho výrazu:

(map fac '(0 1 2 3 4 5 6 7 8 9)) Z=⇒ (1 1 2 6 24 120 720 5040 40320 362880)

Mnoha čtenářům by se ale mohlo zdát, že procedura fac pracuje až přı́liš složitě na to, že počı́tá tak
jednoduchou funkci jako je faktoriál. To je pravda. Během výpočtu jsme zkonstruovali pomocný seznam,
což trvalo n kroků a jeho prvky jsme posléze vynásobili, to trvalo dalšı́ch n kroků. Časová složitost výpočtu
tedy byla O(2n), při výpočtu jsme navı́c konstruovali seznam, který jsme použili pouze jednorázově. Ani
kód procedury nebyl tak čitelný, jak bychom (u jednoduché funkce jakou je faktoriál) očekávali. Lepšı́
variantu procedury fac ukážeme v dalšı́ lekci.

Druhým přı́kladem v této sekci bude procedura pro výpočet prvků Fibonacciho posloupnosti. Fibonacciho
posloupnost je tvořena počátečnı́mi dvěma prvky F0 = 0, F1 = 1 a každý dalšı́ prvek posloupnost vzniká
součtem předchozı́ch dvou, posloupnost tedy vypadá následovně:

0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, . . .

Procedura počı́tajı́cı́ prvky Fibonacciho posloupnosti je uvedená v programu 7.13. Rozeberme si nynı́, jak

Program 7.13. Výpočet prvků Fibinacciho posloupnosti pomocı́ procedur vyššı́ch řádů.

(define fib

(lambda (n)

(cadr

(foldr (lambda (x last)

(list (apply + last) (car last)))

'(1 0)

(build-list n (lambda (x) #f))))))

je procedura fib naprogramovaná. Procedura ve svém těle provádı́ akumulaci pomocı́ foldr. Při této
akumulaci jsou postupně sčı́tány dvě předchozı́ Fibonacciho čı́sla, čı́mž se zı́ská dalšı́ prvek posloupnosti.
Ten je dál použit s předchozı́m prvkem k zı́skánı́ dalšı́ho prvku a tak dále. Akumulace probı́há přes seznam
vytvořený aplikacı́ build-list. Tento seznam má délku n. Samotné prvky seznamu pro nás nebudou mı́t
žádný význam, seznam je použit pouze jako „čı́tač kroků“ určujı́cı́, kolikátý člen Fibonacciho posloupnost

má být nalezen. Při samotné akumulaci je vždy vytvářen dvouprvkový seznam ve tvaru (Fi+1 Fi). Tedy
druhý prvek seznamu je průběžné Fibonacciho čı́slo a prvnı́ prvek seznamu je jeho následnı́k. Pokud máme
k dispozici Fi a Fi+1 můžeme hodnotu Fi+2 stanovit součtem Fi + Fi+1, což provádı́me při akumulaci,
viz explicitnı́ aplikaci procedury sčı́tánı́. Výsledkem akumulace v i-tém kroku je tedy vytvořenı́ seznamu
tvaru (Fi+2 Fi+1) na základě seznamu (Fi+1 Fi), který je navázaný na symbol last (všimněte si, že
argument x je ignorován). Terminátorem akumulace je seznam (F1 F0), to jest seznam (1 0). Pro dané
n je výsledkem akumulace seznam (Fn+1 Fn), stačı́ tedy vrátit jeho druhý prvek, což je požadovaný
výsledek, to je provedeno aplikacı́ cadr na výsledek akumulace. Viz přı́klad použitı́ procedury:

(map fib '(0 1 2 3 4 5 6 7 8 9)) Z=⇒ (0 1 1 2 3 5 8 13 21 34)

Při akumulaci se postupně vytvářı́ dvouprvkové seznamy poslednı́ch dvou uvažovaných Fibonacciho čı́-
sel. Napřı́klad při výpočtu (fib 9) bude argument last při aplikaci procedury předané foldr postupně
nabývat hodnot (1 0), (1 1), (2 1), (3 2), (5 3), (8 5), (13 8), (21 13) a konečně (34 21). Vý-
sledkem akumulace bude tı́m pádem seznam (55 34), jehož druhý prvek je vrácen jako výsledek výchozı́
aplikace (fib 9). Netřeba asi zdůrazňovat, že tento způsob výpočtu Fibonacciho čı́sel je opět dost neefek-
tivnı́. V prvnı́ řadě, kód procedury fib je dost nestravitelný a jeho úplné pochopenı́ již vyžaduje nějakou
chvı́li. Během výpočtu je dále konstruována celá řada „odpadnı́ch seznamů“, napřı́klad n-prvkový seznam
(#f #f · · · , přes který se akumuluje, a jehož hodnoty (paradoxně) nemajı́ žádný význam. Dále v každém
kroku konstruujeme dvouprvkový seznam udržujı́cı́ informaci o poslednı́ch dvou stanovených Fibonacciho
čı́slech, což také výrazně ubı́rá na efektivitě.

Čistota kódu (jeho jednoduchost a čitelnost) a jeho efektivita (výkon) jdou v některých přı́padech proti sobě.
Výše uvedené přı́klady však nejsou ani čistě provedené ani efektivnı́. V dalšı́ sekci se mimo jiné zaměřı́me
na zefektivněnı́ a čistějšı́ naprogramovánı́ procedur fac a fib.

Shrnutı́

V této lekci jsme ze zabývali akumulacı́, což je speciálnı́ postupná aplikace procedur. V jazyku Scheme jsme
uvažovali dodatečné procedury foldr a foldl, pomocı́ nichž lze akumulaci provádět. Ukázali jsme efek-
tivnı́ implementace vybraných procedur pracujı́cı́ch se seznamy, napřı́klad mapovánı́, filtrace a nahrazovánı́
prvků seznamu. Provedli jsme diskusi ohledně jejich časové složitosti i ohledně složitosti jejich původnı́ch
verzı́. Dále jsme se zabývali problematikou rozšiřovánı́ procedur dvou argumentů tak, aby pracovaly s li-
bovolným počtem argumentů. Lekci jsme zakončili ukázkou metody výpočtu faktoriálu a Fibonacciho čı́sel
pomocı́ akumulace.

Pojmy k zapamatovánı́

• akumulace, explicitnı́ aplikace, filtrace,
• zabalenı́ směrem doprava, zabalenı́ směrem doleva,
• rozšı́řenı́ procedur dvou argumentů na libovolné argumenty,
• efektivnı́ implementace procedur,
• výpočet faktoriálu a Fibonacciho čı́sel

Nově představené prvky jazyka Scheme

• procedury foldr, foldl a genuine-foldl.

Kontrolnı́ otázky

1. Čı́m se od sebe lišı́ foldr a foldr?
2. Jak probı́há aplikace foldr?
3. K čemu sloužı́ terminátory?
4. Jak lze využı́t foldr k rozšı́řenı́ procedury dvou argumentů na libovolný počet argumentů?
5. Co pro nás hrálo klı́čovou roli při stanovovánı́ časové náročnosti procedur?

6. Jaký je rozdı́l mezi foldl a genuine-foldl?

Cvičenı́
1. V sekci 7.3 jsme rozšı́řili proceduru min2 na proceduru libovolného počtu argumentů. Stejným způso-

bem rozšiřte proceduru na výběr čı́sla s extrémnı́ absolutnı́ hodnotou abs-min.
2. Napište predikát, který pro posloupnost zjistı́, zda je neklesajı́cı́. Posloupnost bude reprezentovaná

seznamem, jehož prvky jsou čı́sla. Viz přı́klady aplikace:
(nondecreasing? '()) Z=⇒ #t

(nondecreasing? '(1 2 3 4)) Z=⇒ #t

(nondecreasing? '(1 2 4 3)) Z=⇒ #f

(nondecreasing? '(1 4 2 3)) Z=⇒ #f

(nondecreasing? '(4 1 2 3)) Z=⇒ #f

3. Napište procedury after a before, jejichž argumenty budou element 〈elem〉 a seznam 〈l〉. Procedura
after bude vracet seznam prvků za poslednı́m výskytem prvku 〈elem〉 (včetně) v seznamu 〈l〉. Pro-
cedura before zase seznam prvků před prvnı́m výskytem prvku 〈elem〉 (včetně) v seznamu 〈l〉. Viz
přı́klady použitı́:
(after 10 '(1 2 3 4 3 5 6)) Z=⇒ ()

(after 3 '(1 2 3 4 3 5 6)) Z=⇒ (3 5 6)

(after 6 '(1 2 3 4 3 5 6)) Z=⇒ (6)

(before 10 '(1 2 3 4 3 5 6)) Z=⇒ (1 2 3 4 5 6)

(before 1 '(1 2 3 4 3 5 6)) Z=⇒ (1)

(before 3 '(1 2 3 4 3 5 6)) Z=⇒ (1 2 3)

Řešenı́ ke cvičenı́m
1. (define abs-min2

(lambda (x y)

(if (<= (abs x) (abs y)) x y)))

(define abs-min

(lambda args

(foldr abs-min2 (car args) (cdr args))))

2. (define nondecreasing?

(lambda (l)

(foldr (lambda (x y)

(if y

(if (or (equal? y #t) (<= x y))

x

#f)

#f))

#t

l)))

3. (define conseq

(lambda (elem x y)

(if (car y)

y

(cons (equal? x elem)

(cons x (cdr y))))))

(define after

(lambda (elem l)

(let ((found (foldr (lambda (x y) (conseq elem x y)) '(#f . ()) l)))

(if (car found)

(cdr found)

#f))))

(define before

(lambda (elem l)

(let ((found (foldl (lambda (x y) (conseq elem x y)) '(#f . ()) l)))

(if (car found)

(reverse (cdr found))

#f))))

Lekce 8: Rekurze a indukce

Obsah lekce: Tato lekce se věnuje rekurzivnı́m procedurám a výpočetnı́m procesům generovaným rekur-
zivnı́mi procedurami. Dále ukážeme, jak je možné použı́t princip indukce pro dokázánı́ správnosti definice
rekurzivnı́ procedury. Nejprve se zaměřı́me na rekurzi a princip indukce přes přirozená čı́sla. Dále ukážeme
rekurzi a indukci na seznamech, které hrajı́ (nejen) ve funkcionálnı́m programovánı́ klı́čovou roli. Během
výkladu se budeme zabývat vlastnostmi výpočetnı́ch procesů generovaných rekurzivnı́mi procedurami,
jejich složitostı́, průběhem výpočtu a jeho náročnostı́.

Klı́čová slova: koncová rekurze, lineárnı́ rekurze, princip indukce, rekurze, stromová rekurze, střadače.

8.1 Definice rekurzı́ a princip indukce

Pojmy rekurze a indukce jsou jedny z nejdůležitějšı́ch pojmů v informatice a to jak teoretické tak aplikované.
V této úvodnı́ sekci se budeme oběma pojmy zabývat spı́še z matematického pohledu, ale ukážeme jejich
silnou vazbu k funkcionálnı́mu programovánı́ a programovánı́ obecně. Aby nedošlo hned na počátku
k nějakému nedorozuměnı́, upozorněme na fakt, že slovo „rekurze“ má v informatice, ale i v jiných disci-
plı́nách (napřı́klad v logice), mnoho různých významů. My se budeme zabývat rekurzı́ jako metodou definice
funkcı́ (ve smyslu matematických funkcı́, čili zobrazenı́) a procedur. Indukci budeme použı́vat jako obecný
dokazovacı́ princip jı́mž budeme schopni dokazovat vlastnosti rekurzivně definovaných funkcı́ a procedur.

V programátorské terminologii je pod pojmem rekurzivnı́ procedura obvykle myšlena procedura, která ve
svém těle provádı́ aplikaci sebe sama. Tak se na rekurzivnı́ procedury budeme dále v textu dı́vat i my. Ukážeme,
že rekurzı́ (to jest „aplikacı́ sebe sama“) lze vyřešit mnoho problémů z předchozı́ch lekcı́ elegantně (co se
týče jejich naprogramovánı́ a čitelnosti kódu) a efektivně (co se týče jejich výpočetnı́ složitosti). Na úvod
také podotkněme, že rekurzivnı́ procedury (procedury aplikujı́cı́ sebe sama) jsou z technického hlediska
obyčejné procedury. Při úvahách o rekurzivnı́ch procedurách tedy nebudeme muset nijak rozšiřovat modely
vyhodnocovánı́ elementů ani aplikace procedur. Důvodem, proč se těmto „vlastně obyčejným procedurám“
budeme věnovat několik lekcı́ je, že programátoři potřebujı́ obvykle delšı́ dobu na úplné pochopenı́ rekurze
jako principu – od programátora to vyžaduje jistou dávku představivosti a také trpělivosti (zvláště při
počátečnı́m seznamovánı́ se s problematikou a s řešenı́m prvnı́ch přı́kladů).

Nynı́ si ukážeme několik definic pomocı́ rekurze. Abychom si ukázali, že rekurze jako princip nenı́ omezená
jen na „programovánı́ procedur“, budeme si v této sekci ukazovat rekurzivnı́ definice různých zobrazenı́, se
kterými jsme se již setkali v předchozı́ch lekcı́ch. Nebudeme přitom zatı́m ukazovat přı́mou vazbu na jazyk
Scheme. V dalšı́ch sekcı́ch si pak ukážeme souvislost s vytvářenı́m rekurzivnı́ch procedur (ve Scheme).

Uved’me si nejprve několik motivačnı́ch přı́kladů. V sekci 7.6 jsme uvedli proceduru pro výpočet faktoriálu
daného čı́sla. Faktoriál n! čı́sla n jsme popsali poněkud neformálně jako „součin čı́sel od 1 do n“. Mohli
bychom jej však definovat daleko přesněji. Faktoriál lze chápat jako funkci (zobrazenı́), které každému
nezápornému celému čı́slu n přiřazuje hodnotu n! danou následujı́cı́m vztahem:

n! =

{
1 pokud n ≤ 1,
n · (n− 1)! jinak.

Tento vztah řı́ká, že faktoriál nuly a jedničky je roven jedné (prvnı́ řádek definičnı́ho vztahu). Pro n ≥ 2
je faktoriál n! definován na druhém řádku jako čı́slo dané n · (n − 1)!. Slovně řečeno, faktoriál pro n ≥ 2
zı́skáme jako „součin n s faktoriálem n− 1“. V předchozı́ definici jsme vlastně zavedli faktoriál čı́sla n pomocı́
faktoriálu menšı́ho čı́sla. Nejedná se ale o „definici kruhem“, protože faktoriál n je vyjádřen pomocı́ faktoriálu
n− 1 (tedy pomocı́ hodnoty jiného faktoriálu, nikoliv pomocı́ své vlastnı́ hodnoty, což by vedlo „do kruhu“).
Výše uvedená definice n! je prvnı́m přı́kladem rekurzivnı́ definice. V tomto přı́padě tedy rekurzivnı́ definice
zobrazenı́ z N0 (nezáporná celá čı́sla) do N (přirozená čı́sla).

Rozepı́šeme-li hodnoty 0!, 1!, 2!, . . . podle předchozı́ definice, zı́skáme:

0! = 1,
1! = 1,
2! = 2 · 1! = 2 · 1 = 2,

3! = 3 · 2! = 3 · 2 · 1! = 3 · 2 · 1 = 6,
4! = 4 · 3! = 4 · 3 · 2! = 4 · 3 · 2 · 1! = 4 · 3 · 2 · 1 = 24,

...

Všimněte si, že počı́naje třetı́m řádkem můžeme hodnotu faktoriálu jednoznačně spočı́tat na základě znalosti
výsledku z předchozı́ho řádku. To přesně koresponduje s definičnı́m předpisem n! pro n ≥ 2. Stručněji
zapsáno tedy máme:

0! = 1,
1! = 1,
2! = 2 · 1! = 2 · 1 = 2,
3! = 3 · 2! = 3 · 2 = 6,
4! = 4 · 3! = 4 · 6 = 24,

...

Ačkoliv jsme pro n nabývajı́cı́ konkrétnı́ch hodnot n = 0, . . . , 4 ukázali, že výsledky n! jsou jednoznačně
definované (a majı́ očekávané hodnoty), nijak jsme zatı́m neprokázali tento fakt pro libovolné nezáporné
celé n. Je přirozeně jasné, že nelze udělat „ručnı́ výpis n!“ pro každé n, protože nezáporných celých čı́sel
je nekonečně mnoho. Pro ověřenı́ správnosti tedy musı́me sáhnout po nějakém formálnı́m dokazovacı́m
principu13. Správnost našeho zavedenı́ si za chvı́li dokážeme pomocı́ matematické indukce. Ještě předtı́m
však uved’me druhý přı́klad.

Posloupnost F0, F1, F2, . . . Fibonacciho čı́sel, kterou jsme neformálně zavedli v sekci 7.6 jako „posloupnost
začı́najı́cı́ 0 a 1 a jejı́ž každý dalšı́ prvek je součtem předchozı́ch dvou“, bychom nynı́ mohli přesně zavést
pomocı́ definičnı́ho vztahu

Fn =


0 pokud n = 0,
1 pokud n = 1,
Fn−1 + Fn−2 jinak.

Nebo pomocı́ jeho stručnějšı́ ekvivalentnı́ varianty

Fn =

{
n pokud n ≤ 1,
Fn−1 + Fn−2 jinak.

Jedná se opět o přı́klad rekurzivnı́ definice, protože jsme hodnotu Fn (to jest n-té Fibonacciho čı́slo) definovali
pomocı́ hodnot Fn−2 a Fn−1. Je v celku evidentnı́, že každé Fi je jednoznačně definované a že přiřazenı́
n 7→ Fn (pro každé nezáporné celé n) je zobrazenı́ z množiny N0 do množiny N0, přesto i tento fakt dále
dokážeme indukcı́. Dosazenı́m do výše uvedeného definičnı́ho vztahu můžeme vyjádřit Fibonacciho čı́sla
F0, . . . , F4, . . . následovně:

F0 = 0,
F1 = 1,
F2 = F1 + F0 = 0 + 1 = 1,
F3 = F2 + F1 = (F1 + F0) + F1 = (1 + 0) + 1 = 2,
F4 = F3 + F2 = (F2 + F1) + (F1 + F0) = ((F1 + F0) + F1) + (F1 + F0) = ((1 + 0) + 1) + (1 + 0) = 3,

...

Obdobně jako i u faktoriálu můžeme vidět, že počı́naje třetı́m řádkem můžeme každé Fibonacciho čı́slo
stanovit z hodnot předchozı́ch dvou řádků, to jest:

F0 = 0,
F1 = 1,
F2 = F1 + F0 = 0 + 1 = 1,
F3 = F2 + F1 = 1 + 1 = 2,
F4 = F3 + F2 = 2 + 1 = 3,

13Někdo by v tomto okamžiku mohl namı́tat, že „správnost je přece jasná.“ V přı́padě faktoriálu bychom mohli připustit, že
správnost jeho rekurzivnı́ho zavedenı́ je zřejmá. V praxi je však potřeba definovat rekurzivně daleko složitějšı́ zobrazenı́, u kterých
již o správnosti našı́ vlastnı́ definice nemusı́me být „jen tak“ přesvědčeni (správněji: neměli bychom být přesvědčeni – pokud ovšem
netrpı́me obzvláště vyvinutým syndromem „programátorské arogance“), a měli bychom mı́t tedy k dispozici formálnı́ aparát,
kterým správnost prokážeme.

F5 = F4 + F3 = 3 + 2 = 5,
F6 = F5 + F4 = 5 + 3 = 8,...

V tuto chvı́li bychom mohli řı́ct, že pro princip definice rekurzı́ je charakteristické, že n-tá hodnota zobrazenı́
(kromě nulté) může být (obecně ale nemusı́) definována pomocı́ hodnoty přı́slušné některému předchůdci
čı́sla n. Jelikož 0 nemá předchůdce, funkčnı́ hodnota pro nulu musı́ být explicitně definována bez rekurze.
V dalšı́ části této sekce navı́c zjistı́me, že princip rekurze je možné uplatnit nejen pro definice zobrazenı́
z množin nezáporných celých čı́sel (do nějakých jiných množin), nýbrž i pro definice obecných zobrazenı́
z množin hodnot s vhodnou strukturou (napřı́klad seznamů).

Slovo rekurze pocházı́ z latiny a jeho původnı́m významem je „jı́t zpět“, což dobře koresponduje s tı́m, jak
chápeme rekurzi jako princip definice matematických funkcı́ (zobrazenı́): funkčnı́ hodnoty pro některá n
jsou definovány pomocı́ funkčnı́ch hodnot pro čı́sla předcházejı́cı́ n. Při výpočtu funkčnı́ hodnoty pro n
se tedy „jde zpět“ k funkčnı́m hodnotám pro čı́sla ostře menšı́ než n, jejich funkčnı́ hodnoty mohou být
opět definovány pomocı́ funkčnı́ch hodnot předchozı́ch čı́sel, a tak dále. Při výpočtu se tedy postupuje
od funkčnı́ hodnoty pro n směrem k funkčnı́m hodnotám pro čı́sla menšı́ než n.

Nynı́ si ukážeme princip indukce přes přirozená čı́sla (princip matematické indukce), který bude dostačovat při
dokazovánı́ vlastnostı́ rekurzivně definovaných funkcı́ (zobrazenı́) jakými byly výše uvedené n! (faktoriál)
a Fn (Fibonacciho čı́sel). Z praktických důvodů budeme v dalšı́m výkladu přidávat k množině přirozených
čı́sel i nulu a nebudeme to již všude zdůrazňovat.

V následujı́cı́m výkladu budeme pracovat s pojmem „vlastnost přirozeného čı́sla“. Samotný pojem „vlast-
nost“ nebudeme přı́liš formalizovat. Vlastnost budeme značit P . Budeme-li mı́t dánu vlastnost P , pak
budeme vždy předpokládat, že pro každé přirozené čı́slo n platı́: bud’ (i) n má/splňuje vlastnost P (vlastnost
P platı́ pro n), což budeme dále značit P (n), nebo (ii) n nemá/nesplňuje vlastnost P (vlastnost P neplatı́
pro n)14.

Poznámka 8.1. Jako přı́klady vlastnostı́ přirozených čı́sel můžeme uvést vlastnost P , která řı́ká: „n je sudé“,
dále vlastnost P , která řı́ká „n je prvočı́slo“ a podobně. Tyto dvě vlastnosti platı́ pro některá přirozená čı́sla
a pro jiná neplatı́. Třeba vlastnost P : „n je dělitelné jedničkou“ platı́ pro každé přirozené čı́slo. V dalšı́m
textu pro nás budou důležité právě vlastnosti platné pro každé přirozené čı́slo (bude se však jednat o vlast-
nosti netriviálnı́ho charakteru dané rekurzivnı́mi předpisy funkcı́, výrazně složitějšı́ než „n je dělitelné
jedničkou“, a jejich platnost pro každé n budeme muset prokázat). Pokud bude někdy ve slovnı́m popisu
figurovat vı́ce čı́sel, pak by mohlo z vágnı́ho popisu (v přirozeném jazyku) dojı́t k nedorozuměnı́. Napřı́klad
v popisu vlastnosti „čı́slo n je menšı́ nebo rovno čı́slu m“ figurujı́ označenı́ „dvou čı́sel“, nenı́ tedy jasné,
ke kterému se vlastnost vztahuje (přitom je dı́ky použité nerovnosti čı́sel podstatný rozdı́l v tom, jestli
vlastnost vztáhneme k n či k m). V takovém přı́padě budeme vlastnost symbolicky zapisovat ve tvaru

P (n): „ . . . n . . . “,

abychom explicitně vyjádřili, že se v popisu vyjadřuje vlastnost čı́sla označeného n.

V následujı́cı́ větě je prokázána platnost principu indukce přes přirozená čı́sla.

Věta 8.2 (princip indukce přes přirozená čı́sla). K tomu abychom ověřili, že vlastnost P platı́ pro každé n ∈ N0,
stačı́ prokázat platnost následujı́cı́ch dvou bodů:

(i) platı́ P (0),

(ii) pokud platı́ P (i), pak platı́ P (i+ 1).

14Toto neformálnı́ chápánı́ vlastnostı́ budeme však muset použı́vat velice obezřetně. Snadno bychom totiž mohli vytvořit
vlastnost, která by vedla k logickým paradoxům. V tomto kursu však budeme vždy použı́vat vlastnosti, které k nim nevedou. Vı́ce
informacı́ o problematice paradoxů bude studentům předneseno v rámci kursu matematické logiky.

Důkaz. Tvrzenı́ dokážeme sporem. Necht’jsou splněny body (i) a (ii) a zároveň existuje čı́slo n ∈ N0, které
nemá vlastnost P . Ze všech čı́sel, která nemajı́ P můžeme vybrat nejmenšı́ z nich (nejmenšı́ takové čı́slo
vždy existuje, protože množina přirozených čı́sel je dobře uspořádaná a z předpokladu plyne, že existuje
aspoň jedno takové čı́slo). Označme toto čı́slo n0. Pak n0 nemůže být rovno 0, protože bychom porušili
platnost (i). Tı́m pádem n0 ≥ 1. Jelikož jsme n0 zvolili jako nejmenšı́ z čı́sel nemajı́cı́ch P , pak n0 − 1 musı́
mı́t P . Potom ale dle bodu (ii) i n0 musı́ mı́t P , což je spor.

Následujı́cı́ přı́klady ukazujı́ použitı́ principu indukce k prokázánı́ jednoznačnosti a správnost rekurzivnı́ch
definic faktoriálu a posloupnosti Fibonacciho čı́sel.

Přı́klad 8.3. Všimněte si, že principem indukce, viz větu 8.2, nynı́ můžeme snadno prokázat, že výše
uvedená rekurzivnı́ definice n! je skutečně korektnı́, to jest že každému n přiřazuje jednoznačně hodnotu
n!, která je součinem čı́sel od jedné do n. Uvažujme vlastnost P , která řı́ká: „hodnota n! je jednoznačně
definovaná“. Pro n = 0 platı́ P zcela jasně, protože je to dáno prvnı́m řádkem definičnı́ho vztahu, bod (i)
věty 8.2 je tedy pro P splněn triviálně. Ověřı́me bod (ii). Předpokládejme, že n má vlastnost P , tedy hodnota
n! je jednoznačně daná. Pokud je n = 0, pak pro n + 1 je situace opět pokryta prvnı́m řádkem (a indukčnı́
předpoklad v tomto přı́padě na nic nepotřebujeme). Pokud n > 1, pak z toho, že n! je jednoznačně daná
a z druhého řádku odvodı́me

(n+ 1)! = (n+ 1) · (n+ 1− 1)! = (n+ 1) · n!,
což je jednoznačně daná čı́selná hodnota vzniklá vynásobenı́m n! (jednoznačně dané) s čı́slem n + 1. Bod
(ii) taky platı́. Použitı́m principu indukce jsme tedy prokázali jednoznačnost předchozı́ rekurzivnı́ defi-
nice n!. Stejně tak bychom mohli principem indukce dokázat platnost vlastnosti P řı́kajı́cı́ „n! je násobkem
přirozených čı́sel od 1 do n“.

Přı́klad 8.4. Analogicky můžeme principem indukce prokázat, že definice Fn je korektnı́ pro každé n.
Uvažujme pro tento účel vlastnost P (n): „Fi je jednoznačně definované čı́slo pro každé nezáporné i ≤ n“.
Evidentně n = 0 splňuje P , tedy (i) z věty 8.2 pro P platı́. Stejně tak pro n = 1 platı́ P . Z předpokladu,
že P platı́ pro n nynı́ odvodı́me platnost P pro n + 1. To, že P platı́ pro n znamená, že hodnoty všech
Fi, kde i ≤ n, jsou jednoznačně dané. Máme ukázat, že i Fn+1 je jednoznačně daná. Podle druhého řádku
rekurzivnı́ definice Fibonacciho čı́sel platı́:

Fn+1 = Fn+1−1 + Fn+1−2 = Fn + Fn−1.

To jest, Fn+1 je hodnota vzniklá součtem dvou jednoznačně daných hodnot Fn a Fn−1 (jejich jednoznačnost
je daná induktivnı́m předpokladem), tedy i Fn+1 je jednoznačně daná hodnota. Pro P tedy platı́ i bod (ii)
věty 8.2, to jest každé Fn je jednoznačně dané.

Všimněte si, že v rekurzivnı́ch definicı́ch faktoriálu a Fibonacciho čı́sel nemůžeme vypustit meznı́
přı́pady. To jest pro faktoriál přı́pady 0! = 1! = 1 a pro Fibonacciho čı́sla F0 = 0 a F1 = 1. Bez
nich by totiž definice nebyla úplná (a tı́m pádem ani jednoznačná). Dále si všimněte, že při definici
rekurzı́ „jdeme zpět“, to jest vyjadřujeme funkčnı́ hodnoty pomocı́ funkčnı́ch hodnot definovaných pro
předcházejı́cı́ čı́sla. U principu indukce je tomu naopak. Při prokazovánı́ indukcı́ „jdeme dopředu“.

Rekurze a indukce nemusejı́ „jı́t jen přes přirozená čı́sla“. V předchozı́ch ukázkách jsme použili principy
rekurze a indukce dı́ky tomu, že pro každé uvažované nezáporné celé čı́slo n jsme vždy byli schopni rozlišit
dva základnı́ přı́pady:

(i) bud’ platı́ n = 0,

(ii) nebo je n ve tvaru m+ 1, kde m ∈ N0.

To jest bud’ bylo dané nezáporné celé čı́slo nula, nebo bylo následnı́kem jiného nezáporného celého čı́sla.
Principy rekurze a indukce byly možné právě dı́ky tomu, že množina uvažovaných čı́sel N0 (přı́padně
vybavená operacemi jako je sčı́tánı́ a podobně) měla tuto vhodnou strukturálnı́ vlastnost. Existujı́ však i jiné
množiny, u kterých lze přirozeně najı́t strukturálnı́ vlastnosti, které umožňujı́ mı́t principy rekurze a indukce
(v modifikované podobě). Pro nás jako informatiky bude důležitá množina všech seznamů (vybavená dalšı́mi
operacemi) a principy rekurze a indukce, které půjdou „přes seznamy.“

Abychom se nynı́ mohli bavit o rekurzi a indukci na seznamech, ale pořád si (zatı́m) zachovali jistý od-
stup od programovacı́ho jazyka Scheme, zvolı́me následujı́cı́ notaci. Množinu všech neprázdných seznamů
(chápaných jako elementy) budeme označovat L; množinu všech seznamů (chápaných jako elementy) bu-
deme označovat L(); množinu všech elementů budeme označovat E . Máme tedy L ⊂ L() ⊂ E , přitom
L() = L∪{()}. Samotné seznamy, to jest prvky množiny L(), budeme označovat jako obvykle, i když v tuto
chvı́li je pro nás mnohem důležitějšı́ množina všech seznamů L() než konkrétnı́ seznamy. Plně v souladu
s pojetı́m procedur jako matematických funkcı́ (viz sekci 2.5 na straně 56) můžeme nynı́ chápat konstruktor
cons a selektory car a cdr jako následujı́cı́ zobrazenı́:

cons : E × L() → L,
car : L → E ,
cdr : L → L().

To jest, cons zobrazuje dvojice 〈element, seznam〉 na neprázdné seznamy (funkčnı́ hodnota cons(e, l) před-
stavuje seznam vzniklý připojenı́m elementu e na začátek seznamu l); car zobrazuje neprázdné seznamy na
elementy (funkčnı́ hodnota car(l) představuje prvnı́ prvek seznamu l); cdr zobrazuje neprázdné seznamy
na seznamy (funkčnı́ hodnota cdr(l) představuje seznam l bez prvnı́ho prvku). Snadno nahlédneme, že pro-
cedury cons, car a cdr lze skutečně chápat jako reprezentace zobrazenı́ cons, car a cdr. Pomocı́ funkce cons
můžeme vyjádřit strukturálnı́ vlastnost seznamů, kterou budou použı́vat dále uvedené principy rekurze
a indukce: pro každý seznam l ∈ L() platı́, že

(i) bud’ je l prázdný seznam,

(ii) nebo existuje seznam k ∈ L() a element e ∈ E tak, že platı́ l = cons(e, k).

Nynı́ bychom mohli uvažovat dalšı́ procedury, třeba length a append2, a jim odpovı́dajı́cı́ zobrazenı́:

append2: L() × L() → L(),
length: L() → N0.

Zobrazenı́ append2 zobrazuje dvojice seznamů na seznam vzniklý jejich spojenı́m, zobrazenı́ length zobra-
zuje seznamy na jejich délky.

Nynı́ se nám nabı́zı́ alternativnı́ pohled na procedury versus zobrazenı́ (matematické funkce). Většinu
procedur, se kterými se při programovánı́ v Jazyku Scheme setkáváme, lze považovat za konečné repre-
zentace zobrazenı́, to jest konečné reprezentace obecně nekonečných matematických objektů. Funkcionálnı́
jazyk považujeme za čistý, pokud lze každou primitivnı́ proceduru tohoto jazyka chápat jako reprezen-
taci zobrazenı́ a v jazyku nelze vytvořit uživatelsky definovanou proceduru, která by nějaké zobrazenı́
nereprezentovala. Jazyk Scheme z tohoto pohledu čistý nenı́. To je důsledek mimo jiné toho, že s pro-
středı́m manipulujeme jako s elementem prvnı́ho řádu a umožňujeme explicitnı́ vyhodnocenı́ elementů
relativně vzhledem k prostředı́, viz lekci 6. Mezi čisté funkcionálnı́ jazyky patřı́ třeba Haskell a Clean.

Zobrazenı́ length a append2 můžeme velmi snadno nadefinovat rekurzı́ přes seznamy. Při tomto typu re-
kurze jsou funkčnı́ hodnoty f(. . . , l, . . . ), kde l je seznam, vyjádřeny pomocı́ funkčnı́ch hodnot f(. . . , k, . . . ),
kde k je seznam vyjádřitelný z l pomocı́ (aspoň jednoho) použitı́ car a cdr. Abychom zpřesnili pojem „být vyjá-
dřitelný pomocı́ car a cdr“, zavedeme následujı́cı́ pojem.

Definice 8.5 (strukturálnı́ složitost seznamů). Seznam k se nazývá strukturálně jednoduššı́ než seznam l,
pokud existujı́ zobrazenı́ f1, . . . , fk (k ≥ 1) tak, že {f1, . . . , fk} ⊆ {car, cdr} a k = f1(f2(· · · (fk(l)) · · · )). �

Princip definice rekurzı́ přes seznamy je tedy založen na tom, že definujeme (funkčnı́) hodnotu pro prázdný
seznam, který je ze všech seznamů strukturálně nejjednoduššı́ (neexistuje seznam, který by byl struktu-
rálně jednoduššı́ než prázdný seznam), a dále v přı́padě neprázdných seznamů vytvořı́me předpis, který
vyjádřı́ (funkčnı́) hodnoty pro tyto seznamy pomocı́ (funkčnı́ch) hodnot seznamů, které jsou strukturálně
jednoduššı́.

Napřı́klad u délky seznamu můžeme uvažovat takto:

Obrázek 8.1. Schématické zachycenı́ úvahy o spojenı́ dvou seznamů

„Délka prázdného seznamu je nula. Pokud daný seznam nenı́ prázdný, pak je ve tvaru cons(e, k),
kde k je opět seznam. Navı́c k je vyjádřitelný pomocı́ cdr ze seznamu l, protože

k = cdr(cons(e, k)) = cdr(l).

Jelikož je délka seznamu k o jedno menšı́ než délka seznamu l, délku l lze vyjádřit pomocı́ délky
k jako hodnotu 1 + length(k) = 1 + length(cdr(l)).“

Tato úvaha vede na následujı́cı́ rekurzivnı́ definici:

length(l) =

{
0 pokud l je prázdný seznam,
1 + length(cdr(l)) jinak.

Předchozı́ definice vyjadřuje přesně to, jak jsme slovně length popsali. Zcela v souladu s tı́m, jak jsme
v úvodu naznačili, jsme length definovali pro dva přı́pady: nejprve jsme řekli, co je hodnotu length(()) (co
je délkou prázdného seznamu) a pak jsme využili faktu, že druhý prvek každého neprázdného seznamu
je opět seznam a můžeme pro něj uvažovat funkčnı́ hodnotu length a operovat s nı́. Ukažme si funkčnı́
hodnoty length v přı́padě některých seznamů:

length(()) = 0,
length((a)) = 1 + length(cdr((a))) = 1 + length(()) = 1 + 0 = 1,
length((a b)) = 1 + length(cdr((a b))) = 1 + length((b)) = 1 + (1 + length(cdr((b)))) =

= 1 + (1 + length(())) = 1 + (1 + 0) = 2,
length((a b c)) = 1 + length(cdr((a b c))) = 1 + length((b c)) = 1 + (1 + length(cdr((b c)))) =

= 1 + (1 + length((c))) = 1 + (1 + (1 + length(cdr((c))))) =
= 1 + (1 + (1 + length(()))) = 1 + (1 + (1 + 0)) = 3,
...

Pro prázdný, jednoprvkový, dvouprvkový a třı́prvkový seznamy jsou tedy funkčnı́ hodnoty výše defino-
vané length očekávané. Jak tomu bude v přı́padě pro libovolný seznam? Stejně jako v přı́padě faktoriálu
a Fibonacciho čı́sel se o tom nemůžeme přesvědčit ručně tı́m, že „vypı́šeme hodnoty“ length pro každý
seznam (je jich nekonečně mnoho). Opět si ale poradı́me tak, že představı́me vhodný dokazovacı́ princip
a správnost definice pomocı́ něj prokážeme. Předtı́m si ale uvedeme ještě rekurzivnı́ definici append2.

V přı́padě append2můžeme uvažovat takto:

„Pokud spojı́me prázdný seznam s libovolným seznamem k, pak je výsledkem spojenı́ seznam k.
Pokud je l neprázdný seznam, můžeme uvažovat seznam j = cdr(l). Pokud spojı́me j a k, vznikne
nám seznam, který obsahuje všechny prvky z l kromě prvnı́ho, následované prvky ze seznamu k.
Pokud k tomuto seznamu (to jest, k seznamu rovnajı́cı́mu se spojenı́ j a k) připojı́me na začátek
prvnı́ prvek z l (pomocı́ cons), pak jsme zı́skali spojenı́ l a k. Schématicky je úvaha zobrazena na
obrázku 8.1.“

Ačkoliv je předchozı́ úvaha možná o něco málo složitějšı́, vede na následujı́cı́ rekurzivnı́ definici:

append2(l, k) =

{
k pokud l je prázdný seznam,
cons(car(l), append2(cdr(l), k)) jinak.

Předchozı́ definice řı́ká právě to, že (i) spojenı́m prázdného seznamu s druhým seznamem zı́skáme právě
druhý seznam (neutralita prázdného seznamu vůči spojenı́ seznamů); (ii) druhý bod definice append2 řı́ká,
že v přı́padě, kdy je prvnı́ seznam neprázdný, stačı́ spojit prvnı́ seznam bez prvnı́ho prvku s druhým
seznamem a k tomuto výsledku připojit na začátek prvnı́ prvek prvnı́ho seznamu. Všimněte si, že v de-
finici funkčnı́ hodnoty append2(l, k) jde rekurze pouze přes l. To jest append2(l, k) je vyjádřena pomocı́
append2(cdr(l), k), seznam k se neměnı́. Rekurze přes k (druhý ze spojovaných seznamů) by nám při řešenı́
této konkrétnı́ úlohy k ničemu nebyla. Na tomto přı́kladu je již možná trochu vidět, že definice rekurzı́
vyžaduje určitý vhled do problému a cvik. Uved’me si nynı́ přı́klad vyjádřenı́ spojenı́ dvou seznamů (a b)

a (1 2 3) pomocı́ výše definovaného append2:

append2((a b), (1 2 3)) = cons(car((a b)), append2(cdr((a b)), (1 2 3))) =
= cons(a, append2((b), (1 2 3))) =
= cons(a, cons(car((b)), append2(cdr((b)), (1 2 3)))) =
= cons(a, cons(b, append2((), (1 2 3)))) = cons(a, cons(b, (1 2 3))) =
= cons(a, (b 1 2 3)) = (a b 1 2 3).

Nynı́ představı́me princip indukce, který je použitelný pro dokazovánı́ vlastnostı́ zobrazenı́ definovaných
rekurzı́ přes seznamy. Nynı́ již nemůžeme použı́t klasickou matematickou indukci, protože ta „jde přes
čı́sla“. V přı́padě seznamů budeme použı́vat indukci, který jde přes jejich „strukturu“, proto jı́ budeme
řı́kat strukturálnı́ indukce. Analogicky rekurzi přes seznamy budeme řı́kat strukturálnı́ rekurze.

Věta 8.6 (princip strukturálnı́ indukce přes seznamy). K tomu abychom ověřili, že vlastnost P platı́ pro každý
seznam l ∈ L(), stačı́ prokázat platnost následujı́cı́ch dvou bodů:

(i) platı́ P (()), to jest P platı́ pro prázdný seznam,

(ii) pokud platı́ P (l), pak pro každý element e platı́ P (cons(e, l)).

Důkaz. Tvrzenı́ dokážeme opět sporem. Necht’jsou splněny body (i) a (ii) a zároveň existuje seznam l, který
nemá vlastnost P . Ze všech seznamů, které nemajı́ P vybereme seznam s minimálnı́ délkou a označı́me
jej l0. Upozorněme na to, že obecně může existovat vı́ce seznamů se stejnou minimálnı́ délkou, které nemajı́
vlastnost P . Vybraný seznam l0 tedy splňuje vlastnost, že každý (ostře) kratšı́ seznam má vlastnost P .
Zcela evidentně l0 musı́ být neprázdný seznam, jinak by byl porušen bod (i). Jelikož je l0 neprázdný, je
to seznam, který je výsledkem cons(e, l′), pro nějaký element e a seznam l′. Seznam l′ je o jeden element
kratšı́ než seznam l0, má tedy (ostře) menšı́ délku. Tı́m pádem l′ musı́ mı́t vlastnost P . Potom dle bodu (ii)
i l0 = cons(e, l′)musı́ mı́t vlastnost P , což je spor.

Přı́klad 8.7. Zobrazenı́ length : L() → N0 je jednoznačně definované a jeho hodnoty jsou délky seznamů.
To jest pro každý seznam l je length(l) rovno délce seznamu l, tak jak jsme ji doposud chápali. Toto tvrzenı́
můžeme dokázat principem strukturálnı́ indukce z věty 8.6. Vskutku, uvažujme vlastnost P (l): „Délka
seznamu l je rovna length(l).“ Pro prázdný seznam () máme dle prvnı́ho bodu definice length(l) = 0, tedy
délka prázdného seznamu je nula. To jest, prázdný seznam má vlastnost P , bod (i) věty 8.6 je pro P splněn.
Předpokládejme, že l má vlastnost P . To znamená, že length(l) je délka seznamu l. Pro to, abychom ověřili
(ii) musı́me pro každý element e ukázat, že seznam cons(e, l) má vlastnost P . Jelikož je seznam cons(e, l)
neprázdný, z druhého bodu definice length a ze zřejmého faktu cdr(cons(e, l)) = l dostáváme:

length(cons(e, l)) = 1 + length(cdr(cons(e, l))) = 1 + length(l).

To jest length(cons(e, l)) je rovno délce seznamu l zvětšené o 1. Jelikož je cons(e, l) seznam vzniklý z l

197



připojenı́m elementu e na začátek l, dostáváme, že length(cons(e, l)) je délkou seznamu cons(e, l), což
znamená, že cons(e, l)má vlastnost P . Tı́m jsme dokončili důkaz bodu (ii) pro P . Z věty 8.6 tedy okamžitě
dostáváme důkaz správnosti definice length.

Přı́klad 8.8. Správnost definice append2můžeme prokázat podobně jako jsme prokázali správnost definice
length. Nejprve si ale musı́me uvědomit, přes který seznam budeme indukci provádět, protože append2 je
zobrazenı́ typu append2 : L() × L() → L(). Pokud se podı́váme na definici, pak vidı́me, že append2(l, k) je
v netriviálnı́m přı́padě vyjádřen pomocı́ konstrukce obsahujı́cı́ append2(cdr(l), k). Se strukturou druhého
seznamu (to jest seznamu k) se v definici nijak neoperuje. To nám napovı́dá, že indukci bychom měli vést
přes seznam l. Uvažujme tedy vlastnost P (l): „Pro každý seznam k platı́: spojenı́ seznamů l a k (v tomto
pořadı́) je rovno append2(l, k).“ Prokážeme, že každý seznam l má vlastnost P . Pokud je l prázdný, pak
zřejmě append2(l, k) = k, takže bod (i) věty 8.6 pro vlastnost P je triviálně splněn. Předpokládejme, že l má
vlastnost P . To znamená, že pro každý seznam k platı́, že append2(l, k) je seznam vzniklý spojenı́m l a k.
Nynı́ prokážeme, že každý seznam cons(e, l), kde e je libovolný element, má vlastnost P . Seznam cons(e, l)
je neprázdný, tedy dle druhého bodu definice append2 a s využitı́m

car(cons(e, l)) = e,
cdr(cons(e, l)) = l,

můžeme vyjádřit

append2(cons(e, l), k) = cons(car(cons(e, l)), append2(cdr(cons(e, l)), k)) = cons(e, append2(l, k)).

Předchozı́ rovnost řı́ká, že spojenı́ seznamu cons(e, l) se seznamem k je rovno připojenı́ elementu e na
začátek seznamu append2(l, k). Dle indukčnı́ho předpokladu je append2(l, k) výsledek spojenı́ seznamu l
a k. To jest append2(cons(e, l), k) je rovno výsledku připojenı́ elementu e na začátek spojenı́ seznamů l a k.
Jinými slovy, append2(cons(e, l), k) je spojenı́ seznamu cons(e, l) se seznamem k. Máme hotov důkaz bodu
(ii) pro vlastnost P . Z věty 8.6 dostáváme, že výše uvedená definice append2 je korektnı́ definice spojenı́
dvou seznamů.

Poznámka 8.9. Na předchozı́ch důkazech je zajı́mavé, že jsme prokázali vlastnosti nově definovaných
zobrazenı́ pracujı́cı́ch se seznamy, aniž bychom se zabývali tı́m, jak jsou seznamy reprezentovány (jak
vypadajı́ prvky množiny L()). Vše co nám stačilo, byl fakt, že seznam je bud’ prázdný, nebo jej lze chápat
jako funkčnı́ hodnotu cons(e, l). Z pohledu strukturálnı́ indukce je tedy stěžejnı́ právě tato vlastnost, nikoliv
to, zda-li chápeme seznamu jako „elementy jazyka konstruované z párů“ (viz lekci 4 a lekci 5) nebo třeba
nějak úplně jinak. Se seznamy jsme v této sekci pracovali jako s prvky množiny L(), které jsme vyjadřovali
pomocı́ funkčnı́ch hodnot zobrazenı́ cons, car a cdr.

Meznı́ podmı́nky v rekurzivnı́ch definicı́ch length a append2 opět nelze vynechat, protože takové definici
by nebyly kompletnı́. Všimněte si, že analogicky jako v přı́padě rekurze a indukce přes čı́sla, jde princip
strukturálnı́ rekurze směrem „zpět“, protože funkčnı́ hodnoty definovaných funkcı́ jsou vyjádřeny
pomocı́ funkčnı́ch hodnot pro strukturálně jednoduššı́ seznamy. Naopak, princip indukce postupuje od
strukturálně nejjednoduššı́ho seznamu – prázdného seznamu, směrem „dopředu“, to jest ke složitějšı́m
seznamům.

Úkolem této sekce bylo představit principy rekurze a indukce přes čı́sla a přes seznamy. Ukázali jsme, že
rekurzı́ lze definovat důležitá zobrazenı́. S programovánı́m to souvisı́ tak, že analogický princip, jako jsme
použili při definovánı́ zobrazenı́, můžeme použı́t při vytvářenı́ procedur, které tato zobrazenı́ reprezentujı́.
Takovým procedurám budeme řı́kal rekurzivnı́ procedury a blı́že se jim budeme věnovat v dalšı́ch sek-
cı́ch. Princip indukce je pro nás důležitým dokazovacı́m principem, kterým můžeme dokazovat vlastnosti
rekurzivně definovaných zobrazenı́ (a procedur, které je reprezentujı́).

Abychom ještě na závěr sekce demonstrovali sı́lu definic rekurzı́, ukážeme rekurzivnı́ definici zobrazenı́
korespondujı́cı́m s procedurou foldr představenou v předchozı́ lekci. Vzpomeňme, že pomocı́ foldr jsme
byli schopni vytvořit řadu procedur počı́naje length, append2, přes filtračnı́ proceduru filter a tak dále.
Doposud jsme ale neřekli, zda-li je možné proceduru foldr v jazyky Scheme uživatelsky definovat, nebo
jestli musı́ být přı́tomna v jazyku jako primitivnı́ procedura. Odpověd’je, že procedura je definovatelná plně

198



prostředky jazyka, které již máme k dispozici. Definicı́ procedury se ted’ zabývat nebudeme, tu ukážeme
v dalšı́ch částech textu, ale ukážeme rekurzivnı́ definici zobrazenı́ foldr, které je reprezentované procedurou
foldr. Zobrazenı́ foldr lze chápat jako zobrazenı́

foldr : F × E × L() → E ,

kde F je množina všech zobrazenı́ f : E × E → E . Nynı́ můžeme definovat:

foldr(f, t, l) =

{
t pokud l je prázdný seznam,
f(car(l), foldr(f, t, cdr(l))) jinak.

Strukturálnı́ indukcı́ se můžete přesvědčit, že definice je jednoznačná, a že pro seznam l délky n bude funkčnı́
hodnota foldr(f, t, l) zı́skána pomocı́ „postupného zabalenı́“ funkcı́ f tak, jak jsme popsali v předchozı́ lekci.
Snadno potom můžeme vidět, že length a append2můžeme definovat pomocı́ foldr jako

length(l) = foldr(g, 0, l), kde g(x, y) = 1 + y,
append2(l, k) = foldr(cons, k, l).

Všimněte si korespondence předchozı́ch zavedenı́ length a append2 s definicemi procedurlength aappend2
v programech 7.1 a 7.2 na stranách 174 a 175. Tyto ukázky by nám měly dát jakousi představu o tom, že
strukturálnı́ rekurze je skutečně silným nástrojem (který je potřeba umět správně použı́vat).

Rekurze a indukce nejsou jen nějaké „matematické kuriozity v programovánı́ “, jedná se o široce vyu-
žı́vané principy bez nichž by byla naše schopnost řešit problémy výrazně snı́žena. Některé typy úloh
lze bez rekurze řešit jen velmi obtı́žně. Ve funkcionálnı́ch jazycı́ch je tradičně rekurze vedle procedur
vyššı́ch řádů jednou z nejpoužı́vanějšı́ch metod vytvářenı́ procedur (a v důsledku výpočetnı́ch procesů).
Ve funkcionálnı́ch jazycı́ch rekurze de facto nahrazuje cykly, které se použı́vajı́ hlavně v procedurálnı́ch
jazycı́ch. Rekurze je však na rozdı́l od prostých cyklů mnohem mocnějšı́, jak záhy uvidı́me.

V této sekci jsme ukázali řadu důkazů správnosti definic a vlastnostı́ definovaných zobrazenı́. Budeme
v tom pokračovat v omezené mı́ře i v dalšı́ch sekcı́ch. V praxi zpravidla nenı́ potřeba dokazovat správnost
tı́mto detailnı́m způsobem, protože zkušený programátor má již některé typické konstrukce tak řı́kajı́c
„v oku“ a jejich správnost je schopen velmi rychle „vidět“. Zdůrazněme však, že tento nadhled přicházı́
až s určitou programátorskou zkušenostı́, jejı́ž nabytı́ chvı́li trvá. Ani potom bychom však formálnı́ aparát,
který jsme představili v této sekci, neměli považovat za „cosi zbytečného“, ale spı́š za užitečnou pomůcku.

8.2 Rekurze a indukce přes přirozená čı́sla

V této sekci se budeme zabývat rekurzı́ přes přirozená čı́sla (ke kterým z technických důvodů přidáváme
i nulu). Ukážeme, jak souvisı́ rekurzivnı́ definice zobrazenı́, které jsme představili v předchozı́ sekci,
s procedurami, které tato zobrazenı́ reprezentujı́. Jako prvnı́ přı́klad budeme uvažovat rekurzivnı́ definici
n-té mocniny čı́sla. Pro jednoduchost budeme uvažovat pouze přı́pady, kdy n nabývá celočı́selné nezáporné
hodnoty. V tomto přı́padě můžeme pro libovolné x ∈ R definovat xn následujı́cı́m předpisem:

xn =

{
1 pokud n = 0,
x · xn−1 jinak.

Předchozı́ předpis řı́ká, že x0 = 1 a pokud je n ≥ 1, pak je xn = x·xn−1. Principem prezentovaným ve větě 8.2
bychom opět mohli snadno dokázat platnost vlastnostı́ „pro n je hodnota xn jednoznačně definovaná“
a navı́c je zřejmé, že se jedná skutečně o n-tou mocninu x. Upozorněme na fakt, že předchozı́ rekurzivnı́
definice definuje pro x = 0 hodnotu x0 = 1, což je v rozporu s matematickým chápánı́m nutné mocniny
(z matematického pohledu nenı́ hodnota 00 definovaná). Z praktického (programátorského) pohledu je
však vhodné zavést 00 jako 1.

Proceduru expt počı́tajı́cı́ hodnoty xn podle výše uvedeného rekurzivnı́ho předpisu bychom v jazyku
Scheme mohli naprogramovat tak, jak je to uvedeno v programu 8.1. Procedura expt v programu 8.1 je
formalizacı́ předchozı́ho rekurzivnı́ho předpisu v jazyku Scheme. V těle procedury je pomocı́ speciálnı́
formy if vyjádřen podmı́něný výraz: „Pokud je n rovno nule, pak je výsledek umocněnı́ jedna. V opačném
přı́padě je výsledek umocněnı́ roven součinu hodnoty x s hodnotou x umocněnou na n− 1.“

199



Program 8.1. Rekurzivnı́ procedura počı́tajı́cı́ xn.

(define expt

(lambda (x n)

(if (= n 0)

1

(* x (expt x (- n 1))))))

Definice 8.10 (rekurzivnı́ procedura, rekurzivnı́ aplikace). Proceduře budeme řı́kat rekurzivnı́ procedura,
pokud při vyhodnocenı́ jejı́ho těla docházı́ (v některých přı́padech) k aplikaci sebe sama. Aplikaci „sebe
sama“ budeme dále nazývat rekurzivnı́ aplikace procedury. �

Procedura expt je tedy prvnı́m přı́kladem rekurzivnı́ procedury, protože v poslednı́m argumentu předaném
speciálnı́ formě if, který je vyhodnocen v přı́padě, že podmı́nka (prvnı́ argument předaný if) bude
nepravdivá, dojde k aplikaci expt. Jak již ale bylo řečeno, rekurzivnı́ procedury jsou procedury v klasickém
smyslu tak, jak jsme je popsali v úvodnı́ch lekcı́ch tohoto textu (nenı́ tedy na nich nic „speciálnı́ho“).

Poznámka 8.11. Otázkou, kterou bychom se měli zabývat je, proč vlastně aplikace rekurzivnı́ch procedur
„funguje“. Jinými slovy, je z pohledu vyhodnocovacı́ho procesu skutečně v pořádku, že procedura aplikuje
sebe sama? Podı́váme-li se na kód procedury expt v programu 8.1, pak je zřejmé, že procedura vzniklá
vyhodnocenı́m λ-výrazu vznikla v globálnı́m prostředı́. V tomto prostředı́ je i provedena vazba této procedury
na symbol expt. Při aplikaci procedury expt je vytvořeno lokálnı́ prostředı́, jehož předkem je prostředı́
vzniku procedury, tedy globálnı́ prostředı́. V lokálnı́m prostředı́ jsou při aplikaci procedury definovány
vazby symbolů x a n. Pokud při vyhodnocovánı́ těla během aplikaci dojde k vyhodnocenı́ symbolu expt,
pak tento symbol nenı́ nalezen v lokálnı́m prostředı́. Hledánı́m vazby se proto postupuje v nadřazeném
prostředı́, což je globálnı́ prostředı́ – v něm má symbol expt vazbu a tou je právě aplikovaná procedura.
Procedura vzniklá vyhodnocenı́m λ-výrazu v programu 8.1 má tedy skutečně k dispozici „sebe sama“
prostřednictvı́m vazby symbolu expt.

Na rekurzivnı́ proceduře expt si můžeme všimnout dvou částı́:

limitnı́ podmı́nka rekurze je podmı́nka, po jejı́mž splněnı́ je vyhodnocen výraz jež nezpůsobı́ dalšı́ aplikaci
samotné rekurzivnı́ procedury. Na limitnı́ podmı́nku rekurze se lze dı́vat jako na podmı́nku vymezujı́cı́
triviálnı́ přı́pady aplikace procedury, v jejichž přı́padě nenı́ potřeba pro stanovenı́ výsledku aplikace
provádět rekurzivnı́ aplikaci. V přı́padě procedury expt je limitnı́ podmı́nkou rekurze fakt, že n je
rovno nule, v tomto přı́pade je výsledek vyhodnocenı́ čı́slo 1, což koresponduje s faktem x0 = 1. Každá
rekurzivnı́ procedura by měla mı́t alespoň jednu limitnı́ podmı́nku, obecně jich může mı́t i vı́c.

předpis rekurze je část těla procedury, při jejı́mž vyhodnocenı́ docházı́ k rekurzivnı́ aplikaci procedury. Sou-
částı́ předpisu rekurze je vyjádřenı́, jak bude stanoven výsledek aplikace rekurzivnı́ procedury s jistými
argumenty pomocı́ rekurzivnı́ aplikace této procedury s jinými argumenty (obvykle s argumenty, které
jsou ve smyslu konkrétnı́ho použitı́ procedury „jednoduššı́ “). V přı́padě procedury expt je rekurzivnı́
předpis (* x (expt x (- n 1))), což koresponduje s faktem, že xn = x · xn−1 pro n ≥ 1 Rekurziv-
nı́ch předpisů může být v proceduře opět vı́c, ale z principu by měl být aspoň jeden, abychom se mohli
„o rekurzi vůbec bavit“.

Limitnı́ podmı́nku i předpis rekurze lze snadno vyčı́st přı́mo z rekurzivnı́ definice xn tak, jak jsme ji uvedli na
začátku sekce. Při vytvářenı́ rekurzivnı́ch procedur je vždy potřeba si limitnı́ podmı́nky a předpisy rekurze
jasně uvědomit a správně formalizovat (v programu). Absence některé z důležitých částı́ v rekurzivnı́
proceduře, třeba absence limitnı́ podmı́nky, obvykle vede na nekončı́cı́ sériı́ aplikacı́ nebo na vznik chyby.

Nynı́ si uvedeme přı́klad aplikace procedury expt s čı́selnými argumenty 8 a 4 (v tomto pořadı́). Aplikace
procedury s těmito argumenty je znázorněna na obrázku 8.2 (vlevo). Nejprve si všimněte metody, kterou
jsme při znázorněnı́ zvolili. Každou novou aplikaci procedury jsme vyjádřili tak, že do těla výrazu jsme

Obrázek 8.2. Schématické zachycenı́ aplikace procedury expt.

(expt 8 4) vyvolánı́ 1. aplikace procedury
(* 8 (expt 8 3)) navı́jenı́ : vyvolánı́ 2. aplikace
(* 8 (* 8 (expt 8 2))) navı́jenı́ : vyvolánı́ 3. aplikace
(* 8 (* 8 (* 8 (expt 8 1)))) navı́jenı́ : vyvolánı́ 4. aplikace
(* 8 (* 8 (* 8 (* 8 (expt 8 0))))) navı́jenı́ : vyvolánı́ 5. aplikace
(* 8 (* 8 (* 8 (* 8 1)))) dosaženı́ limitnı́ podmı́nky v průběhu 5. aplikace
(* 8 (* 8 (* 8 8))) stav po odvinutı́ 4. aplikace
(* 8 (* 8 64)) stav po odvinutı́ 3. aplikace
(* 8 512) stav po odvinutı́ 2. aplikace
4096 výsledná hodnota vzniklá odvinutı́m 1. aplikace

mı́sto seznamu, jehož vyhodnocenı́ aplikaci způsobilo, zapsali tělo aplikované procedury, v němž jsme zaměnili
formálnı́ argumenty za hodnoty předaných argumentů. V prvnı́m kroku jsme tedy výraz (expt 8 4) nahradili
výrazem (* 8 (expt 8 3)), v dalšı́m kroku jsme výraz (expt 8 3) v (* 8 (expt 8 3)) nahradili
výrazem (* 8 (expt 8 2)) čı́mž vznikl výraz (* 8 (* 8 (expt 8 2))) a tak dále. Jak je z obrázku
patrné, při výpočtu je v prvnı́ fázı́ prováděna rekurzivnı́ aplikace, protože pro hodnoty 4, . . . , 1 navázané
na symbolu n se neuplatňuje limitnı́ podmı́nka rekurze. Této fázi výpočtu, při které docházı́ k postupné
aplikaci rekurzivnı́ho předpisu, se řı́ká „navı́jenı́“. Navı́jenı́ je ukončeno dosaženı́m limitnı́ podmı́nky, viz
obrázek 8.2 (vpravo). V našem přı́padě to je když je na n navázaná hodnota 0. Od tohoto okamžiku se již
procedura expt neaplikuje, ale docházı́ ke „zpětnému dosazovánı́“ vypočtených hodnot a dokončovánı́
jednotlivých aplikacı́ expt, které byly zahájeny během fáze navı́jenı́. Této druhé fázi řı́káme přı́značně
fáze „odvı́jenı́“. Výsledná hodnota vzniká odvinutı́m prvnı́ aplikace expt (prvnı́ aplikace je pochopitelně
odvinuta jako poslednı́, protože se ve fázi odvı́jenı́ pohybujeme zpětně).

Při aplikaci expt s argumenty 8 a 4 došlo de facto k pěti aplikacı́m této procedury (všechny proběhly ve
fázı́ navı́jenı́). Při těchto aplikacı́ch vzniklo tı́m pádem pět prostředı́. Otázkou by možná mohlo být, jestli
to nenı́ nějaký „nadbytečný luxus“. V žádném přı́padě tomu tak nenı́. Během fáze navı́jenı́ je, neformálně
řečeno, budována série odložených výpočtů. Výsledek (expt 8 4) je definován jako součin čı́sla 8 s hodnotou
vzniknou aplikacı́ (expt 8 3). Tento součin však nemůže být dokončen dřı́v, než je dokončena aplikace
expt s argumenty 8 a 3. Do té doby musı́ být někde uloženy hodnoty, které je potřeba mı́t k dispozici
pro dokončenı́ výpočtu, až bude výsledek vyhodnocenı́ (expt 8 3) znám. Tı́mto úložištěm jsou právě
prostředı́. Na obrázku 8.3 jsou zachycena prostředı́ vzniklá při vyhodnocenı́ (expt 8 4), předkem všech
těchto prostředı́ je globálnı́ prostředı́, v němž je na symbol exptnavázána rekurzivně aplikovaná procedura.

Jak jsme již na přı́kladu vysvětlili, průběh výpočetnı́ho procesu generovaného aplikacı́ rekurzivnı́ procedury
můžeme shrnout do dvou důležitých fázı́:

fáze navı́jenı́ je fáze, ve které docházı́ k postupné rekurzivnı́ aplikaci. V průběhu této fáze jsou vytvářena
nová prostředı́ v nichž jsou uloženy informace o vazbách formálnı́ch argumentů rekurzivně aplikované
procedury. Tato prostředı́ v sobě udržujı́ informaci o pomyslném odloženém výpočtu (anglicky deferred
computation). Fáze navı́jenı́ končı́ dosaženı́m limitnı́ podmı́nky rekurze, v jejı́mž přı́padě je vrácena
hodnota vzniklá bez dalšı́ch rekurzivnı́ch aplikacı́ procedury.

fáze odvı́jenı́ Nastává po dosaženı́ limitnı́ podmı́nky rekurze. Během této fáze docházı́ k dokončenı́ vy-
hodnocenı́ těla procedury v prostředı́ch, která vznikla v předchozı́ fázi navı́jenı́. Postupuje se přitom
zpětně, jako prvnı́ je dokončeno vyhodnocenı́ těla v prostředı́ poslednı́ aplikace procedury, následuje
vyhodnocenı́ těla v prostředı́ předposlednı́ aplikace procedury a tak se postupuje až k vyhodnocenı́
těla v prostředı́ prvnı́ aplikace procedury a výsledná hodnota je výsledkem původnı́ aplikace. Během
fáze odvı́jenı́ může v některých přı́padech znovu nastat fáze navı́jenı́ (uvidı́me až na dalšı́ch přı́kladech).

Obrázek 8.3. Prostředı́ vzniklá během vyhodnocenı́ (expt 8 4)

Rekurzivnı́ procedury jsou procedury, které ve svém těle vyvolávajı́ aplikaci sebe sama. Každá rekur-
zivnı́ procedura má svou limitnı́ podmı́nku a předpis rekurze. Aplikaci rekurzivnı́ procedury si můžeme
představit jako proces probı́hajı́cı́ ve dvou základnı́ch fázı́ch, které se mohou vzájemně střı́dat: fáze na-
vı́jenı́ a fáze odvı́jenı́. Na rozdı́l od jazyka Scheme, některé funkcionálnı́ jazyky disponujı́ tak zvaným
lı́ným vyhodnocovánı́m. Jde o princip vyhodnocovánı́ výrazů, který mimo jiné umožňuje definovat rekur-
zivnı́ procedury bez limitnı́ podmı́nky, při jejichž aplikaci výpočetnı́ proces neuvı́zne v sérii nekončı́cı́ch
aplikacı́ (právě dı́ky lı́nému vyhodnocovánı́). Jazyk Scheme nenı́ lı́ný, a každá rekurzivnı́ procedura by
tedy limitnı́ podmı́nku mı́t měla (pokud nechceme záměrně výpočetnı́ proces „vytuhnout“). V dalšı́
části učebnı́ho textu však ukážeme, že lı́né vyhodnocovánı́ můžeme ve Scheme implementovat, takže
předchozı́ tvrzenı́ tak úplně neplatı́.

V tuto chvı́li by nám mělo být jasné, jak vypadajı́ rekurzivnı́ procedury, jak je psát, a také „proč vlastně
fungujı́“. Nynı́ si ukážeme několik dalšı́ch rekurzivnı́ch procedur pracujı́cı́ch s čı́sly. Jako prvnı́ se zamyslı́me
nad zefektivněnı́m stávajı́cı́ procedury expt. Procedura expt uvedená v programu 8.1 je naprogramována
tak, že při své aplikaci redukuje problém nalezenı́ xn na problém nalezenı́ xn−1. Je tedy jasné, že pro
výpočet xn vznikne n+ 1 prostředı́, protože limitnı́ podmı́nka je dána pro n = 0. Časová složitost výpočtu
xn (touto procedurou) je tedy O(n), hodnota x nehraje (teoreticky) roli15. Vzhledem k tomu, že při výpočtu
vznikne n+1 nových prostředı́, můžeme řı́ct, že prostorová složitost výpočtu xn (touto procedurou) je také
O(n) (počet prostředı́ roste lineárně s n). Otázkou je, zda-li bychom proceduru nemohli naprogramovat
efektivněji (z hlediska časového, prostorového nebo z hlediska obou složitostı́).

Při navrhovánı́ rekurzivnı́ch procedur se často uplatňuje princip, který je v informatice označován jako
divide et impera neboli „rozděl a panuj“. Tento princip spočı́vá v tom, že řešenı́ problému je rozloženo na
řešenı́ (obecně několika) podproblémů stejného typu, ale výrazně menšı́ složitosti. Rozloženı́ problému
na menšı́ podproblémy se v programátorské terminologii nazývá dekompozice. Až jsou tyto podproblémy
vyřešeny, je z jejich řešenı́ (v rozumném čase) složeno řešenı́ výchozı́ho problému. Tento způsob nahlı́ženı́
na řešenı́ problémů vede většinou na vytvářenı́ rekurzivnı́ch procedur. Přı́klady použitı́ principu divide et
impera uvidı́me v dalšı́ch sekcı́ch. Už při výpočtu xn si ale ukážeme, že při dekompozici problému můžeme
postupovat výrazně efektivněji než v programu 8.1.

15Samozřejmě, že prakticky hraje roli i hodnota x, protože mocněnı́ velkých celých čı́sel v jejich přesné reprezentaci bude jistě
náročnějšı́ než mocněnı́ malých čı́sel. Tento technický rys však můžeme nynı́ při stanovovánı́ rámcové složitosti přehlédnout,
protože při nı́ jde o to stanovit složitost vzhledem k počtu aplikacı́ násobenı́. Samotnou délku násobenı́ chápeme zjednodušeně
jako konstantnı́.

Výpočet xn bychom mohli urychlit tı́m, že si uvědomı́me, že x2n = xn · xn = (xn)2. To jest, pokud je expo-
nent sudé čı́slo, můžeme problém výpočtu mocniny redukovat na problém výpočtu mocniny s polovičnı́m
exponentem. Intuitivně lze asi vycı́tit, že to je výrazný posun oproti pouhému zmenšenı́ exponentu o jedna,
jak tomu bylo doposud. Hodnotu xn (pro n nezáporné celé čı́slo) bychom tedy mohli definovat takto:

xn =


1 pokud n = 0,
(x

n
2 )2 pokud je n sudé,

x · xn−1 pokud je n liché.

Výše uvedený předpis povede na rekurzivnı́ proceduru s jednou limitnı́ podmı́nkou a se dvěma separátnı́mi
předpisy rekurze – jeden pro přı́pad, kdy je exponent sudý a jeden pro přı́pad, kdy je exponent lichý. Před
uvedenı́m samotné rekurzivnı́ procedury si však dokažme, že naše úprava algoritmu je korektnı́ a vede ke
správným výsledkům. Na základě původnı́ rekurzivnı́ definice xn můžeme indukcı́ prokázat následujı́cı́
tvrzenı́.

Věta 8.12. Pro libovolné x ∈ R a nezáporná celá čı́sla m,n platı́, že xm+n = xm · xn.

Důkaz. Principem indukce prokážeme platnost vlastnosti P (i): „Pro každá dvě nezáporná celá čı́sla m,n
taková, že m + n = i, platı́ xm+n = xm · xn“. Pro i = 0 máme pouze m = 0 a n = 0. Navı́c zřejmě v tomto
přı́padě xm+n = x0+0 = x0 = 1 = 1 · 1 = x0 · x0 = xm · xn, takže bod (i) platı́. Nynı́ prokážeme platnost
bodu (ii). Předpokládejme, že P platı́ pro i. Máme ukázat, že P platı́ pro i+ 1, to jest máme ukázat, že pro
každá dvě nezáporná celá čı́sla m,n taková, že m+n = i+1, platı́ xm+n = xm ·xn. Z rekurzivnı́ definice xn

dostáváme, že xm+n = x · xm+n−1. Jelikož m+ n = i+ 1, pak musı́ být aspoň jedno z čı́sel m a n přirozené.
Předpokládejme, že je to n (přı́pad pro m by se odvodil duálně). Tı́m pádem n− 1 je celé nezáporné čı́slo.
Máme tedy dvě čı́sla, m a n − 1, která jsou celá a nezáporná a platı́ pro ně m + n − 1 = i. Nynı́ můžeme
aplikovat indukčnı́ předpoklad na m a n − 1, to jest dostáváme xm+n−1 = xm · xn−1. S pomocı́ poslednı́
rovnosti tedy vyjádřı́me xm+n takto:

xm+n = x · xm+n−1 = x · xm · xn−1 = xm · x · xn−1 = xm · xn.

Poslednı́ rovnost opět plyne z rekurzivnı́ definice xn. Dohromady jsme tedy prokázali platnost bodu (ii)
pro vlastnost P . Užitı́m principu indukce z věty 8.2 jsme prokázali větu 8.12.

Nynı́ je zřejmé, že i upravená rekurzivnı́ definice xn je korektnı́, protože dvě věty 8.12 máme

(x
n
2 )2 = x

n
2 · x

n
2 = x

n
2+

n
2 = xn

a mohli bychom použı́t tento fakt při důkazu indukcı́. Při prokázánı́ bychom však museli u bodu (ii)
věty 8.2 rozlišit vı́c přı́padů (pro n sudé a n liché). Důkaz ponecháme na laskavém čtenáři. Procedura,
která počı́tá hodnotu xn podle výše uvedeného upraveného vztahu je zobrazena v programu 8.2. Jelikož

Program 8.2. Rychlá rekurzivnı́ procedura počı́tajı́cı́ xn.

(define na2

(lambda (x) (* x x)))

(define expt

(lambda (x n)

(cond ((= n 0) 1)

((even? n) (na2 (expt x (/ n 2))))

(else (* x (expt x (- n 1)))))))

jsme v programu potřebovali odlišit dva různé předpisy rekurze, vyplatilo se nám použı́t speciálnı́ formu
cond mı́sto dvou vnořených speciálnı́m forem if. Jak jsme na tom se složitostı́ nové verze expt? Nejprve
si uved’me přı́klad jejı́ aplikaci s argumenty 2 a 25. Aplikace je schématicky zachycena na obrázku 8.4.
Z přı́kladu je vidět, že vypočtenı́ hodnoty mocniny pro exponent rovný 25 bylo potřeba provést pouze osm
rekurzivnı́ch aplikacı́ procedury expt. Při použitı́ původnı́ verze by jich bylo potřeba rovných dvacet šest.

Obrázek 8.4. Schématické zachycenı́ aplikace rychlé procedury expt.

(expt 2 25) vyvolánı́ 1. aplikace procedury
(* 2 (expt 2 24)) navı́jenı́ : vyvolánı́ 2. aplikace
(* 2 (na2 (expt 2 12))) navı́jenı́ : vyvolánı́ 3. aplikace
(* 2 (na2 (na2 (expt 2 6)))) navı́jenı́ : vyvolánı́ 4. aplikace
(* 2 (na2 (na2 (na2 (expt 2 3))))) navı́jenı́ : vyvolánı́ 5. aplikace
(* 2 (na2 (na2 (na2 (* 2 (expt 2 2)))))) navı́jenı́ : vyvolánı́ 6. aplikace
(* 2 (na2 (na2 (na2 (* 2 (na2 (expt 2 1))))))) navı́jenı́ : vyvolánı́ 7. aplikace
(* 2 (na2 (na2 (na2 (* 2 (na2 (* 2 (expt 2 0)))))))) navı́jenı́ : vyvolánı́ 8. aplikace
(* 2 (na2 (na2 (na2 (* 2 (na2 (* 2 1))))))) dosaženı́ limitnı́ podmı́nky
(* 2 (na2 (na2 (na2 (* 2 (na2 2)))))) stav po odvinutı́ 7. aplikace
(* 2 (na2 (na2 (na2 (* 2 4))))) stav po odvinutı́ 6. aplikace
(* 2 (na2 (na2 (na2 8)))) stav po odvinutı́ 5. aplikace
(* 2 (na2 (na2 64))) stav po odvinutı́ 4. aplikace
(* 2 (na2 4096)) stav po odvinutı́ 3. aplikace
(* 2 16777216) stav po odvinutı́ 2. aplikace
33554432 výsledná hodnota vzniklá odvinutı́m 1. aplikace

Jak můžeme stanovit složitost nové verze procedury expt? Zamysleme se nynı́ nad ideálnı́m přı́padem,
kdy při výpočtu xn je použit pouze rekurzivnı́ předpis pracujı́cı́ se sudým exponentem až na jediný přı́pad
a to pro n = 1. V tomto přı́padě je při prvnı́, druhé, třetı́, . . . aplikaci vypočtena hodnota x0, x1, x2, x4, x8,
x16, x32, . . . Obecně řečeno, při k-té aplikaci (k ≥ 2) je vypočtena hodnota x2

k−2
. Odpověd’ na otázku, při

kolikáté aplikaci je vypočtena hodnota xn tedy vede na řešenı́ rovnice:

xn = x2
k−2

,

jelikož je v našem přı́padě x parametr (zde se dopouštı́me mı́rného zjednodušenı́), stačı́ vyřešit

n = 2k−2,

což je po zlogaritmovánı́ rovno

log n = (k − 2) · log 2,
z čehož vyjádřı́me k následovně:

k = 2 +
log n

log 2
= 2 + log2 n.

Počet kroků po výpočet xn je tedy v přı́padě jediného použitı́ rekurzivnı́ho předpisu pracujı́cı́ho s lichým ex-
ponentem 2+log2 n. Jak se počet kroků promı́tne v přı́padě, kdy dojde k užitı́ vı́ce jak jednoho rekurzivnı́ho
předpisu pro lichý exponent? Uvědomme si nejprve, kolik nejvýš může těchto „patologických situacı́ “ na-
stat. Při aplikaci rekurzivnı́ho předpisu pro lichý exponent je výpočet xn redukován na výpočet xn−1. V tom
přı́padě je ale n− 1 sudé čı́slo, takže nikdy nemohou nastat dvě aplikace tohoto rekurzivnı́ho předpisu po
sobě. To znamená, že v nejhoršı́m přı́padě bude počet kroků nutných k vypočtenı́ xn dvojnásobkem našeho
odhadu 2 + log2 n (aplikace obou rekurzivnı́ch předpisů pro lichý a sudý exponent se budou vzájemně
střı́dat). Dostáváme tı́m tedy, že počet prostředı́, která vznikajı́ během aplikace nové verze procedury expt

roste logaritmicky (při základu 2) vzhledem k n. Snadno již můžeme vidět, že časová i prostorová složitost
jsou shodně O(log2 n), což je výrazně lepšı́ výsledek než původnı́ O(n), který se projevı́ zvlášt’pro velká n.

Při programovánı́ „rychlé verze expt“ bychom si však měli dát pozor na efektivnı́ formalizaci rekurzivnı́ho
vztahu. Kdybychom mı́sto kódu v programu 8.2 použili následujı́cı́ kód

(define expt

(lambda (x n)

(cond ((= n 0) 1)

((even? n) (* (expt x (/ n 2)) (expt x (/ n 2))))

(else (* x (expt x (- n 1))))))),

pak by v rekurzivnı́m předpisu pro sudé n docházelo k výpočtu hodnoty x
n
2 dvakrát. Zbytečně by tedy

docházelo k rekurzivnı́ aplikaci. Ve skutečnosti bychom tedy proceduru nijak neurychlili, ale drasticky
bychom jı́ zpomalili. V programu 8.2 byla hodnota x

n
2 počı́tána pouze jednou a jejı́ druhá mocnina byla

vypočı́tána procedurou na2. V předchozı́m kódu však docházı́ v předpisu rekurze pro sudou větev ke
dvěma rekurzivnı́m aplikacı́m. To jest s každou aplikacı́ původnı́ rychlé procedury se počet aplikacı́ v nové
verzi zdvojnásobı́. Mohli bychom tedy provést hrubý odhad počtu aplikacı́ pro výpočet xn v přı́padě, že
je použit rekurzivnı́ předpis pro lichá n pouze jednou, jako 21+log2 n (hodnota 1 v exponentu reprezentuje
jediné použitı́ rekurzivnı́ho předpisu pro n liché, použitı́ pro n = 0 jsme pro jednoduchost ignorovali), což
je většı́ hodnota něž 2log2 n = n. Procedura provádějı́cı́ dvě rekurzivnı́ aplikace ve svém těle by tedy ve
skutečnosti byla dokonce pomalejšı́ než výchozı́ procedura expt v programu 8.1 (se zvyšujı́cı́m se n by
bylo zpomalenı́ stále vı́ce cı́tit, napřı́klad pro n = 4096 je potřeba pro výpočet celkem 12288 kroků – v této
hodnotě jsou započı́tány i všechny aplikace pro n = 0).

Předchozı́ problém bychom bez použitı́ na2 mohli vyřešit třeba tak, že hodnotu x
n
2 vypočteme pouze

jednou, navážeme ji v lokálnı́m prostředı́ na nějaký symbol a poté ji použijeme dvakrát při výpočtu součinu.
Technicky se ale vlastně jedná o stejné řešenı́ jako při použitı́ uživatelsky definované procedury na2 (v obou
přı́padech vzniká nové prostředı́). Upravený kód by tedy vypadal následovně:

(define expt

(lambda (x n)

(cond ((= n 0) 1)

((even? n) (let ((result (expt x (/ n 2))))

(* result result)))

(else (* x (expt x (- n 1)))))))

Z hlediska řádové složitosti bychom program výrazně neurychlili, kdybychom přidali dalšı́ dvě limitnı́
podmı́nky řešı́cı́ přı́pady mocněnı́ jedničkou a dvojkou:

(define expt

(lambda (x n)

(cond ((= n 0) 1)

((= n 1) x)

((= n 2) (na2 x))

((even? n) (na2 (expt x (/ n 2))))

(else (* x (expt x (- n 1)))))))

Předchozı́ kód by z technického hlediska možná vedl i ke zpomalenı́ výpočtu, protože při každé rekurzivnı́
aplikaci procedury je testováno většı́ množstvı́ podmı́nek.

Někoho by možná mohlo napadnout udělat dalšı́ urychlenı́ algoritmu na bázı́ vyřešenı́ přı́padu pro expo-
nent ve tvaru 3n. Dle věty 8.12 totiž vı́me, že x3n = xn+n+n = xn ·xn ·xn. Nabı́zelo by se tedy naprogramovat
proceduru na3 počı́tajı́cı́ třetı́ mocninu a upravit program 8.2 následovně:

(define expt

(lambda (x n)

(cond ((= n 0) 1)

((= n 1) x)

((= n 2) (na2 x))

((= (modulo n 3) 0) (na3 (expt x (/ n 3))))

((even? n) (na2 (expt x (/ n 2))))

(else (* x (expt x (- n 1)))))))

Přidaná větev řı́ká, že pokud je exponent dělitelný třemi beze zbytku, pak redukujeme problém výpočtu
hodnoty xn na problém výpočtu hodnoty (x

n
3 )3. Analogicky bychom mohli postupovat pro dalšı́ přirozená

čı́sla 4, 5, . . . Z hlediska efektivity jak teoretické tak praktické to však již téměř nic nepřinášı́. V odborných
kruzı́ch panuje názor, že „vylepšovánı́“ programů tı́mto způsobem „nestojı́ za to.“ Uved’me si důvod proč.
Předně, po zařazenı́ nové větve do programu by počet kroků nutných k výpočtu xn nerostl logaritmicky

při základu dvě, ale logaritmicky při základu tři (toto je hodně optimistický odhad, což nám ale nynı́ nevadı́,
protože chceme prokázat, že složitost se výrazně nezlepšı́). Co to ale znamená? Z vlastnostı́ logaritmů vı́me:

log3 n =
log2 n

log2 3
=

1
log2 3

· log2 n ≈ 0.6309 log2 n.

Což jinými slovy znamená, že z hlediska asymptotické složitosti platı́ O(log3 n) = O(log2 n), protože log3 n
se od log2 n lišı́ pouze o multiplikativnı́ konstantu (přibližně) 0.6309. Z tohoto důvodu běžně značı́me třı́du
složitosti bez uvedenı́ základu logaritmu O(log n), protože tato třı́da je ekvivalentnı́ všem O(logk n), kde
k ∈ (1,∞). Řádově má předchozı́ vylepšenı́ programu stejnou složitost jako původnı́ verze v programu 8.2.
Ani z čistě praktického hlediska nenı́ vhodné takové rozšı́řenı́ programu dělat, protože čas, který je ušetřený
přidanou větvı́ je z velké části zkonzumován testovánı́m většı́ho počtu složitějšı́ch podmı́nek.

Na závěr této sekce si uved’me procedury fac a fib pro výpočet faktoriálu a Fibonacciho čı́sel naprogra-
mované pomocı́ rekurzivnı́ch vztahů uvedených v sekci 8.1 na začátku této lekce. Procedury jsou napsány
v programech 8.3 a 8.4. Na prvnı́ pohled je vidět, že rekurzivnı́ procedury fac a fib jsou výrazně čitelnějšı́

Program 8.3. Rekurzivnı́ výpočet faktoriálu.

(define fac

(lambda (n)

(if (<= n 1)

1

(* n (fac (- n 1))))))

Program 8.4. Rekurzivnı́ výpočet prvků Fibonacciho posloupnosti.

(define fib

(lambda (n)

(if (<= n 1)

n

(+ (fib (- n 1))

(fib (- n 2))))))

než procedury uvedené v programech 7.12 a 7.13 vytvořených pomocı́ foldr. Z hlediska výpočtové slo-
žitosti by někoho z nás možná mohlo překvapit, že zatı́mco procedura fib z programu 7.13 na straně 187
vracı́ výsledky „okamžitě“ i pro většı́ vstupnı́ hodnoty (třeba pro n = 1000), rekurzivnı́ procedura 8.4 při
hodnotě n = 1000 „nemá odezvu“. Tento jev bychom si nynı́ neměli ukvapeně vykládat, jako že rekurzivnı́
procedury „nejsou efektivnı́ “ (velmi efektivnı́ proceduru jsme konec konců viděli už v programu 8.2 na
straně 203), spı́š bychom si jej měli vyložit jako důležitou motivaci pro studium vlastnostı́ výpočetnı́ch
procesů generovaných rekurzivnı́mi procedurami. Touto problematikou se budeme věnovat v dalšı́ sekci.

8.3 Výpočetnı́ procesy generované rekurzivnı́mi procedurami

V této sekci se budeme zabývat výpočetnı́mi procesy generovanými rekurzivnı́mi procedurami. Z kva-
litativnı́ho hlediska mohou totiž rekurzivnı́ procedury generovat výpočetnı́ procesy s různou výpočetnı́
náročnostı́. Vše si budeme postupně demonstrovat na přı́kladech. Začneme přı́klady s rekurzivnı́mi verzemi
procedur fac a fib, které jsme ukázali v programech 8.3 a 8.4.

Uvažujme nynı́ rekurzivnı́ verzi procedury fac na výpočet faktoriálu. Pokud tuto proceduru aplikujeme
s argumentem 5, pak bude mı́t výpočetnı́ proces generovaný touto rekurzivnı́ procedurou průběh, který
je zakreslený v obrázku 8.5. Výpočetnı́ proces probı́há analogicky jako u původnı́ verze procedury expt,

Obrázek 8.5. Schématické zachycenı́ aplikace rekurzivnı́ verze fac.

(fac 5) navı́jenı́ : vyvolánı́ 1. aplikace
(* 5 (fac 4)) navı́jenı́ : vyvolánı́ 2. aplikace
(* 5 (* 4 (fac 3))) navı́jenı́ : vyvolánı́ 3. aplikace
(* 5 (* 4 (* 3 (fac 2)))) navı́jenı́ : vyvolánı́ 4. aplikace
(* 5 (* 4 (* 3 (* 2 (fac 1))))) navı́jenı́ : vyvolánı́ 5. aplikace
(* 5 (* 4 (* 3 (* 2 1)))) dosaženı́ limitnı́ podmı́nky
(* 5 (* 4 (* 3 2))) stav po odvinutı́ 4. aplikace
(* 5 (* 4 6)) stav po odvinutı́ 3. aplikace
(* 5 24) stav po odvinutı́ 2. aplikace
120 výsledná hodnota vzniklá odvinutı́m 1. aplikace

kterou jsme představili v programu 8.1, srovnejte s obrázkem 8.2. Rekurzivnı́ procedura pro výpočet
faktoriálu má zřejmě lineárnı́ časovou složitost, protože n! je vypočı́tána v n krocı́ch. Prostorová složitost je
rovněž lineárnı́, protože s každou novou aplikacı́ této procedury vznikne nové prostředı́. Počet vzniklých
prostředı́ během aplikace procedury je tedy lineárně závislý na n a žádné dalšı́ prostorové nároky procedura
nemá. Časová i prostorová složitost výpočetnı́ho procesu generovaného rekurzivnı́ verzı́ fac jsou tedy ve
třı́dě O(n).

Předchozı́ řešenı́ výpočtu faktoriálu je založeno na vyjádřenı́ n! pomocı́ n · (n − 1)!. Hodnotu faktoriálu
bychom ale mohli vyjádřit i jiným rekurzivnı́m předpisem. Konkrétně můžeme rekurzivně zavést zobrazenı́
f : N0 ×N → N tak, že f(n, k) bude mı́t hodnotu faktoriálu n násobenou hodnotou k. Na prvnı́ pohled
se toto zobrazenı́ může jevit jako „nějaká komplikace“ výchozı́ definice faktoriálu, ale jak dál uvidı́me, pro
výpočet faktoriálu bude mnohem efektivnějšı́. Uvažujme tedy předpis definujı́cı́ toto zobrazenı́:

f(n, k) =

{
k pokud n ≤ 1,
f(n− 1, k · n) jinak.

Předpis je opět rekurzivnı́ a opět bychom o něm mohli prokázat, že je jednoznačný (bude plynout z násle-
dujı́cı́ho tvrzenı́, takže to nynı́ prokazovat nebudeme). Na rozdı́l od všech ostatnı́ch předpisů je rekurzivnı́
předpis f zajı́mavý tı́m, že na druhém řádku je hodnota f(n, k) vyjádřena pouze pomocı́ funkčnı́ hodnoty
f(· · ·) (bez prováděnı́ dalšı́ch operacı́, které by byly zapsány „vně výrazu f(· · ·)“). To výrazně zjednodušuje
možnost funkčnı́ hodnotu takto definovaného zobrazenı́ počı́tat. Uved’me si nynı́ několik funkčnı́ch hodnot
f(n, 1) pro n = 0, 1, 2, . . . :

f(0, 1) = 1,
f(1, 1) = 1,
f(2, 1) = f(1, 2) = 2,
f(3, 1) = f(2, 3) = f(1, 6) = 6,
f(4, 1) = f(3, 4) = f(2, 12) = f(1, 24) = 24,
f(5, 1) = f(4, 5) = f(3, 20) = f(2, 60) = f(1, 120) = 120,

...

Z předchozı́ch přı́kladů by měla být jasná intuice skrytá za definicı́ f . Prvnı́ argument sloužı́ jako jakýsi
„čı́tač“, který jde postupně směrem dolů až dojde k jedničce. Přitom se v druhém argumentu postupně
„akumulujı́ “ hodnoty násobků všech hodnot čı́tače. Tedy násobků k ·n ·(n−1) ·(n−2) · · · 2 = n!, akumulace
přitom začı́ná hodnotou k. V předchozı́ch ukázkách tedy k = 1. Toto neformálnı́ tvrzenı́ nynı́ zpřesnı́me
následujı́cı́ větou.

Věta 8.13. Pro každé n ∈ N0, k ∈ N, c ∈ R a f : N0 ×N→ N zavedené předchozı́m rekurzivnı́m předpisem platı́:

(i) c · f(n, k) = f(n, c · k),
(ii) f(n, k) = k · n!,

(iii) f(n, 1) = n!.

Důkaz. (i): Tvrzenı́ prokážeme indukcı́. Z pohledu věty 8.2 nám vlastně jde o prokázánı́ vlastnosti P (n): „pro
každé k ∈ N a c ∈ R platı́ c · f(n, k) = f(n, c · k)“. Pro n ≤ 1máme dle definici c · f(n, k) = c · k = f(n, c · k),
tedy tvrzenı́ platı́. Předpokládejme, že n má vlastnost P . Prokážeme, že i n + 1 má tuto vlastnost. Dle
definičnı́ho vztahu máme c · f(n + 1, k) = c · f((n + 1) − 1, k · (n + 1)) = c · f(n, k · (n + 1)). S užitı́m
indukčnı́ho předpokladu a asociativity násobenı́ dostáváme

c · f(n+ 1, k) = c · f(n, k · (n+ 1)) = f(n, c · (k · (n+ 1))) = f(n, (c · k) · (n+ 1)) = f(n+ 1, c · k),
to jest vlastnost P platı́ pro každé n. Tı́m jsme prokázali bod (i).

(ii): Tvrzenı́ prokážeme indukcı́ a s využitı́m předchozı́ho bodu. Z pohledu věty 8.2 jde ted’ o prokázánı́
vlastnosti P (n): „pro každé k ∈ N platı́ f(n, k) = k ·n!“. Pro n ≤ 1 tvrzenı́ evidentně platı́ pro každé k. Necht’
tvrzenı́ platı́ pro n. Prokážeme jeho platnost pro n+ 1. S využitı́m indukčnı́ho předpokladu, rekurzivnı́ho
vztahu pro n! a bodu (i) tedy máme

k · (n+ 1)! = k · ((n+ 1) · n!) = (k · (n+ 1)) · n! = f(n, k · (n+ 1)) = f(n+ 1, k).

To jest P platı́ pro každé n, tı́m je prokázán bod (ii).

(iii) je speciálnı́m přı́padem (ii) pro k = 1.

Rekurzivnı́ přepis pro f nám umožňuje naprogramovat novou verzi procedury fak. Tato procedura je
napsána v programu 8.5. Přı́sně vzato, s rekurzivně definovaným zobrazenı́m f koresponduje pouze

Program 8.5. Iterativnı́ procedura pro výpočet faktoriálu.

(define fac-iter

(lambda (i accum)

(if (<= i 1)

accum

(fac-iter (- i 1) (* accum i)))))

(define fac

(lambda (n)

(fac-iter n 1)))

pomocná procedura fac-iter. Procedura fac pouze provede aplikaci fac-iter s druhým argumentem
nastaveným na 1. Proceduru fac jsme zavedli proto, že jejı́ uživatelé by se k výpočtu faktoriálu měli stavět
jako k černé skřı́ňce. I když pro jeho výpočet potřebujeme definovat proceduru dvou argumentů, tento fakt
by pro programátory použı́vajı́cı́ pouze fac měl být skryt, což se nám tı́mto rozdělenı́m na dvě procedury
(fac a pomocnou fac-iter) podařilo. V obrázku 8.6 máme zachycen průběh výpočtu nové verze fac pro
hodnotu 5. Na tomto přı́kladu si můžeme všimnout několika zajı́mavých věcı́. Nejprve konstatujme, že

Obrázek 8.6. Schématické zachycenı́ aplikace iterativnı́ verze fac.

(fac 5)

(fac-iter 5 1) navı́jenı́ : vyvolánı́ 1. aplikace
(fac-iter 4 5) navı́jenı́ : vyvolánı́ 2. aplikace
(fac-iter 3 20) navı́jenı́ : vyvolánı́ 3. aplikace
(fac-iter 2 60) navı́jenı́ : vyvolánı́ 4. aplikace
(fac-iter 1 120) navı́jenı́ : vyvolánı́ 5. aplikace
120 dosaženı́ limitnı́ podmı́nky a vrácenı́ hodnoty

je zřejmě vidět, že časovou složitost výpočtu jsme oproti předcházejı́cı́ verzi nijak nezlepšili, pro výpočet
faktoriálu je i nadále potřeba provést n součinů, což vede na časovou složitost v třı́dě O(n). V přı́kladu je ale

zajı́mavé, že fáze odvı́jenı́ prakticky chybı́. Lépe řečeno, fáze odvı́jenı́ je degenerovaná na pouhé postupné vracenı́
téže hodnoty (výsledku). To je důsledkem faktu, že rekurzivnı́ předpis v proceduře fac-iter obsahuje pouze
sebe sama s novými hodnotami. Tato aplikace nenı́ použita jako součást žádného odloženého výpočtu.
Při dosaženı́ limitnı́ podmı́nky je tedy pouze postupně vracena výsledná hodnota, se kterou se již nijak
nemanipuluje.

Otázkou je, zda-li bychom předchozı́ pozorovánı́ o trivialitě fáze odvı́jenı́ dokázali využı́t při efektivnějšı́
aplikaci procedury. Vı́me již, že časovou stránku jsme nevylepšili. Otázkou je, jestli bychom mohli vylepšit
prostorovou náročnost výpočetnı́ho procesu. I v programu 8.6 totiž docházı́ pro vstupnı́ hodnotu n k n apli-
kacı́m fac-iter. Podle principu aplikace uživatelsky definovaných procedur by tedy mělo vzniknout n
nových prostředı́ a prostorová složitost výpočetnı́ho procesu by byla opět O(n). Připomeňme, že v rekur-
zivnı́ verzi fac sloužila prostředı́ k uchovánı́ informacı́ o odloženém výpočtu. Jelikož fac-iter odložený
výpočet nevytvářı́, nabı́zı́ se možnost provádět jejı́ aplikace neustále v jednom prostředı́, pouze s měnı́cı́mi se
vazbami symbolů. Tı́m bychom prostorovou složitost výpočetnı́ho procesu srazili dolů na O(1) (procedura
by pro libovolně velké vstupnı́ n pracovala v konstantnı́m prostoru – což je z programátorského hlediska
„ideálnı́ složitost“). Optimalizaci aplikacı́ některých rekurzivnı́ch procedur v tomto smyslu (nevytvářenı́
nových prostředı́ při každé aplikaci, ale pouze přepisovánı́ vazeb a vyhodnocovánı́ v jednom prostředı́)
nazýváme optimalizace na koncovou rekurzi (anglicky tail recursion optimization, často zkracováno TRO).

Interprety jazyka Scheme dělajı́ optimalizaci na koncovou rekurzi automaticky ve všech situacı́ch, kdy je to
možné. Abychom si tuto optimalizaci blı́že popsali, zavedeme si několik nových pojmů. Jednı́m z hlavnı́ch
pojmů je tak zvaná koncová pozice (speciálnı́ pozice výrazu v těle procedury), ze které lze provést rekurzivnı́
aplikaci procedury bez nutnosti vytvářet nové prostředı́.

Definice 8.14 (koncová pozice, koncová aplikace, koncově rekurzivnı́ procedura). Množina koncových pozic
λ-výrazu Λ je definována následovně:

(i) poslednı́ výraz v těle výrazu Λ je v koncové pozici výrazu Λ;

(ii) je-li (if 〈test〉 〈důsledek〉 〈náhradnı́k〉) v koncové pozici výrazu Λ,
pak 〈důsledek〉 i 〈náhradnı́k〉 (pokud je přı́tomen) jsou v koncové pozici výrazu Λ;

(iii) je-li
(cond (〈test1〉 〈důsledek1〉)

(〈test2〉 〈důsledek2〉)
...

(〈testn〉 〈důsledekn〉)
(else 〈náhradnı́k〉))

v koncové pozici výrazu Λ, pak 〈důsledek1〉, . . . , 〈důsledekn〉 jsou v koncové pozici výrazy Λ a pokud je
přı́tomna i else-větev, pak je i 〈náhradnı́k〉 v koncové pozici Λ;

(iv) je-li (and · · · 〈výraz〉) v koncové pozici výrazu Λ, pak je 〈výraz〉 v koncové pozici výrazu Λ;
totéž platı́ pro speciálnı́ formu or;

(v) je-li (let (· · ·) · · · 〈výraz〉) v koncové pozici výrazu Λ, pak je 〈výraz〉 v koncové pozici výrazu Λ;
totéž platı́ pro všechny ostatnı́ varianty speciálnı́ formy let.

Koncová aplikace procedury vzniklé vyhodnocenı́m λ-výrazuΛ je aplikace vyvolaná z koncové pozice výrazuΛ.
Rekurzivnı́ procedura se nazývá koncově rekurzivnı́, pokud aplikuje sebe sama pouze z koncových pozic (λ-
-výrazu jehož vyhodnocenı́m vznikla). �

Ve standardu jazyka Scheme R5RS a také v jeho IEEE standardu se koncové pozice definujı́ ještě o něco
složitějšı́m způsobem (částečně proto, že jsme ještě nepředstavili úplně všechny speciálnı́ formy jazyka
Scheme). My si ale prozatı́m vystačı́me s předchozı́m chápánı́m.

Přı́klad 8.15. Vrat’me se nynı́ k proceduře fac-iter z programy 8.5. Označme Λ výraz, jehož vyhodno-
cenı́m vzniká procedura, která je potom navázána na symbol fac-iter. Jelikož je v těle výrazu Λ pouze
jediný výraz, je v koncové pozici. Tento výraz je navı́c ve tvaru (if · · · , podle (ii) předchozı́ definice
tedy dostáváme, že i výrazy accum a (fac-iter (- i 1) (* accum i)) jsou v koncových pozicı́ch Λ.

Aplikace iniciovaná vyhodnocenı́m výrazu (fac-iter (- i 1) (* accum i)) je tedy koncová aplikace
procedury fac-iter, protože byla vyvolána z koncové pozice. Jelikož v těle procedury fac-iter již na ji-
ném mı́stě aplikace fac-iter nenı́ vyvolána, procedura fac-iter je tedy koncově rekurzivnı́. Kdybychom
si vzali napřı́klad proceduru fac z programu 8.3 na straně 206, tak tato procedura aplikuje sebe sama
také na jediném mı́stě, ale toto mı́sto nenı́ koncová pozice. V koncové pozici se totiž nacházı́ celý výraz
(* n (fac (- n 1))), nikoliv pouze (fac (- n 1)). Procedura fac z programu 8.3 tedy nenı́ koncově
rekurzivnı́.

Koncově rekurzivnı́ procedury lze ve skutečnosti rozpoznat velmi jednoduše. Rekurzivnı́ procedura
je koncově rekurzivnı́, pokud výsledkem každé jejı́ aplikace pouze nová aplikace sebe sama, která nenı́
součástı́ žádného složeného výrazu (až na konstrukce if, cond a podobně, viz definici 8.14). Tı́m, že
koncové aplikace nejsou již v daném prostředı́ použity k žádnému výpočtu, může se pro provedenı́
aplikace použı́t právě aktuálnı́ prostředı́ a nenı́ potřeba vytvářet prostředı́ nové.

Aplikaci uživatelsky definovaných procedur můžeme nynı́ rozšı́řit o speciálnı́ přı́pad aplikace procedury
z jejı́ koncové pozice. Postup již nebudeme popisovat formálně tak, jak jsme to udělali třeba v definici 2.12
na straně 49, ale vysvětlı́me jej pouze slovně. Pokud je při aplikaci procedury (to jest během vyhodnocenı́
jejı́ho těla) zaznamenán pokus o opětovnou aplikaci z koncové pozice, pak nenı́ vytvářeno nové prostředı́,
pouze se předefinujı́ vazby symbolů v aktuálnı́m prostředı́ tak, aby odpovı́daly hodnotám při nové aplikaci
a v aktuálnı́m prostředı́ (s upravenými vazbami symbolů) je opět vyhodnoceno tělo procedury.

Jelikož je procedura fac-iter z programu 8.5 koncově rekurzivnı́, pak se při výpočtu faktoriálu (pro
libovolně velké n) vytvořı́ pouze dvě prostředı́ – jedno při aplikaci fac z programu 8.5 a druhé při následné
aplikaci fac-iter. V tomto druhém prostředı́ pak již ale probı́há vyhodnocenı́ všech aplikacı́ fac-iter.
Jelikož se v těle fac-iterprovádějı́ operace s konstantnı́ prostorovou složitostı́, pak je zřejmé, že prostorová
složitost tohoto výpočetnı́ho procesu je O(1).

Ačkoliv procedury fac z programů 8.3 a 8.5 můžeme chápat jako reprezentace stejného zobrazenı́, je mezi
nimi z hlediska výpočetnı́ch procesů, které generujı́, výrazný rozdı́l. Tento rozdı́l se promı́tá předevšı́m
to prostorové složitosti obou procedur. Na dalšı́ch přı́kladech uvidı́me řadu výpočetnı́ch procesů, které se
chovajı́ jako výpočetnı́ procesy v přı́padě procedur z obou programů. Proto se vedle rekurzivnı́ch procedur
budeme také bavit rekurzivnı́ch výpočetnı́ch procesech, které tyto procedury generujı́ a budeme rozlišovat
jejich různé typy.

Obecně řečeno, výpočetnı́ proces budeme nazývat rekurzivnı́ výpočetnı́ proces, pokud se bude jednat o proces
generovaný rekurzivnı́ procedurou nebo několika procedurami, které se vzájemně aplikujı́, u kterého lze rozlišit fáze
navı́jenı́ a odvı́jenı́ (přitom fáze odvı́jenı́ může být triviálnı́).

Přı́klad 8.16. V předchozı́ formulaci rekurzivnı́ho výpočetnı́ho procesu je možná nejasné, co je myšleno
„několika vzájemně se aplikujı́cı́mi procedurami“. Napřı́klad pokud budeme uvažovat následujı́cı́ dvě
procedury

(define fac-a (lambda (n) (if (<= n 1) 1 (* n (fac-b (- n 1))))))

(define fac-b (lambda (n) (if (<= n 1) 1 (* n (fac-a (- n 1)))))),

pak snadno vidı́me, že obě jsou modifikace rekurzivnı́ verze procedury pro výpočet faktoriálu. Ovšem, v těle
procedury navázané na fac-a je aplikována procedura navázaná na fac-b a obráceně. Přisně vzato tedy ani
fac-a ani fac-b nejsou rekurzivnı́ procedury. Na druhou stranu je ale zřejmé, že při aplikaci libovolné z nich
(pro n většı́ než jedna) bude výpočetnı́ proces pokračovat aplikacı́ druhé procedury, poté opět aplikacı́ prvnı́
procedury až do bodu, kdy bude dosažena limitnı́ podmı́nka. Výpočetnı́ proces tedy má fázi navı́jenı́. Stejně
tak má fázi odvı́jenı́, která začı́ná splněnı́m limitnı́ podmı́nky. V tomto přı́padě tedy máme dvě procedury,
které se navzájem aplikujı́ a generujı́ rekurzivnı́ výpočetnı́ proces i když nejsou rekurzivnı́. Všimněte si, že
obě procedury počı́tajı́ faktoriál.

V této sekci jsme zatı́m poznali dva typy rekurzivnı́ch výpočetnı́ch procesů:

lineárnı́ rekurzivnı́ proces je rekurzivnı́ proces, který má netriviálnı́ fáze navı́jenı́ a odvı́jenı́. Během fáze
navı́jenı́ je budována „série odložených výpočtů“, přitom počet prostředı́ roste lineárně vzhledem
k velikosti vstupnı́ch argumentů. Po dosaženı́ limitnı́ podmı́nky nastává fáze odvı́jenı́, ve které je
zpětně dokončeno vyhodnocovánı́ výrazů, které bylo započato a „odloženo“ ve fázı́ navı́jenı́. Činnost
lineárně rekurzivnı́ho výpočetnı́ho proces nelze jednoduše přerušit ani nenı́ možné „skočit“ na nějaké
mı́sto rekurze16. Procedura fac z programu 8.3 generuje právě lineárně rekurzivnı́ výpočetnı́ proces.

lineárnı́ iterativnı́ proces je výpočetnı́ proces generovaný koncově rekurzivnı́mi procedurami. Nedocházı́
při něm k vytvářenı́ „odložených výpočtů.“ Každý lineárně iterativnı́ výpočetnı́ proces je během své
činnosti jednoznačně určen

(i) vazbami symbolů v prostředı́ P svého běhu,
(ii) předpisem, jak změnit stav vazeb v P na základě aktuálnı́ch vazeb,

(iii) limitnı́ podmı́nkou ukončujı́cı́ iterativnı́ proces.

Lineárně iterativnı́ výpočetnı́ proces může být během svého výpočtu přerušen a posléze opět obnoven
v mı́stě přerušenı́. Fáze průběhu lineárně iterativnı́ procesu je dána vazbami symbolů v prostředı́
jeho běhu, viz bod (i) z předchozı́ho výpisu. Pokud tyto vazby uchováme a lineárně iterativnı́ proces
opustı́me, je možné jej opět aktivovat s počátečnı́mi hodnotami nastavenými na uschované hodnoty.
Tı́m pádem se iterativnı́ proces „rozběhne“ od mı́sta zastavenı́.

Pro někoho možná nynı́ nastal malý terminologický zmatek, protože se bavı́me o rekurzivnı́ch procedu-
rách a o rekurzivnı́ch výpočetnı́ch procesech. Navı́c některé rekurzivnı́ procedury generujı́ iterativnı́ výpočetnı́
procesy. Ve většině programovacı́ch jazyků (zejména procedurálnı́ch jazyků) rekurzivnı́ procedury ne-
mohou generovat iterativnı́ procesy. Na vytvářenı́ iterativnı́ch procesů jsou v těchto jazycı́ch speciálnı́
prostředky – nejčastěji cykly. Z pohledu jazyka Scheme je tedy iterativnı́ výpočetnı́ proces ekvivalentem
cyklů známých z procedurálnı́ch jazyků.

Přı́klad 8.17. Přerušenı́ iterativnı́ho procesu si lze představit následovně. Vezměme si třeba proces schéma-
ticky ukázaný na obrázku 8.6. Kdybychom na 4. řádku proces „uřı́zli“ a zapamatovali si hodnoty 3 a 20,
což byly aktuálnı́ vazby symbolů v prostředı́, ve kterém byla aplikována procedura fac-iter, pak bychom
mohli kdykoliv iterativnı́ proces „spustit“ počı́naje tı́mto krokem prostě tak, že bychom fac-iter apliko-
vali s danými hodnotami. Výsledkem by byla požadovaná hodnota 120. Naprogramujte tuto proceduru
a prakticky se přesvědčte o jejı́m chovánı́.

Poznámka 8.18. (a) Kvůli zjednodušenı́ terminologie budeme rekurzivnı́m procedurám generujı́cı́m pouze
iterativnı́ výpočetnı́ procesy řı́kat iterativnı́ procedury a procedurám generujı́cı́m pouze lineárně rekurzivnı́
procesy lineárně rekurzivnı́ procedury. Pořád je ale potřeba mı́t na paměti rozdı́l mezi pojmy rekurzivnı́
procedura versus rekurzivnı́ výpočetnı́ proces.

(b) Některé argumenty iterativnı́ch procedur hrajı́ speciálnı́ role. Jednı́m typem argumentů jsou tak zvané
čı́tače. Úkolem čı́tačů je nějakým způsobem počı́tat kroky iterace. Podle stavu čı́tače lze v přı́padě potřeby
zastavit iteraci. Napřı́klad v proceduře fac-iter z programu 8.5 představuje formálnı́ argument i čı́tač,
který je na počátku iterace nastaven na hodnotu n a v každém kroku iterace je snı́žen o jedna. Druhým
typem významných argumentů jsou střadače. Účelem střadačů je nějakým způsobem akumulovat hodnoty
během iterace. Ve výše uvedené proceduře fac-iter je accum střadač, který je na počátku iterace nastaven
na hodnotu 1 a v každém kroku je akumulátor násoben aktuálnı́ hodnotou čı́tače – tı́mto způsobem se
postupně akumulujı́ součiny, které ve výsledku dávajı́ hodnotu faktoriálu.

(c) Některé rekurzivnı́ procedury aplikujı́ sebe sama z několika pozic, některé z nich jsou koncové a některé
koncové nejsou. Takové procedury obecně nejsou iterativnı́, protože při aplikaci z nekoncových pozic je
potřeba vytvořit nové prostředı́. Na druhou stranu, i u těchto procedur jsou koncové aplikace prováděny
v témže prostředı́. Takto se chová třeba následujı́cı́ procedura

16V dalšı́m dı́le toho učebnı́ho textu uvidı́me, že to ve skutečnosti možné je pomocı́ tak zvaných únikových funkcı́ a aktuálnı́ho
pokračovánı́. Jelikož se však jedná o pokročilé speciálnı́ konstrukty, věnujeme se jim až v dalšı́m dı́le textu. Zatı́m bychom se na
„výskok z rekurze“ měli dı́vat jako na něco, co je běžnými programátorskými prostředky neřešitelné.

(define proc

(lambda (x)

(cond ((<= x 0) 0)

((even? x) (+ x (proc (/ x 2))))

(else (proc (- x 1)))))),

která koncově aplikuje sebe sama v přı́padě, že na x je navázané liché čı́slo (viz tělo procedury).Pokud je na
x navázané sudé čı́slo různé od nuly, pak je provedena aplikace z nekoncové pozice a při nı́ je vytvořeno
nové prostředı́.

Na závěr rozboru rekurzivnı́ a iterativnı́ verze procedur pro výpočet faktoriálu dodejme, že z hlediska
programátorské čistoty by bylo vhodné nadefinovat pomocnou proceduru fac-iter interně v proceduře
fac, protože fac-iter byla vytvořena za účelem, že ji bude aplikovat pouze fac. Iterativnı́ procedura
fac s interně definovanou pomocnou procedurou jsou uvedeny v programu 8.6. V programu 8.7 máme

Program 8.6. Iterativnı́ procedura pro výpočet faktoriálu s internı́ definicı́.

(define fac

(lambda (n)

(define iter

(lambda (i accum)

(if (<= i 1)

accum

(iter (- i 1) (* accum i)))))

(iter n 1)))

uvedenou iterativnı́ verzi procedury expt (procedura má opět pomocnou proceduru expt-iter, prová-
dějı́cı́ samotnou iteraci) pracujı́cı́ s časovou složitostı́ O(log n) a prostorovou složitostı́ O(1). Na proceduře

Program 8.7. Iterativnı́ procedura pro výpočet mocniny.

(define expt-iter

(lambda (x n accum)

(cond ((= n 0) accum)

((even? n) (expt-iter (* x x) (/ n 2) accum))

(else (expt-iter x (- n 1) (* accum x))))))

(define expt

(lambda (x n)

(expt-iter x n 1)))

si všimněte toho, že jsme museli mı́rně upravit rekurzivnı́ předpis tak, aby bylo možné jej vyjádřit pomocı́
koncové rekurze. Mı́sto faktu x2n = (xn)2, který jsme využili v programu 8.2 na straně 203, jsme nynı́ využili
vztahu x2n = (x2)n, to jest během iterativnı́ho výpočtu průběžně měnı́me kromě exponentu také základ.
Přesvědčte se sami, že tato úvaha je správná a vede k řešenı́. Na obrázku 8.7 je pro demonstraci uveden
průběh výpočtu 225. Zde si můžeme povšimnout průběžné změny základu (prvnı́ argument), exponentu
(druhý argument) a pomocného střadače (třetı́ argument), jehož hodnota je nakonec vrácena jako výsledek.

Přirozenou otázkou je, zda-li lze vždy rekurzivnı́ výpočetnı́ procesy nahradit iterativnı́mi výpočet-
nı́mi procesy. V terminologii procedur, které tyto procesy generujı́, se tedy ptáme, zda-li pro každou

Obrázek 8.7. Schématické zachycenı́ iterativnı́ verze procedury expt.

(expt 2 25)

(expt-iter 2 25 1) navı́jenı́ : vyvolánı́ 1. aplikace
(expt-iter 2 24 2) navı́jenı́ : vyvolánı́ 2. aplikace
(expt-iter 4 12 2) navı́jenı́ : vyvolánı́ 3. aplikace
(expt-iter 16 6 2) navı́jenı́ : vyvolánı́ 4. aplikace
(expt-iter 256 3 2) navı́jenı́ : vyvolánı́ 5. aplikace
(expt-iter 256 2 512) navı́jenı́ : vyvolánı́ 6. aplikace
(expt-iter 65536 1 512) navı́jenı́ : vyvolánı́ 7. aplikace
(expt-iter 65536 0 33554432) navı́jenı́ : vyvolánı́ 8. aplikace
33554432 dosaženı́ limitnı́ podmı́nky a vrácenı́ hodnoty

proceduru vedoucı́ na rekurzivnı́ proces můžeme napsat proceduru vedoucı́ na iterativnı́ proces. Od-
pověd’ je kladná, ale v některých přı́padech to může být dost těžké a výsledek mnohdy „nestojı́ za
to“ – výsledná procedura generujı́cı́ iterativnı́ proces je výrazně méně čitelná než výchozı́ procedura.
Dı́ky čemu je možné vždy najı́t proceduru vedoucı́ na iterativnı́ proces? Je to tı́m, že rekurzivnı́ aplikaci
a konstrukci odložených výpočtů můžeme plně simulovat pomocı́ zásobnı́ku, který lze v přı́padě jazyka
Scheme udržovat pomocı́ seznamu odložených hodnot čekajı́cı́ch na dalšı́ zpracovánı́.

V programu 8.8 je uvedena iterativnı́ verze procedury expt, která provádı́ simulace fáze navı́jenı́ a odvı́jenı́
původnı́ rekurzivnı́ procedury z programu 8.1 pomocı́ zásobnı́ku. Procedura expt z programu 8.8 má

Program 8.8. Iterativnı́ procedura pro výpočet mocniny s pomocı́ zásobnı́ku.

(define expt

(lambda (x n)

(define expt-stack

(lambda (n accum stack)

(if (= n 0)

(cond ((null? stack) accum)

((car stack) (expt-stack 0 (na2 accum) (cdr stack)))

(else (expt-stack 0 (* accum x) (cdr stack))))

(if (even? n)

(expt-stack (/ n 2) accum (cons #t stack))

(expt-stack (- n 1) accum (cons #f stack))))))

(expt-stack n 1 '())))

v sobě interně definovanou pomocnou proceduru expt-stack třı́ argumentů. Prvnı́m z nich je exponent,
druhým je střadač, který bude na konci výpočtu obsahovat hodnotu mocniny xn a poslednı́m argumentem
je stack. Základ (navázaný na symbol x v prostředı́ aplikace expt) nenı́ potřeba předávat, protože jeho
hodnota je dostupná pro interně definovanou proceduru expt-stack a hodnota x nebude během výpočtu
měněna. Procedura expt-stack, která provádı́ samotný výpočet, pracuje tak, že simuluje fázi navı́jenı́
a odvı́jenı́ na zásobnı́ku. V if-výrazu v těle procedury jsou obě fáze rozlišeny limitnı́ podmı́nkou (= n 0).
Pokud limitnı́ podmı́nky nenı́ dosaženo, je zmenšován exponent (bud’o jedna nebo na polovinu podle toho
jestli je lichý nebo sudý) a na zásobnı́k se pokládajı́ hodnoty #f (přı́znak pro násobenı́ základem v dalšı́ fázı́
výpočtu) a #t (přı́znak pro umocněnı́ na druhou v dalšı́ fázı́ výpočtu). Po dosaženı́ limitnı́ podmı́nky začne
být zpracováván zásobnı́k, což je de facto simulace fáze odvı́jenı́. Pokud je zásobnı́k vyprázdněn, je vrácena
hodnota akumulátoru. Pokud je prvnı́m prvkem na zásobnı́ku #t (přı́znak pro provedenı́ umocněnı́),

provede se dalšı́ iterace s umocněnou hodnotou akumulátoru, v opačném přı́padě je akumulátor vynásoben
základem. Průběh výpočtu (pouze procedura expt-stack) je zobrazen na obrázku 8.8. Jak je z přı́kladu

Obrázek 8.8. Schématické zachycenı́ aplikace expt vytvořené s využitı́m zásobnı́ku.

(expt-stack 25 1 ()) simulace navı́jenı́ : vyvolánı́ 1. aplikace
(expt-stack 24 1 (#f)) simulace navı́jenı́ : vyvolánı́ 2. aplikace
(expt-stack 12 1 (#t #f)) simulace navı́jenı́ : vyvolánı́ 3. aplikace
(expt-stack 6 1 (#t #t #f)) simulace navı́jenı́ : vyvolánı́ 4. aplikace
(expt-stack 3 1 (#t #t #t #f)) simulace navı́jenı́ : vyvolánı́ 5. aplikace
(expt-stack 2 1 (#f #t #t #t #f)) simulace navı́jenı́ : vyvolánı́ 6. aplikace
(expt-stack 1 1 (#t #f #t #t #t #f)) simulace navı́jenı́ : vyvolánı́ 7. aplikace
(expt-stack 0 1 (#f #t #f #t #t #t #f)) ukončenı́ simulace navı́jenı́
(expt-stack 0 2 (#t #f #t #t #t #f)) simulace stavu po odvinutı́ 8. aplikace
(expt-stack 0 4 (#f #t #t #t #f)) simulace stavu po odvinutı́ 7. aplikace
(expt-stack 0 8 (#t #t #t #f)) simulace stavu po odvinutı́ 6. aplikace
(expt-stack 0 64 (#t #t #f)) simulace stavu po odvinutı́ 5. aplikace
(expt-stack 0 4096 (#t #f)) simulace stavu po odvinutı́ 4. aplikace
(expt-stack 0 16777216 (#f)) simulace stavu po odvinutı́ 3. aplikace
(expt-stack 0 33554432 ()) simulace stavu po odvinutı́ 2. aplikace
33554432 výsledná hodnota

a zřejmě i z programu 8.8 patrné, přepisovánı́ rekurzivnı́ch procedur pomocı́ dodatečného zásobnı́ku
v mnoha přı́padech nepřispı́vá k čitelnosti programu, ani výrazně nezlepšuje efektivitu. Iterativnı́ verze
procedury expt z programu 8.8 má časovou složitost O(log n), protože počet kroků je řádově stejný jako
u efektivnı́ rekurzivnı́ varianty z programu 8.2. Prostorová složitost je také O(log n), protože v každém
kroku aplikace je při simulaci navı́jenı́ zvětšen zásobnı́k o jeden prvek. Prostorová složitost je tedy řádově
stejná jako u algoritmu 8.2. Iterativnı́ verze expt pracujı́cı́ se zásobnı́kem tedy nenı́ z pohledu výpočtové
efektivity efektivnějšı́ než lineárně rekurzivnı́ varianta. Iterativnı́ proces generovaný procedurou expt-

iter z programu 8.8 má tedy prostorovou složitost O(log n) (a nikoliv pouze O(1)).

Na programu 8.8 si můžeme uvědomit, že stanovenı́ prostorové složitosti se neodvı́jı́ pouze od počtu aplikacı́
procedur (a tı́m pádem od počtu vzniklých prostředı́), ale musı́me brát v potaz i konstrukci hierarchických
datových struktur jejichž délka je nějakým způsobem závislá na velikosti vstupnı́ch argumentů.

Nynı́ se budeme zabývat třetı́m základnı́ typem rekurzivnı́ch výpočetnı́ch procesů. Doposud jsme se za-
bývali procedurami, které ve svých rekurzivnı́ch předpisech aplikovaly sebe sama právě jednou. Jedinou
výjimkou byla procedura fib z programu 8.4 pro výpočet prvků Fibonacciho posloupnosti pomocı́ jejich
rekurzivnı́ho předpisu. Tuto proceduru jsme „snadno naprogramovali“, ale již na konci sekce 8.2 jsme
konstatovali, že procedura je trestuhodně neefektivnı́. To, jak moc je neefektivnı́, rozebereme nynı́. Zná-
zorněme si nejdřı́v schématicky průběh aplikace této procedury. Jelikož jejı́ rekurzivnı́ předpis obsahuje
dvě aplikace sebe sama, budeme vývoj výpočetnı́ho procesu zachycovat stromem. Uzly ve stromu budou
zastupovat jednotlivé aplikace. Vrcholem stromu bude výchozı́ aplikace. Každý uzel má ve stromu tolik
potomků, kolik je provedeno v rekurzivnı́m předpisu dalšı́ch aplikacı́.

V přı́padě procedury fib z programu 8.4 bude situace pro vstupnı́ hodnotu 5 vypadat tak, jak je znázorněno
na obrázku 8.9. Strom z obrázku 8.9 reprezentuje průběh výpočtu. Je to ale pouze jeho schéma. Tento obrázek
bychom si neměli vykládat tak, že při výpočtu je nějaký takový strom skutečně „sestaven“ a pak se v něm
něco „hledá“. Samotný průběh aplikace je zachycen na obrázku 8.10. Výpočet zahájený vyhodnocenı́m
(fib 5) pokračuje dalšı́ aplikacı́ (fib 4) (za předpokladu, že interpret vyhodnocuje prvky seznamu před
aplikacı́ procedury zleva doprava), dále se pokračuje aplikacı́ (fib 3), (fib 2), a (fib 1). V tomto bodu
a v této fázı́ výpočtu jsme narazili na limitnı́ podmı́nku, nastává fáze odvı́jenı́. V dalšı́m kroku se tedy

Obrázek 8.9. Schématické zachycenı́ aplikace rekurzivnı́ verze fib.

(fib 5)

(fib 4)

(fib 3)

(fib 2)

(fib 1) (fib 0)

(fib 1)

(fib 2)

(fib 1) (fib 0)

(fib 3)

(fib 2)

(fib 1) (fib 0)

(fib 1)

Obrázek 8.10. Postupné prováděnı́ aplikacı́ při použitı́ rekurzivnı́ verze fib.

(fib 5)

(+ (fib 4) (fib 3))

(+ (+ (fib 3) (fib 2)) (fib 3))

(+ (+ (+ (fib 2) (fib 1)) (fib 2)) (fib 3))

(+ (+ (+ (+ (fib 1) (fib 0)) (fib 1)) (fib 2)) (fib 3))

(+ (+ (+ (+ 1 (fib 0)) (fib 1)) (fib 2)) (fib 3))

(+ (+ (+ (+ 1 0) (fib 1)) (fib 2)) (fib 3))

(+ (+ (+ 1 (fib 1)) (fib 2)) (fib 3))

(+ (+ (+ 1 1) (fib 2)) (fib 3))

(+ (+ 2 (fib 2)) (fib 3))

(+ (+ 2 (+ (fib 1) (fib 0))) (fib 3))

(+ (+ 2 (+ 1 (fib 0))) (fib 3))

(+ (+ 2 (+ 1 0)) (fib 3))

(+ (+ 2 1) (fib 3))

(+ 3 (fib 3))

(+ 3 (+ (fib 2) (fib 1)))

(+ 3 (+ (+ (fib 1) (fib 0)) (fib 1)))

(+ 3 (+ (+ 1 (fib 0)) (fib 1)))

(+ 3 (+ (+ 1 0) (fib 1)))

(+ 3 (+ 1 (fib 1)))

(+ 3 (+ 1 1))

(+ 3 2)

5

opět vrátı́me do bodu aplikace (fib 2), abychom zde dokončili „odložený výpočet“. Součástı́ odloženého
výpočtu je však dalšı́ rekurzivnı́ aplikace (fib 0). Po jeho provedenı́ se opět narazı́ na limitnı́ podmı́nku
a výpočet pokračuje v prostředı́ aplikace (fib 2), kde je dokončen výpočet, ten již skutečně dokončen být
může, protože hodnoty 1 a 0 pro součet již byly spočteny předchozı́mi rekurzivnı́mi aplikacemi. Nastává
tedy dále fáze odvı́jenı́ v prostředı́ aplikace (fib 3). Zde je součástı́ odloženého výpočtu dalšı́ rekurzivnı́
aplikaci (fib 1), . . . Fáze navı́jenı́ a odvı́jenı́ se dále střı́dajı́ ve směru šipky nakreslené v obrázku 8.9.
Všimněte si, že neefektivita výpočetnı́ho procesu spočı́vá v opakovaném vypočı́távánı́ týchž hodnot.

Nynı́ můžeme provést hrubý odhad složitosti výpočetnı́ho procesu generovaného rekurzivnı́ procedurou
fib z programu 8.4. Počet kroků, které je pro dané n potřeba k vypočtenı́ výsledku je daný počtem
uzlů, které se nacházejı́ ve stromě zachycujı́cı́m všechny aplikace (protože sčı́tánı́ v jednotlivých aplikacı́ch
probı́há v konstantnı́m čase). V našem přı́padě má každý uzel bud’žádný nebo dva potomky. Nejdelšı́ cesta
od kořene k listu (uzlu bez potomka), tak zvaná hloubka stromu, má pro dané n délku n. Pokud bychom si
nynı́ zjednodušili situaci tı́m, že bychom konstatovali, že strom je vyvážený, pak vı́me, že počet jeho uzlů je
2n − 1. Pesimistický odhad časové složitosti by tedy byl O(2n). Podotkněme, že strom aplikacı́ fib obecně
nenı́ vyvážený a lze udělat mnohem přesnějšı́ odhady časové složitosti výpočty hodnoty fib pro n, ale
důležitým faktem je, že časová složitost narůstá exponenciálně vzhledem k n. Procedura fib z programu 8.4
je tedy použitelná pouze pro malé vstupnı́ hodnoty n (už při n = 30 je prodleva znatelná, protože proběhne
celkem 2692537 aplikacı́). Procedura je tedy prakticky nepoužitelná. S prostorovou složitostı́ na tom nejsme
tak špatně, ta je v třı́dě O(n). Při jejı́m stanovenı́ je klı́čové pozorovánı́, kolik prostoru je spotřebováno
(maximálně) při průchodu stromem z obrázku 8.9 ve směru postupných aplikacı́. Pro dané n má strom
největšı́ hloubku n, tedy spotřebovaný prostor roste lineárně s hloubkou stromu (to jest s rostoucı́m n).

Výpočetnı́ proces generovaný procedurou fib z programu 8.4 nazýváme stromově rekurzivnı́ výpočetnı́ proces.
Charakteristiku procesů toho typu bychom mohli shrnout takto:

stromově rekurzivnı́ proces je výpočetnı́ proces svým průběhem připomı́najı́cı́ lineárnı́ výpočetnı́ proces, jen
s tı́m rozdı́lem, že počet aplikacı́ rekurzivnı́ neroste lineárně s velikostı́ vstupu. Stromově rekurzivnı́
proces je typicky generován procedurami, které ve svých rekurzivnı́ch předpisech vyvolajı́ dvě nebo
vı́ce aplikacı́ sebe sama. Pro stromově rekurzivnı́ výpočetnı́ procesy je charakteristické postupné střı́-
dánı́ fázı́ navı́jenı́ a odvı́jenı́.

V přı́padě stromově rekurzivnı́ch výpočetnı́ch procesů stanovujeme časovou složitost stanovenı́m počtu
aplikacı́ procedury v závislosti na vstupnı́ch datech, což je počet uzlů v pomyslném „stromu“ zachycu-
jı́cı́m průběh celého výpočtu. Při stanovenı́ prostorové složitosti se lze odrazit od hloubky stromu, která
udává maximálnı́ hloubku (zanořenı́) rekurzivnı́ch aplikacı́.

Dobrou zprávou je, že i v přı́padě Fibonacciho čı́sel můžeme naprogramovat proceduru pro jejich výpočet
efektivně. Snadno totiž můžeme najı́t iterativnı́ proceduru, která bude Fibonacciho čı́sla počı́tat v čase
O(n) a v prostoru O(1). Tato procedura bude mı́t jeden argumentem jı́mž bude čı́tač zachycujı́cı́ kolik je
potřeba udělat iteracı́ a dva pomocné argumenty v nichž bude uchovávat poslednı́ dvě vypočtená Fibonacciho
čı́sla. Procedura je uvedena v programu 8.9. Procedura fib opět provede pouze aplikaci pomocné iterativnı́
procedury fib-iter. Procedura fib-iter je z fib aplikována s čı́tačem nastavenı́m na hodnotu vstupnı́ho
čı́sla (poslednı́ argument), prvnı́ dva prvky Fibonacciho posloupnosti jsou 0 a 1 (prvnı́ dva argumenty).
Z kódu procedury fib-iter lze snadno vyčı́st, že iterace je zastavena pokud čı́tač klesne na nulu. Potom
je vrácena hodnota navázaná na prvnı́m symbolu (prvnı́ pomocný argument). Rekurzivnı́ předpis řı́ká, že
v každém kroku je snı́žen čı́tač o jedna, a záznam o poslednı́ch dvou Fibonacciho čı́slech je změněn tak, aby
se v dalšı́m kroku opět jednalo o (dalšı́) dvě poslednı́ Fibonacciho čı́sla v posloupnosti. Na obrázku 8.11 je
ukázka aplikace procedury fib z programu 8.9 pro vstupnı́ hodnotu 5. Časová složitost výpočtu je zřejmě
ve třı́dě O(n), prostorová v O(1). Procedur fib-iter v programu 8.9 bychom opět mohli definovat jako
internı́ ve fib, abychom provedli odstı́něnı́ pomocné procedury fib-iter od uživatele procedury fib, viz
program 8.10.

Program 8.9. Iterativnı́ procedura pro výpočet Fibonacciho čı́sel.

(define fib-iter

(lambda (a b i)

(if (<= i 0)

a

(fib-iter b (+ a b) (- i 1)))))

(define fib

(lambda (n)

(fib-iter 0 1 n)))

Obrázek 8.11. Schématické zachycenı́ aplikace iterativnı́ verze fib.

(fib 5)

(fib-iter 0 1 5) navı́jenı́ : vyvolánı́ 1. aplikace
(fib-iter 1 1 4) navı́jenı́ : vyvolánı́ 2. aplikace
(fib-iter 1 2 3) navı́jenı́ : vyvolánı́ 3. aplikace
(fib-iter 2 3 2) navı́jenı́ : vyvolánı́ 4. aplikace
(fib-iter 3 5 1) navı́jenı́ : vyvolánı́ 5. aplikace
(fib-iter 5 8 0) navı́jenı́ : vyvolánı́ 6. aplikace
5 dosaženı́ limitnı́ podmı́nky a vrácenı́ hodnoty

V této sekci jsme ukázali, že různé rekurzivnı́ procedury generujı́ kvalitativně různé výpočetnı́ procesy
s různou náročnostı́ na výpočetnı́ zdroje počı́tače. S tı́mto faktem je vždy potřeba dopředu počı́tat.
Při programovánı́ bychom se měli soustředit jak na efektivitu procedur tak na čistotu jejich provedenı́
a na jejich celkovou čitelnost. Efektivita a čistota provedenı́ jdou leckdy proti sobě, u konkrétnı́ch
aplikacı́ je tedy potřeba najı́t přijatelný kompromis. Uživatele obvykle zajı́má „rychlost programu“,
což hovořı́ pro většı́ důraz na efektivitu. Na druhou stranu uživatele také často zajı́má „rozšiřitelnost,
přizpůsobitelnost a přenositelnost programu“, což naopak vede k důrazu na čisté provedenı́ programu
(vysoce optimalizované programy jsou zpravidla těžko čitelné a jejich rozšiřitelnost je tı́m pádem těžšı́
nebo prakticky nemožná).

Poznámka 8.19. Stromová rekurze je často použı́vaným nástrojem při zpracovánı́ hierarchických dat. Touto
problematikou se budeme zabývat předevšı́m v dalšı́ lekci. V této sekci jsme si představili stromově re-
kurzivnı́ výpočetnı́ procesy jako jeden typ procesů generovaných rekurzivnı́mi procedurami. Obecně nelze
řı́ct, jak by se na prvnı́ pohled možná mohlo zdát, že stromově rekurzivnı́ procesy jsou „neefektivnı́ “, nebo
že vždy vedou na exponenciálnı́ časovou složitost. Tak tomu nenı́. Složitost totiž vždy stanovujeme vzhle-
dem k velikosti vstupnı́ch dat. Pokud budou vstupnı́ data reprezentována (nějakou) hierarchickou datovou
strukturou s n uzly a tyto uzly budou nějakou procedurou procházeny s použitı́m stromové rekurze jeden
po druhém, pak bude časová složitost této procedury lineárnı́, i když šlo o proceduru generujı́cı́ stromově
rekurzivnı́ výpočetnı́ proces.

8.4 Jednorázové použitı́ procedur

U mnoha přı́kladů v předchozı́ lekci jsme při vytvářenı́ procedury řešı́cı́ daný problém vytvořili pomocnou
proceduru, kterou jsme z hlavnı́ procedury aplikovali pouze jednou. Napřı́klad tomu tak bylo v programech 8.9
a 8.10. Třeba v programu 8.9 byla pomocná iterativnı́ procedura fib-iter použita jednorázově v těle

Program 8.10. Iterativnı́ procedura pro výpočet Fibonacciho čı́sel s internı́ definicı́.

(define fib

(lambda (n)

(define iter

(lambda (a b i)

(if (<= i 0)

a

(iter b (+ a b) (- i 1)))))

(iter 0 1 n)))

procedury fib. Na žádném dalšı́m mı́stě programu (mimo tělo samotné procedury fib-iter) již k aplikaci
procedury fib-iter nedocházı́. V předchozı́ sekci byla vždy pomocná procedura iterativnı́, obecně se ale
můžeme do podobné situace dostat i v přı́padě, kdy pomocná procedura generuje lineárnı́ nebo stromově
rekurzivnı́ výpočetnı́ proces.

V jazyku Scheme máme k dispozici aparát pro stručnějšı́ definici rekurzivnı́ procedury, která je po své
definici jednorázově použita (s danými argumenty). K tomuto účelu sloužı́ rozšı́řenı́ speciálnı́ formy let,
kterému řı́káme pojmenovaný let. Popis tohoto rozšı́řenı́ let následuje.

Definice 8.20 (speciálnı́ forma let, pojmenovaná verze). Speciálnı́ forma pojmenovaný let se použı́vá se třemi
nebo vı́ce argumenty ve tvaru

(let 〈jméno〉 ((〈symbol1〉 〈výraz1〉)
(〈symbol2〉 〈výraz2〉)

...

(〈symboln〉 〈výrazn〉))
〈tělo1〉
〈tělo2〉
...

〈tělok〉),

kde n ≥ 0, k ≥ 1; 〈symbol1〉, . . . , 〈symboln〉 jsou vzájemně různé symboly, 〈jméno〉 je symbol (obecně se může
jednat i o některý ze symbolů 〈symbol1〉, . . . , 〈symboln〉), 〈výraz1〉, . . . , 〈výrazn〉 jsou libovolné výrazy jejichž
vyhodnocenı́m vznikajı́ počátečnı́ vazby přı́slušných symbolů, a 〈tělo1〉, . . . , 〈tělok〉 jsou výrazy tvořı́cı́ tělo.
Aplikace pojmenované verze speciálnı́ formy let s výše uvedenými argumenty probı́há tak, že interpret
tuto formu nahradı́ následujı́cı́m kódem

(let ()

(define 〈jméno〉
(lambda (〈symbol1〉 〈symbol2〉 · · · 〈symboln〉)
〈tělo1〉
〈tělo2〉

...

〈tělok〉))
(〈jméno〉 〈výraz1〉 〈výraz2〉 · · · 〈výrazn〉)),

který je vyhodnocen v aktuálnı́m prostředı́. �

Na prvnı́ pohled se pojmenovaný let od tradičnı́ formy let lišı́ tı́m, že jeho prvnı́ argument nenı́ seznam
(vazeb), ale symbol. Ve zbylé části již ze syntaktického hlediska rozdı́l nenı́. Z předchozı́ho popisu je vidět,
že výskyty pojmenovaného let v programu jsou během vyhodnocovánı́ nahrazovány jiným kódem, ve
kterém je použita tradičnı́ speciálnı́ forma let k vytvořenı́ nového prostředı́ (bez vazeb, všimněte si, že let

má prázdný seznam vazeb). V těle speciálnı́ formy let je potom klasickým způsobem definována procedura,

která je navázána na symbol 〈jméno〉. Tělo procedury je tvořeno tělem pojmenovaného let a argumenty
procedury jsou symboly, které v pojmenovaném let označovaly jména nových vazeb. Tato procedura je
potom jednorázově aplikována s argumenty jimiž jsou hodnoty vzniklé vyhodnocenı́m výrazů předaných
pojmenovanému let.

Pokud použijeme pojmenovaná let (to jest let, kde prvnı́m argumentem je symbol – jméno) aniž bychom
dané jméno použili v těle této speciálnı́ formy, pak má pojmenovaný let stejný efekt jako klasický let:

(let proc ((x (* 2 5))

(y 20))

(+ x y)) Z=⇒ 30

což je dobře vidět, když si uvědomı́me, že předchozı́ kód je ekvivalentnı́:

(let ()

(define proc

(lambda (x y)

(+ x y)))

(proc (* 2 5) 20)) Z=⇒ 30

Z definice pojmenovaného let uvedené v předchozı́m přı́kladě je dobře vidět, že prostředı́ vyhodnocenı́
těla je prostředı́ aplikace procedury navázané na symbol 〈jméno〉. Navı́c v těle má symbol 〈jméno〉 aktuálnı́
vazbu jı́ž je právě aplikovaná procedura. Pomocı́ symbolu 〈jméno〉 a jeho vazby je tedy možné provádět
aplikaci „sebe sama“. Napřı́klad následujı́cı́ kód ukazuje provede sečtenı́ čı́sel od 0 do 10 pomocı́ rekurzivnı́
aplikace (jednorázové) procedury vytvořené touto speciálnı́ formou:

(let proc ((n 10))

(if (= n 0)

0

(+ n (proc (- n 1))))) Z=⇒ 55

Předchozı́ použitı́ pojmenovaného let je během vyhodnocovánı́ přepsáno na

(let ()

(define proc

(lambda (n)

(if (= n 0)

0

(+ n (proc (- n 1))))))

(proc 10))

Následujı́cı́ přı́klad ukazuje rekurzivnı́ verzi procedury fib pro výpočet prvků Fibonacciho posloupnosti,
která je definována pomocı́ pojmenovaného let a jednorázově aplikována s hodnotou 30:

(let fib ((n 30))

(if (<= n 1)

n

(+ (fib (- n 1))

(fib (- n 2))))) Z=⇒ 832040

Jméno uvedené v pojmenovaném let se může shodovat s některým ze symbolů pojmenovávajı́cı́m vazby,
i když to nenı́ účelné. V takovém přı́padě je totiž vazba symbolu 〈jméno〉 překryta hodnotou předávaného
argumentu, viz přı́klad:

(let f ((x (* 2 5))

(y 20)

(f 100))

(list x y f)) Z=⇒ (10 20 100),

což je opět patrné, pokud si předchozı́ pojmenovaný let rozepı́šeme:

(let ()

(define f

(lambda (x y f)

(list x y f)))

(f (* 2 5) 20 100)) Z=⇒ (10 20 100)

V programech 8.11 a 8.12 jsou uvedeny verze programů 8.6 a 8.10 ze stran 212 a 218, ve kterých byla interně
definovaná a jednorázově aplikovaná procedura vytvořena pomocı́ pojmenovaného let.

Program 8.11. Iterativnı́ procedura pro výpočet faktoriálu s internı́ definicı́ pomocı́ pojmenovaného
let.

(define fac

(lambda (n)

(let iter ((i n)

(accum 1))

(if (<= i 1)

accum

(iter (- i 1) (* accum i))))))

Program 8.12. Iterativnı́ procedura pro výpočet Fib. čı́sel s internı́ definicı́ pomocı́ pojmenovaného let.

(define fib

(lambda (n)

(let iter ((a 0)

(b 1)

(i n))

(if (<= i 0)

a

(iter b (+ a b) (- i 1))))))

Použitı́ pojmenovaného let je vhodné v přı́padě, kdy potřebujeme okamžitě a jednorázově aplikovat
právě definovanou pomocnou proceduru, která je rekurzivnı́. Z hlediska použitı́ pojmenovaného let

je přitom jedno, jaký rekurzivnı́ výpočetnı́ proces tato pomocnı́ procedura generuje.

8.5 Rekurze a indukce na seznamech

Nynı́ se budeme blı́že zabývat rekurzivnı́mi procedurami pracujı́cı́mi se seznamy. Po teoretické stránce
jde opět o klasické uživatelsky definované procedury, jak jsme se představili již v lekci 2. Rekurzivnı́
procedury pracujı́cı́ se seznamy mohou stejně jako procedury pracujı́cı́ nad čı́sly generovat kvalitativně
různé rekurzivnı́ výpočetnı́ procesy. Opět se jedná o lineárně rekurzivnı́ výpočetnı́ proces, lineárně iterativnı́
výpočetnı́ proces a stromově rekurzivnı́ výpočetnı́ proces.

Jako prvnı́ si ukážeme rekurzivnı́ variantu procedury length pro výpočet délky seznamu. Připomeňme, že
tuto proceduru jsme již implementovali hned dvakrát. Prvnı́ verze použı́vala map a apply, viz program 6.1
na straně 144. Druhou verzi jsme implementovali pomocı́ foldr, viz program 7.1 na straně 174. Varianta
pomocı́ foldr byla efektivnějšı́ než verze použı́vajı́cı́ map a apply. V programu 8.13 máme uvedenu
rekurzivnı́ verzi procedury length, která je co se týče efektivity stejně kvalitnı́ jako verze z programu 7.1.
Limitnı́ podmı́nkou výpočtu délky seznamu je prázdný seznam, ten má délku nula. Pokud je seznam

Program 8.13. Výpočet délky seznamu pomocı́ rekurze.

(define length

(lambda (l)

(if (null? l)

0

(+ 1 (length (cdr l))))))

Program 8.14. Iterativnı́ výpočet délky seznamu.

(define length

(lambda (l)

(let iter ((l l)

(steps 0))

(if (null? l)

steps

(iter (cdr l) (+ steps 1))))))

neprázdný, pak je rekurzivnı́m předpisem problém výpočtu délky tohoto seznamu redukován na problém
nalezenı́ délky seznamu bez prvnı́ho prvku (rekurzivnı́ aplikace) a přičtenı́ jedničky (započtenı́ prvnı́ho
prvku výchozı́ho seznamu). Všimněte si, že rekurzivnı́ procedura z programu 8.13 je v podstatě jen přepisem
rekurzivnı́ho definičnı́ho vztahu ze sekce 8.1.

Procedura length z programu 8.13 má prostorovou složitost O(n), kde n je délka seznamu, protože při
každé rekurzivnı́ aplikaci je vytvořeno nové prostředı́ (rekurzivnı́ aplikace length nenı́ koncová). Snadno
bychom ale mohli length naprogramovat iterativně. K tomuto účelu bychom potřebovali (jednorázově
použı́vanou) pomocnou proceduru s dvěma argumenty. Jednı́m by byl seznam, který by sloužil jako čı́tač –
– při každém kroku bychom z něj odebı́rali jeden prvek. Druhý argument by sloužil jako čı́selný střadač,
který by počı́tal počet kroků. Iterativnı́ verze length je uvedena v programu 8.14. Tato iterativnı́ verze má
prostorovou složitost O(1), během aplikace jsou vytvořena vždy jen dvě prostředı́ a nejsou konstruovány
žádné hierarchické datové struktury. Průběh aplikace iterativnı́ verze length pro konkrétnı́ vstupnı́ seznam
ve tvaru (a 10.2 b #f) je zobrazen na obrázku 8.12. Jak je z tohoto obrázku patrné, průběh iterativnı́ho
procesu se ničı́m nelišı́ od průběhu iterativnı́ch procesů uvedených v předchozı́ sekci. Na věci nic neměnı́
to, že nynı́ zpracováváme seznam a limitnı́ podmı́nka je formulována pro seznam. V dalšı́ch ukázkách v této
sekci již tedy nebudeme podrobně znázorňovat průběhy jednotlivých výpočetnı́ch procesů a ponecháme je
na laskavém čtenáři.

Nynı́ se budeme věnovat rekurzivnı́mu spojenı́ dvou seznamů. V úvodu této lekce jsme ukázali rekur-
zivnı́ definici zobrazenı́ append2. V programu 8.15 je definována procedura append2, která reprezentuje
rekurzivně definované append2 ze sekce 8.1. Limitnı́ podmı́nkou je, pokud je prvnı́ ze spojovaných se-
znamů prázdný, v tom přı́padě je spojenı́ rovno druhému seznamu. V opačném přı́padě je problém spojenı́
dvou seznamů redukován na problém spojenı́ prvnı́ho seznamu bez prvnı́ho prvku s druhým seznamem
(rekurzivnı́ aplikace) a následné připojenı́ prvnı́ho prvku prvnı́ho seznamu na začátek výsledku před-
chozı́ho spojenı́. Proceduru append2 jsme již také několikrát implementovali. Poprvé to byla neefektivnı́
implementace v programu 5.2 na straně 123, která byla založená na build-list a přı́stupu k prvkům se-
znamu pomocı́ list-ref. Dále jsme provedli implementaci pomocı́ foldr, viz program 7.2 na straně 175.
Z hlediska efektivity, je rekurzivnı́ verze append2 shodná s append2 vytvořené pomocı́ foldr. Samotnou
proceduru foldr bychom mohli rovněž snadno naprogramovat jako rekurzivnı́ proceduru. Tı́mto a dalšı́mi
problémy se budeme zabývat v následujı́cı́ lekci.

Ne všechny procedury, se kterými jsme se setkali, však lze efektivně naprogramovat pomocı́ foldr. Při-
nejmenšı́m to nenı́ z programátorského pohledu nijak „přı́močaré“. Takovou procedurou je třeba procedura

Obrázek 8.12. Schématické zachycenı́ aplikace iterativnı́ verze length.

(length (a 10.2 b #f))

(iter (a 10.2 b #f) 0) navı́jenı́ : vyvolánı́ 1. aplikace
(iter (10.2 b #f) 1) navı́jenı́ : vyvolánı́ 2. aplikace
(iter (b #f) 2) navı́jenı́ : vyvolánı́ 3. aplikace
(iter (#f) 3) navı́jenı́ : vyvolánı́ 4. aplikace
(iter () 4) navı́jenı́ : vyvolánı́ 5. aplikace
4 dosaženı́ limitnı́ podmı́nky a vrácenı́ hodnoty

Program 8.15. Spojenı́ dvou seznamů pomocı́ rekurze.

(define append2

(lambda (l1 l2)

(if (null? l1)

l2

(cons (car l1) (append2 (cdr l1) l2)))))

list-ref, která pro daný seznam a index vracı́ prvek seznamu nacházejı́cı́ se na pozici určené indexem.
Snadno lze ale naprogramovat iterativnı́ proceduru, která n-tý prvek seznamu vracı́. Viz proceduru v pro-
gramu 8.16. Tato procedura skutečně vždy generuje iterativnı́ výpočetnı́ proces, protože jejı́ jediná rekur-

Program 8.16. Rekurzivnı́ verze proceduru vracejı́cı́ prvek na dané pozici v seznamu.

(define list-ref

(lambda (l index)

(if (= index 0)

(car l)

(list-ref (cdr l) (- index 1)))))

zivnı́ aplikace se nacházı́ v koncové pozici. Limitnı́ podmı́nkou procedury je, pokud je index nulový. V tom
přı́padě vracı́me prvnı́ prvek seznamu, který lze zı́skat v konstantnı́m čase pomocı́ car. Pokud je index ne-
nulový, pak vı́me, že hledaný prvek se nacházı́ hlouběji v seznamu. Hledánı́ n-tého prvku seznamu je tedy
redukováno na hledánı́ prvku na pozici n − 1 v seznamu zkráceném o prvnı́ prvek. Procedura list-ref

má časovou složitost O(n), kde n je pozice (index) v seznamu. Časová složitost tedy nenı́ závislá na délce
vstupnı́ho seznamu, ale na pozici, ze které chceme prvek vrátit. Prostorová složitost je O(1). Z hlediska
složitosti jde o významný posun oproti verzi list-ref z programu 6.4 na straně 146, která měla při efek-
tivnı́ implementaci procedur filter (pracujı́cı́ v čase a prostoru O(n)) a length (z programu 8.14) časovou
složitost O(4n) a prostorovou složitost O(3n) (zdůvodněte si podrobně proč).

Analogicky jako proceduru list-ref bychom mohli naprogramovat obecnějšı́ proceduru list-tail,
která pro daný seznam a čı́selný argument n vrátı́ seznam bez prvnı́ch n prvků. Viz program 8.17. Jedná
se opět o iterativnı́ proceduru s časovou složitostı́ O(n), kde n je počet vypuštěných prvků, a prostorovou
složitostı́ O(1). Z definice procedury list-tail je zřejmé, že pomocı́ nı́ bychom mohli definovat list-ref:

(define list-ref

(lambda (l index)

(car (list-tail l index))))

Mapovánı́ přes jeden seznam bychom mohli přı́močaře naprogramovat pomocı́ rekurzivnı́ procedury.
V předchozı́ch částech textu jsme již dvě implementace procedury map1 (mapovánı́ přes jediný seznam)

Program 8.17. Procedura vracejı́cı́ seznam bez zadaného počtu prvnı́ch prvků.

(define list-tail

(lambda (l skip)

(if (= skip 0)

l

(list-tail (cdr l) (- skip 1)))))

viděli. Prvnı́ z nich byla v programu 5.3 na straně 125, využı́vala těžkopádně build-list a list-ref a byla
hodně neefektivnı́. Dalšı́ implementace byla ukázána v programu 7.4 na straně 176 a použı́vala foldr. Tato
verze již byla efektivnı́ a efektivnějšı́ verzi z principu vytvořit nelze – při mapovánı́ totiž vytvářı́me nový
seznam délky n, takže časová i prostorová složitost nemohou být řádově lepšı́ než O(n). Rekurzivnı́ verze
map1 však může být čitelnějšı́ než verze vytvořená pomocı́ foldr. Rekurzivnı́ verzi map1 máme zobrazenu
v programu 8.18. Meznı́m přı́padem je opět přı́pad pro prázdný seznam. V tomto přı́padě je výsledkem

Program 8.18. Mapovacı́ procedura pracujı́cı́ s jednı́m seznamem pomocı́ rekurze

(define map1

(lambda (f l)

(if (null? l)

'()

(cons (f (car l))

(map1 f (cdr l))))))

mapovánı́ procedury přes prázdný seznam opět prázdný seznam. V přı́padě, že je daný seznam neprázdný,
je výsledkem mapovánı́ seznam vzniklý připojenı́m modifikace prvnı́ho prvku na výsledek mapovánı́ přes
seznam bez prvnı́ho prvku.

Jednou z procedur, kterou by bylo bez rekurze těžké vytvořit, je procedura build-list sloužı́cı́ k vytvářenı́
seznamů dané délky jež obsahujı́ prvky dané jako výsledky aplikace procedury jednoho argumentu (argu-
mentem je pozice prvku v seznamu). Naprogramovat proceduru build-list pomocı́ rekurze je přı́močaré.
Triviálnı́m přı́padem je vytvořenı́ seznamu délky nula – ten je vždycky prázdný. V opačném přı́padě lze
redukovat problém vytvořenı́ n-prvkového seznamu na problém vytvořenı́ seznamu o n − 1 prvcı́ch na
jehož počátek je připojen nový prvek. Rekurzivnı́ procedura je uvedena v programu 8.19. V proceduře

Program 8.19. Rekurzivnı́ verze procedury na vytvářenı́ seznamů.

(define build-list

(lambda (n f)

(let build-next ((i 0))

(if (= i n)

'()

(cons (f i) (build-next (+ i 1)))))))

build-list je pomocı́ pojmenovaného let vytvořena jednorázově aplikovaná rekurzivnı́ procedura vy-
tvářejı́cı́ seznam. Tuto proceduru jsme vytvořili proto, že jsme potřebovali dalšı́ argument pomocı́ nějž
si předáváme informaci o pozici prvku, který chceme vytvořit. V limitnı́ podmı́nka vyjadřuje zastavenı́
navı́jenı́ (konstrukce seznamu) v přı́padě, že již jsme na pozici vyskytujı́cı́ se „za poslednı́m požadova-
ným prvkem“. Dokud nenı́ limitnı́ podmı́nky dosaženo, jsou rekurzivně konstruovány prvky pomocı́ cons
a pomocı́ předané procedury navázané na f. Implementace build-list z programu 8.19 má časovou

i prostorovou složitost O(n), kde n je délka seznamu, který chceme vytvářet. Jedná se tedy o maximálně
efektivnı́ implementaci (lepšı́ řádové složitosti již nelze dosáhnout).

V programu 8.19 jsme výsledný seznam konstruovali jakoby odpředu. To jest od prvku s nejnižšı́m indexem
až k prvku s nejvyššı́m indexem. Mohli bychom postupovat i obráceně. To by mělo zdánlivou výhodu
v tom, že bychom nemuseli definovat pomocnou proceduru. Řešenı́ je uvedeno v programu 8.20. Při

Program 8.20. Neefektivnı́ verze procedury na vytvářenı́ seznamů.

(define build-list-rev

(lambda (n f)

(if (= n 0)

'()

(append2 (build-list-rev (- n 1) f)

(list (f (- n 1)))))))

pozorném studiu programu záhy zjistı́me, že program je ale velmi neefektivnı́. Je to způsobeno použı́vánı́m
append2 při každé rekurzivnı́ aplikaci. Pomocı́ append2 spojujeme seznamy délky n− 1 s jednoprvkovým
seznamem – tı́m realizujeme připojenı́ nově vytvořeného prvku na konec seznamu. Nejefektivnějšı́ verze
append2 potřebuje n kroků pro spojenı́ prvnı́ho seznamu délky n s libovolným seznamem. Tı́m pádem
procedura build-list-rev potřebuje postupně n − 1, n − 2, . . . , 1 kroků k postupnému zařazenı́ všech
vytvářených prvků na konec seznamu. Součtem prvků aritmetické posloupnosti tedy zı́skáme celkem
n·(n−1)
2 kroků. Časová složitostbuild-list-rev je tedy kvadratická vzhledem k délce vytvářeného seznamu.

I prostorová složitost je v této třı́dě, protože během použı́vánı́ append2 jsou postupně dokola konstruovány
nové seznamy délek n, n − 1, . . . , 1. Celkově vzato je tedy procedura build-list-rev velmi neefektivnı́.
Na této proceduře je taky dobré uvědomit si, že i když procedura generuje lineárně rekurzivnı́ výpočetnı́
proces, jejı́ časové složitost nenı́ lineárnı́ vzhledem ke velikosti vstupu. Slovo „lineárnı́“ v pojmu „lineárně
rekurzivnı́ výpočetnı́ proces“ se totiž vztahuje k počtu rekurzivnı́ch aplikacı́ procedury. Vždy je ale potřeba
ještě brát v potaz, co se (z hlediska časové náročnosti) děje v každé z aplikacı́.

Samotné vytvářenı́ seznamů počı́naje poslednı́m prvkem však nenı́ zavrženı́hodná myšlenka. Mı́sto rekur-
zivnı́ verze konstrukce je však vhodné použı́vat ke konstrukci seznamu iterativnı́ proceduru, která bude
nově vytvářené prvky postupně „střádat“ do nového seznamu. V programu 8.21 je uvedena procedura
build-list-iter vytvářejı́cı́ seznam právě tı́mto způsobem. Procedura build-list-iter opět použı́vá

Program 8.21. Rekurzivnı́ verze procedury na vytvářenı́ seznamů.

(define build-list-iter

(lambda (n f)

(let iter ((i (- n 1))

(accum '()))

(if (< i 0)

accum

(iter (- i 1) (cons (f i) accum))))))

pomocnou proceduru vytvořenou pomocı́ pojmenovaného let. Tato pomocı́ procedura má dva argumenty,
jednı́m je čı́tač jdoucı́ od n − 1 (poslednı́ pozice v konstruovaném seznamu) k 0 (prvnı́ pozice v konstruo-
vaném seznamu). V každém kroku konstrukce je tento čı́tač zmenšen o jedna. Iterace končı́ pokud je čı́tač
menšı́ než hodnota nula (všechny prvky již byly vytvořeny). Druhým argumentem pomocné procedury
je akumulátor, ke kterému jsou postupně (zepředu) přidávány nově vytvořené prvky. Po dosaženı́ limitnı́
podmı́nky je vrácena hodnota akumulátoru. Časová a prostorová složitost tohoto řešenı́ je O(n). Oproti
verzi z programu 8.19 je rekurzivnı́ aplikace prováděna v jednom prostředı́, protože jde o iteraci. I přesto je
ale prostorová složitost O(n), protože dojde k vytvořenı́ n-prvkového seznamu.

Nynı́ se budeme věnovat efektivnı́ proceduře pro převracenı́ prvků seznamu. Jde nám tedy o efektivnı́
implementaci procedury reverse. Stejně jako v přı́padě vytvářenı́ seznamů „odzadu“, které jsme viděli
v procedurách build-list-rev a build-list-iter, bude existovat několik řešenı́ lišı́cı́ch se ve své
efektivitě. Nejprve si ukážeme řešenı́, které nenı́ přı́liš efektivnı́. V programu 8.22 je uvedena lineárně
rekurzivnı́ procedura reverse. Tato procedura je založena na faktu, že převrácenı́ prázdného seznamu

Program 8.22. Neefektivnı́ verze převracenı́ prvků v seznamu.

(define reverse

(lambda (l)

(if (null? l)

'()

(append (reverse (cdr l))

(list (car l))))))

je triviálnı́ a pokud máme převrátit neprázdný seznam, pak stačı́ převrátit zbytek tohoto seznamu bez
prvnı́ho prvku a potom nakonec takového seznamu připojit původnı́ prvnı́ prvek. Procedura reverse

z programu 8.22 tedy trpı́ stejným neduhem jako lineárně rekurzivnı́ procedura build-list-rev – při
každé rekurzivnı́ aplikaci je použita procedura append2 s lineárnı́ časovou složitostı́. Převrácenı́ seznamu
tı́mto způsobem má tedy opět řádově kvadratickou časovou i prostorovou složitost.

Efektivnı́ verzi převrácenı́ seznamu bychom mohli vytvořit pomocı́ iterativnı́ procedury provádějı́cı́ spojenı́
dvou seznamů tak, že ve výsledku je prvnı́ z těchto dvou seznamů spojen v opačném pořadı́. Proceduru
provádějı́cı́ tento typ spojenı́ nazveme rev-append2 a jejı́ kód je uveden v programu 8.23. Procedura

Program 8.23. Iterativnı́ verze převrácenı́ prvků v seznamu.

(define rev-append2

(lambda (l1 l2)

(if (null? l1)

l2

(rev-append2 (cdr l1)

(cons (car l1) l2)))))

(define reverse

(lambda (l)

(rev-append2 l '())))

rev-append2 použı́vá svůj prvnı́ argument (prvnı́ ze spojovaných seznamů) jako čı́tač a druhý argument
(druhý ze spojovaných seznamů) jako akumulátor. Pokud je prvnı́ ze spojovaných seznamů prázdný, je
vrácen druhý seznam. V opačném přı́padě je z prvnı́ho seznamu odebrán jeho prvnı́ prvek, který je přidán
na začátek druhého seznamu. Viz následujı́cı́ přı́klady použitı́ procedury rev-append2:

(rev-append2 '() '()) Z=⇒ ()

(rev-append2 '() '(1 2 3 4)) Z=⇒ (1 2 3 4)

(rev-append2 '(a) '(1 2 3 4)) Z=⇒ (a 1 2 3 4)

(rev-append2 '(a b) '(1 2 3 4)) Z=⇒ (b a 1 2 3 4)

(rev-append2 '(a b c) '(1 2 3 4)) Z=⇒ (c b a 1 2 3 4)

(rev-append2 '(a b c) '()) Z=⇒ (c b a)

Nynı́ je jasné, že efektivnı́reversemůžeme naprogramovat jako proceduru, která pouze aplikuje proceduru
rev-append2 s druhým seznamem, který je prázdný, viz program 8.23. Procedura rev-append2 a tı́m
pádem i procedura reverse má lineárnı́ časovou i prostorovou složitost.

Nynı́, na konci sekce, uvedeme implementaci třı́děnı́ nazývaného mergesort. Tato metoda je založena na
takzvaném slévánı́ již setřı́děných seznamů. Ze dvou setřı́děných seznamů je totiž možné snadno vytvořit
jeden setřı́děný seznam. Stačı́ nám totiž postupně odebı́rat vždy menšı́ z prvnı́ch prvků těchto seznamů a
konstruovat z nich výsledný seznam. Předved’me si to na konkrétnı́ch seznamech:

prvnı́ seznam: (1 7 8) (7 8) (7 8) (7 8) (8) ()

druhý seznam: (2 3 9 10) (2 3 9 10) (3 9 10) (9 10) (9 10) (9 10)

výsledek: () (1) (1 2) (1 2 3) (1 2 3 7) (1 2 3 7 8)

prvnı́ seznam: () ()

druhý seznam: (10) ()

výsledek: (1 2 3 7 8 9) (1 2 3 7 8 9 10)

Proceduru na slévánı́ dvou setřı́děných seznamů můžeme napsat třeba takto:

(define merge

(lambda (s1 s2)

(cond ((null? s1) s2)

((null? s2) s1)

((<= (car s1) (car s2)) (cons (car s1) (merge (cdr s1) s2)))

(else (cons (car s2) (merge s1 (cdr s2)))))))

Jedná se tedy o rekurzı́vnı́ proceduru. Jejı́ limitnı́ podmı́nkou je to, že je aspoň jeden ze seznamů prázdný. Je-
li této podmı́nky dosaženo, je vrácen zbývajı́cı́ (potenciálně neprázdný) seznam. Nenı́-li limitnı́ podmı́nky
dosaženo, redukujeme problém na připojenı́ prvku procedurou cons a slévánı́ seznamů, z nichž je jeden
původnı́ a jeden vznikne z druhého původnı́ho odebránı́m prvnı́ho prvku.

V uvedené definici procedury merge použı́váme k porovnávánı́ prvnı́ch prvků seznamů predikát <=. To sa-
mozřejmě nenı́ jediné uspořádánı́, podle kterého můžeme chtı́t slévat. Jednoduchým rozšı́řenı́m procedury
merge na proceduru vyššı́ho řádu, které budeme mimo setřı́děných seznamů předávat navı́c predikát urču-
jı́cı́ uspořádánı́, dosáhneme výrazného zobecněnı́. Proceduru navı́c napı́šeme tak, aby tento nový argument
byl volitelný. Následuje definice rozšı́řené procedury.

(define merge

(lambda (s1 s2 . pred?)

(let ((<= (if (null? pred?) <= (car pred?))))

(let merge ((s1 s1)

(s2 s2))

(cond ((null? s1) s2)

((null? s2) s1)

((<= (car s1) (car s2)) (cons (car s1) (merge (cdr s1) s2)))

(else (cons (car s2) (merge s1 (cdr s2)))))))))

V těle nové verze procedury merge vytvářı́me lokálnı́ prostředı́ s vazbou na symbol <=. V přı́padě, že je
proceduře merge předán volitelný argument (seznam navázaný na symbol pred?) je tedy neprázdný, bude
na symbol <= navázán tento argument, tedy prvnı́ prvek seznamu pred?. V opačném přı́padě ponecháme
symbolu <= jeho původnı́ význam. V tomto lokálnı́m prostředı́ pak pomocı́ pojmenovaného let vytvářı́me
a jednorázově aplikujeme proceduru merge, která se nijak nelišı́ od jejı́ předchozı́ verze.

Novou proceduru merge tedy můžeme použı́t se dvěma argumenty, v tom přı́padě se bude chovat stejně
jako ta původnı́:

(merge '(1 5 6 7 9) '(2 2 3 4 4 4 10)) Z=⇒ (1 2 2 3 4 4 4 5 6 7 9 10)

nebo se třemi argumenty. Tehdy je třetı́m argumentem určeno uspořádánı́, v jakém by měly být setřı́děny
slévané seznamy, a jakým způsobem probı́há výběr „menšı́ho prvku“ při samotném slévánı́:

(merge '(9 7 6 5 1) '(10 4 4 4 3 2 2) >=) Z=⇒ (10 9 7 6 5 4 4 4 3 2 2 1)

Právě to nám umožňuje použı́t tuto proceduru ke slévánı́ jiných seznamů než čı́selných. Můžeme napřı́klad
slévat seznamy obsahujı́cı́ racionálnı́ čı́sla v reprezentaci představené v sekci 4.6:

(merge (list (make-r 1 3) (make-r 2 3))

(list (make-r 1 2) (make-r 2 2))

r<=) Z=⇒ ((1 . 3) (1 . 2) (2 . 3) (1 . 1))

Procedura slévánı́ seznamů je tedy základem třı́dı́cı́ metody mergesort. Celý popis této metody je následujı́cı́:

Vstup: seznam (a predikát uspořádánı́)

Výstup: setřı́děný seznam

Postup:
• Pokud je seznam prázdný nebo jednoprvkový, je již setřı́děný a je výstupem.

• Jinak seznam rozdělı́me na dvě části (seznamy), každou z nich setřı́dı́me algoritmem mergesort.

• Tyto dva setřı́děné seznamy slijeme do jednoho, výsledný setřı́děný seznam je pak výstupem.

Samotné slévánı́ bychom tedy měli implementované. Zbývá nám implementace procedury na rozdělenı́
seznamu na dvě části. Při dělenı́ nám nezáležı́ na zachovánı́ pořadı́ prvků, protože výsledné seznamy
budou stejně setřı́děny metodou mergesort. Můžeme proto během jednoho průchodu projı́t všechny prvky
seznamu a střı́davě je dávat do dvou různých seznamů. Tı́m bude zajištěna časová složitost O(n), kde n je
délka seznamu. Implementace takové procedury by vypadala takto:

(define divide-list

(lambda (l)

(let divide

((l l)

(1st '())

(2nd '()))

(if (null? l)

(cons 1st 2nd)

(divide (cdr l) 2nd (cons (car l) 1st))))))

Pomocı́ pojmenovaného let jsme nadefinovali iterativnı́ proceduru divide. Jejı́ argumenty 1st a 2nd

majı́ roli střadačů. Při každém kroku přidáváme prvnı́ prvek seznamu l do jednoho z nich. Přitom je
stále zaměňujeme, všimněte si, že při rekurzivnı́m volánı́ předáváme proceduře divide střadač 2nd jako
argument 1st. Tı́m je dosaženo rovnoměrného rozdělenı́. Iterace končı́ po průchodu celým seznamem l,
pak vracı́me tečkový pár obou střadačů. Viz přı́klady aplikace této procedury.

(divide-list '()) Z=⇒ (())

(divide-list '(1)) Z=⇒ (() 1)

(divide-list '(1 2 3 5 6 7)) Z=⇒ ((6 4 2) 7 5 3 1)

Nynı́ tedy napı́šeme proceduru mergesort.

(define mergesort

(lambda (l . pred?)

(let ((<= (if (null? pred?) <= (car pred?))))

(if (or (null? l) (null? (cdr l)))

l

(let ((divided-list (divide-list l)))

(merge (mergesort (car divided-list) <=)

(mergesort (cdr divided-list) <=)

<=))))))

Jedná se o přı́mý přepis předpisu uvedeného výše. Procedurou mergesortmůžeme třı́dit libovolné seznam
na základě predikátu určujı́cı́ho uspořádánı́ jejich prvků. Viz následujı́cı́ přı́klady:

(mergesort '(2 5 3 4 1)) Z=⇒ (1 2 3 4 5)

(mergesort '(2 5 3 4 1) >=) Z=⇒ (5 4 3 2 1)

Následujı́cı́ přı́klad ukazuje třı́děnı́ seznamu racionálnı́ch čı́sel reprezenrovaných páry:

(map r->number

(mergesort (list (make-r 2 1)

(make-r 1 2)

(make-r 3 2))

r<=)) Z=⇒ (1 3/2 2)

8.6 Reprezentace polynomů

V této sekci se budeme věnovat reprezentaci polynomů a procedurami manipulujı́cı́mi s polynomy v této
reprezentaci. Jedná se o komplexnı́ přı́klad na kterém předvedeme práci se seznamy, použitı́ rekurze,
a použitı́ procedur apply, eval a akumulačnı́ch procedur představených v předcházejı́cı́ lekci.

Polynomy jedné proměnné budeme reprezentovat pomocı́ seznamů čı́sel. Čı́slo na pozici i v našı́ reprezen-
taci polynomu bude představovat koeficient členu stupně i. Tedy napřı́klad polynom

5x3 − 2x2 + 1
bude reprezentován seznamem (1 0 -2 5). Konstantnı́ polynomy jsou tedy reprezentovány jednoprvko-
vými čı́selnými seznamy, napřı́klad nulový a jednotkový polynom definujeme takto:

(define the-zero-poly '(0))

(define the-unit-poly '(1))

U reprezentacı́ polynomu přitom nepovolujeme nadbytečné nuly na konci seznamu. Třeba seznamy

(1 2 0), (1 2 0 0), (1 2 0 0 0)

nejsou reprezentacemi polynomů. Z tohoto pravidla vyjı́máme nulový polynom, který je vždy reprezento-
ván jednoprvkovým seznamem (0). Dále prázdný seznam nenı́ reprezentacı́ polynomu.

Prvnı́ procedurou v této sekci je konstruktor polynomu poly. Jedná se o proceduru libovolného počtu
argumentů. Těmito argumenty jsou čı́sla – koeficienty polynomu, seřazené vzestupně podle řádu. Procedura
polyvracı́ seznam reprezentujı́cı́ polynom se zadanými koeficienty. Činnost procedury si dobře uvědomı́me
na přı́kladech jejı́ aplikace:

(poly) Z=⇒ (0) reprezentace polynom 0
(poly 0) Z=⇒ (0) reprezentace polynom 0
(poly 0 0) Z=⇒ (0) reprezentace polynom 0
(poly -1 2) Z=⇒ (-1 2) reprezentace polynomu 2x− 1
(poly -1 0 2) Z=⇒ (-1 0 2) reprezentace polynomu 2x2 − 1
(poly -1 0 2 0 0 0) Z=⇒ (-1 0 2) reprezentace polynomu 2x2 − 1
(poly -1 0 2 0 0 0 3) Z=⇒ (-1 0 2 0 0 0 3) reprezentace polynomux 3x6 + 2x2 − 1

Nynı́ se podı́váme na implementaci této procedury:

(define poly

(lambda coeffs

(let ((result

(let cons-poly

((coeffs coeffs)

(zero-accum '()))

(cond ((null? coeffs) '())

((= (car coeffs) 0) (cons-poly (cdr coeffs)

(cons 0 zero-accum)))

(else (append zero-accum

(list (car coeffs))

(cons-poly (cdr coeffs) '())))))))

(if (null? result)

the-zero-poly

result))))

V těle procedury poly definujeme pomocı́ pojmenovaného let rekurzivnı́ proceduru cons-poly, která
bere dva argumenty. Prvnı́m argumentem je seznam zbývajı́cı́ch koeficientů coeffs. Druhý argument
je seznam zero-accum, který má funkci střadače. V tom si procedura cons-poly pamatuje počet nul
od poslednı́ho nenulového koeficientu. Důvodem, proč tento střadač potřebujeme, je to, že reprezentace
polynomu nesmı́ obsahovat nuly na konci. Pokud při zpracovánı́ seznamu koeficientů narazı́me na nulu,
nemůžeme v daném okamžiku vědět, jestli všechny dalšı́ koeficienty nebudou také nulové. Proto nulové
koeficienty střádáme do seznamu zero-accum a použijeme je až při nalezenı́ nenulového koeficientu.
Limitnı́ podmı́nkou procedury cons-poly je skutečnost, že je seznam nezpracovaných koeficientů prázdný.
Pokud je splněna, je vrácen prázdný seznam. V opačném přı́padě mohou nastat dvě situace. Bud’to je dalšı́
nezpracovaný koeficient (hlava seznamu coeffs) roven nule, a pak přidáme nulu do střadače zero-accum
a pokračujeme zpracovánı́m zbytku koeficientů procedurou cons-poly. Nebo je nenulový a v tom přı́padě
přidáme všechny nuly ze střadače zero-accum a tento nenulový koeficient do seznamu, který vznikne
aplikacı́ procedury cons-poly seznam coeffs bez prvnı́ho prvku a prázdný střadač.
Výsledkem aplikace procedury cons-poly bude seznam, který nemá na konci nuly. Může se však stát, že
tento seznam bude prázdný. To nastane v přı́padě, že všechny prvky seznamu coeffs budou čı́sla 0. Pokud
tedy bude výsledkem prázdný seznam, budeme vracet reprezentaci nulového polynomu.

Pomocı́ právě nadefinovaného konstruktoru poly a procedury apply můžeme vytvořit proceduru konver-
tujı́cı́ seznam na reprezentaci polynomu:

(define list->poly

(lambda (l) (apply poly l)))

Dále vytvořı́me predikáty constant-poly? a zero-poly?. Jejich definice je velice jednoduchá proto kód
nebudeme nijak komentovat. Tyto predikáty využijeme při implementaci dalšı́ch procedur v této sekci.

(define constant-poly?

(lambda (p)

(and (pair? p)

(number? (car p))

(null? (cdr p)))))

(define zero-poly?

(lambda (p)

(equal? p the-zero-poly)))

Nynı́ nadefinujeme proceduru poly-degree. Tato procedura bude pro zadaný polynom vracet jeho stupeň.

(define poly-degree

(lambda (p)

(if (constant-poly? p)

0

(+ 1 (poly-degree (cdr p))))))

Proceduru poly-degree jsme vytvořili jako jednoduchou rekurzivnı́ proceduru. Limitnı́ podmı́nkou je
konstantnost polynomu. Je-li této podmı́nky dosaženo, je výsledkem nula. V opačném přı́padě redukujeme
problém na přičtenı́ jedničky a nalezenı́ stupně polynomu, který vznikne z původnı́ho polynomu odebránı́m
prvnı́ho prvku z jeho reprezentace (což je v podstatě vydělenı́ polynomu polynomem x). Uvedený kód je
velice podobný programu 8.13 na straně 221, kde jsme implementovali proceduru zjišt’ujı́cı́ délku seznamu.

Procedura poly-degree tedy zjišt’uje stupeň polynomu, jak se můžeme přesvědčit na následujı́cı́ch ukáz-
kách použitı́ této procedury:

(poly-degree (poly)) Z=⇒ 0

(poly-degree (poly 0)) Z=⇒ 0

(poly-degree (poly 0 0)) Z=⇒ 0

(poly-degree (poly -1 2)) Z=⇒ 1

(poly-degree (poly -1 0 2)) Z=⇒ 2

(poly-degree (poly -1 0 2 0 0 0)) Z=⇒ 2

229



(poly-degree (poly -1 0 2 0 0 0 3)) Z=⇒ 5

Dalšı́ dvě procedury budou provádět násobenı́ polynomu speciálnı́mi polynomy. Konkrétně procedura
poly*xn bude násobit vstupnı́ polynom polynomem xn a procedura poly*c bude násobit vstupnı́ polynom
konstantnı́m polynomem c. Násobenı́m polynomem xn je vzhledem k našı́ reprezentaci přidánı́ n nul na
začátek seznamu. Následuje implementace procedury poly*xn provádějı́cı́ násobenı́ polynomem xn:

(define poly*xn

(lambda (p n)

(if (= n 0)

p

(cons 0 (poly*xn p (- n 1))))))

Proceduru poly*xn jsme tedy napsali jako rekurzivnı́ proceduru. Limitnı́ podmı́nkou je rovnost čı́sla n

nule. V takovém přı́padě vracı́me původnı́ polynom. V opačném přı́padě přidáváme nulu na začátek
reprezentace součinu polynomu s polynomem xn−1. Ve skutečnosti bychom měli ještě provádět test na
zjištěnı́, jestli polynom pnenı́ nulový a v takovém přı́padě vracet vždy nulový polynom. Úpravu necháváme
na čtenáři. Následujı́ přı́klady volánı́:

(poly*xn (poly 1 0 -2) 0) Z=⇒ (1 0 -2)

(poly*xn (poly 1 0 -2) 1) Z=⇒ (0 1 0 -2)

(poly*xn (poly 1 0 -2) 2) Z=⇒ (0 0 1 0 -2)

(poly*xn (poly 1 0 -2) 3) Z=⇒ (0 0 0 1 0 -2)

Násobenı́ polynomu konstantou c bude velice jednoduché. V přı́padě, že je tato konstanta rovna nule,
vracı́me nulový polynom. Jinak stačı́ mapovat proceduru násobenı́ konstantou na seznam reprezentujı́cı́
polynom. Definice takové procedury by vypadala takto:

(define poly*c

(lambda (p c)

(if (= c 0)

the-zero-poly

(map (lambda (x) (* c x)) p))))

Procedura poly*c tedy vracı́ polynom vynásobený konstantou:

(poly*c (poly 1 0 -2) 0) Z=⇒ (0)

(poly*c (poly 1 0 -2) 1) Z=⇒ (1 0 -2)

(poly*c (poly 1 0 -2) 2) Z=⇒ (2 0 -4)

(poly*c (poly 1 0 -2) 3) Z=⇒ (3 0 -6)

Dalšı́ část sekce budeme věnovat operacı́m nad polynomy. Přesněji budeme implementovat procedury
sčı́tánı́, odčı́tánı́ násobenı́ a dělenı́ polynomů. Nejdřı́ve tedy sčı́tánı́:

(define poly2+

(lambda (p1 p2)

(list->poly

(let add ((p1 p1)

(p2 p2))

(cond ((null? p1) p2)

((null? p2) p1)

(else (cons (+ (car p1) (car p2))

(add (cdr p1) (cdr p2)))))))))

V těle procedury poly2+ jsme vytvořili pomocı́ pojmenovaného let rekurzivnı́ proceduru add. Procedura
realizuje jednoduché sčı́tánı́ členů polynomů po odpovı́dajı́cı́ch si složkách (konstantnı́ člen prvnı́ho po-
lynomu je přičten ke konstantnı́mu členu druhého polynomu, stejně tak pro lineárnı́, kvadratický a dalšı́
členy). Jediným problémem, který je potřeba ošetřit, jsou obecně různé stupně sčı́taných polynomů, což se
projevı́ v různě dlouhých vstupnı́ch seznamech.

Limitnı́ podmı́nkou procedury poly2+ je skutečnost, že jeden ze vstupnı́ch seznamů je prázdný. V tom
přı́padě je vrácen druhý seznam. Prázdný seznam sice nenı́ reprezentacı́ polynomu, ale může se stát že při

rozkladu problému bude proceduře add předán prázdný seznam. Problém na nalezenı́ součtu polynomů
zde rozkládáme na problém nalezenı́ součtu jednoduššı́ch polynomů, které vzniknou z původnı́ch ode-
bránı́m prvnı́ho prvku jejich reprezentace (aplikacı́ procedury car), a přidánı́ součtu těchto prvnı́ch prvků
na začátek (aplikacı́ procedury cons). Výsledný seznam ale nemusı́ být platnou reprezentacı́ polynomu.
Napřı́klad budou-li vstupem procedury add reprezentace polynomů x+ 2 a −x− 1, tedy vstupem budou
seznamy (2 1) a (-1 -1), bude výsledkem jejı́ aplikace seznam (1 0). To je dvouprvkový seznam končı́cı́
nulou a tedy nenı́ platnou reprezentacı́ seznamu. Proto na výsledek, který vrátı́ procedura add, aplikujeme
ještě proceduru list->poly, kterou jsme nadefinovali výše. Viz přı́klady použitı́:

(poly2+ '(0) '(1 2 -3)) Z=⇒ (1 2 -3)

(poly2+ '(1 2 3) '(0)) Z=⇒ (1 2 3)

(poly2+ '(1 2 3) '(1 2 -3)) Z=⇒ (2 4)

(poly2+ '(1 2 3) '(-1 2 -3)) Z=⇒ (0 4)

(poly2+ '(1 2 3) '(-1 -2 -3)) Z=⇒ (0)

Uvedená implementace procedury poly2+ je na jednu stranu elegantnı́, ale druhou stranu nenı́ přı́liš
efektivnı́. Provádı́me v nı́ vlastně dva průchody seznamem. V prvnı́m ze dvou seznamů reprezentujı́cı́ch
polynomy vytvářı́me seznam součtů koeficientů. Druhý průchod je proveden aplikacı́ pomocné procedury
list->poly, kdy z takto vytvořeného seznamu vytvořı́me reprezentaci polynomu. Tyto dvě činnosti mů-
žeme provádět současně a tak kód zefektivnit. Program definice procedury poly2+ by pak mohl vypadat:

(define poly2+

(lambda (p1 p2)

(let ((result

(let add ((p1 p1)

(p2 p2)

(zero-accum '()))

(cond ((and (null? p1) (null? p2)) '())

((null? p1) (append zero-accum p2))

((null? p2) (append zero-accum p1))

(else (let ((sum (+ (car p1) (car p2))))

(if (= sum 0)

(add (cdr p1) (cdr p2) (cons 0 zero-accum))

(append zero-accum

(list sum)

(add (cdr p1) (cdr p2) '())))))))))

(if (null? result)

the-zero-poly

result))))

Tato definice poly2+ se lišı́ od předchozı́ předevšı́m v proceduře add, která nynı́ bere o jeden argument
navı́c. Tento argument zero-accum je pomocný a procedura add jej bude použı́vat k zapamatovánı́ si počtu
nul od poslednı́ho nenulového koeficientu, podobně jako tomu je u implementace procedury list->poly.
Procedura add je rekurzivnı́ a stejně jako v předchozı́ implementaci je limitnı́ podmı́nkou skutečnost, že je
jeden ze vstupnı́ch seznamů prázdný. V tom přı́padě vracı́me druhý z nich. V přı́padě, že limitnı́ podmı́nka
nebyla splněna, je vypočten součet prvnı́ch prvků vstupnı́ch seznamů. S tı́mto čı́slem pak procedura zacházı́
podobně jako procedura list->poly. Tedy pokud je součet nulový, je rekurzivně volána procedura add, se
seznamy bez prvnı́ch prvků a s pomocným seznamem zero-accum, do kterého přidáme nuly. Je-li naopak
součet nenulový, je procedura add aplikována na ocasy seznamu, ale jako argument zero-accum je předán
prázdný seznam. Nuly ze seznamu zero-accum a vypočtený součet prvnı́ch členů je připojen na začátek
seznamu, který bude výsledkem této aplikace procedury add. Výsledkem aplikace procedury add bude
vždy reprezentace polynomu, až na jednu výjimkou. Tou je prázdný seznam. A to vyřešı́me jednoduchou
podmı́nkou.

Jelikož je sčı́tánı́ polynomů monoidálnı́ operace, můžeme pomocı́ procedury poly2+ a akumulačnı́ pro-
cedury foldr nadefinovat proceduru na sčı́tánı́ libovolného množstvı́ polynomů:

(define poly+ (lambda polys (foldr poly2+ the-zero-poly polys)))

Tuto proceduru tedy můžeme použı́t ke sčı́tánı́ jakéhokoli počtu polynomů. Viz přı́klady aplikace:

(poly+) Z=⇒ (0)

(poly+ '(1 2 3)) Z=⇒ (1 2 3)

(poly+ '(1 2 3) '(10 20 30)) Z=⇒ (11 22 33)

(poly+ '(1 2 3) '(10 20 30) '(-11 0 -33)) Z=⇒ (0 22)

(poly+ '(1 2 3) '(10 20 30) '(-11 0 -33) '(0 -22)) Z=⇒ (0)

Dále můžeme pomocı́ procedury sčı́tánı́ dvou polynomů implementovat proceduru odčı́tánı́ jednoho po-
lynomu od druhého. Odečtenı́ polynomu je vlastně přičtenı́m opačného polynomu, ve smyslu vynásobenı́
konstantou −1. Implementace je tedy velice přı́močará:

(define poly-

(lambda (p1 p2)

(poly+ p1 (poly*c p2 -1))))

Procedura poly2* bude vracet součin dvou polynomů:

(define poly2*

(lambda (p1 p2)

(if (or (zero-poly? p1) (zero-poly? p2))

the-zero-poly

(let mul ((p1 p1)

(p2 p2))

(if (null? p1)

'()

(poly2+ (poly*c p2 (car p1))

(poly*xn (mul (cdr p1) p2) 1)))))))

Nejdřı́ve jsme vyřešili speciálnı́ přı́pad, tedy to, že alespoň jeden z polynomů byl nulový. K samotnému
násobenı́ jsme pak využili procedur poly2+, poly*c a poly*xn, které jsme nadefinovali výše v této sekci.

(poly2* '(0) '(1 2 3)) Z=⇒ (0)

(poly2* '(1 2 3) '(0)) Z=⇒ (0)

(poly2* '(1) '(1 2 3)) Z=⇒ (1 2 3)

(poly2* '(0 1) '(1 2 3)) Z=⇒ (0 1 2 3)

(poly2* '(-2 0 1) '(1 2 3)) Z=⇒ (-2 -4 -5 2 3)

Stejně jako sčı́tánı́ polynomů i násobenı́ polynomů je monoidálnı́ operace. Proto můžeme podobným způ-
sobem jako jsme dřı́ve implementovali proceduru poly+ vytvořit proceduru poly*, která bude počı́tat
součin libovolného počtu polynomů:

(define poly*

(lambda polys

(foldr poly2* the-unit-poly polys)))

A použı́vat ji s libovolným počtem argumentů:

(poly*) Z=⇒ (1)

(poly* '(1 2 3) '(0 1)) Z=⇒ (0 1 2 3)

(poly* '(1 2 3) '(0 1) '(0 1)) Z=⇒ (0 0 1 2 3)

(poly* '(1 2 3) '(0 1) '(0 1) '(2)) Z=⇒ (0 0 2 4 6)

Poslednı́ zbývajı́cı́ operacı́ je dělenı́ polynomů, představované procedurou poly/. Kód si uvedeme včetně
jednoduchých komentářů.

(define poly/

(lambda (p1 p2)

;; pomocná procedura: vrat’poslednı́ prvek neprázdného seznamu
(define last

(lambda (l)

(if (null? (cdr l))

(car l)

(last (cdr l)))))

;; vrat’ seznam (podı́l zbytek)
(let poly/

((p1 p1)

(p2 p2))

;; rozlišenı́ situace podle tvaru dělitele
(cond

;; nulový dělitel
((zero-poly? p2) 'divided-by-zero)

;; dělitel je konstantnı́ polynom
((= (poly-degree p2) 0) (list (poly*c p1 (/ 1 (car p2))) '()))

;; stupeň dělence je menšı́ než stupeň dělitele
((< (poly-degree p1) (poly-degree p2)) (list '() p1))

;; stupeň dělence je většı́ nebo roven stupni dělitele
(else (let* ((leader (poly*xn (list (/ (last p1) (last p2)))

(- (poly-degree p1) (poly-degree p2))))

(rest (poly/ (poly- p1 (poly* leader p2)) p2)))

(list (poly+ leader (car rest)) (cadr rest))))))))

Tato procedura nám tedy vracı́ dvouprvkový seznam, jehož prvnı́m prvkem je výsledek dělenı́ a druhým
prvkem je zbytek po tomto dělenı́. Neřešili jsme skutečnost, že tento výsledný seznam může obsahovat
prázdný seznam, což nenı́ reprezentace polynomu. Úpravu opět ponecháváme na čtenáři. Viz přı́klady
použitı́ procedury:

(poly/ '(1 2 3) '(0)) Z=⇒ „CHYBA: Pokus o dělenı́ nulou.“
(poly/ '(1 2 3) '(1)) Z=⇒ ((1 2 3) ())

(poly/ '(1 2 3) '(2)) Z=⇒ ((1/2 1 1 1/2) (0))

(poly/ '(1 2 3) '(10)) Z=⇒ ((1/10 1/5 3/10) (0))

(poly/ '(1 2 3) '(0 1)) Z=⇒ ((2 3) (1))

(poly/ '(1 2 3) '(0 2)) Z=⇒ ((1 1 1/2) (1))

(poly/ '(1 2 3) '(5 2)) Z=⇒ ((-11/4 1 1/2) (59/4))

(poly/ '(1 2 3) '(5 2 3)) Z=⇒ ((1) (-4))

(poly/ '(1 2 3) '(5 2 3 4)) Z=⇒ ((0) (1 2 3))

Pomocı́ procedury poly/ jednoduše definujeme procedury poly-quotient a poly-modulo na zjišt’ovánı́
podı́lu a zbytku po dělenı́:

(define poly-quotient

(lambda (p q)

(car (poly/ p q))))

(define poly-modulo

(lambda (p q)

(cadr (poly/ p q))))

Též můžeme napsat proceduru zjišt’ujı́cı́ hodnotu polynomu v daném bodě k, ta je totiž totožná se zbytkem
po dělenı́ polynomem k − x:

(define poly-value

(lambda (p k)

(car (poly-modulo p (poly (- k) 1)))))

Popřı́padě můžeme s využitı́m curryingu vytvářet z polynomů polynomiálnı́ funkce:

(define poly-function

(lambda (p)

(lambda (k)

(poly-value p k))))

Procedura poly-sexpr bude brát dva argumenty – polynom 〈p〉 a symbol 〈var〉 – a bude převádět poly-
nom na vyhodnotitelný S-výraz. Navı́c bude eliminovat násobenı́ nulou a upravovat násobenı́ jedničkou
a přičı́tánı́ nuly. Viz přı́klady aplikace této procedury:

(poly-sexpr '(0) 'x) Z=⇒ 0

(poly-sexpr '(2) 'x) Z=⇒ 2

(poly-sexpr '(0 1) 'x) Z=⇒ x

(poly-sexpr '(2 1) 'x) Z=⇒ (+ 2 x)

(poly-sexpr '(0 0 1) 'x) Z=⇒ (* x x)

(poly-sexpr '(0 1 1) 'x) Z=⇒ (+ x (* x x))

(poly-sexpr '(0 2 1) 'x) Z=⇒ (+ (* 2 x) (* x x))

(poly-sexpr '(3 0 1) 'x) Z=⇒ (+ 3 (* x x))

(poly-sexpr '(3 1 1) 'x) Z=⇒ (+ 3 x (* x x))

(poly-sexpr '(3 2 1) 'x) Z=⇒ (+ 3 (* 2 x) (* x x))

Nynı́ se podrobně podı́váme na následujı́cı́ definici této procedury:

(define poly-sexpr

(lambda (p var)

(if (constant-poly? p)

(car p)

(let* ((result

(let build ((p (cdr p))

(degree 1))

(cond ((null? p) '())

((= (car p) 0) (build (cdr p) (+ degree 1)))

(else

(cons (if (and (= (car p) 1) (= degree 1))

var

(append (if (= (car p) 1)

'(*)

(list '* (car p)))

(build-list degree

(lambda (i) var))))

(build (cdr p) (+ degree 1)))))))

(result (if (= (car p) 0) result (cons (car p) result))))

(cond ((null? result) 0)

((null? (cdr result)) (car result))

(else (cons '+ result)))))))

V těle procedury poly-sexpr jsme nejdřı́ve vyřešili speciálnı́ přı́pad polynomů, kterým je konstantnı́ poly-
nom. S-výraz odpovı́dajı́cı́ takovému polynomu je právě jediný prvek z jeho reprezentace. Nenı́-li polynom
konstantnı́, procházı́me procedurou build jeho reprezentaci – bez konstantnı́ho členu – a konstruujeme
podle nı́ nový seznam, který obsahuje mı́sto každého nenulového prvku 〈coef〉 seznam

(* 〈coef〉 〈var〉 · · · 〈var〉).

Tento seznam obsahuje tolikrát symbol 〈var〉, kolik je stupeň zpracovávaného členu polynomu. Stupeň ak-
tuálnı́ho polynomu si přitom předáváme pomocı́ argumentu degreeprocedury build. Toto „nahrazovánı́ “

má výjimku. Výjimkou jsou členy jejichž koeficient je 1, v takovém přı́padě do seznamu nedáváme 〈coef〉
(řešı́me násobenı́ jedničkou), pokud má navı́c tento člen stupeň 1, nevytvářı́me seznam, ale nahrazujeme
jej pouze symbolem 〈var〉. Na začátek takto vzniklého seznamu přidáme konstantnı́ člen polynomu, je-li
nenulový. Pokud je takto vzniklý seznam jednoprvkový, vracı́me jeho jediný prvek. V opačném přı́padě
vracı́me tento seznam s přidaným symbolem + na jeho začátku.

Poznámka 8.21. Ve zbytku této sekce budeme v přı́kladech použitı́ implementovaných procedur použı́vat
k výpisu reprezentace polynomů výsledky aplikace procedury poly-sexpr.

Pomocı́ této procedury můžeme napsat proceduru nalezenı́ hodnoty polynomu v daném bodě. Jednoduše
vytvořı́me let-blok, který bude vázat symbol x, a jehož těle bude S-výraz, který vytvořı́me pomocı́ pro-
cedury poly-sexpr, ze zadaného polynomu a symbolu x. Takto zkonstruovaný let-blok vyhodnotı́me
pomocı́ procedury eval. Následuje definice právě popsané procedury:

(define poly-sexpr-value

(lambda (p x)

(eval (list 'let (list (list 'x x))

(poly-sexpr p 'x)))))

Podobným způsobem jako jsme definovali proceduru poly-sexpr-value můžeme napsat i proceduru,
která polynom převádı́ na proceduru jednoho argumentu, která představuje polynomiálnı́ funkci. Princip
je stejný, jen výraz, který vznikne aplikacı́ poly-sexpr „neobalı́me“ let-blokem, ale λ-výrazem, a ten pak
vyhodnotı́me aplikacı́ procedury eval. Definice této procedury by pak vypadala takto:

(define poly-sexpr-function

(lambda (p)

(eval (list 'lambda (list 'x)

(poly-sexpr p 'x)))))

Na závěr této sekce naimplementujeme procedury symbolické derivace a symbolické integrace polynomů.
Nejdřı́ve tedy symbolická derivace realizovaná procedurou poly-diff:

(define poly-diff

(lambda (p)

(let ((result

(let diff ((p (cdr p))

(n 1))

(if (null? p)

p

(cons (* (car p) n)

(diff (cdr p) (+ n 1)))))))

(if (null? result)

the-zero-poly

result))))

V kódu definujeme přes pojmenovaný let rekurzivnı́ proceduru diff. Jejı́m prvnı́m argumentem je se-
znam, druhým je pak čı́tač, ve kterém si procedura pamatuje kolikátá jejı́ aplikace právě probı́há. Toto
čı́slo vlastně souhlası́ s aktuálně zpracovávaným prvkem seznamu reprezentujı́cı́ho polynom a tedy také se
stupněm odpovı́dajı́cı́ho členu tohoto polynomu. Výsledkem aplikace procedury diff na čı́selný seznam
je seznam, jehož prvky jsou prvky původnı́ho seznamu vynásobené jejich pozicemi (tentokrát výjimečně
počı́taných od 1). Této proceduře ale nepředáváme přı́mo derivovaný polynom, ale jeho reprezentaci bez
prvnı́ho prvku (odebránı́ prvnı́ho prvku z reprezentace polynomu, znamená odebránı́ konstantnı́ho členu
tohoto polynomu a snı́ženı́ stupně ostatnı́ch prvků). Výsledek aplikace procedury diff na takový seznam,
je-li neprázdný, odpovı́dá reprezentaci polynomu, který je derivacı́ původnı́ho. Pokud je vrácen prázdný
seznam (což se stane v přı́padě, že jsme derivovali konstantnı́ polynom), vracı́me nulový polynom. Viz
aplikace této procedury:

(poly-sexpr (poly-diff '(5)) 'x) Z=⇒ 0

(poly-sexpr (poly-diff '(5 2)) 'x) Z=⇒ 2

(poly-sexpr (poly-diff '(5 2 4)) 'x) Z=⇒ (+ 2 (* 8 x))

(poly-sexpr (poly-diff '(0 0 0 -7)) 'x) Z=⇒ (* -21 x x)

(poly-sexpr (poly-diff '(5 2 4 -7)) 'x) Z=⇒ (+ 2 (* 8 x) (* -21 x x))

Poslednı́ proceduru, kterou napı́šeme v této sekci, je procedura na symbolickou integraci polynomu. V jejı́m
těle definujeme proceduru int. Ta procházı́ seznam podobně jako procedura diff, jen mı́sto seznamu
násobků prvků původnı́ho seznamu jejich pozicemi (bráno od 1), vracı́ seznam podı́lů prvků a jejich pozic.
Jako konstantnı́ člen pak přidáváme pro jednoduchost nulu. Následuje definice této procedury:

(define poly-int

(lambda (p)

(let ((result

(let int ((p p)

(n 1))

(if (null? p)

p

(cons (/ (car p) n)

(int (cdr p) (+ n 1)))))))

(if (zero-poly? result)

the-zero-poly

(cons 0 result)))))

Viz aplikace procedury poly-int:

(poly-sexpr (poly-int '(5)) 'x) Z=⇒ (* 5 x)

(poly-sexpr (poly-int '(5 2)) 'x) Z=⇒ (+ (* 5 x) (* x x))

(poly-sexpr (poly-int '(5 2 4)) 'x) Z=⇒ (+ (* 5 x) (* x x) (* 4/3 x x x))

(poly-sexpr (poly-int '(0 0 0 -7)) 'x) Z=⇒ (* -7/4 x x x x)

(poly-sexpr (poly-int '(5 2 4 -7)) 'x) Z=⇒ (+ (* 5 x) (* x x) (* 4/3 x x x)

(* -7/4 x x x x))

Shrnutı́

V této lekci jsme se zabývali rekurzivnı́mi procedurami a výpočetnı́mi procesy generovanými těmito pro-
cedurami. Nejprve jsme uvedli rekurzi a princip indukce přes přirozená čı́sla. Dále jsme se zabývali struktu-
rálnı́ rekurzı́ a strukturálnı́ indukcı́. Ukázali jsme také obecné principy rekurze a indukce. Principy rekurze
a indukce byly v obou přı́padech (rekurze přes čı́sla a přes seznamy) těsně vázané na strukturálnı́ vlastnosti
prvků množin, se kterými jsme pracovali. Pro každé nezáporné celé čı́slo jsme vždy mohli udělat úvahu,
že bud’ je čı́slo rovno nule, nebo je následnı́kem nějakého jiného čı́sla. Stejně tak každý seznam je bud’
prázdný, nebo vzniká z jiného seznamu připojenı́m nového prvnı́ho prvku. Pomocı́ rekurze jsme definovali
zobrazenı́ mezi množinami čı́sel a seznamů. Pomocı́ indukce jsme prokazovali správnost a vlastnosti takto
zavedených zobrazenı́. Dále jsme ze zabývali rekurzı́ a indukcı́ z pohledu programovánı́. Představili jsme
rekurzivnı́ procedury jako procedury aplikujı́cı́ sebe sama. Každá rekurzivnı́ procedura, kterou jsme uva-
žovali, měla svůj rekurzivnı́ předpis (předpisy) a limitnı́ podmı́nku (podmı́nky). Vytvářenı́ rekurzivnı́ch
procedur je přı́močaré, pokud si dobře uvědomı́me, jak má vypadat limitnı́ podmı́nka (vztahujı́cı́ se ke
krajnı́mu přı́padu) a jak vypadá rekurzivnı́ předpis, který vyjadřuje redukci problému na problém (nebo
několik problémů) o menšı́m rozsahu. Dále jsme zjistili, že rekurzivnı́ procedury generujı́ různé výpočetnı́
procesy. Zabývali jsme se třemi typy procesů generovaných rekurzivnı́mi procedurami, lineárně rekurziv-
nı́m výpočetnı́m procesem, lineárně iterativnı́m výpočetnı́m procesem a stromově rekurzivnı́m výpočetnı́m
procesem. U každého z nich jsme uvedli několik metod stanovenı́ složitosti výpočetnı́ho procesu. Pro efek-
tivnı́ prováděnı́ iterativnı́ho procesu jsme upravili aplikaci uživatelsky definovaných procedur vzhledem
k aplikaci z koncových pozic. Ukázali jsme rovněž užitečný aparát, pomocı́ nějž je možné vytvářet jed-
norázově aplikovatelné rekurzivnı́ procedury. Lekci jsme uzavřeli sériı́ přı́kladů rekurzivnı́ch procedur
pracujı́cı́ch se seznamy.

Pojmy k zapamatovánı́

• princip indukce, princip rekurze, aplikace sebe sama,
• rekurzivnı́ definice, rekurzivnı́ definice zobrazenı́,
• matematická indukce, indukce přes přirozená čı́sla,
• strukturálně jednoduššı́ seznamy, strukturálnı́ indukce, strukturálnı́ rekurze,
• čistý funkcionálnı́ jazyk,
• rekurzivnı́ procedura, rekurzivnı́ aplikace procedury,
• limitnı́ podmı́nka rekurze, předpis rekurze, série aplikacı́,
• fáze navı́jenı́, odložený výpočet, deferred computation,
• dosaženı́ limitnı́ podmı́nky, fáze odvı́jenı́,
• dekompozice, divide et impera, rozděl a panuj,
• rekurzivnı́ výpočetnı́ proces, lineárnı́ rekurzivnı́ výpočetnı́ proces,
• koncová pozice, koncová aplikace, koncová rekurze, koncově rekurzivnı́ procedura,
• optimalizace na koncovou rekurzi, tail recursion optimization,
• degenerovaná fáze odvı́jenı́, lineárnı́ iterativnı́ výpočetnı́ proces, cyklus,
• lineárně rekurzivnı́ procedura, iterativnı́ procedura,
• čı́tač, střadač,
• jednorázová aplikace,
• simulace navı́jenı́ a odvı́jenı́ pomocı́ zásobnı́ku,
• stromově rekurzivnı́ výpočetnı́ proces.

Nově představené prvky jazyka Scheme

• speciálnı́ forma: pojmenovaný let.

Kontrolnı́ otázky

1. K čemu sloužı́ princip rekurze?
2. K čemu byste použili indukci?
3. Jak lze dokázat správnost rekurzivnı́ch definic?
4. Jaký je rozdı́l mezi matematickou indukcı́ a strukturálnı́ indukcı́?
5. Co to znamená, že jeden seznam je strukturálně jednoduššı́ než jiný seznam?
6. Je Scheme čistý funkcionálnı́ jazyk?
7. Co jsou to rekurzivnı́ procedury?
8. Co řı́ká princip „divide et impera“?
9. Jaký je rozdı́l mezi rekurzivnı́mi procedurami a procedurami, se kterými jsme pracovali v předchozı́ch lekcı́ch?

10. Jaký je rozdı́l mezi rekurzivnı́ procedurou a rekurzivnı́m výpočetnı́m procesem?
11. Jaké typy rekurzivnı́ch výpočetnı́ch procesů znáte?
12. Co je charakteristické pro lineárně rekurzivnı́ výpočetnı́ proces?
13. Jak je definována koncová pozice?
14. Co je to koncová aplikace a koncově rekurzivnı́ procedura?
15. Jak je ve většině programovacı́ch jazyků realizována iterace?
16. Má každý lineárně iterativnı́ výpočetnı́ proces lineárnı́ časovou složitost?
17. Co jsou a k čemu sloužı́ čı́tače?
18. Co jsou a k čemu sloužı́ střadače?
19. V jakých fázı́ch probı́hajı́ jednotlivé rekurzivnı́ výpočetnı́ procesy?
20. Z jakého důvodu je v rekurzivnı́ch procedurách přı́tomná limitnı́ podmı́nka?
21. Co máme na mysli pod pojmem jednorázová aplikace rekurzivnı́ procedury?

22. Jakým způsobem lze vždy nahradit rekurzivnı́ proceduru iterativnı́ procedurou?
23. U kterých výpočetnı́ch procesů roste exponenciálně počet prostředı́ vzniklých během aplikace?
24. Který z rekurzivnı́ch výpočetnı́ch procesů je možné snadno „zastavit“ a „rozběhnout“?

Cvičenı́

1. Naprogramujte proceduru perrin jednoho argumentu 〈n〉, která vracı́ 〈n〉-tý člen Pn Perinovy po-
sloupnosti. Perinova posloupnost je definována následujı́cı́m předpisem:

Pn =


3 pokud n = 0,
0 pokud n = 1,
2 pokud n = 2,
Pn−2 + Pn−3 jinak.

Proceduru implementujte

(a) jako rekurzivnı́ proceduru

(b) jako iterativnı́ proceduru

(c) pomocı́ pojmenovaného let

2. Napište rekurzivnı́ proceduru vracejı́cı́ aritmetický průměr čı́sel v seznamu.
3. Napište rekurzivnı́ verzi procedury list-indices, kterou jsme definovali v programu 6.5.
4. Implementujte Euklidův algoritmus na výpočet největšı́ho společného dělitele.
5. Naprogramuje proceduru reverse pomocı́ pojmenovaného let.
6. Doplňte sadu procedur pro práci s polynomy ze sekce 8.6 o:

(a) predikát poly?, který zjišt’uje, zda je jeho argument reprezentacı́ seznamu

(b) konstruktor poly-roots libovolného počtu argumentů vytvářejı́cı́ polynom podle výčtu všech
jeho kořenů.

(c) proceduru poly-gcd, která počı́tá největšı́ společný dělitel dvou polynomů, a která bude imple-
mentaci Euklidova algoritmu na polygomech.

Úkoly k textu

1. Zamyslete se, proč prvnı́ argument speciálnı́ formy if nenı́ v koncové pozici. Co by nefungovalo?

2. Určete složitosti procedur uvedených v sekci 8.6.

3. Napište proceduru pascal, která přijı́má dva čı́selné argumenty 〈x〉, 〈y〉 a vracı́ prvek pascalova
trojúhelnı́ka na pozici (〈x〉, 〈y〉). Viz přı́klady volánı́:

(pascal 0 0) Z=⇒1 (pascal 5 1) Z=⇒5 (pascal 5 3) Z=⇒10 (pascal 7 3) Z=⇒35

Určete složitost vašeho řešenı́.

Řešenı́ ke cvičenı́m

1. (define perrin

(lambda (n)

(cond ((= n 0) 3)

((= n 1) 0)

((= n 2) 2)

(else (+ (perrin (- n 2))

(perrin (- n 3)))))))

(a) (define perrin

(lambda (n)

(define iter

(lambda (a b c i)

(if (<= i 0)

a

(iter b c (+ a b) (- i 1)))))

(iter 3 0 2 n)))

(b) (define perrin

(lambda (n)

(let iter ((a 3) (b 0) (c 2) (i n))

(if (<= i 0)

a

(iter b c (+ a b) (- i 1))))))

2. (define arit-means

(lambda numbers

(if (null? numbers)

#f

(let iter ((l numbers)

(accum 0)

(nr 0))

(if (null? l)

(/ accum nr)

(iter (cdr l) (+ accum (car l)) (+ nr 1)))))))

3. (define list-indices

(lambda (l elem)

(let find-occurrs

((l l)

(i 0))

(cond ((null? l) '())

((equal? elem (car l)) (cons i (find-occurrs (cdr l) (+ i 1))))

(else (find-occurrs (cdr l) (+ i 1)))))))

4. (define gcd

(lambda (a b)

(if (= b 0)

a

(gcd b (modulo a b)))))

5. (define reverse

(lambda (l)

(let iter ((l l)

(accum '()))

(if (null? l)

accum

(iter (cdr l) (cons (car l) accum))))))

6. (a) (define poly?

(lambda (p)

(or (constant-poly? p)

(and (pair? p)

(let poly-test ((p p)

(last-zero? #f))

(if (null? p)

(not last-zero?)

(if (and (pair? p) (number? (car p)))

(poly-test (cdr p) (= (car p) 0))

#f)))))))

(b) (define poly-roots

(lambda roots

(if (null? roots)

the-unit-poly

(poly* (poly (- (car roots)) 1)

(apply poly-roots (cdr roots))))))

(c) (define poly-gcd

(lambda (p1 p2)

(let ((m (poly-modulo p1 p2)))

(if (zero-poly? m)

p2

(poly-gcd p2 m)))))

240



Lekce 9: Hloubková rekurze na seznamech

Obsah lekce: V této lekci pokračujeme v problematice rekurzivnı́ch procedur. Nejprve uvedeme několik
metod zastavenı́ rekurze. Dále ukážeme, že rekurzivnı́ procedury je možné vytvářet i bez speciálnı́ formy
define. V dalšı́ části lekce představı́me speciálnı́ formu letrec jakožto dalšı́ variantu formy let umož-
nujı́cı́ definovat lokálnı́ rekurzivnı́ procedury. V poslednı́ části se budeme věnovat hloubkové rekurzi na
seznamech a hloubkovým akumulačnı́m procedurám.

Klı́čová slova: hloubková rekurze na seznamech, speciálnı́ forma letrec, y-kombinátor, zastavenı́ rekurze.

9.1 Metody zastavenı́ rekurze

V předchozı́ lekci jsme uvedli množstvı́ definic rekurzivnı́ch procedur. Limitnı́ podmı́nku rekurzivnı́ch
procedur a výraz vyhodnocený po jejı́m dosaženı́ jsme přitom vždy vyjadřovali pomocı́ speciálnı́ch forem
if nebo cond. Tyto speciálnı́ formy jsme tedy v rekurzivnı́ch programech použı́vali k „zastavenı́ rekurze“,
to jest k zastavenı́ postupných aplikacı́ téže procedury. Existujı́ však i dalšı́ metody, kterými lze zastavit
rekurzi. V této kapitole se budeme věnovat právě těmto metodám, a přı́kladům jejich použitı́.

Rekurzi je možné zastavit

• pomocı́ speciálnı́ch forem if a cond
Tato možnost byla bohatě prezentována v předcházejı́cı́ lekci. Nynı́ jen pro srovnánı́ s dalšı́m bodem,
uvedeme (bez dalšı́ho komentáře) vlastnı́ implementaci predikátu list?, který zjišt’uje, zda je jeho
argument seznam:

(define list?

(lambda (l)

(if (null? l)

#t

(and (pair? l)

(list? (cdr l))))))

• pomocı́ speciálnı́ch forem and a or
V definicı́ch 2.22 a 2.23 na stranách 64 a 66 jsme popisovali, jak probı́há aplikace těchto speciálnı́ch
forem. V obou přı́padech je pro nás důležitým rysem to, že tyto speciálnı́ formy vyhodnocujı́ své argu-
menty sekvenčně jeden po druhém. Navı́c platı́, že vyhodnocenı́ každého z argumentů je podmı́něno
výsledkem vyhodnocenı́ předcházejı́cı́ho argumentu. Napřı́klad pokud se jeden z argumentů pro and

vyhodnotı́ na „nepravda“, dalšı́ argumenty již nebudou vyhodnocovány. Analogické situace platı́ pro
or v přı́padě, kdy se argument vyhodnotı́ na „pravda“, viz lekci 2.

Jako přı́klad použitı́ těchto speciálnı́ch forem k zastavenı́ rekurze uved’me nejdřı́ve predikát list?,
který jsme definovali předchozı́m bodě pomocı́ speciálnı́ formy if:

(define list?

(lambda (l)

(or (null? l)

(and (pair? l)

(list? (cdr l))))))

V těle λ-výrazu máme tedy použitu speciálnı́ formu or. Jejı́m prvnı́m argumentem (null? l). Po
vyhodnocenı́ tohoto prvnı́ho argumentu máme dvě možnosti. Bud’to se vyhodnotil na pravdu (seznam
navázaný na l je prázdný), a pak je pravda výsledkem aplikace speciálnı́ formy or a také výsledkem
aplikace predikátu list?, nebo se vyhodnotil na nepravdu (seznam navázaný na l je neprázdný) a pak
pokračujeme vyhodnocenı́m dalšı́ho argumentu. Tı́m je seznam

(and (pair? l) (list? (cdr l)))

a protože se jedná o poslednı́ argument, bude výsledek jeho vyhodnocenı́ také výsledkem aplikace

speciálnı́ formy or a tedy i aplikace predikátu list?. Tato část kódu už je shodná s alternativnı́ větvı́
speciálnı́ formy if v programu uvedeném v předchozı́m bodě.

Dále můžeme napsat třeba efektivnı́ verze predikátů forall a exists, které jsme poprvé definovali
v lekci 6. Jejich definice už nebudeme podrobně popisovat, jedná se o podobný přepis jako v přı́padě
predikátu list?:

(define forall

(lambda (f l)

(or (null? l)

(and (f (car l))

(forall f (cdr l))))))

(define exists

(lambda (f l)

(and (not (null? l))

(or (f (car l))

(exists f (cdr l))))))

Na prvnı́ pohled by se mohlo zdát, že zastavenı́ rekurze využı́vajı́cı́ speciálnı́ formy and a or lze
použı́t pouze při programovánı́ predikátů (procedur vracejı́cı́ch jako výsledky aplikace pravdivostnı́
hodnoty). Jako protipřı́klad můžeme uvést definici procedury fak na výpočet faktoriálu:

(define fak

(lambda (n)

(or (and (= n 0) 1)

(* n (fak (- n 1))))))

Zatı́mco u předchozı́ch přı́kladů na zastavenı́ rekurze pomocı́ speciálnı́ch forem and a or, které byly
definicemi predikátů, je tato definice poněkud nepřehledná. Třeba v přı́padě predikátu list? jsme
mohli kód intuitivně čı́st jako: „Argument l je seznam, pokud je to prázdný seznam nebo je to pár a
současně druhý prvek tohoto páru je zase seznam.“ U definice procedury fak toto provést nelze. Při
programovánı́ je kromě samotné funkčnosti programu potřeba dbát i na jeho čitelnost a výše uvedené
rekurzivnı́ procedura počı́tajı́cı́ faktoriál, která zastavuje rekurzi pomocı́ or je spı́š odstrašujı́cı́ přı́klad.

• i když „zdánlivě nemá limitnı́ podmı́nku“.
V některých přı́padech je limitnı́ podmı́nka rekurze v proceduře „skrytá“, to jest je obsažena až „hlouběji
v těle“ procedury. Jako přı́klad uvažujme třeba proceduru depth-map, která pro zadanou proceduru
a seznam, jehož prvky mohou být opětně seznamy, vytvořı́ seznam, který má stejnou strukturu jako
původnı́ seznam, ale jeho prvku jsou výsledky aplikace zadané procedury na původnı́ prvky seznamu.
Tedy napřı́klad:

(depth-map - '(1 (2 () (3)) (((4))))) Z=⇒ (-1 (-2 () (-3)) (((-4))))

Definici takto popsané procedury bychom mohli napsat napřı́klad takto:

(define depth-map

(lambda (f l)

(map (lambda (x)

(if (list? x)

(depth-map f x)

(f x)))

l)))

Procedura na každý prvek daného seznamu aplikuje bud’to sama sebe nebo zadanou proceduru – to
podle toho, jestli se jedná o seznam. K rekurzivnı́ aplikaci „sebe sama“ tedy nedojde, pokud seznam
nebude obsahovat dalšı́ seznamy. Přı́má formulace této limitnı́ podmı́nko ovšem nikde v kódu nenı́,
ale je jaksi rozložena v použitı́ procedury map a v jejı́m prvnı́m argumentu. Dalšı́m přı́kladům na tento
typ limitnı́ podmı́nky se budeme věnovat v sekci 9.5.

Jak již jsme v předchozı́ lekco naznačili, teoreticky máme rovněž možnost „nezastavovat rekurzi“, to jest
psát rekurzivnı́ procedury bez limitnı́ podmı́nky. Zatı́m to ale nemá přı́liš velký smysl. Programy pak
generujı́ nekonečný výpočetnı́ proces, který sestává z nekončı́cı́ série aplikacı́ jedné procedury. Třeba
následujı́cı́ kód je definicı́ procedury, jejı́ž aplikace způsobı́ nekončı́cı́ sérii aplikacı́:

(define loop-to-infinity

(lambda ()

(loop-to-infinity)))

Při pokusu aplikovat loop-to-infinity (bez argumentu) dojde k „zacyklenı́ výpočtu“.

Již v lekci 1 jsme naznačili, že bychom si s konceptem „if jako procedura“ nevystačili. Hlavnı́m důvodem
k tomuto tvrzenı́ je fakt, že if jako procedura nezastavı́ rekurzi. Přı́činou by právě byl jejı́ procedurálnı́ cha-
rakter, který způsobuje, že před samotnou aplikacı́ docházı́ k vyhodnocenı́ všech předaných argumentů.
Vyhodnoceny by tedy byly obě „větvě“ rekurzivnı́ procedury (vžraz následujı́cı́ za limitnı́ podmı́nkou
i předpis rekurze) bez ohledu na výsledek vyhodnocenı́ podmı́nky. Při použitı́ „if jako procedury“ k za-
stavenı́ rekurze ale jedna z větvı́ obsahuje rekurzivnı́ volanı́ procedury, a tak docházı́ k nekonečné smyčce
aplikacı́ této procedury. Viz následujı́cı́ přı́klad, který zachycuje pokus o vytvořenı́ procedury na výpočet
faktoriálu pomocı́ „if jako procedury“:

(define if-proc

(lambda (condition expr altex)

(if condition expr altex)))

(define fak-loop

(lambda (n)

(if-proc (= n 1)

1

(* n (fak-loop (- n 1))))))

Tı́mto způsobem definovaná procedura fak-loop bude vždy cyklit.

9.2 Rekurzivnı́ procedury definované pomocı́ y-kombinátoru

V lekci 2 jsme vysvětlili vznik uživatelských procedur. Uživatelské procedury vznikajı́ vyhodnocovánı́m λ-
-výrazů a každou uživatelskou proceduru lze chápat jako trojici hodnot: seznam argumentů, tělo, a prostředı́
vzniku. Připomeňme, že na vzniku procedur se nijak nepodı́lı́ speciálnı́ forma define. V této a předchozı́
lekci jsme vytvářeli procedury, které ve svém těle aplikovaly samy sebe. Tento typ „sebeaplikace“ bylo
možné provést, protože symbol, na který byla procedura navázána pomocı́ define, se nacházel v prostředı́
jejı́ho vzniku. Na prvnı́ pohled se tedy může zdát, že v přı́padě rekurzivnı́ch procedur hraje define

významnou roli. Tato role sice nespočı́vá v samotném vytvořenı́ procedury (tu má pořád na starost speciálnı́
formalambda), ale v umožněnı́ odkázat se z těla procedury na sebe sama prostřednictvı́m hodnoty navázané
na symbol (jméno procedury) pomocı́ define. V této kapitole ukážeme, že define nenı́ z pohledu aplikace
rekurzivnı́ch procedur potřeba, což může být jistě pro řadu čtenářů překvapujı́cı́ závěr.

Procedury vytvořené vyhodnocenı́m λ-výrazů mohou být přı́mo aplikovány s danými argumenty. Na-
přı́klad vyhodnocenı́ výrazu ((lambda (x) (* 2 x)) 10) vede na jednorázovou aplikaci procedury
jež vynásobı́ svůj argument s dvojkou. Otázkou je, zda-li jsme schopni definovat „jednorázovou rekur-
zivnı́ proceduru“, aniž bychom provedli definici vazby pomocı́ speciálnı́ formy define. Je zřejmé, že
pokud má být daná procedura rekurzivnı́, musı́ mı́t k dispozici „sebe sama“ prostřednictvı́m vazby
některého symbolu. Na prvnı́ pohled by se tedy použitı́ define mohlo zdát jako nevyhnutelné.

Našim cı́lem tedy bude vytvořit pojmenovanou rekurzivnı́ proceduru bez použitı́ define. Nejprve si
ukažme, kudy cesta nevede (a proč). Jako modelovou proceduru si vezmeme rekurzivnı́ proceduru na

výpočet faktoriálu. Jelikož je našı́m úkolem tuto proceduru vytvořit jako pojmenovanou, leckoho možná
napadne „pouze nahradit define pomocı́ let“ následujı́cı́m způsobem:

(let ((fak (lambda (n)

(if (= n 0)

1

(* n (fak (- n 1)))))))

(fak 6))

Vyhodnocenı́ předchozı́ho kódy však končı́ chybou, jejı́ž vznik by nám měl být v tuto chvı́li jasný. Předchozı́
kód je totiž ekvivalentnı́ programu:

((lambda (fak)

(fak 6))

(lambda (n)

(if (= n 0)

1

(* n (fak (- n 1))))))

Symbol fak, který se nacházı́ v těle procedury vzniklé vyhodnocenı́m λ-výrazu (lambda (n) · · · ) zřejmě
nemá žádný vztah k symbolfak, který je vázaný v λ-výrazu(lambda(fak) · · · ). Uvědomte si, že procedura
vytvořená vyhodnocenı́m λ-výrazu (lambda (n) · · · ) vznikla v globálnı́m prostředı́ (kde fak nemá vazbu).
Z naprosto stejného důvodu by chybou skončil i následujı́cı́ program, který se od předchozı́ho lišı́ použitı́m
let* mı́sto let (vzhledem k tomu, že v přı́padě vytvořenı́ jedné vazby se let* a let shodujı́ se navı́c jedná
o týž program jako v předchozı́m přı́padě).

(let* ((fak (lambda (n)

(if (= n 0)

1

(* n (fak (- n 1)))))))

(fak 6))

Problém zavedenı́ rekurzivnı́ procedury bez define vyřešı́me tak, že uvažované rekurzivnı́ proceduře
předáme sebe sama prostřednictvı́m nového argumentu. Napřı́klad v přı́padě procedury pro výpočet faktoriálu
by situace vypadala následovně:

(lambda (fak n)

(if (= n 0)

1

(* n (fak fak (- n 1)))))

Všimněte si, že pokud proceduru vytvořenou vyhodnocenı́m předchozı́ho λ-výrazu aplikujeme s prvnı́m
argumentem jı́mž bude ta sama procedura, pak bude zcela legitimně probı́hat rekurzivnı́ volánı́, protože
v těle procedury bude na symbol fak, který je nynı́ jednı́m z argumentů, navázána právě volaná procedura.
Pochopitelně, při rekurzivnı́m volánı́ musı́ být procedura opět předávána, což se v kódu promı́tlo do tvaru
volánı́ (fak fak (- n 1)). Nynı́ tedy zbývá vyřešit poslednı́ problém, jak zavolat předchozı́ proceduru tak,
aby na svém prvnı́m argumentu byla tatáž procedura navázána. K vyřešenı́ tohoto problému použijeme
takzvaný y-kombinátor. Předtı́m než jej obecně popı́šeme si všimněte, že pokud se nám podařı́ následujı́cı́
proceduru

(lambda (y)

(y y 6))

aplikovat s argumentem jı́mž bude procedura vzniklá vyhodnocenı́m (lambda (fak n) · · · ), pak v těle
procedury vzniklé vyhodnocenı́m (lambda (y) (y y 6)) proběhne aplikace procedury vzniklé vyhodno-
cenı́m (lambda (fak n) · · · ), přitom na prvnı́m argumentu bude navázána právě aplikovaná procedura
a druhým argumentem bude čı́slo 6, což přesně povede na požadovaný rekurzivnı́ výpočet. Výše popsanou
aplikaci však můžeme provést jednoduše spojenı́m předchozı́ch dvou částı́ dohromady tak, jak to ukazuje
program 9.1.

Program 9.1. Rekurzivnı́ procedura pro výpočet faktoriálu aplikovaná pomocı́ y-kombinátoru.

((lambda (y)

(y y 6))

(lambda (fak n)

(if (= n 0)

1

(* n (fak fak (- n 1))))))

Kód v programu 9.1 tedy způsobı́ aplikaci rekurzivnı́ procedury pro výpočet faktoriálu jejı́mž prvnı́m
argumentem je samotná procedura pro výpočet faktoriálu a druhým argumentem je hodnota, pro kterou
chceme faktoriál počı́tat (v tomto konkrétnı́m přı́padě hodnota 6). Zápis předchozı́ho kódy bychom mohli
zjednodušit pomocı́ speciálnı́ formy let následovně:

(let ((y (lambda (fak n)

(if (= n 0)

1

(* n (fak fak (- n 1)))))))

(y y 6))

Principiálně se ale jedná o totéž jako v předchozı́m přı́padě.

Nynı́ můžeme obecně popsat y-kombinátor a jeho použitı́ při zavedenı́ rekurzivnı́ch procedur. Pod pojmem
y-kombinátor máme na mysli právě λ-výraz v následujı́cı́m tvaru:

(lambda (y)

(y y 〈argument1〉 〈argument2〉 · · · 〈argumentn〉))

Argumenty 〈argument1〉 · · · 〈argumentn〉 v předchozı́m výrazu reprezentujı́ argumenty, které chceme předat
volané (rekurzivnı́) proceduře navázané na symbol y. Prvnı́m předaným argumentem je ovšem hodnota
navázaná na y, tedy samotná procedura. Proceduru vzniklou vyhodnocenı́m předchozı́ho λ-výrazu (y-
-kombinátoru) zavoláme s jediným argumentem jı́mž bude procedura n + 1 argumentů. y-kombinátor je
tedy část programu, která je odpovědná za aplikaci rekurzivnı́ procedury, konkrétně za aplikaci procedury
spojené s předánı́m prvnı́ho argumentu jı́mž je sama procedura. Následujı́cı́ program demonstruje použitı́
y-kombinátoru při aplikaci rekurzivnı́ procedury dvou argumentů (spojenı́ dvou seznamů):

((lambda (y)

(y y list-a list-b))

(lambda (append2 s1 s2)

(if (null? s1)

s2

(cons (car s1) (append2 append2 (cdr s1) s2)))))

Předchozı́ kód provede spojenı́ seznamů navázaných na symbolech list-a a list-b.

V předchozı́ch přı́kladech jsme pomocı́ y-kombinátoru provedli vždy jen jednu aplikaci procedury. Nic
ale nebránı́ tomu, abychom rekurzivnı́ proceduru vytvořenou pomocı́ y-kombinátoru lokálně pojmenovali
prostřednictvı́m vazby vytvořené speciálnı́ formou let a pak ji použili opakovaně. Viz následujı́cı́ přı́klad:

(let ((append2

(lambda (list-a list-b)

(let ((y (lambda (append2 s1 s2)

(if (null? s1)

s2

(cons (car s1)

(append2 append2 (cdr s1) s2))))))

(y y list-a list-b)))))

· · · výrazy · · ·
(append2 '(1 2 3) '(a b c))

· · · výrazy · · ·)

Lokálnı́ definicı́ rekurzivnı́ch procedur se budeme zabývat i v dalšı́ části této lekce. Praktické použitı́
y-kombinátorů ukážeme v lekci 12.

Pozornı́ čtenáři si jistě pamatujı́, že v sekci 2.5 jsme uvedli následujı́cı́ kód

((lambda (y) (y y)) (lambda (x) (x x))),

který rovněž způsobı́ nekončı́cı́ sérii aplikacı́ téže procedury. V druhé lekci, kde byl tento kód poprvé
uveden, jsme si celou situaci možná nebyli schopni zcela představit. Nynı́ se ale na výše uvedený kód
můžeme dı́vat jako na kód, ve kterém je použit y-kombinátor pro rekurzivnı́ aplikaci procedury bez
argumentu. Ve skutečnosti tedy použitá proceduru „jeden argument má“, dı́ky němuž má k dispozici
sebe sama.

9.3 Lokálnı́ definice rekurzivnı́ch procedur

V předchozı́ sekci jsme ukázali, že pomocı́ speciálnı́ch forem let a let* nemůžeme lokálně „definovat“
rekurzivnı́ procedury. Můžeme však běžným způsobem použı́t internı́ definice. Jako přı́klad uved’me
proceduru na výpočet kombinačnı́ho čı́sla

(
n
k

)
podle vzorce(

n

k

)
=

n!
k! · (n− k)!

Proceduru pro výpočet faktoriálu ale budeme definovat lokálně. Mohli bychom ji napřı́klad napsat takto:

(define comb

(lambda (n k)

(define fak (lambda (n)

(if (<= n 1) 1 (* n (fak (- n 1))))))

(/ (fak n) (fak k) (fak (- n k)))))

Uvedená definice má ale jeden závažný nedostatek. Tı́mto nedostatkem je skutečnost, že při každém volánı́
procedury comb vytvářı́me stále znovu proceduru pro výpočet faktoriálu. To je zbytečné, protože vytvořenı́
této procedury je nezávislé na argumentech procedury comb. To můžeme snadno vyřešit tak, že nebudeme
proceduru fak definovat v těle procedury comb, ale proceduru comb vytvořı́me až lokálnı́m prostředı́ ve
kterém bude definována procedura fak. Kód definice procedury comb by pak vypadal takto:

(define comb

(let ()

(define fak (lambda (n)

(if (<= n 1) 1 (* n (fak (- n 1))))))

(lambda (n k)

(/ (fak n) (fak k) (fak (- n k))))))

Podobně bychom mohli proceduru comb napsat pomocı́ dalšı́ varianty speciálnı́ formy let, a to speciálnı́
formy letrec. Nynı́ uvedeme definici procedury comb napsanou s použitı́m této speciálnı́ formy a vzápětı́
tuto speciálnı́ formu podrobně popı́šeme.

(define comb

(letrec ((fak (lambda (n)

(if (<= n 1) 1 (* n (fak (- n 1)))))))

(lambda (n k)

(/ (fak n) (fak k) (fak (- n k))))))

Ted’ tedy k speciálnı́ formě letrec:

Definice 9.1 (Speciálnı́ forma letrec). Speciálnı́ forma letrec se použı́vá ve stejném tvaru jako speciálnı́
forma let*, tedy ve tvaru

(letrec ((〈symbol1〉 〈element1〉)
(〈symbol2〉 〈element2〉)
...

(〈symboln〉 〈elementn〉)
〈výraz1〉
〈výraz2〉
...

〈výrazm〉),

kde n je nezáporné čı́slo a m je přirozené čı́slo; 〈symbol1〉; 〈symbol2〉, . . . 〈symboln〉 jsou symboly; 〈element1〉,
〈element2〉, . . . 〈elementn〉 a 〈výraz1〉, 〈výraz2〉, . . . 〈výrazm〉 jsou libovolné výrazy. Celý výraz nazýváme
letrec-blok, výrazy 〈výraz1〉, 〈výraz2〉, . . . 〈výrazm〉 nazýváme souhrnně tělem letrec-bloku.

letrec-blok vyhodnocuje stejným způsobem jako výraz

(let ((〈symbol1〉 undefined)
(〈symbol2〉 undefined)
...

(〈symboln〉 undefined))

(define 〈symbol1〉 〈element1〉)
(define 〈symbol2〉 〈element2〉)

...

(define 〈symboln〉 〈elementn〉)
〈výraz1〉
〈výraz2〉
...

〈výrazm〉),

kde undefined je speciálnı́ element jazyka zastupujı́cı́ „nedefinovanou hodnotu“. �

Přı́klad 9.2. letrec-blok ve tvaru

(letrec ((x 10)

(y (+ x 2)))

(list x y)) Z=⇒ (10 12)

se přepisuje na následujı́cı́ výraz, který je vyhodnocen:

(let ((x undefined)
(y undefined))

(define x 10)

(define y (+ x 2))

(list x y)) Z=⇒ (10 12)

Poznámka 9.3. (a) Standard R5RS jazyka Scheme přesně nezavádı́, jak chápat nedefinovanou hodnotu.
Nedefinovaná hodnota, který hraje roli ve speciálnı́ formě letrec je ve většině interpretů jazyka Scheme
zastoupena speciálnı́m elementem jazyka. Obvykle je tento element odlišný od elementu „nespecifikovaná
hodnota“, který jsme zavedli již v prvnı́ lekci (připomeňme, že napřı́klad vyhodnocenı́m (if #f #f)

zı́skáme element zastupujı́cı́ nespecifikovanou hodnotu). Vzhledem ke způsobu jakým se přepisujı́ letrec-
bloky, můžeme napsat výraz, jehož výsledek vyhodnocenı́ bude právě nedefinovaná hodnota použitá jako
počátečnı́ vazba symbolů v letrec-blocı́ch:

(letrec ((x x)) x) Z=⇒ undefined

(b) Speciálnı́ forma letrec je stejně jako ostatnı́ dřı́ve uvedené varianty speciálnı́ formy let nadbytečná
v tom smyslu, že jsou nahraditelné jiným kódem.

(c) Na rozdı́l od použitı́ speciálnı́ formy a let* je možné odkazovat se ve 〈element1〉, 〈element2〉, . . . 〈elementn〉
nejen „dozadu“ (na vazby definované výše) ale i „dopředu“. Napřı́klad následujı́cı́ kód bude fungovat:

(letrec ((x y)

(y 10))

(list x y)) Z=⇒ (undefined 10)

Výsledkem jeho vyhodnocenı́ je seznam (undefined 10). Uvedený kód se totiž přepı́še na let-blok

(let ((x undefined)
(y undefined))

(define x y)

(define y 10)

(list x y)) Z=⇒ (undefined 10)

Během definice (define x y) je na symbol y navázána nedefinovaná hodnota. Proto i po vyhodnocenı́
této definice bude na symbol x navázána nedefinovaná hodnota. Až poté je na symbol y navázáno čı́slo 10.
Tělo (list x y) původnı́ho letrec-bloku je tedy vyhodnoceno na výraz (undefined 10).

9.4 Implementace vybraných rekurzivnı́ch procedur

V předchozı́ch lekcı́ch jsme ukázali implementace mnoha procedur pracujı́cı́ch se seznamy. Nynı́ se k nim
vrátı́me a naimplementujeme je s použitı́m rekurze. Nejdřı́ve ale uvedeme implementace samotných aku-
mulačnı́ch procedur foldr a foldl (pro jeden seznam):

(define foldr

(lambda (f basis l)

(if (null? l)

basis

(f (car l) (foldr f basis (cdr l))))))

Jedná se tedy o rekurzivnı́ proceduru, jejı́ž limitnı́ podmı́nkou je prázdnost seznamu. Pokud je tato splněna
vracı́ se terminátor basis. Jinak je aplikována procedura f na prvnı́ prvek seznamu a výsledek rekurzivnı́ho
volánı́ procedury foldr se zkráceným seznamem. Nynı́ předejme k definici procedury foldl:

(define foldl

(lambda (f basis l)

(let iter ((l l)

(accum basis))

(if (null? l)

accum

(iter (cdr l)

(f (car l) accum))))))

V těle této procedury definujeme a aplikujeme pomocnou proceduru iter. Tato iterativnı́ procedura přijı́má
dva argumenty – seznam doposud nezpracovaných prvkůl (na začátku celý seznam) a výsledek akumulace
zpracovaných prvků accum (na začátku terminátor basis). Pokud je seznam l prázdný, je vrácen accum,
jinak voláme proceduru iter se zkráceným seznamem l a s nabalenı́m jeho prvnı́ho prvku na argument
accum.

V lekci 6 jsme implementovali mnoho procedur pomocı́ procedur pro akumulaci na seznamech foldr

a foldl. Se znalostı́ toho, jak jsou tyto dvě procedury implementovány, je velmi snadné přepsat ukázkové
procedury z lekce 6 na rekurzivnı́ procedury, které ve svém těle nevolajı́ akumulačnı́ procedury. Přepis

je přitom v podstatě mechanickou záležitostı́. V této sekci proto uvedeme jen tři přı́klady, a to proceduru
append na spojovánı́ (libovolného množstvı́) seznamů, procedura mapovánı́ procedury přes libovolný
počet seznamů map a procedura compose na skládánı́ (libovolného množstvı́) funkcı́.

Nejprve tedy uvedeme definici rekurzivnı́ procedury append. Jako prvnı́ nadefinujeme proceduru append2

na spojenı́ dvou seznamů a pomocı́ nı́ napı́šeme rozšı́řenı́ na libovolný počet argumentů. Viz program 9.2.
Procedura append zastavuje rekurzi v přı́padě, že je počet jejich argumentů nulový nebo jednotkový. Pak je
spojenı́ triviálnı́. Jinak dva argumenty spojı́me pomocnou procedurou append2 a aplikujeme pak append

na menšı́ počet argumentů.

Program 9.2. Spojovanı́ seznamů append naprogramované rekurzivně

(define append2

(lambda (l1 l2)

(if (null? l1) l2

(cons (car l1) (append2 (cdr l1) l2)))))

(define append

(lambda lists

(cond ((null? lists) '())

((null? (cdr lists)) (car lists))

(else (apply append

(append2 (car lists) (cadr lists))

(cddr lists))))))

Proceduru append bychom mohli napsat i bez použitı́ pomocné procedury pro spojenı́ dvou seznamů
append2. Viz program 9.3. V něm postupně opětně zastavujeme rekurzi v přı́padě, že je seznam seznamů
určených ke spojenı́ prázdný nebo jednoprvkový. Problém pak rozkládáme na spojenı́ méně seznamů
(pokud je prvnı́ seznam prázdný), nebo na spojenı́ stejného počtu seznamů, s tı́m, že prvnı́ z nich je
zkrácený o jeden prvek (a tedy strukturálně jednoduššı́) a připojenı́ prvku na začátek seznamu.

Program 9.3. Spojovanı́ seznamů append naprogramované rekurzivně bez použitı́ pomocné procedury

(define append

(lambda lists

(cond ((null? lists) '())

((null? (car lists)) (apply append (cdr lists)))

(else (cons (caar lists)

(apply append

(cons (cdar lists) (cdr lists))))))))

V předchozı́ lekci jsme ukázali implementaci rekurzivnı́ procedury map1, (viz program 8.18). Nynı́ pomocı́
nı́ naimplementujeme proceduru mapovánı́ procedury na libovolný počet seznamů map. Následuje kód
této procedury:

(define map

(lambda (f . lists)

(if (null? (car lists))

'()

(cons (apply f (map1 car lists))

(apply map (cons f (map1 cdr lists)))))))

Poslednı́ procedurou, kterou v této sekci napı́šeme, je procedura na skládánı́ libovolného počtu funkcı́
(procedur reprezentujı́cı́ch zobrazenı́). Takovou proceduru jsme již implementovali v programu 7.10 na
straně 184 pomocı́ akumulačnı́ procedury foldl. Nynı́ uděláme totéž bez nı́. Viz program 9.4.

Program 9.4. Skládánı́ funkcı́ compose bez použitı́ procedury foldl

(define compose

(lambda functions

(let iter ((flist functions)

(f (lambda (x) x)))

(if (null? flist)

f

(iter (cdr flist)

(lambda (x)

((car flist) (f x))))))))

V programu 9.4 tedy pomocı́ pojmenovaného let definujeme a aplikujeme pomocnou proceduru iter

dvou argumentů. Prvnı́ argumentem je seznam funkcı́ ke skládánı́ a druhý je akumulačnı́ argument,
kde si pamatujeme prozatı́mnı́ složenı́ funkcı́ (na začátku identita) Tato procedura je rekurzivnı́ a při
splněnı́ limitnı́ podmı́nky, kterou je test prázdnosti seznamu skládaných funkcı́, akumulačnı́ argument.
Při nesplněnı́ limitnı́ podmı́nky je rekurzivně volána procedura iter se seznamem funkcı́ bez prvnı́ho
prvku a složenı́ akumulačnı́ho argumentu s prvnı́ funkcı́ ze seznam funkcı́. Viz následujı́cı́ aplikace této
procedury:

(define s '(0 1 2 3 4))

(map (compose) s) Z=⇒ (0 1 2 3 4)

(map (compose (lambda (x) (* 2 x))) s) Z=⇒ (0 2 4 6 8)

(map (compose (lambda (x) (* 2 x)) (lambda (x) (* x x))) s) Z=⇒ (0 4 16 36 64)

(map (compose (lambda (x) (* x x)) (lambda (x) (* 2 x))) s) Z=⇒ (0 2 8 18 32)

9.5 Hloubková rekurze na seznamech

Doposud jsme vı́ce méně zpracovávali seznamy „prvek po prvku“. V této sekci se budeme zabývat im-
plementacı́ rekurzivnı́ch procedur, které při zpracovánı́ seznamu budou aplikovat samy sebe na ty prvky
seznamů, které jsou opět seznamy. Typická úloha patřı́cı́ do tohoto druhu úloh je napřı́klad spočı́tánı́ ato-
mických prvků (to jest těch prvků celé struktury, které nejsou seznamy). Dále třeba linearizace seznamu,
tedy vytvořenı́ seznamu atomických prvků.

Nynı́ napı́šeme a rozebereme možnou implementaci uvedených úloh (počet atomů a linearizace), poté
se zaměřı́me na podobnost obou definic a navrhneme zobecněnı́. Definici procedury linearizace seznamu
linearize provedeme takto:

(define linearize

(lambda (l)

(cond ((null? l) '())

((list? (car l)) (append (linearize (car l)) (linearize (cdr l))))

(else (cons (car l) (linearize (cdr l)))))))

V této rekurzivnı́ proceduře je limitnı́ podmı́nkou rekurze prázdnost zpracovávaného seznamu. Je-li této
podmı́nky dosaženo, vracı́ procedura prázdný seznam. V opačném přı́padě (tedy pokud seznam nenı́
prázdný), zkoumáme jeho prvnı́ prvek. Jestliže je tento prvek opět seznam, zlinearizujeme jej procedurou
linearize a procedurou append ji připojı́me ke zlinearizovanému zbytku seznamu. Je-li prvnı́ prvek
atomický, přidáme jej k linearizaci zbytku seznamu. Tı́mto postupem vytvořı́me lineárnı́ seznam, který
obsahuje všechny atomické prvky z původnı́ho seznamu. Viz ukázky použitı́:

(linearize '()) Z=⇒ ()

(linearize '(1)) Z=⇒ (1)

(linearize '((1))) Z=⇒ (1)

(linearize '(1 ((2))) Z=⇒ (1 2)

(linearize '(1 ((2)) () (3 (4) 5))) Z=⇒ (1 2 3 4 5)

Ted’ se zaměřme na druhou uvedenou typickou úlohu, kterou je zjištěnı́ počtu atomických prvků v dané
hierarchické struktuře. Následuje kód implementace této procedury.

(define atoms

(lambda (l)

(cond ((null? l) 0)

((list? (car l)) (+ (atoms (car l)) (atoms (cdr l))))

(else (+ 1 (atoms (cdr l)))))))

Limitnı́ podmı́nku tvořı́, stejně jako v předchozı́ proceduře, test na prázdnost seznamu. Pokud je tedy
seznam prázdný, je vrácena nula. Jinak se zajı́máme o to, jestli je prvnı́ prvek tohoto seznamu opětně
seznam. Jestliže ano, sečteme jeho atomy a atomy ve zbytku seznamu. Jestliže prvnı́ prvek nenı́ seznam,
a tedy je to atomický prvek, přičteme za něj jedničku k počtu atomů ve zbytku seznamu. Viz přı́klady
aplikace této procedury:

(atoms '()) Z=⇒ 0

(atoms '(1)) Z=⇒ 1

(atoms '((a))) Z=⇒ 1

(atoms '(1 ((2))) Z=⇒ 2

(atoms '(1 () 2 (a (() (b (c))) (d) e))) Z=⇒ 7

Pozorný čtenář si jistě povšiml nemalé podobnosti uvedených definic. V obou přı́padech zastavujeme
rekurzi v přı́padě prázdného seznamu. Když byl seznam neprázdný, zajı́malo nás, jestli byl jeho prvnı́ prvek
opět seznam nebo ne. V přı́padě seznamu jsme rekurzivně volali proceduru pro tento prvnı́ prvek a také
pro zbytek seznamu. Výsledek této aplikace jsme pak zkombinovali pomocı́ dalšı́ procedury (append, +).
V přı́padě atomického prvku jsme tento prvek zpracovali, a zkombinovali jej (vlastně stejnou procedurou
jako v předchozı́ větvi) s výsledkem rekurzivnı́ho vyvolánı́ procedury na zbytek seznamu. Konkrétně
u procedury atoms jsme zpracovali atomický prvek tak, že jsme jej zaměnili za jedničku, tu jsme pak sečetli
aplikacı́ procedury +. U procedury linearize jsme uvedli kód:

(else (cons (car l) (linearize (cdr l))))

Stejný význam by měl kód, napsaný takto:

(else (append (list (car l)) (linearize (cdr l)))).

Zpracovánı́ prvku je tedy v tomto přı́padě vytvořenı́ jednoprvkového seznamu aplikacı́ procedury list

a procedurou kombinace je procedura append. Tato podobnost nám umožňuje vytvořit zobecněnı́ těchto
procedur – hloubkovou akumulačnı́ proceduru depth-accum.

(define depth-accum

(lambda (combine nil modifier l)

(cond ((null? l) nil)

((list? (car l)) (combine (depth-accum combine nil modifier (car l))

(depth-accum combine nil modifier (cdr l))))

(else (combine (modifier (car l))

(depth-accum combine nil modifier (cdr l)))))))

Sjednotili jsme tedy předchozı́ dvě procedury do jedné obecnějšı́. Rozdı́ly v procedurách jsme zahrnuli do
dalšı́ch argumentů této procedury: combine pro různé způsoby kombinace výsledků z rekurzivnı́ch volánı́
(popř. ze zpracovánı́ atomických prvků), nil pro různé návratové hodnoty při splněnı́ limitnı́ podmı́nky
a modifier pro různá zpracovánı́ atomických prvků. Napřı́klad aplikace procedur atoms a linearize

bychom mohli nahradit aplikacı́ procedury depth-accum takto:

(depth-accum + 0 (lambda (x) 1) '(a (b c (d)) e)) Z=⇒ 5

(depth-accum append '() list '(a (b c (d)) e)) Z=⇒ (a b c d e)

Pomocı́ procedury depth-accum samozřejmě můžeme definovat dalšı́ hloubkově rekurzivnı́ procedury.
Ted’ uvedeme několik dalšı́ch procedur tohoto typu, k jejich vytvořenı́ použijeme právě proceduru depth-

accum. Prvnı́ z nich bude procedura depth-count zjišt’ujı́cı́ (hloubkově) počet výskytů daného atomického
prvku v zadané struktuře. Viz jejı́ definici:

(define depth-count

(lambda (atom l)

(depth-accum + 0 (lambda (x)

(if (equal? atom x)

1 0))

l)))

Terminátorem je v tomto přı́padě čı́slo nula a kombinačnı́ procedurou je procedura sčı́tánı́ +. To je vlastně
stejné jako u implementace procedury atoms. Na rozdı́l od procedury atoms ale nezpracováváme každý
atom tak, že jej zaměňujeme za čı́slo jedna, ale bud’to za čı́slo jedna nebo za čı́slo nula v závislosti na tom,
jestli je tento atom stejný jako zadaný element. Pro porovnánı́ elementů jsme použili proceduru equal?.
Následujı́ ukázky použitı́ takto vytvořené procedury.

(depth-count 1 '()) Z=⇒ 0

(depth-count 1 '(1)) Z=⇒ 1

(depth-count 1 '((1))) Z=⇒ 1

(depth-count 1 '(1 (1))) Z=⇒ 2

(depth-count 'a '(1 () 2 (a (() (b (a))) (d) e))) Z=⇒ 2

Dále vytvořı́me predikát depth-find, který zjišt’uje, zda je daný atom přı́tomen ve struktuře. K jejı́ imple-
mentaci opět použijeme proceduru hloubkové akumulace na seznamu depth-accum. Predikát depth-find
můžeme implementovat náhledově:

(define depth-find

(lambda (x l)

(depth-accum (lambda (x y) (or x y))

#f

(lambda (y) (equal? x y))

l)))

Proceduru depth-accum jsme tedy použili těmito argumenty: procedura napodobujı́cı́ speciálnı́ formu or

pro dva argumenty (taktéž bychom zde mı́sto vytvářenı́ nové procedury mohli využı́t proceduru or-proc,
se kterou jsme se již setkali v lekci 6). Terminátorem je pravdivostnı́ hodnota nepravda a zpracovánı́m
atomického prvku zde rozumı́me vrácenı́ pravdivostnı́ hodnoty podle toho, zda je shodný s hledaným
elementem. Následujı́ ukázky aplikace této procedury.

(depth-find 'a '(1 () 2 (a (() (b (a))) (d) e))) Z=⇒ #t

(depth-find 'b '(1 () 2 (a (() (b (a))) (d) e))) Z=⇒ #t

(depth-find 'x '(1 () 2 (a (() (b (a))) (d) e))) Z=⇒ #f

Poslednı́m užitı́m procedury depth-accum (ve stávajı́cı́ podobě), kterou si v této sekci ukážeme bude
procedura hloubkové filtrace atomů daných vlastnosti, které zachovává hloubkovou strukturu seznamu.
Pro jeho definici viz program 9.5. Terminátorem je zde prázdný seznam, atomické prvky zpracováváme
tak, že pokud splňujı́ zadaný predikát x, aplikujeme na ně modifikátor modifier, jinak je ponecháváme
stejné. Procedurou pro kombinaci výsledků je pak konstruktor páru cons. Viz přı́klad aplikace:

(define s '(1 () 2 (a (() (b (a))) (d) e)))

(depth-replace number? - s) Z=⇒ (-1 () -2 (a (() (b (a))) (d) e))

(depth-replace symbol? (lambda (x) #f) s) Z=⇒ (1 () 2 (#f (() (#f (#f))) (#f) #f))

Procedury linearize a atoms, jejichž definice jsme uvedli na začátku této sekce, bychom samozřejmě
mohli napsat i jinak. Napřı́klad použitı́ procedur apply a map nám umožňuje znatelně zkrátit kód těchto
procedur. Definice procedury linearize by vypadalo takto:

Program 9.5. Implementace procedury hloubkového nahrazovanı́ atomů depth-replace

(define depth-replace

(lambda (prop? modifier l)

(depth-accum cons

'()

(lambda (z)

(if (prop? z)

(modifier z)

z))

l)))

(define linearize

(lambda (l)

(if (list? l)

(apply append (map linearize l))

(list l))))

Procedura linearize nejdřı́ve zkontroluje, zda je jejı́ argument seznam. Jestliže ne, vytvořı́me jednoprv-
kový seznam obsahujı́cı́ tento prvek. V přı́padě, že ano, pomocı́ procedury map aplikuje sama sebe na každý
jeho prvek – dostáváme tak seznam, jehož prvky jsou lineárnı́ seznamy (bud’to zlinearizované seznamy
z původnı́ho seznamu nebo jednoprvkové seznamy vytvořené z atomických prvků). Tyto seznamy spojı́me
v jeden aplikacı́ procedury append.

Obdobným způsobem napı́šeme proceduru atoms:

(define atoms

(lambda (l)

(if (list? l)

(apply + (map atoms l))

1)))

Stejně jako v definici procedury linearize jsme nejdřı́ve zjistili, zda je argumentem procedury seznam.
Pokud ne, vracı́me čı́slo 1. Jinak aplikujeme na každý prvek tohoto seznamu pomocı́ procedury map.
Výsledný seznam čı́sel pak sečteme aplikacı́ procedury sčı́tánı́.

Nynı́ se můžeme zaměřit na zobecněnı́ inspirované těmito dvěma definicemi. Podruhé naprogramujeme
proceduru depth-accum zobecňujı́cı́ procedury atoms a linearize tı́mto způsobem:

(define depth-accum

(lambda (combine modifier l)

(if (list? l)

(apply combine (map (lambda (x) (depth-accum combine modifier x)) l))

(modifier l))))

Prvky, ve kterých je kódy procedur atoms a linearize lišily jsme shrnuli do argumentů combine a modi-
fier. V porovnánı́ s předchozı́m řešenı́m nám tedy odpadl terminátor nil.

Funkci předchozı́ch procedur atoms a linearize bychom mohli vyjádřit pomocı́ procedury depth-accum

tak jak ukazujı́ následujı́cı́ přı́klady:

(depth-accum append list '(a (b c (d)) e))

(depth-accum + (lambda (x) 1) '(a (b c (d)) e))

Takto definovaná procedura depth-accummá ale dva nedostatky. Procedura pro kombinaci prvků musı́ být
procedurou libovolného počtu argumentů. To proto, že ji aplikujeme pomocı́ procedury apply na seznam
jehož délku předem neznáme. Druhým nedostatkem je skutečnost že procedura depth-accum funguje
i když jejı́m poslednı́m argumentem nebude seznam.

Zatı́mco s druhým nedostatkem bychom se mohli docela klidně smı́řit, prvnı́ nedostatek nám nové řešenı́
jaksi degraduje v porovnánı́ s předchozı́m. Napřı́klad nemůžeme přı́mo přepsat proceduru depth-replace

z programu 9.5, protože tam jsme ke kombinaci použı́vali proceduru cons, která přijı́má přesně dva argu-
menty. Tento nedostatek bychom mohli odstranit napřı́klad tak, že bychom z libovolné monoidálnı́ operace
vytvářeli přı́slušnou operaci libovolného počtu argumentů. To bychom mohli provádět třeba pomocı́ takto
nadefinované procedury:

(define arbitrary

(lambda (f nil)

(lambda l

(foldr f nil l))))

Takto můžeme pomocı́ depth-accum přepsat proceduru depth-replace, kterou jsme definovali v pro-
gramu 9.5. V kódu tak z procedury cons vytvořı́me proceduru libovolného množstvı́ argumentů. K tomu
ale potřebujeme doplnit neutrálnı́ prvek. Ta hraje v podstatě stejnou úlohu jako terminátor v předchozı́
verzi. Takže se tak připravujeme o výhodu menšı́ho počtu argumentů.

(define depth-replace

(lambda (prop? modifier l)

(depth-accum (arbitrary cons '())

(lambda (z) (if (prop? z) (modifier z) z))

l)))

V závěru této sekce se podı́váme na praktické využitı́ hloubkové rekurze. Představme si, máme tabulku
naplněnou čı́selnými údaji a že máme zpracovávat výrazy zadané uživatelem. Tyto uživatelem zadané
výrazy obsahujı́ odkazy do datové tabulky, což jsou páry ve tvaru (řádek . sloupec). Pro přesnějšı́ představu
uved’me přı́klad takové tabulky a přı́klad takového výrazu:

(define tab

'((1 0 1 0 0 1 1 0 3 0)

(0 2 2 1 0 1 0 4 0 5)

(2 1 0 1 5 0 2 1 3 1)

(3 0 2 1 5 0 4 1 1 1)

(2 1 2 0 5 1 3 1 1 2)

(0 1 3 0 0 0 0 1 1 2)))

(+ 1 (3 . 2) (* 2 (2 . 4)))

Vyhodnocenı́ uvedeného výrazu přirozeně skončı́ chybou. Našı́m prvnı́m záměrem tedy bude proceduru
query->sexprnahrazujı́cı́ tyto páry ve tvaru(řádek . sloupec) seznamy(table-ref řádek sloupec). K tomu
použijeme proceduru depth-replace, kterou jsme definovali v programu 9.5.

(define query->sexpr

(lambda (expr)

(depth-replace pair?

(lambda (x)

(let ((row (car x))

(col (cdr x)))

(list 'table-ref row col)))

expr)))

Viz přı́klad použitı́:

(query->sexpr '(+ 1 (3 . 2) (* 2 (2 . 4))))

Z=⇒ (+ 1 (table-ref 3 2) (* 2 (table-ref 2 4)))

Proceduru query->sexpr ted’ využijeme k vytvořenı́ procedury vyššı́ho řádu query->proc, která přijı́má
jeden argument, kterým je procedura, která je použita jako procedura dereference do datové tabulky.
Provedeme to tak, že výraz, který je vrácen procedurou query->sexpr „obalı́me“ kontextem

(lambda (table-ref) · · ·)
a takto vzniklý λ-výraz vyhodnotı́me procedurou eval. Viz následujı́cı́ kód:

(define query->proc

(lambda (expr)

(eval (list 'lambda '(table-ref) (query->sexpr expr)))))

Zde uvádı́me přı́klad aplikace:

((query->proc '(+ 1 (3 . 2) (* 2 (2 . 4))))

(lambda (row col) (list-ref (list-ref tab row) col))) Z=⇒ 13

Pomocı́ této procedury query->proc můžeme definovat proceduru pro vyhodnocenı́ výrazu s odkazy
do tabulky vzhledem ke konkrétnı́ tabulce. Procedura eval-in-table bude brát dva argumenty: výraz
a tabulku. Ve svém těle pak aplikacı́ procedury query->proc vytvořı́ proceduru (viz výše), kterou aplikuje
na proceduru pro přı́stup k údaji v tabulce. Procedura pro přı́stup je přitom jen dvojnásobná aplikace
procedury list-ref. Viz definici procedury eval-in-table a po nı́ následujı́cı́ přı́klady použitı́:

(define eval-in-table

(lambda (expr table)

((query->proc expr)

(lambda (row col)

(list-ref (list-ref table row) col)))))

(eval-in-table '(+ 1 2) tab) Z=⇒ 3

(eval-in-table '(+ 1 (1 . 7)) tab) Z=⇒ 5

(eval-in-table '(+ 1 (3 . 2) (* 2 (2 . 4))) tab) Z=⇒ 13

Proceduru table-ref přitom můžeme realizovat různými způsoby. Dokonce můžeme opustit původně
zamýšlený formát tabulky, jako seznam seznamů, pracovat napřı́klad s lineárnı́m seznamem. Tuto variantu
zachycuje následujı́cı́ program:

(define eval-in-linear-list

(lambda (expr l cols)

((query->proc expr)

(lambda (row col)

(list-ref l (+ (* cols row) col))))))

(eval-in-linear-list '(list (0 . 0) 'blah (1 . 7) 'halb (2 . 5))

'(10 20 30 40 50 60 70 80 90 100

11 21 31 41 51 61 71 81 91 101

12 22 32 42 52 62 72 82 92 102)

10) Z=⇒ (10 blah 81 halb 62)

Shrnutı́

V této lekci jsme pokračovali v problematice rekurzivnı́ch procedur. Ukázali různé způsoby specifikace
limitnı́ podmı́nky v rekurzivnı́ch procedurách a vysvětlili jsme, proč by „if jako procedura“ nebyla pou-
žitelná pro zastavovánı́ rekurze. Dále jsme se zabývali otázkou, zda je speciálnı́ forma define nutná pro
vytvářenı́ rekurzı́vnı́ch procedur, a ukázali jsme že rekurzi můžeme zajistit i pomocı́ konstruktu nazývaného
y-kombinátor. V dalšı́ části lekce jsme představili speciálnı́ formu letrec, která umožňuje definovat lokálnı́
rekurzivnı́ procedury. V závěru lekce jsme se pak věnovali hloubkové rekurzi na seznamech a hloubkovým
akumulačnı́m procedurám.

Pojmy k zapamatovánı́

• metody zastavenı́ rekurze

• y-kombinátor
• hloubková rekurze na seznamech
• atom
• linearizace seznamu

Nově představené prvky jazyka Scheme
• element nedefinovaná hodnota
• speciálnı́ forma letrec

Kontrolnı́ otázky
1. Jakými způsoby lze zastavit rekurzi?
2. Proč if jako procedura nezastavı́ rekurzi?
3. Jakou má roli speciálnı́ forma define při psanı́ rekurzivnı́ch procedur.
4. Co je to y-kombinátor?
5. Jak se použı́vá speciálnı́ forma letrec?
6. Na jaké výrazy se přepisujı́ letrec-bloky?
7. Co je to nedefinovaná hodnota?
8. Co se myslı́ hloubkovou rekurzı́ na seznamech?

Cvičenı́
1. pomocı́ y-kombinátoru naprogramujte následujı́cı́ procedury:

• proceduru pro výpočet n-tého Fibonacciho čı́sla fib
• mapovánı́ procedury přes jeden seznam map1

• linearizace seznamu linearize

2. Implementujte následujı́cı́ procedury bez použitı́ procedury depth-accum:
• depth-filter – hloubková filtrace na seznamu
• depth-map – hloubkové mapovánı́ procedury přes seznam
• atom? – predikát zjišt’ujı́cı́, jestli je element atomem seznamu

3. Implementujte procedury z předchozı́ho úkolu pomocı́ prvnı́ verze depth-accum.
4. Implementujte procedury z předchozı́ho úkolu pomocı́ druhé verze depth-accum.

Úkoly k textu

1. Popište rozdı́l mezi speciálnı́ formou letrec a speciálnı́ formou let+ z úkolů k textu lekce 2. Napište
kód, jehož výsledek vyhodnocenı́ se bude lišit při použitı́ těchto speciálnı́ch forem.

2. Implementuje některou proceduru provádějı́cı́ hloubkovou rekurzi na seznamech (napřı́klad depth-

count, depth-find nebo depth-accum) pomocı́ y-kombinátoru. Existuje nějaký principiálnı́ rozdı́l
při použitı́ y-kombinátoru pro hloubkovou a běžnou rekurzi? Vysvětlete proč ano či proč ne.

Řešenı́ ke cvičenı́m
1. • ;; fib

(lambda (n)

((lambda (y)

(y y n))

(lambda (fib n)

(if (<= n 1) n

(+ (fib fib (- n 1))

(fib fib (- n 2)))))))

• ;; map1

(lambda (f l)

((lambda (y)

(y y f l))

(lambda (map1 f l)

(if (null? l)

'()

(cons (f (car l))

(map1 map1 f (cdr l)))))))

• ;; linearize

(lambda (l)

((lambda (y)

(y y l))

(lambda (lin l)

(cond ((null? l) '())

((list? (car l)) (append (lin lin (car l)) (lin lin (cdr l))))

(else (cons (car l) (lin lin (cdr l))))))))

2. (define depth-filter

(lambda (p? l)

(cond ((null? l) '())

((list? (car l)) (cons (depth-filter p? (car l))

(depth-filter p? (cdr l))))

((p? (car l)) (cons (car l) (depth-filter p? (cdr l))))

(#t (depth-filter p? (cdr l))))))

• (define depth-map

(lambda (f l)

(cond ((null? l) '())

((list? (car l)) (cons (depth-map f (car l))

(depth-map f (cdr l))))

(#t (cons (f (car l)) (depth-map f (cdr l)))))))

• (define atom?

(lambda (a l)

(if (null? l) #f

(or (if (list? (car l))

(atom? a (car l))

(equal? (car l) a))

(atom? a (cdr l))))))

3. • (define depth-filter

(lambda (pred? l)

(depth-accum (lambda (x y)

(if (null? x)

y

(cons x y)))

'()

(lambda (x) (if (pred? x) x '()))

l)))

• (define depth-map

(lambda (f l)

(depth-accum cons '() f l)))

• (define atom?

(lambda (a l)

(depth-accum (lambda (x y) (or x y)) #f (lambda(x) (equal? a x)) l)))

4. • (define depth-filter

(lambda (pred? l)

(depth-accum (arbitrary (lambda (x y)

(if (null? x)

y

(cons x y))) '())

(lambda (x) (if (pred? x) x '()))

l)))

• (define depth-map

(lambda (f l)

(depth-accum list f l)))

• (define atom?

(lambda (a l)

(depth-accum (arbitrary (lambda (x y) (or x y)) #f)

(lambda(x) (equal? a x)) l)))

Lekce 10: Kombinatorika na seznamech, reprezentace stromů a množin

Obsah lekce: Tato lekce obsahuje pokročilejšı́ přı́klady sloužı́cı́ k procvičenı́ práce s hierarchickými daty.
V prvnı́ části lekce se budeme zabývat reprezentacı́ stromů pomocı́ seznamů a jejich prohledávánı́m do
hloubky a do šı́řky. Dále ukážeme reprezentaci množin pomocı́ uspořádaných seznamů a pomocı́ binárnı́ch
vyhledávacı́ch stromů. Závěr lekce je věnován kombinatorice na seznamech.

Klı́čová slova: kombinatorika na seznamech, reprezentace množin, stromy.

10.1 Reprezentace n-árnı́ch stromů

Prvnı́ sada přı́kladů se týká n-árnı́ch stromů (dále v této sekci budeme použı́vat pojmenovánı́ strom).
Nejdřı́ve popı́šeme reprezentaci těchto struktur, poté nadefinujeme přı́slušný konstruktor a selektory. V zá-
věru sekce pak budeme realizovat algoritmy pro průchod stromem do hloubky a do šı́řky.

Za strom budeme považovat jakýkoli seznam, který, pokud je neprázdný, má za prvky ocasu dalšı́ stromy.
Hlavu seznamu přitom nazýváme hodnota a prvky ocasu nazýváme podstromy nebo též větve stromu. Prvky
stromu rozumı́me hodnotu stromu a prvky jeho větvı́. Stromy, které nemajı́ podstromy nebo jsou všechny
jejich podstromy prázdné, nazýváme listy.

Napřı́klad strom nakreslený na obrázku 10.1 je reprezentován seznamem ve tvaru

(a (b) (c (e) (f (g) (h) (i (j)))) (d)).

Větve tohoto stromu jsou reprezentovány seznamy (b), (c (e) (f (g) (h) (i (j)))) a (d). Hodnota
stromu je a. Prvky stromu jsou a, b, c, d, e, f, g, h, i a j.

Obrázek 10.1. Přı́klad n-árnı́ho stromu

a

b

c

e

f

g

h

i

j

d

Následuje definice konstruktoru a selektorů pro uvažovanou reprezentaci stromů:

(define make-tree

(lambda (val . subtrees)

(cons val subtrees)))

(define get-val car)

(define get-branches

(lambda (x)

(if (null? x) '() (cdr x))))

Dále uvádı́me predikát tree? zjišt’ujı́cı́ zda je jeho argument reprezentace stromu. Predikát tedy zjistı́, jestli
se jedná o seznam a v přı́padě že ano, zjistı́ jestli jsou jeho prvky (vyjma prvnı́ho) též reprezentace stromu.
Implementace je velmi přı́močará, využı́váme v nı́ predikátu forall, který jsme zavedli v lekci 6.

(define tree?

(lambda (x)

(and (list? x)

(forall tree? (cdr x)))))

Dalšı́ procedury, které naimplementujeme v této sekci, jsou selektory vracejı́cı́ konkrétnı́ větve. Přesněji
selektor left-branch vracejı́cı́ k-tou větev zleva a selektor right-branch vracejı́cı́ k-tou větev zprava.
Přitom budeme chtı́t, aby pro k, které bude záporné, nebo věšı́ nebo rovno počtu větvı́ vracel selektor
prázdný strom (prázdný seznam). Následuje definice selektorů:

(define left-branch

(lambda (tree k)

(if (or (< k 0) (null? tree))

'()

(let iter ((branches (cdr tree))

(k k))

(cond ((null? branches) '())

((= k 0) (car branches))

(else (iter (cdr branches) (- k 1))))))))

Procedura left-branch nejprve ověřı́, zda nebylo zadáno záporné čı́slo a zda zadaný strom nenı́ prázdný.
Pokud ano, vracı́ prázdný strom. V opačném přı́padě pomocı́ iterativnı́ procedury iter, postupně „odjı́má“
podstromy, dokud nedojde na konec seznamu podstromů – pak vracı́ prázdný strom – nebo dokud nenajde
požadovaný podstrom. Proceduru right-branch pak lze vytvořit jednoduše pomocı́ procedury left-

branch. Viz jejı́ následujı́cı́ definici:

(define right-branch

(lambda (tree k)

(left-branch tree (- (length tree) k 2))))

Nynı́ budeme realizovat algoritmy průchodu stromu do hloubky a průchodu stromu do hloubky. Průcho-
dem stromu přitom myslı́me proceduru, která postupně „zpracuje“ všechny prvky stromu. Pod pojmem
„zpracovánı́ prvků“ budeme mı́t na mysli konstrukci seznamu prvků v podstromech v tom pořadı́, v jakém
je navštěvujeme. Dále uvedené algoritmy se lišı́ právě v pořadı́, v jakém prvky stromu zpracovávajı́. Při
průchodu stromu do hloubky je strom zpracováván „po větvı́ch“, při průchodu stromu do šı́řky je strom
zpracováván „po patrech“. Viz ilustraci na obrázku 10.2. Procedura dfs uvedená v programu 10.1 pro

Obrázek 10.2. Ukázka průchodu do šı́řky (uprostřed) a do hloubky (vpravo).

průchod stromu do hloubky vracı́ prázdný seznam pro prázdný strom. Jinak vezme větve stromu a projde
je do hloubky (aplikacı́ sebe sama), výsledné seznamy spojı́ a přidá k nim hodnotu stromu.

Procedura bfs uvedená v programu 10.2 pro průchod stromu do šı́řky využı́vá pomocnou proceduru
aux, která si ve svém argumentu pamatuje seznam stromů určených k průchodu. Pokud je tento seznam
prázdný, vracı́ prázdný seznam. Jinak vezme hodnoty za všech stromů v seznamu. V dalšı́m kroku je pak
tento seznam tvořen podstromy těchto stromů. Tı́m jakoby „sestoupı́me o jedno patro“.

Program 10.1. Prohledávánı́ stromu do hloubky.

(define dfs

(lambda (tree)

(if (null? tree) '()

(cons (get-val tree)

(apply append (map dfs (get-branches tree)))))))

Program 10.2. Prohledávánı́ stromu do šı́řky.

(define bfs

(lambda (tree)

(let aux ((tree (list tree)))

(if (null? tree) '()

(append (map get-val tree)

(aux (apply append (map get-branches tree))))))))

10.2 Reprezentace množin pomocı́ uspořádaných seznamů

V sekci 6.5 jsme ukázali, jak se dajı́ reprezentovat konečné množiny pomocı́ seznamů neobsahujı́cı́ch
vı́cenásobné výskyty prvků (duplicity). V této části předvedeme jinou možnost reprezentace množin. Za
množiny budeme považovat seznamy bez vı́cenásobných výskytů prvků, které jsou navı́c uspořádané.
Tedy napřı́klad seznam (5 8 2 4 9), který byl v sekci 6.5 platnou reprezentacı́ množiny, nenı́ uspořádaný
(podle uspořádánı́ daným predikátem <=), a proto nenı́ reprezentacı́ množiny pro tuto sekci. Platnou
reprezentacı́ této množiny by pak byl seznam (2 4 5 8 9).

Toto zpřı́sněnı́ reprezentace nám umožnı́ psát efektivnějšı́ procedury, než jsou ty ze sekce 6.5. Na druhou
stranu ale musı́me mı́t dané nějaké rozumné lineárnı́ uspořádánı́ (rozumné ve smyslu složitosti – ověřenı́,
zda-li je jeden prvek menšı́ nebo roven než druhý musı́ mı́t přijatelnou časovou a prostorovou složitost).
Také kód procedur pro tuto reprezentaci bude mnohem méně přehledný, což může vést ke vzniku chyb při
jejich implementaci. Proto je nutné provádět důsledné testovánı́.

U následujı́cı́ch procedur (in?, cons-set, union a inter) budeme uvažovat jen čı́selné množiny a uspo-
řádánı́ dané predikátem <=. U každé z těchto procedur nejdřı́ve naznačı́me, jak pracujı́, uvedeme jejich
definici s popisem kódu a přı́klady aplikace.

Prvnı́ procedura – predikát in? – bude využı́vat uspořádánı́ seznamu k rychlejšı́mu zjištěnı́ nepřı́tomnosti
daného prvku v množině. Tato procedura bude postupně procházet přes prvky seznamu dokud nenarazı́
na prvek, který je shodný s hledaným prvkem, dokud neprojde všechny prvky, nebo dokud nenarazı́ na
prvek, který je většı́ než ten hledaný. Právě poslednı́ přı́pad zastavenı́ je přı́nosem zpřı́sněnı́ reprezentace
množiny. Viz následujı́cı́ definici predikátu in?:

(define in?

(lambda (x set)

(cond ((or (null? set) (< x (car set))) #f)

((= x (car set)) #t)

(else (in? x (cdr set))))))

V této rekurzı́vnı́ proceduře zastavujeme rekurzi při zjištěnı́, že je seznam prázdný, nebo že je prvnı́ prvek
seznamu většı́ než hledaný prvek. V tom přı́padě je jisté, že hledaný prvek v množině nenı́, a tak je vrácena
nepravda #f. Dále zastavujeme rekurzi, pokud je prvnı́ prvek seznamu shodný s hledaným prvek, a tehdy
je vrácena pravda #t. Zastavenı́ rekurze je v kódu zahrnuto v prvnı́ch dvou větvı́ch speciálnı́ formy cond.

V přı́padě, že nenı́ splněna ani jedna z těchto limitnı́ch podmı́nek, je predikát in rekurzivně aplikován na
seznam bez prvnı́ho prvku. Následujı́ přı́klady aplikace:

(in? 3 '()) Z=⇒ #f

(in? 3 '(1 2)) Z=⇒ #f

(in? 3 '(1 2 3 4)) Z=⇒ #t

(in? 3 '(1 2 4)) Z=⇒ #f

Druhou procedurou je konstruktor množiny cons-set. Ten má zatřı́dit zadaný element do dané množiny,
pokud už tam tento element nenı́. Ta bude procházet seznam reprezentujı́cı́ množinu podobným způsobem
jako predikát in?. Přı́nosem zpřı́sněnı́ reprezentace je opětně to, že můžeme dřı́ve zjistit nepřı́tomnost
prvku. Implementace této procedury se bude jen málo lišit od implementace predikátu in?. Viz definici:

(define cons-set

(lambda (x set)

(cond ((null? set) (list x))

((= x (car set)) set)

((<= (car set) x) (cons (car set) (cons-set x (cdr set))))

(else (cons x set)))))

V těle této procedury zastavujeme rekurzi v přı́padě, že je množina prázdná a v tom přı́padě vracı́me
jednoprvkový seznam obsahujı́cı́ přidaný prvek. Dále zastavujeme rekurzi při zjištěnı́, že je prvnı́ prvek
seznamu shodný s přidávaným elementem a tehdy vracı́me původnı́ seznam bez jakékoli změny, protože je
v něm přidávaný prvek už obsažen. Poslednı́ limitnı́ podmı́nkou je skutečnost, že prvnı́ prvek seznamu je
většı́, než přidávaný element. Pokud je tato podmı́nka splněna, znamená to, že jsme našli mı́sto, kam má být
prvek přidán. Pokud nenı́ splněna žádná z limitnı́ch podmı́nek, pak rozkládáme problém na přidánı́ prvku
do seznamu (konstruktorem cons), a přidávánı́ prvku do menšı́ množiny (tedy množiny reprezentované
strukturálně jednoduššı́m seznamem). Následujı́ přı́klady použitı́ procedury cons-set.

(cons-set 3 '()) Z=⇒ ()

(cons-set 3 '(1 2)) Z=⇒ (1 2 3)

(cons-set 3 '(1 2 3 4)) Z=⇒ (1 2 3 4)

(cons-set 3 '(1 2 4)) Z=⇒ (1 2 3 4)

Dalšı́ dvě procedury uvedené v této sekci představujı́ množinové operace sjednocenı́ a průniku. Protože se
jedná o složitějšı́ procedury, ukážeme nejdřı́ve to, jak pracujı́ na konkrétnı́m přı́padě a pak až se zaměřı́me
na konkrétnı́ implementaci.

U operace sjednocenı́ množin zpracováváme současně dva seznamy reprezentujı́cı́ množiny. S těmito se-
znamy provádı́me slévánı́, podobně jako tomu bylo u procedury merge v sekci 8.5. Jediným rozdı́lem je to,
že sledujeme i možnost, že by při průběžném zpracovánı́ seznamů nastala skutečnost, že tyto seznamy majı́
stejný prvnı́ prvek. V tom přı́padě zařadı́me tento prvek do výsledného seznamu a odebereme jej z obou
seznamů. Tı́m zamezı́me vzniku duplicit ve výsledném seznamu, které by nastaly při běžném slévánı́. Viz
přı́klad:

seznam A seznam B prvky zařazené do A ∪B
(2 4 6 7) (1 2 3 4 5) 1

(2 4 6 7) (2 3 4 5) 2

(4 6 7) (3 4 5) 3

(4 6 7) (4 5) 4

(6 7) (5) 5

(6 7) () 6 7, výsledek A ∪B: (1 2 3 4 5 6 7)

Následuje definice procedury union:

(define union

(lambda (set-A set-B)

(cond ((null? set-A) set-B)

((null? set-B) set-A)

((= (car set-A) (car set-B))

(cons (car set-A)

(union (cdr set-A) (cdr set-B))))

((<= (car set-A) (car set-B))

(cons (car set-A)

(union (cdr set-A) set-B)))

(else (cons (car set-B) (union set-A (cdr set-B)))))))

Limitnı́ podmı́nkou je test na prázdnost alespoň jednoho ze seznamů. Pokud je tato podmı́nka splněna,
vracı́me druhý ze seznamů (ten, který nenı́ prázdný). To je vyjádřeno prvnı́mi dvěma větvemi speciálnı́
formy cond. Jinak rozlišujeme tyto možnosti:

• Oba seznamy začı́najı́ stejným prvkem. Pak tento společný prvek přidáme na začátek sjednocenı́
množin bez tohoto prvku. To je zahrnuto v třetı́ větvi formy cond.

• Jeden seznam začı́ná menšı́m prvkem než druhý seznam. V tom přı́padě sjednotı́me seznamy bez
tohoto nejmenšı́ho prvku, a tento prvek přidáme do výsledku. Tato možnost je vyjádřena poslednı́mi
dvěma větvemi formy cond.

Viz přı́klady aplikace:

(union '() '()) Z=⇒ ()

(union '() '(1 2 3 4 5)) Z=⇒ (1 2 3 4 5)

(union '(2 4 6 7) '()) Z=⇒ (2 4 6 7)

(union '(2 4 6 7) '(1 2 3 4 5)) Z=⇒ (1 2 3 4 5 6 7)

Podobně pracuje i procedura inter na výpočet průniku množin. Prvky do výsledné množiny ale zařazu-
jeme jen v přı́padě, že se vyskytujı́ v obou seznamech. Takové se nám při průběžném zpracovánı́ dostanou
na začátek seznamu. Viz následujı́cı́ přı́klad:

seznam A seznam B prvky zařazené do A ∩B
(2 4 6 7) (1 2 3 4 5)

(2 4 6 7) (2 3 4 5) 2

(4 6 7) (3 4 5)

(4 6 7) (4 5) 4

(6 7) (5)

(6 7) () výsledek A ∩B: (2 4)

Následuje implementace procedury inter:

(define inter

(lambda (set-A set-B)

(cond ((or (null? set-B) (null? set-A)) '())

((= (car set-A) (car set-B))

(cons (car set-A)

(inter (cdr set-A) (cdr set-B))))

((<= (car set-A) (car set-B)) (inter (cdr set-A) set-B))

(else (inter set-A (cdr set-B))))))

Procedura inter zastavuje rekurzi v přı́padě, že alespoň jeden ze seznamů je prázdný. V takovém přı́padě
je výsledkem prázdná množina. Pokud oba seznamy začı́najı́ stejným elementem, přidáme tento společný
element do průniku množin bez tohoto elementu. Jinak pouze „odjı́máme“ nejmenšı́ prvky ze seznamů.

Viz přı́klady aplikace:

(inter '() '()) Z=⇒ ()

(inter '() '(1 2 3 4 5)) Z=⇒ ()

(inter '(2 4 6 7) '()) Z=⇒ ()

(inter '(2 4 6 7) '(1 2 3 4 5)) Z=⇒ (2 4)

10.3 Reprezentace množin pomocı́ binárnı́ch vyhledávacı́ch stromů

V této sekci ukážeme dalšı́ možnost, jak reprezentovat množiny. Jedná se o reprezentaci pomocı́ binárnı́ch
vyhledávacı́ch stromů. Binárnı́m vyhledávacı́m stromem vzhledem k uspořádánı́ ≤ rozumı́me:

• prázdný strom

• strom, který má právě dva podstromy. Tyto podstromy nazýváme pravý podstrom a levý podstrom. Jsou-li
tyto podstromy neprázdné, musı́ platit, že hodnota levého podstromu je menšı́ vzhledem k uspořádánı́
≤ než hodnota stromu a hodnota pravého podstromu je většı́ vzhledem k uspořádánı́ ≤ než hodnota
stromu.

Binárnı́ vyhledávacı́ stromy budeme reprezentovat stejně jako n-árnı́ stromy uvedené v sekci 10.1. Tomu
budou odpovı́dat i konstruktory a selektory pro binárnı́ strom.

(define make-tree

(lambda (value left right)

(list value left right)))

(define make-leaf

(lambda (value)

(make-tree value '() '())))

(define tree-value car)

(define tree-left cadr)

(define tree-right caddr)

Pomocı́ těchto struktur budeme reprezentovat množiny. Napı́šeme predikát in? zjišt’ujı́cı́, zda je zadaný
prvek v množině, a konstruktor cons-treepřidávajı́cı́ prvek do množiny. Při implementaci těchto procedur
budeme uvažovat stromy reprezentujı́cı́ množiny čı́sel a uspořádánı́ dané predikátem <=.

Procedura in? zastavuje rekurzi v přı́padě, že zkoumaný strom je prázdný – pak vracı́ nepravdu – nebo
v přı́padě, že hodnota stromu je shodná s hledaným prvkem – pak vracı́ pravdu. Pokud ani jedna z těchto
limitnı́ch podmı́nek podmı́nek nenı́ splněna, pokračuje hledánı́m prvku v levém nebo v pravém podstromu,
to podle porovnánı́ hledaného prvku a hodnoty stromu.

(define in?

(lambda (x tree)

(cond ((null? tree) #f)

((= x (tree-value tree)) #t)

((< x (tree-value tree)) (in? x (tree-left tree)))

(else (in? x (tree-right tree))))))

Procedura konstrukce množiny reprezentované stromem také zastavuje v přı́padě, že je strom prázdný.
Tehdy vracı́ jednoprvkový strom vytvořený procedurou make-leaf. Zastavuje také, pokud je hodnota
stromu rovna přidávanému prvku. V tom přı́padě je prvek už ve stromu přı́tomen a ten se jeho přidánı́m
nezměnı́, a je tedy vrácen původnı́ strom. Jinak je vytvořen nový strom do jehož levého nebo pravého
podstromu (v závislosti na porovnánı́ vkládaného prvku a hodnoty stromu) je vložen přidávaný prvek
(použitı́m procedury cons-tree).

(define cons-tree

(lambda (x tree)

(cond ((null? tree) (make-leaf x))

((= x (tree-value tree)) tree)

((< x (tree-value tree))

(make-tree (tree-value tree)

(cons-tree x (tree-left tree))

(tree-right tree)))

(else

Program 10.3. Výpočet potenčnı́ množiny.

(define power-set

(lambda (set)

(if (null? set)

'(())

(append (map (lambda (x)

(cons (car set) x))

(power-set (cdr set)))

(power-set (cdr set))))))

Program 10.4. Efektivnějšı́ výpočet potenčnı́ množiny.

(define power-set

(lambda (set)

(if (null? set)

'(())

(let ((power-rest (power-set (cdr set))))

(append (map (lambda (x)

(cons (car set) x))

power-rest)

power-rest)))))

(make-tree (tree-value tree)

(tree-left tree)

(cons-tree x (tree-right tree)))))))

10.4 Kombinatorika na seznamech

V této poslednı́ sekci předvedeme několik přı́kladů na výpočet kombinatorických úloh nad seznamy.
Budeme napřı́klad hledat všechny podmnožiny množiny reprezentované seznamem bez vı́cenásobných
výskytů prvků, všechny permutace, všechny k-prvkové kombinace nebo k-prvkové kombinace prvků
zadané množiny.

Výpočet potenčnı́ množiny 2A (množiny všech podmnožin množiny A) vytvořı́me podle tohoto rekurziv-
nı́ho přepisu:

2A =

{
{∅} pokud A = ∅,
2{a2,... } ∪

⋃{
{a1} ∪B|B ∈ 2{a2,... }

}
pokud A = {a1, a2, . . . } je neprázdná.

Pokud zadaná množina A = {a1, a2, . . . } nenı́ prázdná, vypočteme nejdřı́ve potenčnı́ množinu jejı́ pod-
množiny {a2, . . . }. Do výsledné množiny pak zařadı́me všechny prvky této potenčnı́ množiny „jakoby dva-
krát“. Jednou to budou přı́mo tyto množiny, podruhé tyto množiny s přidaným prvkem a1. Program 10.3
je přı́mým přepisem právě uvedeného popisu. Všimněte si, že se v programu 10.3 vyskytuje dvakrát výraz
(power-set (cdr set)). Výpočet můžeme zefektivnit pomocı́ lokálnı́ vazby tak, jak je uvedeno v pro-
gramu 10.4. Důsledkem, že potenčnı́ množinu podmnožiny počı́táme jen jednou, bude nejen kratšı́ čas
výpočtu, ale také struktura výsledného seznamu bude vypadat jinak (viz obrázek 10.3, vlevo je výsledek
pro původnı́ proceduru, vpravo je výsledek pro novou proceduru). Z čistě funkcionálnı́ho pohledu na
seznamy, který jsme doposud uplatňovali, je to však principiálně jedno.

Dále se budeme zabývat výpočtem všech permutacı́ n prvků. Budeme to provádět tak, že postupně projdeme
všech n prvků seznamu. Pro každý vytvořı́me seznam permutacı́ majı́cı́ch na začátku právě tento prvek. Ty

Obrázek 10.3. Výsledek aplikace staré a vylepšené verze power-set pro argument (a b).

. .

a . b ()

. .

a ()

. .

b ()

() () . .

a . b ()

. .

a ()

. . () (). .

.

. . . . . .

.

. . . .. .

Program 10.5. Výpočet všech permutacı́ prvků množiny.

(define permutation

(lambda (l)

(if (null? l)

'(())

(let perm

((base l)

(p-set (cdr l)))

(if (null? base)

'()

(append

(map (lambda (x)

(cons (car base) x))

(permutation p-set))

(perm (cdr base)

(cdr (append p-set (list (car base)))))))))))

vytvořı́me tak, ze najdeme (rekurzı́) permutace zbývajı́cı́ch prvků a na jejich začátek přidáme uvažovaný
prvek. Tyto seznamy spojı́me. Následujı́cı́ schéma zachycuje hledanı́ všech permutacı́ prvků seznamu (1

2 3):

uvažovaný prvek: 1 2 3

zbývajı́cı́ prvky: (2 3) (3 1) (1 2)

permutace zbývajı́cı́ch prvků: (2 3) (3 2) (3 1) (1 3) (1 2) (2 1)

permutace začı́najı́cı́ daným prvkem: (1 2 3) (1 3 2) (2 3 1) (2 1 3) (3 1 2) (3 2 1)

Program 10.5 obsahuje definici právě popsané procedury. Jednotlivé prvky zadaného seznamu l jsou pro-
cházeny pomocnou procedurou perm. Tato procedura si v argumentu p-set pamatuje seznam zbylých
prvků (to jest prvků, které jsou různé od právě zpracovávaného). Z prvků tohoto seznamu vytvořı́ per-
mutace (pomocı́ rekurzivnı́ho volánı́ procedury permutation), připojı́ na jejich začátek aktuálnı́ prvek,
a seznam takto vytvořených permutacı́ připojı́ k permutacı́m začı́najı́cı́m ostatnı́mi prvky (procedura perm
pokračuje zpracovánı́m dalšı́ho prvku).

Můžeme také generovat přı́mo konkrétnı́ permutace, aniž bychom museli vytvářet všechny. A to přes jejich
index. K tomu použijeme takzvanou faktoradickou soustavu čı́sel. Faktoradická čı́selná soustava je soustava
o proměnlivém základu založeném na faktoriálu:

základ: 7 6 5 4 3 2 1 0

hodnota pozice: 7! 6! 5! 4! 3! 2! 1! 0!

v dekadické soustavě: 5040 720 120 24 6 2 1 1

Prvnı́ch šest čı́sel faktoradické soustavy je ukázáno v prostřednı́m sloupci tabulky 10.4.

Vztah faktoradických čı́sel a permutacı́ n prvků je velice jednoduchý. My ted’popı́šeme, jak z faktoradického

Obrázek 10.4. Faktoradická čı́sla a permutace.

0 020100 (0,1,2)
1 021100 (0,2,1)
2 120100 (1,0,2)
3 121100 (1,2,0)
4 220100 (2,0,1)
5 221100 (2,1,0)

čı́sla zı́skat permutaci:

Z n prvků vytvořı́me tzv. seznam kandidátů. Čı́slice, která je v n-ciferné reprezentaci (přı́padně doplněné
počátečnı́mi nulami) faktoradického čı́sla nejvı́ce vlevo označuje index prvku v seznamu kandidátů, který
dosadı́me na nejlevějšı́ neobsazené pozici v n-tici permutace. Vybraný prvek odebereme ze seznamu kan-
didátů a z reprezentace faktoradického čı́sla odebereme nejlevějšı́ čı́slici. Zbytek pozic n-tice permutace
vyplnı́me jako permutaci zbývajı́cı́ch kandidátů podle zkráceného faktoradického čı́sla.

Přı́klad 10.1. Předved’me popsaný postup na následujı́cı́m přı́kladě
220100 {a, b, c} [c, , ]
0100 {a, b} [c, a, ]
00 {b} [c, a, b]

Permutace prvků a, b, c odpovı́dajı́cı́ faktoradickému čı́slu 220100 je permutace [c, a, b]. Dalšı́ přı́klady jsou
v tabulce 10.4.

Při implementaci budeme využı́vat procedur, které jsme naprogramovali dřı́ve. Konkrétně se jedná o pro-
cedurufak (viz programy 7.12, 8.3, 8.5 a 8.6 na stranách 187, 206, 208 a 212 a procedururemove z programu6.3
na straně 146, respektive jejı́ efektivnı́ implementaci).

Vyjdeme z jednoduché procedury na převod čı́sla dec v dekadické soustavě na čı́slo v bas-adické soustavě:

(define prevod

(lambda (dec bas)

(let iter ((dec dec)

(res '()))

(if (= dec 0) res

(iter (quotient dec bas) (cons (modulo dec bas) res))))))

Nynı́ uděláme dvě úpravy (viz program 10.6):

1. na mı́sto pevně zadaného základu bude procedura přijı́mat proceduru basf, která pro index (pozici,
počı́tanou zleva) vracı́ jejı́ základ. To proto, abychom mohli převádět čı́sla z dekadické soustavy na
soustavy s proměnlivým základem.

2. procedura bude vracet přesně l čı́sel (výsledek bude ořezán, nebo budou doplněny počátečnı́ nuly).
Čı́slo l bude předáváno proceduře jako dalšı́ parametr.

Program 10.6 obsahuje definici procedury pro převod prevod s právě popsanými úpravami a definici
procedury index->perm realizujı́cı́ výše popsané hledánı́ permutace náležejı́cı́ k danému indexu.

Viz přı́klady aplikace:

(index->perm 0 '(1 2 3)) Z=⇒ (1 2 3)

(index->perm 1 '(1 2 3)) Z=⇒ (1 3 2)

(index->perm 2 '(1 2 3)) Z=⇒ (2 1 3)

(index->perm 3 '(1 2 3)) Z=⇒ (2 3 1)

(index->perm 4 '(1 2 3)) Z=⇒ (3 1 2)

(index->perm 5 '(1 2 3)) Z=⇒ (3 2 1)

Program 10.6. Výpočet permutace prvků množiny pomocı́ faktoradických čı́sel.

(define prevod

(lambda (dec basf l)

(let iter ((dec dec)

(res '())

(i 1))

(if (> i l) res

(let ((bas (basf i)))

(iter (quotient dec bas)

(cons (modulo dec bas) res)

(+ i 1)))))))

(define index->perm

(lambda (i set)

(let perm ((set set)

(fnum (prevod i n! (length set))))

(if (null? set) '()

(cons (list-ref set (car fnum))

(perm (remove set (car fnum))

(cdr fnum)))))))

Program 10.7. Výpočet všech kombinacı́ prvků množiny.

(define combination

(lambda (k set)

(cond ((null? set) '())

((= k 0) '(()))

((= k 1) (map list set))

(else (append (map (lambda (x)

(cons (car set) x))

(combination (- k 1) (cdr set)))

(combination k (cdr set)))))))

Procedura combination přijı́má dva argumenty – čı́slo k a množinu S – a vracı́ seznam všech k-prvkových
kombinacı́ prvků množiny S (to jest seznam všech různých k-tic, jejichž složky jsou vzájemně různé prvky
množiny S). Pracuje tak, že procházı́ seznam reprezentujı́cı́ množinu S a pro každý prvek vytvořı́ dva
seznamy kombinacı́:

1. kombinace obsahujı́cı́ tento prvek. Ty najde tak, že vytvořı́ k − 1-prvkové kombinace doposud neuva-
žovaných prvků a pak do nich přidá právě zpracovávaný prvek.

2. kombinace neobsahujı́cı́ tento prvek. Hledajı́ se tedy k-prvkové kombinace doposud neuvažovaných
prvků.

Tyto seznamy se pak spojı́ dohromady. Viz program 10.7

Malou úpravou kódu z programu 10.7 dostaneme proceduru, která vracı́ seznam kombinacı́ s opaková-
nı́m. Program 10.8 se lišı́ od definice procedury combination-dup jen v tom detailu, že při rekurzivnı́m
volánı́ procedury, jehož výsledkem má být zbytek kombinace, neodebı́ráme už dosazený prvek z množiny
kombinovaných prvků.

Program 10.8. Výpočet všech kombinacı́ s opakovánı́m.

(define combination-dup

(lambda (k set)

(cond ((null? set) '())

((= k 0) '(()))

(else (append (map (lambda (x)

(cons (car set) x))

(combination-dup (- k 1) set))

(combination-dup k (cdr set)))))))

Shrnutı́

Tato lekce obsahuje pokročilejšı́ přı́klady sloužı́cı́ k procvičenı́ práce s hierarchickými daty. V prvnı́ části lekce
se budeme zabývat reprezentacı́ stromů pomocı́ seznamů a jejich prohledávánı́m do hloubky a do šı́řky.
Dále ukážeme reprezentaci množin pomocı́ uspořádaných seznamů a pomocı́ binárnı́ch vyhledávacı́ch
stromů. Závěr lekce je věnován kombinatorice na seznamech.

Lekce 11: Kvazikvotovánı́ a manipulace se symbolickými výrazy

Obsah lekce: Tato lekce obsahuje několik klasických přı́kladů zpracovánı́ symbolických dat se kterými
se lze setkat i v jiné v literatuře. Nejprve se budeme zabývat kvazikvotovánı́m, což je obecnějšı́ metoda
kvotovánı́ umožňujı́cı́ programátorům snadněji definovat některé seznamy. Dále se budeme zabývat ná-
sledujı́cı́mi okruhy problémů: zjednodušovánı́ aritmetických výrazů, symbolická derivace, konverze mezi
symbolickými výrazy zapsanými v prefixové, infixové, postfixové a polské bezzávorkové notaci. Nakonec
ukážeme metodu vyhodnocovánı́ výrazů v polské bezzávorkové notaci.

Klı́čová slova: kvazikvotovánı́, infixová notace, postfixová notace, polská bezzávorková notace.

11.1 Kvazikvotovánı́

V sekci 4.5 jsme ukázali speciálnı́ formu quote, která vracela svůj argument v nevyhodnocené podobě.
Nynı́ představı́me zobecněnı́ této speciálnı́ formy – quasiquote. Speciálnı́ forma quasiquote, v porovnánı́
se speciálnı́ formou quote, umožňuje navı́c určit podvýrazy jejı́ho argumentu, které budou vyhodnoceny
běžným způsobem. Bez tohoto určenı́ funguje speciálnı́ formaquasiquote stejným způsobem jako speciálnı́
forma quote. Viz následujı́cı́ přı́klady:

(quasiquote 15) Z=⇒ 15

(quasiquote symbol) Z=⇒ symbol

(quasiquote (x . 3)) Z=⇒ (x.3)

(quasiquote (1 2 (3 4) 5)) Z=⇒ (1 2 (3 4) 5)

Zatı́mco speciálnı́ forma quote „slepě“ vracı́ svůj argument, speciálnı́ forma quasiquote argument projde a
vyhledá v něm podvýrazy, které jsou jednoprvkové seznamy začı́najı́cı́ symbolem unquote. Tyto podvýrazy
jsou nahrazeny vyhodnocenı́m jejich druhého prvku. Nejlépe to bude vidět na přı́kladech:

(define x 3)

(quasiquote (unquote x)) Z=⇒ 3

(quasiquote (unquote (+ 1 2))) Z=⇒ 3

(quasiquote (1 x (unquote (+ 1 x 2)))) Z=⇒ (1 x 6)

Přı́klad 11.1. (a) Někdy se symbol unquote nemusı́ nacházet viditelně na začátku seznamu. V následujı́cı́
ukázce tento symbol nenı́ prvnı́m prvkem seznamu (1 unquote (+ 1 2)), ale až druhým. Tento seznam
je tečkový pár (1.(unquote (+ 1 2))), jehož druhým prvkem je seznam začı́najı́cı́ symbolem unquote:

(quasiquote (1 unquote (+ 1 2))) Z=⇒ (1 . 3)

je totéž jako

(quasiquote (1 . (unquote (+ 1 2)))) Z=⇒ (1 . 3).

(b) Symbol unquote musı́ být prvnı́m prvkem seznamu, nikoli jen páru. Navı́c seznam, jehož prvnı́ prvek je
symbol unquote, musı́ obsahovat právě jeden dalšı́ prvek. Následujı́cı́ výrazy neobsahujı́ správné použitı́
symbolu unquote v argumentu speciálnı́ formy quasiquote:

(quasiquote (1 (unquote . 2)))

(quasiquote (1 (unquote)))

(quasiquote (1 (unquote 2 3)))

Výsledky vyhodnocenı́ těchto výrazů standard jazyka Scheme R5RS neurčuje. Naopak přı́mo řı́ká, že v ta-
kových přı́padech docházı́ k nepředpovı́datelnému chovánı́. V našem abstraktnı́m interpretu necháme vy-
hodnocenı́ takových výrazů skončit chybou „CHYBA: Nekorektnı́ argument speciálnı́ formy quasiquote“.

Dalšı́ symbol se speciálnı́m významem pro speciálnı́ formu quasiquote je symbol unquote-splicing.
Pokud argument speciálnı́ formy quasiquote obsahuje seznam ve tvaru (unquote-splicing 〈arg〉), je
vyhodnocen výraz 〈arg〉. Předpokládá se, že se tento argument vyhodnotı́ na seznam. Celý výraz

(unquote-splicing 〈arg〉)
je pak nahrazen tı́mto výsledným seznamem bez otevı́rajı́cı́ a uzavı́rajı́cı́ závorky. Viz přı́klad:

(quasiquote (1 (unquote (build-list 3 +)) #t)) Z=⇒ (1 (0 1 2) #t)

(quasiquote (1 (unquote-splicing (build-list 3 +)) #t)) Z=⇒ (1 0 1 2 #t)

(quasiquote (1 (unquote (map - (list 1 2 3))) #t)) Z=⇒ (1 (-1 -2 -3) #t)

(quasiquote (1 (unquote-splicing (map - (list 1 2 3))) #t)) Z=⇒ (1 -1 -2 -3 #t)

Odstraněny jsou přitom jen vnějšı́ závorky. Pokud seznam vzniklý vyhodnocenı́m výrazu 〈arg〉 má jako
prvky dalšı́ seznamy, nebudou k nim náležejı́cı́ výrazy odstraněny. Tedy napřı́klad:

(quasiquote (1 2 (unquote-splicing (map list (list 1 2 3))))) Z=⇒ (1 2 (1) (2) (3))

V sekci 4.5 jsme zavedli použitı́ uvozovky ' jako syntaktický cukr pro speciálnı́ formu quote. Také pro
speciálnı́ formu quasiquote je k dispozici zkrácený zápis. Namı́sto výrazu (quasiquote 〈arg〉) můžeme
ekvivalentně psát jen `〈arg〉. Dále namı́sto výrazů (unquote 〈arg〉) a (unquote-splicing 〈arg〉) je možno
psát pouze ,〈arg〉 a ,@〈arg〉 (v tomto pořadı́). Napřı́klad:

`1 Z=⇒ 1

`symbol Z=⇒ symbol

`(x . 3) Z=⇒ (x.3)

`(1 2 ,(+ 2 3) #f) Z=⇒ (1 2 5 #f)

`(1 2 ,(build-list 3 +) #f) Z=⇒ (1 2 (0 1 2) #f)

`(1 2 ,@(build-list 3 +) #f) Z=⇒ (1 2 0 1 2 #f)

Podvýrazy argumentu speciálnı́ formy quasiquote mohou přirozeně opětně obsahovat použitı́ této speci-
álnı́ formy:

`(1 2 ,(list `(3))) Z=⇒ (1 2 ((3)))

`(1 2 ,(map (lambda (x) `(,x)) `(1 2 3))) Z=⇒ (1 2 ((1) (2) (3)))

Nynı́ ukážeme několik přı́kladů s použitı́m kvotovánı́ a kvazikvotovánı́. Všechny jsou uvedeny s použitı́m
syntaktického cukru, i po jeho odstraněnı́. Speciálnı́ forma quote svůj argument vracı́ bez jakékoli změny,
bez ohledu na to, jestli se uvnitř něho vyskytujı́ symboly unquote a unquote.

'`(1 2 3)=(quote (quasiquote (1 2 3)))

Z=⇒ (quasiquote (1 2 3))

`'(1 2 3)=(quasiquote (quote (1 2 3)))

Z=⇒ (quote (1 2 3))

``(1 2 3)=(quasiquote (quasiquote (1 2 3)))

Z=⇒ (quasiquote (1 2 3))

'`(1 2 ,(+ 1 2))=(quote (quasiquote (1 2 (unquote (+ 1 2)))))

Z=⇒ (quasiquote (1 2 (unquote (+ 1 2)))

`'(1 2 ,(+ 1 2))=(quasiquote (quote (1 2 (unquote (+ 1 2)))))

Z=⇒ (quote (1 2 3))

Upozorněme, že na symboly unquote a unquote-splicing nejsou navázány ani procedury, ani speciálnı́
formy. Tı́m, že na tyto symboly navážeme nějakou hodnotu, nezměnı́me vyhodnocovánı́ speciálnı́ formy
quasiquote. Viz následujı́cı́ přı́klady:

unquote Z=⇒ „CHYBA: Symbol unquote nemá vazbu.“
(define unquote (lambda (x) (+ 1 x)))

`(,10)=(quasiquote (unquote 10)) Z=⇒ (10)

`(,,10)=(quasiquote (unquote (unquote 10))) Z=⇒ (11)

Standard jazyka Scheme R5RS neřı́ká jakým způsobem má interpret zacházet se symboly unquote

a unquote-splicing. Některé interpretry, jako je napřı́klad DrScheme, se odlišujı́ od našeho abstrakt-
nı́ho interpretru v aplikaci speciálnı́ formy quasiquote. Výše popsaným způsobem se chovajı́ napřı́-
klad interpretry Bigloo nebo Guile. Napřı́klad v DrScheme po navázánı́ hodnoty na unquote přestane
správně fungovat vyhodnocovánı́ kvazikvotovaných výrazů.

11.2 Zjednodušovánı́ aritmetických výrazů

V této a následujı́cı́ch sekcı́ch uvedeme procedury pro zpracovánı́ symbolických výrazů. V této sekci půjde
o vytvořenı́ procedury simplify, která zjednodušuje aritmetické výrazy.

Aritmetické výrazy budeme v tomto přı́padě reprezentovat S-výrazy které jsou čı́sla, symboly nebo dvou-
prvkové nebo třı́prvkové seznamy, jejichž prvnı́ prvek je jeden ze symbolů +, -, * a / označujı́cı́ch operaci
a zbylé prvky jsou aritmetické výrazy. Čı́sla přitom reprezentujı́ čı́selné hodnoty, které označujı́, symboly
reprezentujı́ „proměnné“, a výrazy ve tvaru seznamu reprezentujı́ provedenı́ operace sčı́tánı́, odčı́tánı́, ná-
sobenı́ a dělenı́ nad danými operandy. Napřı́klad seznam (+ (* 3 x) 2) reprezentuje aritmetický výraz
ve tvaru 3x+ 2.

Kód procedury simplify uvedený v programu 11.1 nejdřı́ve popı́šeme a poté se budeme věnovat úpravám
tohoto řešenı́.

V těle procedury simplifydefinujeme několik pomocných procedur. Prvnı́ z nich je predikát ==, který zjistı́,
zdali jsou oba jeho argumenty čı́sla a zdali jsou si rovna (při porovnánı́ predikátem =). Dále definujeme
proceduru div-by-zero, pomocı́ niž bude vrácen přı́znak dělenı́ nulou a to v přı́padě, že by měl být
zjednodušován výraz, ve kterém se dělı́ nulou. V těle jsou dále uvedeny pomocné procedury zjednodušujı́cı́
určitý typ výrazu. V programu 11.1 máme uvedenu jen jednu z nich – proceduru simplify-add zajišt’ujı́cı́
zjednodušovánı́ výrazů, které jsou ve tvaru součtu. Dalšı́ procedury, které by se v kódu nacházely na mı́stě
výpustky, jsou uvedeny v programu 11.2. Jedná se o procedury simplify-min, simplify-mul a konečně
simplify-div zajišt’ujı́cı́ zjednodušovánı́ výrazů ve tvaru rozdı́lu, součinu a podı́lu.

Procedura simplify zjednodušuje výraz podle jeho charakteru. Jedná-li se o čı́slo nebo o symbol, jedná
se o atomický výraz a zjednodušit už nejde. Pokud se jedná o seznam začı́najı́cı́ symbolem operace, zjed-
nodušı́ se jeho operandy rekurzivnı́m volánı́m procedury simplify. V přı́padě, že se po zjednodušenı́
jedná o dvouprvkový seznam obsahujı́cı́ (mimo symbolu operace) čı́slo, je zjednodušenı́m přı́mo výsle-
dek vyhodnocenı́ tohoto výrazu. Totéž platı́ pro třı́prvkový seznam obsahujı́cı́ symbol operace a dvě
čı́sla. V ostatnı́ch přı́padech aplikujeme jednu ze zjednodušujı́cı́ch pomocných procedur (to jest procedur
simplify-add, . . . , simplify-div) v závislosti na symbolu operace, kterým zjednodušovaný výraz začı́ná.
Viz přı́klady aplikace:

(simplify '(+ 10)) Z=⇒ 10

(simplify '(+ 1 2)) Z=⇒ 3

(simplify '(+ (* 2 3) (+ y (+ 2 x)))) Z=⇒ (+ 6 (+ y (+ 2 x)))

(simplify '(- x 0)) Z=⇒ x

(simplify '(- 0 x)) Z=⇒ (- x)

(simplify '(/ 0 x)) Z=⇒ 0

(simplify '(/ x 0)) Z=⇒ division-by-zero

(simplify '(/ x 1)) Z=⇒ x

(simplify '(/ 1 x)) Z=⇒ (/ x)

(simplify '(/ (+ 2 x) (* 2 0.5))) Z=⇒ (+ 2 x)

Elegantnějšı́ho řešenı́ můžeme dosáhnout tak, že vytvořı́me tabulku, ve které budeme mı́t uložené pomocné
procedury na zjednodušovánı́ a k nim přı́slušné symboly operacı́. Implementovat bychom ji mohli napřı́klad
tak, jak je uvedeno v programu 11.3.

V proceduře simplify pak můžeme větvenı́ cond zkrátit třeba takto:

(if (and (number? x) (number? y)) (eval expr)

(apply (cdr (assoc op simplification-table)) expr))

Zde využı́váme procedury assoc, kterou jsme doposud neuvedli. Tato procedura přijı́má dva argumenty 〈e〉
a 〈l〉. Prvnı́ argument 〈e〉 je libovolný element a druhý argument 〈l〉 je seznam tečkových párů. Procedura
vracı́ ten nejlevějšı́ tečkový pár ze seznamu 〈l〉, jehož prvnı́ prvek je shodný s elementem 〈e〉. Viz ilustrativnı́
přı́klady použitı́ assoc:

(define s '((a . 10) (b . (a b c)) (20 30) (b . 666)))

(assoc 'a s) Z=⇒ (a . 10)

Program 11.1. Procedura pro zjednodušovánı́ aritmetických výrazů.

(define simplify

(lambda (expr)

;; zjisti, zda-li jsou oba argumenty stejna cisla

(define ==

(lambda (x y)

(and (number? x) (number? y) (= x y))))

;; osetreni deleni nulou

(define div-by-zero

(lambda () 'division-by-zero))

;; zjednoduseni pro pricitani

(define simplify-add

(lambda (op x y)

(cond ((== x 0) y) zjednodušenı́ využı́vajı́cı́ 0 + y = y
((== y 0) x) zjednodušenı́ využı́vajı́cı́ x+ 0 = 0
((equal? x y) `(* 2 ,x)) zjednodušenı́ využı́vajı́cı́ x+ x = 2 · x
(else (list op x y)))))

...

(cond

((number? expr) expr)

((symbol? expr) expr)

((and (list? expr) (member (car expr) '(+ * - /)))

(let* ((op (car expr))

(expr (map simplify expr)))

(if (null? (cddr expr))

(if (number? (cadr expr))

(eval expr)

expr)

(let ((x (cadr expr))

(y (caddr expr)))

(cond ((and (number? x) (number? y)) (eval expr))

((equal? op '+) (simplify-add op x y))

((equal? op '*) (simplify-mul op x y))

((equal? op '-) (simplify-min op x y))

(else (simplify-div op x y))))))))))

Program 11.2. Internı́ definice v proceduře pro zjednodušovánı́ výrazů.

;; zjednoduseni pro nasobeni

(define simplify-mul

(lambda (op x y)

(cond ((== x 0) 0) zjednodušenı́ využı́vajı́cı́ 0 · y = 0
((== y 0) 0) zjednodušenı́ využı́vajı́cı́ x · 0 = 0
((== x 1) y) zjednodušenı́ využı́vajı́cı́ 1 · y = y
((== y 1) x) zjednodušenı́ využı́vajı́cı́ x · 1 = y
(else (list op x y)))))

;; zjednoduseni pro odcitani

(define simplify-min

(lambda (op x y)

(cond ((== x 0) (list op y)) zjednodušenı́ využı́vajı́cı́ 0− y = −y
((== y 0) x) zjednodušenı́ využı́vajı́cı́ x− 0 = x
((equal? x y) 0) zjednodušenı́ využı́vajı́cı́ x− x = 0
(else (list op x y)))))

;; zjednoduseni pro deleni

(define simplify-div

(lambda (op x y)

(cond ((== x 0) 0) zjednodušenı́ využı́vajı́cı́ 0y = 0
((== x 1) (list op y)) zjednodušenı́ využı́vajı́cı́ 1y = y−1

((== y 0) (div-by-zero)) dělenı́ nulou je nedefinované
((== y 1) x) zjednodušenı́ využı́vajı́cı́ x

1 = x
((equal? x y) 1) zjednodušenı́ využı́vajı́cı́ x

x = 1
(else (list op x y)))))

Program 11.3. Tabulka procedur pro zjednodušovánı́ výrazů.

(define simplification-table

;; pomocne procedury == a div-by-zero

(let ((== (lambda (x y)

(and (number? x) (number? y) (= x y))))

(div-by-zero (lambda () 'division-by-zero)))

;; vlastni tabulka

`((+ . ,(lambda (op x y)

(cond ((== x 0) y)

((== y 0) x)

((equal? x y) `(* 2 ,x))

(else (list op x y)))))

(* . ,(lambda (op x y)

(cond ((== x 0) 0)

((== y 0) 0)
...

Program 11.4. Procedura pro vyhledávánı́ v asociačnı́m seznamu.

(define assoc

(lambda (key alist)

(cond ((null? alist) #f)

((equal? key (caar alist)) (car alist))

(else (assoc key (cdr alist))))))

(assoc 'b s) Z=⇒ (b a b c)

(assoc 20 s) Z=⇒ (20 30)

(assoc 'c s) Z=⇒ #f

Proceduru assoc by bylo jednoduché implementovat, jak ukazuje program 11.4.

Nynı́ se zaměřı́me na zobecněnı́ procedury simplify tak, aby bylo možné zjednodušovat i aritmetické
výrazy obsahujı́cı́ použitı́ + a * pro n operandů. Při sčı́tánı́ (násobenı́) vyfiltrujeme všechna čı́sla a se-
čteme (vynásobı́me) je. Zbylých, to jest nečı́selných prvků, vytvořı́me seznam. To provedeme napřı́klad
následovně s využitı́m procedur filter a remove (viz napřı́klad programy 6.2 a 6.3 na straně 146.)

(let ((value (apply (eval op) (filter number? (cdr expr))))

(compound (remove number? (cdr expr))))

· · ·)

Tyto hodnoty pak zpracujeme následujı́cı́ procedurou simplify-addmul:

(define simplify-addmul

(lambda (op compound value)

(cond ((and (equal? op '*) (= value 0)) 0)

((null? compound) value)

((= value (eval `(,op))) (if (null? (cdr compound))

(car compound)

(cons op compound)))

(else `(,op ,value ,@compound)))))

Procedura simplify-addmul vracı́ nulu, pokud se jedná o operaci * a čı́selná hodnota navázaná na for-
málnı́ argument value je nulová. A vracı́ přı́mo čı́selnou hodnotu, pokud je seznam složených výrazů
(navázaný na formálnı́ argument compound), prázdný. Pokud je čı́selná hodnota rovna neutrálnı́mu prvku
přı́slušné operace, což zjistı́me vyhodnocenı́m výrazu (= value (eval `(,op)), je vrácen původnı́ vý-
raz bez čı́selné hodnoty, nebo složený výraz ze seznamu compound, pokud je tento seznam jednoprvkový.
V ostatnı́ch přı́padech už nelze výraz zjednodušit.

Výsledný kód s uvedenými dvěma změnami by pak vypadal tak jak je to uvedeno v programu 11.5.

Následujı́ přı́klady aplikace:

(simplify '(+)) Z=⇒ 0

(simplify '(+ 10)) Z=⇒ 10

(simplify '(+ 1 2)) Z=⇒ 3

(simplify '(+ 1 2 x 3 y 5 6)) Z=⇒ (+ 17 x y)

(simplify '(+ (* 2 3) y (+ 2 x))) Z=⇒ (+ 6 y (+ 2 x))

(simplify '(- (* 1 x) (+ 2 -3 x 1))) Z=⇒ 0

(simplify '(- x 0)) Z=⇒ x

(simplify '(- 0 x)) Z=⇒ (- x)

(simplify '(/ 0 x)) Z=⇒ 0

(simplify '(/ x 0)) Z=⇒ division-by-zero

(simplify '(/ x 1)) Z=⇒ x

(simplify '(/ 1 x)) Z=⇒ (/ x)

Program 11.5. Vylepšená procedura pro zjednodušovánı́ aritmetických výrazů.

;; tabulka pro zjednodusovani

(define simplification-table

(let ((== (lambda (x y)

(and (number? x) (number? y) (= x y))))

(div-by-zero (lambda () 'division-by-zero))

(simplify+* (lambda (op . rest)

(let ((value (apply (eval op) (filter number? rest)))

(compound (remove number? rest)))

(simplify-addmul op compound value)))))

`((+ . ,simplify+*)

(* . ,simplify+*)

(- . ,(lambda (op x y)

(cond ((== x 0) (list op y))

((== y 0) x)

((equal? x y) 0)

(else (list op x y)))))

(/ . ,(lambda (op x y)

(cond ((== x 0) 0)

((== x 1) (list op y))

((== y 0) (div-by-zero))

((== y 1) x)

((equal? x y) 1)

(else (list op x y))))))))

;; vlastni procedura provadejici zjednodusovani

(define simplify

(lambda (expr)

(cond

((number? expr) expr)

((symbol? expr) expr)

((list? expr)

(let ((op (car expr))

(expr (map simplify expr)))

(if (forall number? (cdr expr))

(eval expr)

(let ((simplifier (assoc op simplification-table)))

(if simplifier

(apply (cdr simplifier) expr)

(error "No record for such operation"))))))

(else (error "Incorrect input expression")))))

(simplify '(/ (+ 2 x) (+ 1 x (* 2 1/2)))) Z=⇒ 1

Procedura simplify samozřejmě ještě nenı́ (zdaleka) dokonalá. Na závěr sekce ukážeme několik přı́kladů
aplikace procedury simplify, které ukazujı́ jejı́ nedostatky:

(simplify '(+ x (- x))) Z=⇒ (+ x (- x)) ideálnı́ zjednodušenı́: 0

(simplify '(* x (/ 1 x))) Z=⇒ (* x (/ x)) ideálnı́ zjednodušenı́: 1

(simplify '(+ 1 (+ 1 (+ 1 x)))) Z=⇒ (+ 1 (+ 1 (+ 1 x) ideálnı́ zjednodušenı́: (+ x 3)

(simplify '(+ x x x x)) Z=⇒ (+ x x x x) ideálnı́ zjednodušenı́: (* x 4)

11.3 Symbolická derivace

V této sekci se budeme zabývat realizacı́ procedury diff na nalezenı́ symbolické derivace aritmetického
výrazu (podle dané proměnné). Pro jednoduchost se omezı́me jen na aritmetické výrazy s binárnı́mi ope-
racemi.

Jak na to tedy půjdeme: Proměnné budeme derivovat na čı́slo 1, konstanty na čı́slo 0, složitějšı́ výrazy
pak podle jejı́ho derivačnı́ho předpisu. Tento předpis budeme vybı́rat na základě operace, kterou výraz
začı́ná. Předpis přitom budeme reprezentovat procedurou dvou argumentů, kterými budou operandy
derivovaného výrazu. Procedura předpisu bude vracet výraz odpovı́dajı́cı́ derivaci výrazu. Napřı́klad pro
operaci +, to bude procedura, která vznikne vyhodnocenı́m λ-výrazu

(lambda (x y) `(+ ,(derive x) ,(derive y))),

která formalizuje známý derivačnı́ vztah (f + g)′ = f ′ + g′. Operace a k nim přı́slušné předpisy budeme
uchovávat v tabulce obdobně jako v sekci 11.2. Tato tabulka bude opět reprezentovaná seznamem řádků.
Těmito řádky budou tečkové páry, tabulka tedy bude organizována jako asociačnı́ seznam. Tento přı́stup
umožňuje rozšiřovat proceduru pro derivovánı́ pouhým přidávánı́m řádků do této tabulky. Napřı́klad
kdybychom chtěli derivovat i výrazy obsahujı́cı́ podvýrazy ve tvaru (sin x), stačilo by přidat do kódu
řádek

(sin . ,(lambda (x) `(* (cos ,x) ,(derive x))),

což formalizuje vztah (sin f)′ = (cos f ′) · f ′ (nezapomeňte, že f nemusı́ být pouze proměnná). Definice
procedury diff je uvedena v programu 11.6.

Procedura diff tedy pro aritmetický výraz a proměnnou vracı́ jeho derivaci podle této proměnné. Viz
přı́klady aplikace této procedury:

(diff 'x 'x) Z=⇒ 1

(diff 'x 'y) Z=⇒ 0

(diff '(+ x x) 'x) Z=⇒ (+ 1 1)

(diff '(+ x x) 'y) Z=⇒ (+ 0 0)

(diff '(* x y) 'x) Z=⇒ (+ (* x 0) (* 1 y))

(diff '(+ (* x 2) (* x x)) 'x) Z=⇒ (+ (+ (* x 0) (* 1 2)) (+ (* x 1) (* 1 x)))

(diff '(* x (* x x)) 'x) Z=⇒ (+ (* x (+ (* x 1) (* 1 x))) (* 1 (* x x)))

Výsledný výraz můžeme zjednodušit pomocı́ procedury simplify, kterou jsme naspali v předchozı́ sekci.

(simplify (diff 'x 'x)) Z=⇒ 1

(simplify (diff 'x 'y)) Z=⇒ 0

(simplify (diff '(+ x x) 'x)) Z=⇒ 2

(simplify (diff '(+ x x) 'y)) Z=⇒ 0

(simplify (diff '(* x y) 'x)) Z=⇒ y

(simplify (diff '(+ (* x 2) (* x x)) 'x)) Z=⇒ (+ (+ x x) 2)

(simplify (diff '(* x (* x x)) 'x)) Z=⇒ (+ (* x (+ x x)) (* x x))

Poznámka 11.2. Všimněte si, že v těle procedury diff v programu 11.6 jsme definovali pomocnou pro-
ceduru derive v jejı́mž těle je použita tabulka pravidel. Na druhou stranu ale taky platı́, že procedury,
které se nacházejı́ v tabulce pravidel, použı́vajı́ pomocnou proceduru derive. To je ale zcela v pořádku

Program 11.6. Procedura pro symbolickou derivaci.

(define diff

(lambda (expr var)

(define variable?

(lambda (expr)

(equal? expr var)))

(define constant?

(lambda (expr)

(or (number? expr)

(and (symbol? expr)

(not (variable? expr))))))

(define table

`((+ . ,(lambda (x y) `(+ ,(derive x) ,(derive y))))

(- . ,(lambda (x y) `(- ,(derive x) ,(derive y))))

(* . ,(lambda (x y) `(+ (* ,x ,(derive y)) (* ,(derive x) ,y))))

(/ . ,(lambda (x y) `(/ (- (* ,(derive x) ,y) (* ,x ,(derive y)))

(* ,y ,y))))))

(define derive

(lambda (expr)

(cond ((variable? expr) 1)

((constant? expr) 0)

(else (apply (cdr (assoc (car expr) table))

(cdr expr))))))

(derive expr)))

a kód je naprosto funkčnı́. Někoho by možná mohlo zmást, že v těle procedury máme několik procedur,
které se na sebe vzájemně odkazujı́. Mohla by se tedy nabı́dnout otázka, zdali je to vůbec přı́pustné a jestli
k vůli tomu při aplikaci diff nemůže dojı́t k chybě. K chybě nedojde proto, že veškeré aplikace procedur
derive a procedur v tabulce pravidel probı́hajı́ až v okamžiku, kdy jsou v lokálnı́m prostředı́ procedury
diff zavedeny všechny lokálnı́ vazby symbolů variable?, constant?, table a derive. To je důsledkem
toho, že při vzniku procedury se nevyhodnocuje jejı́ tělo. Jinými slovy, během vzniku procedury vůbec
nevadı́, že v jejı́m těle je uveden výraz obsahujı́cı́ symbol, který je (zatı́m) bez vazby. Podstatné je, že vazby
budou existovat v okamžiku jejı́ aplikace, což je v přı́padě programu 11.6 splněno.

Pomocı́ procedury diff můžeme napsat proceduru vyššı́ho řádu, která vracı́ proceduru jako proceduru
jednoho argumentu reprezentujı́cı́ matematickou funkci. Provedeme to tak, že zkonstruujeme λ-výraz, jehož
tělem je výsledek aplikace procedur simplify a diff na zadaný výraz. Ten vyhodnotı́me procedurou eval.

(define diff-procedure

(lambda (expr var)

(eval `(lambda (,var)

,(simplify (diff expr var))))))

Procedura diff-procedure na rozdı́l od procedury derivace, kterou jsme implementovali v programu 2.7
na straně 61 vracı́ přesnou reprezentaci derivované funkce. Viz následujı́cı́ přı́klad použitı́:

(define f (diff-procedure '(* x x) 'x))

(f 1) Z=⇒ 2

(f 10) Z=⇒ 20

(f 15) Z=⇒ 30

Na druhou stranu procedura derivace pro přibližnou derivaci z programu 2.7 je univerzálnějšı́ v tom,
že jı́ můžeme předat jako argument rovnou proceduru. Procedura diff pracuje nad symbolickými výrazy
a pokud by byla derivované funkce zastoupena procedurou obsahujı́cı́ operace, které nejsou v tabulce
pravidel pro derivaci, museli bychom tabulku rozšı́řit.

Programovacı́ jazyky FORTRAN a LISP vznikly prakticky současně (FORTRAN je o něco málo staršı́).
Přesto je filosofie obou jazyků naprosto odlišná a dá se řı́ct, že oba dva jazyky předurčily vývoj mnoha
dalšı́ch rodin programovacı́ch jazyků. Zatı́mco FORTRAN byl programovacı́ jazyk sloužı́cı́ pro nume-
rické výpočty, tedy jediná data, která bylo možné ve FORTRANu zpracovávat, byla čı́sla, LISP byl již
od počátku jazykem zpracovávajı́cı́m symbolická data, viz [MC60]. Za symbolická data považujeme data
reprezentujı́cı́ symbolické výrazy, tedy čı́sla, symboly a seznamy. To je výrazný posun proti pouhému
„numerickému zpracovánı́ čı́sel“. Vznik LISPu byl motivován potřebou mı́t k dispozici programovacı́
jazyk, ve kterém bude možné pracovat s procedurami reprezentujı́cı́mi rekurzivnı́ funkce nad sym-
bolickými daty. Za zmı́nku stojı́, že motivačnı́ přı́klad se symbolickými derivacemi byl představen již
v prvnı́m publikovaném dokumentu o jazyku LISP, kterým je článek [MC60].

11.4 Infixová, postfixová a bezzávorková notace

V této sekci se budeme zabývat problémem převodu výrazů zapsaných v různých notacı́ch. Již v prvnı́
lekci jsme konstatovali, že jazyk Scheme použı́vá prefixovou notaci, ale běžně se také použı́vajı́ jiné notace.
Nejčastějšı́ je infixová, méně časté jsou postfixová notace a postfixová bezzávorková notace (tak zvaná polská
reverznı́ notace), viz sekci 1.2. Při psanı́ praktických aplikacı́ se můžeme setkat s problémem převodu výrazů
mezi těmito notacemi. Tı́m se tedy budeme zabývat nynı́.

Prvnı́ a nejdůležitějšı́ je si vždy uvědomit strukturu výrazu. Máme-li napřı́klad symbolický výraz

(+ (* 2 x) (- (/ (+ x 2) z)) 5),

pak jej z hlediska operacı́ a operandů můžeme zachytit uzlově ohodnoceným stromem tak, jak je tomu
v obrázku 11.1 vlevo. Z hlediska datové reprezentace je seznam (+ (* 2 x) (- (/ (+ x 2) z)) 5)

Obrázek 11.1. Struktura výrazu (+ (* 2 x) (- (/ (+ x 2) z)) 5)

Obrázek 11.2. Fyzická struktura seznamu (+ (* 2 x) (- (/ (+ x 2) z)) 5)

reprezentován strukturou párů, která je nakreslena v obrázku 11.2. Problém převodu prefixového výrazu
do ostatnı́ch notacı́ je vlastně problémem průchodu stromem naznačeným v obrázku 11.1, který je fyzicky
reprezentovaný strukturou z obrázku 11.2. Pokud budeme tento strom procházet do hloubky, to jest ve
směru tak, jak to naznačuje šı́pka v obrázku 11.1 vpravo, projdeme postupně všechny uzly reprezentujı́cı́
„podvýrazy“ přitom pro daný výraz je vždy jako prvnı́ zpracován jeho nejlevějšı́ podvýraz. Jelikož se pre-
fixová, infixová a postfixová notace lišı́ jen pozicı́ (symbolu) operace, převod můžeme provést jednoduše
tak, že během průchodu strukturou budeme konstruovat jejı́ „kopii“ až na to, že v každém nelistovém uzlu
budeme vždy dávat symbol pro operaci na požadované mı́sto: tedy bud’ před, mezi, nebo za všechny ope-
randy. To v podstatě v grafové terminologii odpovı́dá pre-order, in-order a post-order zpracovánı́ nelistových
uzlů stromu.

V programu 11.7 je uvedena procedura prefix->postfix pro převod výrazu v prefixové notaci do post-
fixové notace. Při splněnı́ prvnı́ch třı́ podmı́nek v cond-výrazu je převod triviálnı́. V přı́padě, že vstupnı́
výraz je seznam, je nejprve rekurzivně aplikována procedura prefix->postfix, což způsobı́ převod všech
podvýrazů do postfixové notace. Z výsledku jejich převodu je vytvořen výsledný výraz připojenı́m symbolu
pro operaci za poslednı́ z něj. Viz přı́klady použitı́ procedury:

(prefix->postfix '(* 2 x)) Z=⇒ (2 x *)

(prefix->postfix '(- (/ (+ x 2) z))) Z=⇒ (((x 2 +) z /) -)

Upravenı́m procedury pro převod do postfixové notace můžeme vytvřit proceduru prefix->polish pro
převod prefixových výrazů do polské reverznı́ bezzávorkové notace tak, jak je to ukázáno v programu 11.8.
Jedinou změnou oproti myšlence použité v programu prefix->postfix je to, že nynı́ kromě připojenı́
symbolu operace až za poselnı́ operand navı́c odstraňujeme z výrazu veškeré závorky (kromě vnějšı́ch). To
v podstatě odpovı́dá operaci linearlizace, kterou jsme probrali v lekci 9, viz sekci 9.5. Vskutku, bezzávorko-

Program 11.7. Procedura pro převod výrazů do postfixové notace.

(define prefix->postfix

(lambda (S-expr)

(cond ((number? S-expr) S-expr)

((symbol? S-expr) S-expr)

((null? S-expr) S-expr)

((list? S-expr)

(append (map prefix->postfix (cdr S-expr))

(list (car S-expr))))

(else "incorrect expression"))))

Program 11.8. Procedura pro převod do postfixové (polské) bezzávorkové notace.

(define prefix->polish

(lambda (S-expr)

(cond ((number? S-expr) (list S-expr))

((symbol? S-expr) (list S-expr))

((null? S-expr) (list S-expr))

((list? S-expr)

(append (apply append (map prefix->polish (cdr S-expr)))

(list (car S-expr))))

(else "incorrect expression"))))

vou notaci bychom z prefixové mohli zı́skat pouhou linearizacı́ přı́slušného seznamu. Na druhou stranu,
procedura z programu 11.8 pracuje efektivněji (jednoprůchodově). Viz přı́klady použitı́ procedury:

(prefix->polish 'x) Z=⇒ (x)

(prefix->polish 20) Z=⇒ (20)

(prefix->polish '(* 2 x)) Z=⇒ (2 x *)

(prefix->polish '(- (/ (+ x 2) z))) Z=⇒ (x 2 + z / -)

(prefix->polish '(* 2 (+ 3 5))) Z=⇒ (2 3 5 + *)

(prefix->polish '(* (+ 2 3) 5)) Z=⇒ (2 3 + 5 *)

Na předchozı́m přı́kladu si povšimněte dvou věcı́. Výsledný výraz je vždy ve tvaru jediného lineárnı́ho
seznamu a to i v přı́padě, že vstupnı́ výraz byl atom, to je rozdı́l oproti závorkované postfixové notaci.
Na poslednı́ch dvou řádcı́ch předchozı́ho přı́kladu je vidět, jak se do bezzávorkové notace promı́tá jiné
uzávorkovánı́ dvou aritmetických výrazů.

Analogicky, jako jsme naprogramovali procedury pro převod z prefixové do postfixových notacı́, bychom
mohli naprogramovat i procedury pro opačné převody. Jejich naprogramovánı́ je vı́ce méně rutinnı́ a ne-
cháme jej na laskavém čtenáři. Mnohem obtı́žnějšı́ je však manipulace s infixovými výrazy. Nejprve uved’me
proceduru pro převod prefixových výrazů do infixu. Procedura prefix->infix vykonávajı́cı́ tento převod
je uvedena v programu 11.9. Převod čı́sel a symbolů do infixu je opět triviálnı́. V přı́padě seznamů mu-
sı́me ošetřit několik situacı́. Ve vnitřnı́m cond-výrazu nejprve ošetřujeme situaci, kdy je převáděný seznam
jednoprvkový, to jest obsahuje pouze symbol pro operaci a žádné operandy. V tomto přı́padě je převod
opět triviálnı́. Dále ošetřujeme situaci, kdy máme vstupnı́ operaci s jednı́m operandem. Ten převedeme do
infixu, ale symbol pro operaci pı́šeme pořád před něj, to odpovı́dá napřı́klad hodnotám−2, 12 reprezentova-
ným seznamy (- 2) a (/ 2), a podobně. V ostatnı́ch přı́padech musı́me za každý operand vložit symbol
pro operaci. Navı́c do sebe vnořı́me závorky tak, aby každá operace měla pouze dva operandy. K tomuto
účelu můžeme s výhodnou použı́t proceduru foldl, viz lekci 7 věnujı́cı́ se akumulaci. Následujı́cı́ přı́klady
ukazujı́ použitı́ procedury.

Program 11.9. Procedura pro převod výrazů do infixové notace.

(define prefix->infix

(lambda (S-expr)

(cond ((null? S-expr) S-expr)

((number? S-expr) S-expr)

((symbol? S-expr) S-expr)

((pair? S-expr)

(let* ((op (car S-expr))

(tail (cdr S-expr))

(len (length tail)))

(cond ((= len 0) S-expr)

((= len 1) (list op (prefix->infix (car tail))))

(else (foldl (lambda (expr collected)

(list collected op expr))

(prefix->infix (car tail))

(map prefix->infix (cdr tail)))))))

(else "incorrect expression"))))

(prefix->infix '(-)) Z=⇒ (-)

(prefix->infix '(- 2)) Z=⇒ (- 2)

(prefix->infix '(- 2 3)) Z=⇒ (2 - 3)

(prefix->infix '(- 2 3 4)) Z=⇒ ((2 - 3) - 4)

(prefix->infix '(- (/ (+ x 2) z))) Z=⇒ (- ((x + 2) / z))

Převod výrazu z obrázku 11.1 by dopadl takto:

(prefix->infix '(+ (* 2 x) (- (/ (+ x 2) z)) 5))

Z=⇒ (((2 * x) + (- ((x + 2) / z))) + 5)

Poznámka 11.3. Podotkněme, že proceduraprefix->infixnenı́ zdaleka dokonalá, nerespektuje napřı́klad
přirozenou asociativitu operacı́. Všechny výrazy ve tvaru x1 } · · ·} xn chápe jako výrazy tvaru

(· · · ((x1 } x2)} x3) · · ·} xn),

což nemusı́ být vždy žádoucı́. V některých přı́padech vyžadujeme závorkovánı́ opačně, někdy jej lze úplně
vynechat. Úpravy procedury ponecháváme na čtenáři.

Na závěr sekce poznamenejme, že pokud jsme řekli, že převod do infixové notace je obtı́žnějšı́ než převod
na postfixovou notaci, který je vı́ce méně rutinnı́, pak musı́me řı́ct, že převod z infixové notace (třeba do prefi-
xové) je ještě výrazně složitějšı́. Zvlášt’je tomu tak právě v přı́padě, kdy se vynechávajı́ závorky z důvodu
asociativity operacı́, nebo když chceme do našich úvah započı́st pravidla pro priority operacı́ (pravidla typu
„násobenı́ “ má přednost před „sčı́tánı́m“, tedy infixový výraz (2 * 3 + 5) znamená v prefixové notaci
(+ (* 2 3) 5) a nikoliv (* 2 (+ 3 5))). Problém rozpoznávánı́ struktury výrazů je obecně obtı́žný.
V informatice se tı́mto obecným problémem zabývá samotná disciplı́na – teorie formálnı́ch jazyků a auto-
matů, která je přednášena jako jeden ze základnı́ch kursů na všech informatických oborech vysokých škol.
Proto ponecháme převody z infixové notace zatı́m stranou. Co bychom si ale z této sekce měli odnést je
poznatek, že jednou z hlavnı́ch výhod prefixové notace je jejı́ jednoduchost a tı́m pádem snadná strojová
zpracovatelnost. Zvolenı́ prefixové notace symbolických výrazů jako základnı́ notace pro dialekty LISPu
lze tedy bez nadsázky označit jako geniálnı́ tah.

11.5 Vyhodnocovánı́ výrazů v postfixové a v bezzávorkové notaci

V této sekci se budeme snažit pro výrazy použı́vané v předchozı́ sekci sestavit vhodné evaluátory. Pro
výrazy v prefixovém tvaru to činit nemusı́me, protože evaluátor již je nám k dispozici v podobě procedury

Program 11.10. Procedura vyhodnocujı́cı́ postfixové výrazy.

(define postfix-eval

(lambda (expr)

(cond

((number? expr) expr)

((symbol? expr) (eval expr))

(else

(let iter ((expr expr)

(args '()))

(cond ((null? expr) '())

((null? (cdr expr))

(apply (postfix-eval (car expr))

(reverse args)))

(else (iter (cdr expr)

(cons (postfix-eval (car expr)) args)))))))))

eval a jedná se samotný evaluátor jazyka Scheme. Zaměřı́me se tedy na konstrukci procedur provádějı́cı́
vyhodnocovánı́ výrazů v postfixové a bezzávorkové reverznı́ notaci. Infixovou notacı́ se opět kvůli složitosti
nebudeme zabývat (studenti se s problematikou bı́že setkajı́ také v kursu překladačů).

Vyhodnocovánı́ závorkovaných postfixových výrazů bychom mohli provést analogicky jako vyhodnoco-
vánı́ výrazů v prefixové notaci, pouze musı́me počı́tat s tı́m, že výraz, jenž se má vyhodnotit na proceduru
(nebo na speciálnı́ formu), stojı́ v seznamu jako poslednı́. Neformálně můžeme popsat vyhodnocovánı́
výrazů v postfixové notaci následovně:

• čı́slo se vyhodnotı́ na svou hodnotu,

• symbol se vyhodnotı́ na svou vazbu,

• je-li daný výraz seznam, tak se začnou vyhodnocovat jeho prvky jeden po druhém, až se vyhodnotı́
poslednı́ z nich, pak se ověřı́, jestli se (poslednı́) vyhodnotil na proceduru. Pokud ano, je procedura
aplikována s argumenty jimiž jsou elementy vzniklé vyhodnocenı́m předchozı́ch prvků seznamu.

Postup můžeme formalizovat procedurou postfix-eval uvedenou v programu 11.10. Tato procedura
obsahuje pomocnou koncově rekurzivnı́ proceduru iter, která je v jejı́m těle jednorázově aplikována
(pomocı́ pojmenovaného let). Tato pomocná procedura se stará o postupné procházenı́ prvků v seznamu,
které jsou během průchodu postupně vyhodnocovány. Výsledky jejich vyhodnocenı́ se postupně akumulujı́
v seznamu navázaném na args. Při dosaženı́ poslednı́ho prvku je aplikována procedura zı́skaná vyhod-
nocenı́m poslednı́ho elementu. Argumenty předané při aplikaci procedury jsou právě elementy, které byly
akumulovány v seznamu navázaném na args.

(postfix-eval '(60 (10 20 +) /)) Z=⇒ 2

(postfix-eval '((10 20 +) 60 /)) Z=⇒ 1/2

(postfix-eval '(10 20 +)) Z=⇒ 30

Je samozřejmě možné vyhodnocovat i výrazy obsahujı́cı́ procedury pro manipulaci s páry:

(postfix-eval '((10 20 cons) (30 40 cons) list)) Z=⇒ ((10 . 20) (30 . 40))

Upozorněme na fakt, že vyhodnocovacı́ proces implementovaný v programu 11.10 je oproti vyhodnoco-
vacı́mu procesu jazyka Scheme silně zjednodušený. Vůbec se napřı́klad nepočı́tá s použitı́m speciálnı́ch
forem. Viz přı́klad:

(postfix-eval '(20 ((x) x lambda))) Z=⇒ „CHYBA: Symbol x nemá vazbu“

Důvodem chyby je právě pořadı́, v jakém vyhodnocujeme argumenty. Pokud postupujeme v seznamu od
jeho počátku až ke konci, právě až na konci zjistı́me, zda-li se daný výraz vyhodnotil na proceduru nebo

na speciálnı́ formu. To jest zavedenı́ speciálnı́ch forem v našem modelu vyhodnocovánı́ by nemělo smysl,
protože ještě před tı́m, než zjistı́me typ aplikovaného elementu, jsou již vyhodnoceny všechny argumenty.

Nynı́ obrátı́me našı́ pozornost na vyhodnocovánı́ bezzávorkových výrazů. To by mohlo být na prvnı́ pohled
složitějšı́, protože ve výrazech chybějı́ závorky explicitně určujı́cı́ „strukturu výrazů“. I v tomto přı́padě lze
ale navrhnout jednoznačný vyhodnocovacı́ proces a implementovat jej.

Před tı́m, než začneme konstruovat evaluátor výrazů, si musı́me objasnit jeden nepřı́jemný rys, který
souvisı́ s odstraněnı́m závorek z postfixových výrazů. V bezzávorkové notaci totiž již obecně neexistuje
jednoznačný přepis prefixových výrazů, pokud bereme v potaz symboly pro operace s libovolným početem
operandů, viz přı́klad:

(+ 2 3 (* 4 5)) Z=⇒ 25

(prefix->polish '(+ 2 3 (* 4 5))) Z=⇒ (2 3 4 5 * +)

(+ 2 (* 3 4 5)) Z=⇒ 62

(prefix->polish '(+ 2 (* 3 4 5))) Z=⇒ (2 3 4 5 * +)

Zde vidı́me, že dva různé výrazy majı́ stejnou reverznı́ bezzávorkovou reprezentaci. Než přistoupı́me ke
konstrukci vyhodnocovacı́ho procesu, musı́me tuto situaci vyřešit. Jinak bychom nevěděli, zda-li vyhodno-
tit výraz (2 3 4 5 * +) na hodnotu 25 nebo na hodnotu 62 (nebo ještě nějak úplně jinak). Jelikož problém
nejednoznačnosti spočı́vá v tom, že neznáme počet operandů pro operace, nabı́zı́ se dvě metody řešenı́:

(i) Každou operaci uvažovat vždy pouze s pevnou aritou, to jest s pevným počtem operandů. Toto řešenı́
je jednoduché a my se jej přidržı́me. Zavedeme tabulku symbolů, ve které budeme mı́t pro každou
operaci zaznamenán počet jejı́ch operandů.

(ii) Do výrazů budeme před každý symbol operace vždy zapisovat počet operandů, na které se vztahuje.
Napřı́klad ve výrazu (+ 2 3 (* 4 5)) má „+“ tři operandy a „*“ pouze dva. V reverznı́ bezzávor-
kové notaci bychom tedy výraz zapsali ve tvaru (2 3 4 5 2 * 3 +), přitom červeně jsou zdůrazněny
hodnoty reprezentujı́cı́ počty operandů. Na druhou stranu výraz (+ 2 (* 3 4 5)) by byl reprezento-
ván seznamem (2 3 4 5 3 * 2 +). Oba výrazy jsou tedy různé a k nejednoznačnostem nedocházı́.
Výrazy kódované tı́mto způsobem jsou pochopitelně delšı́, ale zase v nich můžeme použı́vat operace
s libovolnými operandy.

Nynı́ vezmeme v úvahu úmluvu (i) a budeme se soustředit na implementaci. Jako prvnı́ budeme definovat
tabulku symbolů, viz definici v programu 11.11. V této tabulce je pro každý symbol uvažované operace

Program 11.11. Jednoduchá tabulka s prostředı́m vazeb pro postfixový evaluátor.

(define env

`((+ 2 ,+)

(- 2 ,-)

(/ 2 ,/)

(* 2 ,*)

(n 1 ,-)

(^ 2 ,expt)))

třı́prvkový záznam ve formě seznamu (〈symbol〉 〈arita〉 〈procedura〉), kde 〈symbol〉 je jméno operace, 〈arita〉
je čı́slo určujı́cı́ počet operandů dané operace a 〈procedura〉 je procedura použı́vajı́cı́ se při „aplikaci operace“.
Všimněte si, že v tabulce 11.11 jsme definovali unárnı́ operaci označenou n, což je unárnı́ minus. Pro unárnı́
minus již nemůžeme zvolit symbol „-“, který je vyhrazen pro odčı́tánı́ dvou čı́sel (uvědomte si, že arita „-“
je dána pevně na jedinou hodnotu).

Vyhodnocovánı́ postfixových výrazů budeme provádět pomocı́ dodatečné datové struktury – zásobnı́ku.
Samotná reprezentace zásobnı́ku je z našeho pohledu primitivnı́, protože jej můžeme reprezentovat přı́mo
seznamem (cons „přidává“ prvek na vrchol zásobnı́ku, car vracı́ prvek na vrcholu, a cdr „odebı́rá“ prvek
z vrcholu zásobnı́ku). Během vyhodnocovánı́ tedy budeme mı́t k dispozici vstupnı́ výraz reprezentovaný

lineárnı́m seznamem čı́sel a symbolů a zásobnı́k. V každém kroku se ze vstupnı́ho výrazu odebere nejlevějšı́
„slovo“ (čı́slo nebo symbol) a zpracuje se. Zásobnı́k je na počátku prázdný. Podle typu slova na vstupu
budeme rozlišovat dvě situace:

(i) Na vstupu je čı́slo. V tomto přı́padě přesuneme čı́slo na vrchol zásobnı́ku a zpracujeme následujı́cı́
vstupnı́ slovo.

(ii) Je-li na vstupu symbol f označujı́cı́ operaci, odstranı́me jej ze vstupu a vyzvedneme ze zásobnı́ku
tolik prvků, jaká je arita symbolu f (aritu nejdeme v tabulce symbolů, kterou jsme již zavedli) a apli-
kujeme přı́slušnou proceduru s těmito argumenty (procedura je opět k nalezenı́ v tabulce). Výsledek
vyhodnocenı́ dáme na vrchol zásobnı́ku.

Vyhodnocovánı́ končı́ vypotřebovánı́m vstupu. V tom přı́padě je výsledek uložen na vrcholu zásobnı́ku
(nebo za výsledek vyhodnocovánı́ můžeme považovat celý zásobnı́k).

Přı́klad 11.4. (a) Uvažujme výraz (10 20 +). Průběh jeho vyhodnocovánı́ daný předchozı́m postupem,
můžeme zaznamenat v tabulce se dvěma sloupci. Prvnı́ sloupec ukazuje stav vstupnı́ho výrazu, ze kterého
jsou postupně zepředu odebı́rány prvky, a druhý sloupec představuje aktuálnı́ stav zásobnı́ku. Každý
řádek tabulky pak koresponduje s jednı́m elementárnı́m krokem výpočetu. V přı́padě našeho výrazu bude
výpočet vypadat takto:

vstup zásobnı́k
(10 20 +) ()

(20 +) (10)

(+) (20 10)

() (30)

Výsledek vyhodnocenı́ je tedy 30.

(b) V přı́padě výrazu '(10 20 30 + *) probı́há vyhodnocovánı́ takto:

vstup zásobnı́k
(10 20 30 + *) ()

(20 30 + *) (10)

(30 + *) (20 10)

(+ *) (30 20 10)

(*) (50 10)

() (500)

Výsledek vyhodnocenı́ je 500.

(c) Konečně, v přı́padě '(10 20 + 30 *) probı́há vyhodnocovánı́ takto:

vstup zásobnı́k
(10 20 + 30 *) ()

(20 + 30 *) (10)

(+ 30 *) (20 10)

(30 *) (30)

(*) (30 30)

() (900)

Výsledek vyhodnocenı́ je 900.

Proceduru provádějı́cı́ vyhodnocovánı́ pomocı́ zásobnı́ku můžeme naprogramovat tak, jak je to uvedeno
v programu 11.12. Zde je uvedena procedura polish-eval, která jako argumenty akceptuje daný výraz,
který bude vyhodnocen, a tabulku symbolů. V našem přı́padě budeme vždy použı́vat tabulku z pro-
gramu 11.11. Nic však nebránı́ tomu, abychom zavedli jinou tabulku. Procedura ve svém těle použı́vá
pomocnou internı́ iterativnı́ proceduru iter, která má dva argumenty: seznam reprezentujı́cı́ vstupnı́ vý-
raz a seznam reprezentujı́cı́ zásobnı́k (na počátku prázdný). Ve svém těle procedura dělá přesně to, co jsme
řekli v předchozı́ch paragrafech. Pomocná procedura list-pref uvedená v programu 11.11 sloužı́ k zı́s-
kánı́ prvnı́ch n prvků ze seznamu: procedura sloužı́ k vyzvednutı́ vı́ce hodnot ze zásobnı́ku před aplikacı́
procedury odpovı́dajı́cı́ symbolu operace.

Program 11.12. Procedura vyhodnocujı́cı́ výrazy v reverznı́ bezzávorkové notaci.

(define list-pref

(lambda (n l)

(if (<= n 0)

'()

(cons (car l)

(list-pref (- n 1) (cdr l))))))

(define polish-eval

(lambda (expr env)

(let iter ((input expr)

(stack '()))

(if (null? input)

stack

(let ((word (car input))

(tail (cdr input)))

(if (not (symbol? word))

(iter tail (cons word stack))

(let ((func (assoc word env)))

(if (not func)

(error "Symbol not bound")

(let ((arity (cadr func))

(proc (caddr func)))

(iter tail

(cons (apply proc

(reverse (list-pref arity stack)))

(list-tail stack arity))))))))))))

(polish-eval '(10 20 +) env) Z=⇒ (30)

(polish-eval '(10 20 30 +) env) Z=⇒ (50 10)

(polish-eval '(10 20 30 + *) env) Z=⇒ (500)

(polish-eval '(10 20 30 * +) env) Z=⇒ (610)

(polish-eval '(10 20 + 30 *) env) Z=⇒ (900)

(polish-eval '(10 20 * 30 +) env) Z=⇒ (230)

(polish-eval '(10 30 n +) env) Z=⇒ (-20)

(polish-eval '(10 n 30 +) env) Z=⇒ (20)

(polish-eval '(10 n 30 n +) env) Z=⇒ (-40)

(polish-eval '(10 n 30 n + n) env) Z=⇒ (40)

(polish-eval '(10 8 2 / ^) env) Z=⇒ (10000)

Zásobnı́kové vyhodnocovánı́ postfixových bezzávorkových výrazů se použı́vá daleko častěji, než jak
bychom možná intuitivně čekali. Na tomto stylu vyhodnocovánı́ výrazů je založeno celé jedno minoritnı́
paradigma – zásobnı́kové paradigma (dost často se však za samostatné paradigma nepovažuje). Typickým
zástupcem zásobnı́kového jazyka je FORTH. Jazyk PostScript, který umı́ interpretovat každá trochu
lepšı́ tiskárna, a který je v současnosti de facto standardem, pokud jde přes přenositelné formáty popisu
tiskové strany, je rovněž dialektem jazyka FORTH upraveným právě pro použitı́ popisu obsahu tiskové
strany. Virtuálnı́ stroje zpracovávajı́cı́ programy v bajtkódu (viz prvnı́ lekci) jsou vesměs naprogramo-
vány jako zásobnı́kové vyhodnocovacı́ programy zpracovávajı́cı́ programy v zásobnı́kových jazycı́ch.
Existujı́ samozřejmě i hardwarové zásobnı́kové procesory, které jsou součástı́ malých počı́tačů a tiskáren.

Shrnutı́

V této lekci jsme se zabývali zpracovánı́m symbolických výrazů reprezentovaných seznamy skládajı́cı́mi se
z čı́sel, symbolů a dalšı́ch seznamů reprezentujı́cı́ symbolické výrazy. Nejprve jsme pro zjednodušenı́ práce
zavedli rozšı́řenı́ kvotovánı́ – tak zvané kvazikvotovánı́, což je obecnějšı́ metoda kvotovánı́ umožňujı́cı́
programátorům snadněji definovat některé seznamy. V lekci jsme se potom věnovali zjednodušovánı́ arit-
metických výrazů, tyto aritmetické výrazy byly reprezentovány seznamy. Ukázali jsme několik procedur
pro zjednodušovánı́ s různou vnitřnı́ strukturou a s různými schopnostmi. Dalšı́m přı́kladem bylo počı́tánı́
symbolických derivacı́. Potom jsme naši pozornost přesunuli na reprezentaci výrazů v infixové, postfixové,
a reverznı́ bezzávorkové notaci. Naprogramovali jsme řadu procedur pro konverzi výrazů v prefixové
notaci na ostatnı́ notace a zpět. Poukázali jsme na fakt, že infixová notace je z hlediska strojového zpra-
covánı́ dost komplikovaná. V poslednı́ sekci jsme se věnovali konstrukcı́ procedur vyhodnocujı́cı́ výrazy
v postfixové a reverznı́ bezzávorkové notaci. Navrhli jsme několik modelů jejich vyhodnocovánı́ a uká-
zali jejı́ch silné a slabé stránky. Vyhodnocovánı́ reverznı́ch bezzávorkových výrazů jsme naprogramovali
pomocı́ manipulace s dodatečným zásobnı́kem, který byl fyzicky reprezentován seznamem.

Pojmy k zapamatovánı́

• kvazikvotovánı́

Nově představené prvky jazyka Scheme

• speciálnı́ forma quasiquote
• procedura assoc

Kontrolnı́ otázky

1. Co je kvazikvotovánı́? jak se lišı́ od kvotovánı́?
2. Jak jsme naprogrogramovali zjednodušovánı́ aritmetických výrazů?
3. Jak jsme naprogrogramovali symbolickou derivaci?

4. Jak jsme naprogrogramovali převod mezi jednotlivými notacemi?

Cvičenı́

1. Bez použitı́ interpretru určete výsledky vyhodnocenı́ následujı́cı́ch výrazů:
(quasiquote symbol)

`(symbol)

(car ``symbol)

`(+ 1)

`(,+ 1 2)

`(1 ,1)

`(1 ,,1)

`(,+ . ,-)

`(1 + 2 ,@3)

(quote (,@()))

(quasiquote quasiquote)

(quasiquote (+ 1 (unquote +)))

(quasiquote (1 2 ,(+ 1 2)))

`,(+ 1 2)

`,`,(+ 1 2)

`(1 2 ,@(build-list 5 (lambda (x) (* x x))) 3)

(quasiquote (1 2 (unquote-splicing (map list '(1 2 3))) 5))

unquote-splicing

```()

`'+

(quote unquote)

2. Upravte kód procedury diff v programu 11.6 tak, aby bylo možné derivovat výrazy, ve kterých je
součet a součin použit na libovolné množstvı́ argumentů.

Úkoly k textu

1. Popište, jak nahradit použitı́ speciálnı́ formy quasiquote.

2. Rozšiřte proceduru simplify tak, aby neměla nedostatky uvedené na konci sekce 11.2.

3. Naprogramujte proceduru vyhodnocovánı́ pro infixové výrazy s pevně daným počtem argumentů
(nejvýše dva).

Řešenı́ ke cvičenı́m

1. symbol, (symbol), quasiquote, (+ 1), („procedura sčı́tánı́“ 1 2)
(1 1) chyba („procedura sčı́tánı́“ . „procedura odčı́tánı́“) chyba ((unquote-splicing ()))
quasiquote (+ 1 „procedura sčı́tánı́“) (1 2 3) 3 3 (1 2 0 1 4 9 16 3)
(1 2 (1) (2) (3) 5) chyba (quasiquote (quasiquote ())) ’+ unquote

2. Oproti programu 11.6 stačı́ změnit lokálnı́ definici tabulky takto:
(define table

(let ((2*->* (lambda (x) (if (and (pair? x) (equal? '*2 (car x)))

`(* ,@(cdr x))

x))))

`((+ . ,(lambda l `(+ ,@(map (lambda(x) (derive x)) l))))

(- . ,(lambda (x y) `(- ,(derive x) ,(derive y))))

(* . ,(lambda l (derive (foldr (lambda (x a) `(*2 ,x ,a)) 1 l))))

(*2 . ,(lambda (x y) `(+ (* ,(2*->* x) ,(derive y)) (* ,(derive x) ,(2*->* y)))))

(/ . ,(lambda (x y) `(/ (- (* ,(derive x) ,y) (* ,x ,(derive y))) (* ,y ,y)))))))

Lekce 12: Čistě funkcionálnı́ interpret Scheme

Obsah lekce: V této lekci se nejprve budeme zabývat automatickým přetypovánı́m a generickými procedu-
rami. Pro generické procedury zavedeme jednoduchou metodu jejich aplikace prostřednictvı́m vyhledávánı́
procedur pomocı́ vzorů uvedených v tabulkách. Ve zbytku lekce ukážeme implementaci čistě funkcionálnı́
podmnožiny jazyka Scheme. Zaměřı́me se na implementaci datové reprezentace elementů jazyka pomocı́
manifestovaných typů a na implementaci vyhodnocovacı́ho procesu.

Klı́čová slova: generická procedura, koerce, manifestovaný typ, přetypovánı́, tabulka generických procedur.

12.1 Automatické přetypovánı́ a generické procedury

V úvodnı́ sekci této lekce uděláme malou odbočku od hlavnı́ho zaměřenı́ lekce jı́mž bude konstrukce
interpretu jazyka Scheme. V této sekci se budeme zabývat generickými procedurami. Většina procedur, které
jsme v jazyku Scheme použı́vali, byly těsně svázané s konkrétnı́m datovým typem. Napřı́klad procedura
append prováděla spojovánı́ seznamů a nebylo ji možné použı́t s argumenty jiných typů než jsou seznamy.
Toto chovánı́ je v mnoha situacı́ch žádoucı́.

Někdy je ale potřeba jednoduše použı́vat tutéž proceduru s argumenty různých datových typů. Napřı́klad
procedura pro sčı́tánı́ + je schopna pracovat s čı́sly v přesné reprezentaci (racionálnı́ zlomky) a s čı́sly
v přibližné reprezentaci (čı́sla s pohyblivou desetinnou tečkou). Proceduru sčı́tánı́ je dokonce možné použı́t
s argumenty různých typů současně, to jest s některými argumenty (čı́sly) v přesné reprezentaci a s někte-
rými argumenty (čı́sly) v přibližné reprezentaci. Procedura pro sčı́tánı́ tedy musı́ provádět dı́lčı́ operace,
které jsou podmı́něné typem elementů. V některých přı́padech tedy musı́ provést před samotným součtem
jistou „konverzi datových typů“, aby mohla aritmetickou operaci provést.

Procedurám, které svou činnost řı́dı́ podle typů svých argumentů, řı́káme generické procedury. Přesněji ře-
čeno, za generické procedury považujeme ty procedury, které podle typů argumentů provádějı́ aplikace
jiných procedur a provádějı́ přı́padně dodatečnou konverzi argumentů (elementů) na elementy vyža-
dovaných typů. Napřı́klad procedura sčı́tánı́ v jazyku Scheme je generická procedura. Pokud je sčı́tánı́
aplikováno s racionálnı́mi argumenty, provede se sčı́tánı́ přı́mo v přesná reprezentaci a výsledkem je opět
čı́slo v přesné reprezentaci. Pokud by byl byt’jen jeden z argumentů v přibližné reprezentaci, pak procedura
provede nejprve konverzi všech argumentů do přibližné reprezentace a provede sečtenı́ v přibližné repre-
zentaci. Výsledkem takového sečtenı́ je čı́slo v přibližné reprezentaci. To by nás nemělo překvapit, protože
při práci s interpretem jazyka Scheme jsme si mohli všimnout následujı́cı́ho chovánı́ +:

(+ 1/2 10) Z=⇒ 21/2

(+ 0.5 10) Z=⇒ 10.5

(+ 1/2 2/3) Z=⇒ 7/6

(+ 0.5 0.666) Z=⇒ 1.166

Automatickým konverzı́m na elementy jiných typů, ke kterým může docházet během použı́vánı́ gene-
rických procedur, se řı́ká koerce. Koerce je tedy obecně řečeno implicitnı́ přetypovánı́. Skoro ve všech pro-
gramovacı́ch jazycı́ch jsou k dispozici nějaké prostředky pro explicitnı́ přetypovánı́, to jest programátorem
vynucenou změnu typu elementu. Přı́kladem může být třeba převod čı́sla na řetězec znaků. V jazyku Scheme,
jako ve většině programovacı́ch jazyků, existuje datový typ „řetězec znaků“. S řetězci lze dělat běžné ope-
race, jimiž se podrobně nebudeme zabývat, protože to nenı́ našı́m hlavnı́m cı́lem (zájemce odkazuji na
specifikaci R5RS jazyka Scheme, viz [R5RS]). Jednou z operacı́ s řetězci je jejich spojovánı́. K tomu sloužı́
procedura string-append. Spojovánı́ řetězců je demonstrováno následujı́cı́mi přı́klady:

(string-append) Z=⇒ ""

(string-append "Ahoj" "svete") Z=⇒ "Ahojsvete"

(string-append "Ahoj " "svete") Z=⇒ "Ahoj svete"

(string-append "a" "b" "c") Z=⇒ "abc"

Převod čı́sel na řetězce se provádı́ pomocı́ procedury number->string, které pro dané čı́slo vracı́ řetězec
znaků obsahujı́cı́ externı́ reprezentaci čı́sla. Pro ilustraci viz následujı́cı́ ukázky použitı́ procedury:

(number->string 10.2) Z=⇒ "10.2"

(number->string (/ -1 2)) Z=⇒ "-1/2"

(number->string (sqrt -1)) Z=⇒ "0+1i"

Předchozı́ operaci bychom de facto mohli chápat jako explicitnı́ přetypovánı́. Na programátorovu žádost byl
element čı́slo „převeden“ na element řetězec, se kterým se dál může pracovat. Naproti tomu výše uvedené
implicitnı́ přetypovánı́ (koerci) provádějı́ automaticky některé (generické) procedury. Je zajı́mavé, že koerce je
někdy chápána jako potenciálnı́ zdroj chyb a některé jazyky koerci vůbec neumožňujı́. Jednı́m z takových
jazyků je napřı́klad funkcionálnı́ jazyk ML.

Ve zbytku této sekce si ukážeme modelový přı́klad, jak lze naprogramovat uživatelsky definované gene-
rické procedury. Vyjdeme z nám dobře známé operace sčı́tánı́ čı́sel a obohatı́me ji tak, aby mohla sloužit
i ke spojovánı́ řetězců. Umožnı́me navı́c, abychom mohli tuto novou verzi generického sčı́tánı́ použı́vat
s argumenty různých typů (tedy s čı́sly i s řetězci). Toto zobecněnı́ sčı́tánı́ je ve skutečnosti docela praktické.
Umožňuje nám snadno vytvářet textový výstup, v němž jsou některé hodnoty dopočtené, viz ukázku:

(let ((x 10)

(y 20))

(+ "Soucin " x " a " y

" je " (* x y) ".")) Z=⇒ "Soucin 10 a 20 je 200."

Jako prvnı́ vytvořı́me pomocný predikát match-type? sloužı́cı́ k testovánı́ typů argumentů. Predikát je

Program 12.1. Porovnávánı́ datového typu argumentu se vzorem.

(define match-type?

(lambda (preds args)

(or (and (null? preds) (null? args))

(and (pair? preds)

(pair? args)

((car preds) (car args))

(match-type? (cdr preds) (cdr args))))))

uveden v programu 12.1. Prvnı́m argumentem procedury match-type? je seznam predikátů a druhým
argumentem je seznam elementů. Výsledek aplikace match-type? pro dané argumenty je „pravda“, právě
když jsou oba dva seznamy předané jako argumenty stejně dlouhé a elementy z druhého seznamu od-
povı́dajı́ po řadě typům daným predikáty v prvnı́m seznamu. Použitı́ match-type? je demonstrováno
následujı́cı́m přı́kladem.

(match-type? `(,number? ,number?) '(10.2 "Ahoj")) Z=⇒ #f

(match-type? `(,number? ,string?) '(10.2 "Ahoj")) Z=⇒ #t

(match-type? `(,string? ,number?) '(10.2 "Ahoj")) Z=⇒ #f

(match-type? `(,string? ,string?) '(10.2 "Ahoj")) Z=⇒ #f

Predikát match-type? bude použit při testovánı́ shody argumentů se vzorem v tabulce určujı́cı́ chovánı́
generické procedury. Pro každou generickou proceduru budeme vždy uvažovat tabulku metod17, což bude
tabulka ve speciálnı́m tvaru určujı́cı́m vzory a procedury pro aplikaci, pokud argumenty budou odpovı́dat
danému vzoru.

Pokud se nynı́ pro ilustraci zaměřı́me pouze na dva argumenty, můžeme nově implementované generické
sčı́tánı́ popsat tabulkou uvedenou v programu 12.2. Po vyhodnocenı́ výrazu z programu 12.2 dojde k navá-
zánı́ seznamu párů na symbol table-generic. Seznam se skládá z párů jejichž prvnı́ prvek lze chápat jako
identifikátor typu konkrétnı́ operace a druhý prvek je samotná operace, která se má provést. Napřı́klad
prvnı́ řádek bychom mohli čı́st tak, že v přı́padě sčı́tánı́ dvou čı́sel je použita původnı́ operace „sčı́tánı́ čı́sel“

17Pojem „metoda“ je často použı́ván v objektovém programovánı́. My budeme za metody považovat procedury spolu se vzorem
jejich použitı́, které budou součástı́ tabulky určujı́cı́ činnost generické procedury.

Program 12.2. Konkrétnı́ tabulka metod generické procedury.

(define table-generic

(let ((+ +))

`(((+ ,number? ,number?) . ,+)

((+ ,number? ,string?) . ,(lambda (x y)

(string-append (number->string x) y)))

((+ ,string? ,number?) . ,(lambda (x y)

(string-append x (number->string y))))

((+ ,string? ,string?) . ,string-append))))

(všimněte si vazby + v let-výrazu, pro připomenutı́ viz lekci 3). Na druhém řádku tabulky je uvedeno,
že v přı́padě sčı́tánı́ čı́sla a řetězce se provede konverze čı́sla na řetězec a výsledek se spojı́ s předaným
řetězcem. Při aplikaci poslednı́ jmenované procedury vlastně z hlediska uživatele generické procedury
(kterou vytvořı́me dále) docházı́ ke koerci (přetypovánı́ čı́sla na řetězec).

Program 12.3. Vyhledánı́ přı́slušné operace v tabulce metod generické procedury.

(define table-lookup

(lambda (table op args)

(cond ((null? table) #f)

((not (equal? (caaar table) op)) (table-loopkup (cdr table) op args))

((match-type? (cdaar table) args) (cdar table))

(else (table-lookup (cdr table) op args)))))

Procedura table-lookup uvedená v programu 12.3 má na starost vyhledávat přı́slušnou operaci v ta-
bulce metod generické procedury. Procedura table-lookup bere jako argumenty tabulku metod generické
procedury, symbol identifikujı́cı́ generickou proceduru (v jedné tabulce metod mohou být záznamy pro
vı́ce generických operacı́), a seznam argumentů, podle kterých se v tabulce vyhledává. Procedura itera-
tivně procházı́ tabulku a při nalezenı́ prvnı́ shody vzoru metody s argumenty (zde se použı́vá predikát
match-type?) je vrácena procedura jež je součástı́ metody.

(table-lookup table-generic '+ '(10 20)) Z=⇒ primitivnı́ procedura „sčı́tánı́ čı́sel“
(table-lookup table-generic '+ '(10 "svete")) Z=⇒ uživ. def. procedura z 2. řádku tabulky
(table-lookup table-generic '+ '("Ahoj" 20)) Z=⇒ uživ. def. procedura z 3. řádku tabulky
(table-lookup table-generic '+ '("Ahoj" "svete")) Z=⇒ primitivnı́ procedura „spojenı́ řetězců“
(table-lookup table-generic '+ '(10 #t)) Z=⇒ #f

Aplikace generických procedur bude prováděna pomocı́ procedury apply-generic, viz program 12.4.
Procedura apply-generic provede vyhledánı́ metody v tabulce navázané na table-generic podle vzoru

Program 12.4. Aplikace generické procedury.

(define apply-generic

(lambda (op . args)

(let ((proc (table-lookup table-generic op args)))

(if proc

(apply proc args)

(error "No method for these types")))))

a provede následnou aplikaci procedury jež je součástı́ shodujı́cı́ se metody. Pro většı́ pohodlı́ při apli-
kaci generické procedury provedeme navázánı́ nově vytvořené procedury na symbol +. Kód provádějı́cı́
tuto vazbu je uveden v programu 12.5. Pomocná procedura foldr1 pracuje stejně jako foldr s jednı́m

Program 12.5. Generická procedura pro sčı́tánı́.

(define foldr1

(lambda (f term l)

(cond ((null? l) term)

((null? (cdr l)) (car l))

(else (f (car l) (foldr1 f term (cdr l)))))))

(define +

(lambda args

(foldr1 (lambda (x y)

(apply-generic '+ x y))

0

args)))

seznamem, pouze s tı́m rozdı́lem, že hodnota navázaná na symbol term je vrácena pouze v přı́padě, že
seznam předaný foldr1 jako třetı́ argument je prázdný. V přı́padě, že seznam je jednoprvkový, je vrácen
tento prvek – to je jediný rozdı́l oproti původnı́ verzi foldr. V programu 12.5 jsme použili foldr1 k na-
programovánı́ nové procedury libovolných argumentů, která je posléze navázána na symbol +. Procedura
foldr1 je zde použita k zobecněnı́ aplikace generické procedury ze dvou argumentů na libovolné množstvı́
argumentů. Touto problematikou jsme se již zabývali v sekci 7.3. Novou generickou proceduru + je nynı́
možné použı́vat následujı́cı́mi způsoby:

(+) Z=⇒ 0

(+ 10) Z=⇒ 10

(+ 10 20) Z=⇒ 30

(+ 10 20 30) Z=⇒ 60

(+ 10 "x" 30) Z=⇒ "10x30"

(+ 10 20 30 "") Z=⇒ "102030"

(+ 10 20 "" 30) Z=⇒ "102030"

(+ 10 "" 20 30) Z=⇒ "1050"

(+ "" 10 20 30) Z=⇒ "60"

Všimněte si, že pokud uvádı́me jako argumenty pouze čı́sla, generická procedura + se chová stejně jako
původnı́ procedura sčı́tánı́. Zbývá odpovědět na otázku, proč jsme při vytvořenı́ generické procedury
+ použili foldr1 mı́sto klasického foldr. Je to z důvodu praktičnosti. Při použitı́ foldr mı́sto foldr1

bychom totiž dostali nepřirozený výsledek v situaci, kdy je poslednı́ ze sčı́taných elementů řetězec:

(+ "") Z=⇒ "0"

(+ "Ahoj " "svete") Z=⇒ "Ahoj svete0"

(+ "Faktorial " 4 " je " 24 ".") Z=⇒ "Faktorial 4 je 24.0"

Naproti tomu verze se foldr1 se chová přirozeně:

(+ "") Z=⇒ ""

(+ "Ahoj " "svete") Z=⇒ "Ahoj svete"

(+ "Faktorial " 4 " je " 24 ".") Z=⇒ "Faktorial 4 je 24."

12.2 Systém manifestovaných typů

Jelikož nám v této lekci jde o konstrukci interpretu jazyka Scheme, jako jeden z prvnı́ch problémů musı́me
vyřešit, jak budeme reprezentovat jednotlivé elementy jazyka: čı́sla, pravdivostnı́ hodnoty, symboly, páry,

procedury (primitivnı́ a uživatelsky definované), speciálnı́ formy a dalšı́. V této sekci nejprve naznačı́me
obecnou reprezentaci elementů při nı́ž užijeme tak zvanou manifestaci typů. Každý element se bude skládat
ze dvou základnı́ch částı́:

(i) identifikátor typu elementu,

(ii) data charakterizujı́cı́ element.

Identifikátor typu bude sloužit k jednoznačnému určenı́ o jaký element se jedná. Pomocı́ tohoto identifi-
kátoru tedy rozlišı́me, zda-li napřı́klad daný element reprezentuje pár nebo proceduru. Jako identifikátory
typů budeme použı́vat symboly jejichž jména budou typ elementu označovat. Druhou částı́ každého ele-
mentu jsou data představujı́cı́ „hodnotu elementu“.

Poznámka 12.1. (a) Pokud si napřı́klad představı́me dvě čı́sla -13 a 27, pak jejich vnitřnı́ reprezentace
(v našem konstruovaném interpretu Scheme) budou obsahovat shodný identifikátor typu „čı́slo“. Oba
elementy budou mı́t ale různou datovou část – v prvnı́m přı́padě bude datová část uchovávat čı́selnou
hodnotu −13, v druhém přı́padě 27.

(b) Otázkou je, proč v elementech potřebujeme identifikátory jejich typů a zda-li bychom se bez nich
mohli obejı́t. Identifikátory skutečně potřebujeme, abychom mohli správně rozlišovat datové typy elementů
(a v důsledku abychom byli schopni správně vyhodnocovat elementy). V některých přı́padech totiž pouze
na základě znalosti „datové části elementu“ nejsme schopni určit jeho typ. Vezměme si napřı́klad tečkový
pár (2 . 3). Mohli bychom se na něj dı́vat jako na dvojici čı́selných hodnot. Ta může reprezentovat třeba
zlomek 23 , nebo komplexnı́ čı́slo 2 + 3i. Z pohledu syntaxe a sémantiky jazyka, viz sekci 1.2, je identifikátor
typu sémantická informace určujı́cı́ význam datové části elementu (datová část elementu by sama o sobě neměla
žádný význam).

Pod pojmem manifestovaný typ máme tedy na mysli přı́tomnost identifikátoru typu „v datech“. Při repre-
zentaci elementů budeme použı́vat manifestaci typů, jak jsme to naznačili v úvodu této sekce.

Pro práci s elementy a manifestovanými typy si vytvořı́me sadu pomocných procedur, které jsou uvedeny
v programu 12.6. Procedura curry-make-elem sloužı́ k vytvářenı́ konstruktorů elementů s manifestovaným

Program 12.6. Systém manifestovaných typů.

(define curry-make-elem

(lambda (type-tag)

(lambda (data)

(cons type-tag data))))

(define get-type-tag car)

(define get-data cdr)

(define curry-scm-type

(lambda (type)

(lambda (elem)

(equal? type (get-type-tag elem)))))

typem. Procedura curry-make-elem bere jako argument symbol (formálnı́ argument type-tag) označujı́cı́
daný typ. Symbolům označujı́cı́m typy se někdy řı́ká „visačky“ nebo „tagy“ (z anglického tags). Procedura
curry-make-elem vracı́ konstruktor jı́mž je procedura jednoho argumentu. Tento argument představuje
hodnotu, kterou chceme vložit do datové části elementu. Na fyzické úrovni je každý element reprezentován
párem, jehož prvnı́ složka je visačka a druhá složka je tvořena hodnotou elementu. Všimněte si, že procedura
curry-make-elempouze rozkládá proceduru dvou argumentůconsna dvě procedury jednoho argumentu,
tento princip jsme popsali v sekci 2.4 jako currying. Procedury get-type-tag a get-data pro daný element

vracı́ jeho visačku respektive jeho datovou složku. Poslednı́ procedura v programu 12.6 je curry-scm-type.
Tato procedura pro daný identifikátor typu vracı́ predikát testujı́cı́ zda-li daný element je tohoto typu či
nikoliv. Viz následujı́cı́ přı́klad použitı́.

Nejprve vytvořı́me konstruktory a predikáty testujı́cı́ typ pro dva různé datové typy (racionálnı́ a komplexnı́
čı́sla):

(define make-frac (curry-make-elem 'fraction))

(define make-cplx (curry-make-elem 'complex-number))

(define frac? (curry-scm-type 'fraction))

(define cplx? (curry-scm-type 'complex-number))

Předchozı́ procedury můžeme použı́vat následujı́cı́m způsobem:

(define a (make-frac '(2 . 3)))

(define b (make-cplx '(2 . 3)))

a Z=⇒ (fraction 2 . 3)

b Z=⇒ (complex-number 2 . 3)

(get-data a) Z=⇒ (2 . 3)

(get-data b) Z=⇒ (2 . 3)

(frac? a) Z=⇒ #t

(frac? b) Z=⇒ #f

(cplx? a) Z=⇒ #f

(cplx? b) Z=⇒ #t

Elementy jazyka Scheme budeme reprezentovat jako hodnoty s manifestovaným typem. Typ elementu
je manifestován pomocı́ visačky, což je identifikátor typu. Typ elementu je potřeba rozeznávat, protože
samotná datová část elementu jednoznačně neurčuje typ (sémantiku) elementu. V dalšı́ sekci uvidı́me,
že manifestace typu nám umožnı́ rozlišovat od sebe napřı́klad vnitřnı́ reprezentaci párů a uživatelsky
definovaných procedur. Samotný princip manifestace typů je samozřejmě použitelný i při řešenı́ jiných
problémů než je reprezentace elementů jazyka Scheme.

12.3 Datová reprezentace elementů jazyka Scheme

V této sekci ukážeme implementaci jednotlivých elementů jazyka Scheme, které budeme potřebovat při
vytvořenı́ jeho interpretu. Budeme postupovat od nejjednoduššı́ch elementů ke složitějšı́m.

Před tı́m, než začneme, je potřeba udělat několik terminologických poznámek. Jelikož se zabýváme imple-
mentacı́ interpretu jazyka Scheme v jazyku Scheme, pracujeme vlastně se dvěma interprety současně. Prvnı́m
z interpretů je pro nás ten interpret, který použı́váme při vývoji programu. Tı́mto programem je (druhý)
interpret jazyka Scheme. V této sekci se tedy budeme zabývat reprezentacı́ elementů nově vytvářeného
interpretu Scheme, nikoliv reprezentacı́ elementů v interpretu, který při vytvářenı́ použı́váme. Abychom
zjednodušili terminologii, zavedeme nynı́ pojmy metainterpret a interpret jazyka Scheme:

• metainterpret jazyka Scheme je již existujı́cı́ interpret, který použı́váme pro vytvářenı́ dalšı́ch programů
(mimo jiné našeho nového interpretu jazyka Scheme),

• interpret jazyka Scheme je (meta)program pro metainterpret jazyka Scheme, který provádı́ interpretaci
jisté podmnožiny jazyka Scheme.

Podobně jako rozlišujeme pojmy metainterpret a interpret můžeme odlišovat dalšı́ pojmy, se kterými jsme
se doposud setkali. Tak třeba metajazyk (jazyk interpretovaný metainterpretem) a jazyk (jazyk interpretovaný
interpretem), metaelement (element metajazyka) a element (element jazyka), metaprogram (program v metaja-
zyku) a program (program v jazyku). Abychom situaci ještě zjednodušili, budeme někdy předponu „meta“
vynechávat, a to v přı́padě, kdy bude jasné, že se bavı́me o původnı́m interpretu jazyka Scheme nebo o sou-
visejı́cı́ch pojmech.

Pokud budeme hovořit o „metapojmech“, budeme tı́m mı́t vždy na mysli pojmy vztažené k metainter-
pretu jazyka Scheme, to jest k interpretu, který použı́váme k vývoji našeho nového interpretu podmno-
žiny jazyka Scheme. Na programy, které dosud vytvářı́me se taky musı́me dı́vat ze dvou úhlů pohledu.
Metaprogramy jsou programy pro výchozı́ metainterpret, tedy náš nový interpret jazyka Scheme je
sám o sobě metaprogram. Programy pro nově vytvářený interpret nazýváme v souladu s předchozı́
úmluvou pouze „programy“.

Nynı́ již obrat’me naši pozornost k reprezentaci elementů jazyka Scheme. Mezi nejjednoduššı́ elementy patřı́
bezpochyby čı́sla. Při jejich implementaci využijeme toho, že se snažı́me vytvořit „Scheme ve Scheme“, tedy
nový interpret Scheme v již existujı́cı́m metainterpretu Scheme. Dı́ky tomu můžeme de facto převzat veškeré
aritmetické (meta) procedury, jak uvidı́me dále. Při implementaci také nebudeme rozlišovat jednotlivé typy
čı́sel (přesnou a nepřesnou reprezentaci čı́sel, racionálnı́ čı́sla, komplexnı́ čı́sla a tak dále). Pro čı́sla tedy
zavedeme pouze jejich konstruktor a predikát testujı́cı́ typ „čı́slo“ následovně:

(define make-number (curry-make-elem 'number))

(define scm-number? (curry-scm-type 'number))

Analogicky jednoduchá bude reprezentace symbolů. Hodnotou symbolu (jakožto elementu jazyka) je pro
nás jeho „jméno“, které můžeme ztotožnit s řetězcem znaků. Abychom situaci ještě vı́ce zjednodušili,
nebudeme k označenı́ jmen symbolů použı́vat řetězce znaků, ale metasymboly dostupné v metainterpretu
jazyka Scheme. Konstruktory symbolů a predikát testujı́cı́ typ tedy zavedeme:

(define make-symbol (curry-make-elem 'symbol))

(define scm-symbol? (curry-scm-type 'symbol))

Nynı́ se zaměřı́me na speciálnı́ elementy jazyka, které se vyhodnocovaly na sebe sama. Jednalo se o prav-
divostnı́ hodnoty, prázdný seznam, a element zastupujı́cı́ nedefinovanou hodnotu. Pro tyto elementy nepotřebu-
jeme vytvářet konstruktory, protože pravdivostnı́ hodnoty jsou pouze dvě, prázdný seznam je pouze jeden
a stejně tak pouze jeden je element zastupujı́cı́ nedefinovanou hodnotu.

V přı́padě pravdivostnı́ch hodnot tedy vytvořı́me dva nové elementy zastupujı́cı́ nepravdu a pravdu. Tyto
elementy navážeme na symboly scm-false a scm-true. Dále vytvořı́me predikát testujı́cı́ typ „pravdi-
vostnı́ hodnota“. Viz následujı́cı́ kód.

(define scm-false ((curry-make-elem 'boolean) #f))

(define scm-true ((curry-make-elem 'boolean) #t))

(define scm-boolean? (curry-scm-type 'boolean))

Prázdný seznam bude nově vytvořený element navázaný na the-empty-list:

(define the-empty-list ((curry-make-elem 'empty-list) '()))

(define scm-null? (lambda (elem) (equal? elem the-empty-list)))

A konečně stejným způsobem vytvořı́me i element zastupujı́cı́ nedefinovanou hodnotu:

(define the-undefined-value ((curry-make-elem 'undefined) '()))

(define scm-undefined? (lambda (elem) (equal? elem the-undefined-value)))

Všimněte si toho, že předchozı́ predikáty scm-null? a scm-undefined? využı́vajı́ toho, že prázdný seznam
a nedefinovaná hodnota jsou unikátnı́ elementy daného typu. V tomto přı́padě je tedy možné naprogramo-
vat predikáty testujı́cı́ typ jako predikáty testujı́cı́ rovnost s daným elementem.

Nynı́ se budeme zabývat tečkovými páry. Nejprve podotkněme, že pro to, abychom v našem interpretu
mohli uvažovat seznamy, nenı́ nutné (a ani vhodné) vytvářet elementy jazyka typu „seznam“. Plně si
vystačı́me s dále navrženými páry a prázdným seznamem, který jsme definovali výše. To je zcela v souladu
s tı́m, jak jsme zavedli seznamy pomocı́ párů v lekci 5. Páry pro nás tedy budou speciálnı́ elementy typu
„pár“, jejichž datovou složkou budou tvořit dva elementy v pevně daném pořadı́. Fyzicky budeme páry
reprezentovat pomocı́ metapárů ve tvaru

(pair.(〈prvnı́〉. 〈druhý〉)),

kde 〈prvnı́〉 je element představujı́cı́ prvnı́ složku páru a 〈druhý〉 je element představujı́cı́ druhou složku
páru. Pro páry vytvořı́me jejich konstruktor, dva selektory a predikát testujı́cı́ typ „pár“. V programu 12.7
je uveden konstruktor páru make-pair. Jedná se o proceduru dvou argumentů, která vznikla v prostředı́

Program 12.7. Reprezentace tečkových párů.

(define make-pair

(let ((make-physical-pair (curry-make-elem 'pair)))

(lambda (head tail)

(make-physical-pair (cons head tail)))))

(define scm-pair? (curry-scm-type 'pair))

(define pair-car

(lambda (pair)

(if (scm-pair? pair)

(car (get-data pair))

(error "CAR: argument must be a pair"))))

(define pair-cdr

(lambda (pair)

(if (scm-pair? pair)

(cdr (get-data pair))

(error "CDR: argument must be a pair"))))

v němž je na symbol make-physical-pair navázána procedura vytvářejı́cı́ element s manifestovaným
typem „pár“. Samotná procedura make-pair provádı́ pouze jednu aplikaci make-physical-pair při nı́ž
jsou obě dvě složky spojeny do metapáru pomocı́ cons. Opět jsme tedy zvolili strategii, že k reprezentaci
párů nám sloužı́ metapáry a konstruktor páru make-pair je vytvořen pomocı́ konstruktoru metapáru cons.
Pro objasněnı́ uved’me následujı́cı́ přı́klady použitı́ make-pair:

(define a (make-number 1))

(define b (make-number 2))

(make-pair a b) Z=⇒ (pair (number . 1) number . 2)

(make-pair a the-empty-list) Z=⇒ (pair (number . 1) empty-list)

(make-pair the-empty-list b) Z=⇒ (pair (empty-list) number . 2)

V programu 12.7 je dále uveden predikát scm-pair? testujı́cı́, zda-li je daný element typu „pár“. Dále jsou
zde uvedeny selektory pair-car a pair-cdr sloužı́cı́ k přı́stupu k prvnı́, přı́padně druhé, složce párů.
Jejich implementace je přı́močará.

Přı́klad 12.2. Nynı́ si již můžeme udělat základnı́ představu o tom, jak bude vypadat reprezentace složitěj-
šı́ch elementů jazyka pomocı́ metaelementů. Napřı́klad pár (10 . ahoj), to jest pár, jehož prvnı́ složkou
je čı́slo a druhou složkou je symbol bude v metainterpretu fyzicky reprezentován metaelementem vyobra-
zeným na obrázku 12.1. Dále napřı́klad λ-výraz (10 . ahoj) bude v metainterpretu reprezentován tak,
jak ukazuje obrázek 12.2. Jak je na prvnı́ Pohled zřejmé, reprezentace tohoto relativně malého seznamu je
dost velká. To je jakási daň, kterou musı́me zaplatit za visačky jednoznačně určujı́cı́ typy elementů.

Dalšı́ elementy jazyka, jejichž reprezentaci popı́šeme, jsou prostředı́. Koncept prostředı́ byl představen již
v lekci 1 a dále zpřesněn v lekci 2. Prostředı́ potřebujeme k udržovánı́ vazeb mezi symboly a elementy a kvůli
schůdné implementaci lexikálnı́ho rozsahu platnosti: každé prostředı́, kromě globálnı́ho, má ukazatele na
svého lexikálnı́ho předka (prostředı́ svého vzniku). Datová část prostředı́ tedy musı́ obsahovat jednak
tabulku vazeb mezi symboly a elementy a jednak (ukazatel na) dalšı́ prostředı́. Elementy typu prostředı́
tedy budeme reprezentovat metapáry ve tvaru

Obrázek 12.1. Fyzická reprezentace páru (10 . ahoj) pomocı́ metaelementů.

pair . . .

number 10

symbol ahoj. . ..

Obrázek 12.2. Fyzická reprezentace seznamu (lambda (x) (+ x 1)) pomocı́ metaelementů.

(environment.(〈předek〉. 〈tabulka〉)),

kde 〈předek〉 je bud’to element prostředı́ nebo element „nepravda“ (dané prostředı́ nemá předka) a 〈tabulka〉
je tabulka vazeb mezi symboly a elementy udržovaná jako internı́ reprezentace seznamu ve tvaru

((〈symbol1〉 〈element1〉)
(〈symbol2〉 〈element2〉)

...

(〈symboln〉 〈elementn〉)).

Pro práci s prostředı́mi budeme potřebovat několik základnı́ch procedur. V prvnı́ řadě to bude konstruktor
prostředı́ make-env a dva selektory get-pred a get-table, které pro dané prostředı́ vracejı́ prostředı́
předka nebo tabulku vazeb. Tyto procedury jsou uvedeny v programu 12.8. V tomto programu je dále
uveden predikát testujı́cı́ typ elementu „prostředı́“. Všimněte si, že základnı́ konstruktory a selektory
prostředı́ jsou de facto stejné jako konstruktory a selektory párů, viz program 12.7. V přı́padě prostředı́ ale
platı́, že jeho datové složky nejsou libovolné elementy, ale přesně vymezené elementy (prvnı́ element je
předek, tedy „nepravda“ nebo opět prostředı́ a druhý element je vždy tabulka vazeb).

Pro pohodlnou manipulaci s prostředı́m zavedeme dalšı́ procedury. V našem novém interpretu totiž musı́me
mı́t nějak definováno prostředı́ počátečnı́ch vazeb, musı́me mı́t tedy k dispozici procedury, kterými prostředı́
vytvořı́me. Dalšı́ prostředı́ již budou vznikat při aplikaci uživatelsky definovaných procedur tak, jak jsme to
vysvětlili v lekci 2. V programu 12.9 je uvedeno několik procedur, které využijeme při definici počátečnı́ch
prostředı́. Predikát global? je pro daný element pravdivý, právě když je element prostředı́, které nemá
předka. Jedná se tedy o predikát testujı́cı́, zda-li je předaný element globálnı́ prostředı́. Pomocná procedura
assoc->env provede převod asociačnı́ho metaseznamu na tabulku vazeb realizovanou asociačnı́m seznamem.
Při bližšı́m pohledu je vidět, že assoc->env je v podstatě jen jednoduchá rekurzivnı́ procedura, která
převádı́ metaseznam metapárů na seznam párů ve vnitřnı́ reprezentaci (nového interpretu), přitom také
provádı́ vytvářenı́ nových symbolů z metasymbolů. Pro objasněnı́ ještě uved’me přı́klad jejı́ho použitı́:

(define s `((a . ,(make-number 10)) (b . ,(make-number 20))))

(assoc->env s) Z=⇒ (pair (pair (symbol . a) number . 10)

pair (pair (symbol . b) number . 20)

empty-list)

Konečně procedura make-global-env z programu 12.9 sloužı́ k vytvořenı́ globálnı́ho prostředı́. Jejı́ použitı́
uvidı́me v jedné z dalšı́ch sekcı́.

Program 12.8. Reprezentace prostředı́.

(define make-env

(let ((make-physical-env (curry-make-elem 'environment)))

(lambda (pred table)

(make-physical-env (cons pred table)))))

(define scm-env? (curry-scm-type 'environment))

(define get-table

(lambda (elem)

(if (scm-env? elem)

(cdr (get-data elem))

(error "GET-TABLE: argument must be an environment"))))

(define get-pred

(lambda (elem)

(if (scm-env? elem)

(car (get-data elem))

(error "GET-PRED: argument must be an environment"))))

Při implementaci vyhodnocovacı́ho procesu budeme potřebovat hledat vazby v prostředı́ch. K tomuto účelu
naprogramujeme dvě pomocné procedury. Procedura scm-assoc z programu 12.10 provádı́ prakticky totéž
co standardnı́ procedura assoc (viz standard R5RS jazyka Scheme [R5RS]). Jediný rozdı́l je v tom, že scm-
assoc pracuje s asociačnı́mi seznamy (a nikoliv metaseznamy) v jejich internı́ reprezentaci. Procedura pro
daný klı́č a asociačnı́ seznam prohledává daný asociačnı́ seznam a vracı́ prvnı́ pár, jehož prvnı́ složka odpo-
vı́dá klı́či. Pokud žádný takový pár neexistuje, je vrácena „nepravda“. Proceduru scm-assoc tedy můžeme
použı́t k vyhledánı́ vazby v tabulce vazeb nějakého prostředı́. Druhá procedura v programu 12.10 je pro-
cedura lookup-env, což je komplexnějšı́ procedura, která vyhledává vazbu symbolů v prostředı́ nebo vracı́
element navázaný na symbol not-found, pokud nenı́ vazba nalezena. Formálnı́ argument pojmenovaný
search-nonlocal? sloužı́ jako přepı́nač, kterým lze ovlivňovat, co se má stát, pokud vazba nenı́ nalezena
v tabulce aktuálnı́ho prostředı́. Pokud je při aplikaci lookup-env na search-nonlocal? navázána hodnota
„pravda“, pak se při nenalezenı́ vazby v tabulce lokálnı́ho prostředı́ postupuje k nadřazenému prostředı́.
V přı́padě, že je na search-nonlocal? navázána „nepravda“, pak je při nenalezenı́ vazby v tabulce lokál-
nı́ho prostředı́ okamžitě vrácena hodnota navázaná na not-found. Procedura lookup-env bude použı́vána
k hledánı́ vazeb symbolů přı́mo během vyhodnocovánı́ elementů.

Nynı́ se budeme zabývat reprezentacı́ procedur a speciálnı́ch forem. Reprezentace primitivnı́ch procedur,
tedy procedur, které budou přı́mo zabudované v našem novém interpretu, bude jednoduchá. Vytvořı́me
jejich konstruktor a predikát testujı́cı́ typ „primitivnı́ procedura“ následovně:

(define make-primitive (curry-make-elem 'primitive))

(define scm-primitive? (curry-scm-type 'primitive))

Datovou složkou primitivnı́ procedury bude metaprocedura, tedy nějaká procedura, která je přı́mo součástı́
konstruovaného interpretu a která bude při aplikaci prováděna metainterpretem jazyka Scheme. V tuto
chvı́li připomeňme, že už v prvnı́ lekci, kdy jsme primitivnı́ procedury zavedli, jsem upozornili na fakt,
že se nebudeme zabývat tı́m, jak jsou vytvořené. Obecně můžeme řı́ct, že primitivnı́ procedury jsou vždy
vytvořeny pomocı́ nějakých metaprocedur. Primitivnı́ metaprocedury (primitivnı́ procedury v metainter-
pretu) jsou rovněž vytvořeny pomocı́ nějakých „metametaprocedur“, které jsou, v přı́padě interpretu jazyka
Scheme vzniklého kompilacı́, zapsané v kódu stroje (a zdrojové kódy těchto „metametaprocedur“ budou
naprogramovány nejspı́š v nějakém vyššı́m programovacı́m jazyku, třeba v jazyku C, což je oblı́bený jazyk
pro tvorbu interpretů a překladačů).

Program 12.9. Konstruktor pro globálnı́ prostředı́.

(define assoc->env

(lambda (l)

(if (null? l)

the-empty-list

(make-pair (make-pair (make-symbol (caar l))

(cdar l))

(assoc->env (cdr l))))))

(define make-global-env

(lambda (alist-table)

(make-env scm-false (assoc->env alist-table))))

(define global?

(lambda (elem)

(and (scm-env? elem)

(equal? scm-false (get-pred elem)))))

At’tak či onak, nynı́ se musı́me zabývat tı́m, jak primitivnı́ procedury vytvářet, protože se zabýváme kon-
strukcı́ interpretu jazyka. Některé primitivnı́ procedury budeme programovat jako uživatelsky definované
metaprocedury. Řadu důležitých primitivnı́ch procedur, napřı́klad aritmetické procedury, ale můžeme vytvo-
řit jednoduchým „trikem“ a to tak, že pouze zabalı́me metaproceduru, která je protějškem dané procedury,
do pomocného programu, který z daných elementů vyzvedne jejich datovou část, aplikuje metaproceduru
s takto zı́skanými hodnotami a nakonec výsledkem aplikace (tedy metaelement) převede do internı́ repre-
zentace. Na tuto problematiku se blı́že podı́váme v dalšı́ch sekcı́ch.

V lekci 2 jsme uvedli, že uživatelsky definované procedury chápeme jako trojice hodnot 〈〈parametry〉, 〈tělo〉,P〉,
kde 〈parametry〉 je seznam formálnı́ch argumentů, 〈tělo〉 je element reprezentujı́cı́ tělo procedury a P je
prostředı́ vzniku procedury. V lekci 3 jsme rozšı́řili procedury tak, že jsme umožnili, aby v jejich těle bylo
přı́tomno vı́c výrazů. Toto rozšı́řenı́ jsme učinili z důvodu pohodlného zavedenı́ internı́ch definic. Jeli-
kož se ale jedná o rys, který nenı́ čistě funkcionálnı́ (při vytvářenı́ definic docházı́ k vedlejšı́mu efektu
jı́mž je modifikace prostředı́), budeme se dále zabývat konceptem procedur tak, jak jsme jej představili
v lekci 2 (jeden výraz v těle). Uživatelsky definovaná procedura je tedy element jazyka, který v sobě
agreguje tři hodnoty (seznam formálnı́ch argumentů, prostředı́ a tělo). Reprezentaci uživatelsky defino-
vaných procedur jakožto elementů našeho jazyka můžeme tedy provést zcela přı́močaře tak, jak je to
ukázáno v programu 12.11. Procedura make-procedure je konstruktor uživatelsky definovaných procedur
akceptujı́cı́ tři argumenty: prostředı́, seznam argumentů a tělo. Tyto tři položky jsou v elementu fyzicky
uloženy do třı́prvkového metaseznamu. Dále máme k dispozici tři selektory procedure-environment,
procedure-arguments a procedure-body vracejı́cı́ prostředı́, seznam argumentů a tělo pro daný element
typu „uživatelsky definovaná procedura“. Nakonec jsme opět zavedli predikát testujı́cı́ daný typ. Jelikož
primitivnı́ procedury a uživatelsky definované procedury chápeme souhrnně jako procedury, vytvořı́me
navı́c dodatečný predikát testujı́cı́, zda-li je element procedura (primitivnı́ nebo uživatelsky definovaná):

(define scm-procedure?

(lambda (elem)

(or (scm-primitive? elem)

(scm-user-procedure? elem))))

Poslednı́m typem uvažovaných elementů jazyka budou speciálnı́ formy. Speciálnı́ formy budou imple-
mentované opět pomocı́ metaprocedur. Pro každou speciálnı́ formu, kterou bude náš interpret obsahovat,
budeme vytvářet speciálnı́ uživatelsky definovanou metaproceduru. Zatı́m tedy vytvořı́me pouze základnı́
reprezentaci elementů typu „speciálnı́ forma“:

Program 12.10. Vyhledávánı́ vazeb v prostředı́.

(define scm-assoc

(lambda (key alist)

(cond ((scm-null? alist) scm-false)

((equal? key (pair-car (pair-car alist))) (pair-car alist))

(else (scm-assoc key (pair-cdr alist))))))

(define lookup-env

(lambda (env symbol search-nonlocal? not-found)

(let ((found (scm-assoc symbol (get-table env))))

(cond ((not (equal? found scm-false)) found)

((global? env) not-found)

((not search-nonlocal?) not-found)

(else (lookup-env (get-pred env) symbol #t not-found))))))

Program 12.11. Reprezentace uživatelsky definovaných procedur.

(define make-procedure

(let ((make-physical-procedure (curry-make-elem 'procedure)))

(lambda (env args body)

(make-physical-procedure (list env args body)))))

(define procedure-environment (lambda (proc) (car (get-data proc))))

(define procedure-arguments (lambda (proc) (cadr (get-data proc))))

(define procedure-body (lambda (proc) (caddr (get-data proc))))

(define scm-user-procedure? (curry-scm-type 'procedure))

(define make-specform (curry-make-elem 'specform))

(define scm-specform? (curry-scm-type 'specform))

Nynı́ se dostáváme do bodu, kdy si můžeme dovolit malou epistemickou úvahu. Je zajı́mavé, že při
našem exkurzu programovánı́m v jazyku Scheme jsme postupovali od procedur vyššı́ch řádů směrem
k párům. U párů pro uvedli, že je můžeme plně vyjádřit pomocı́ uživatelsky definovaných procedur
vyššı́ch řádů. Nynı́ postupujeme zdánlivě obráceně. Uživatelsky definované procedury máme úplně
reprezentované pomocı́ (meta) párů.

12.4 Vstup a výstup interpretu

Konstruovaný interpret jazyka Scheme musı́ mı́t k dispozici základnı́ vstupnı́ a výstupnı́ části, konkrétně
reader (proceduru realizujı́cı́ načı́tánı́ vstupnı́ch symbolických výrazů a jejich převod do internı́ reprezentace)
a printer (proceduru, která se stará o vytištěnı́ externı́ reprezentace elementů).

Pro načtenı́ symbolického výrazu můžeme použı́t primitivnı́ proceduru read, se kterou jsme se již setkali
v lekci 5. Po načtenı́ však musı́me ještě provést konverzi hodnoty zı́skané aplikacı́ read do našı́ internı́
reprezentace. To jest všechny načtené symboly, čı́sla a seznamy je potřeba převést na přı́slušné elementy
„symbol“, „čı́slo“ a „páry“ tak, jak jsme je představili v předchozı́ sekci. Tuto konverzi pro nás bude provádět
nově vytvořená procedura expr->intern, viz program 12.12. Samotný reader je již možné naprogramovat

Program 12.12. Převod do internı́ reprezentace a implementace readeru.

(define expr->intern

(lambda (expr)

(cond ((symbol? expr) (make-symbol expr))

((number? expr) (make-number expr))

((and (boolean? expr) expr) scm-true)

((boolean? expr) scm-false)

((null? expr) the-empty-list)

((pair? expr) (make-pair (expr->intern (car expr))

(expr->intern (cdr expr))))

((eof-object? expr) #f)

(else (error "READER: Syntactic error.")))))

(define scm-read

(lambda ()

(expr->intern (read))))

pomocı́ read a procedury expr->intern tak, jak je to uvedeno v programu 12.12. Dodejme, že pokud
bychom neměli v jazyku Scheme k dispozici proceduru read, museli bychom rovněž naprogramovat
samotné načı́tánı́ vstupnı́ch výrazů. V přı́padě jazyka Scheme by to nebylo přı́liš obtı́žné, ale tématicky tento
problém spadá do jiných kurzů. Laskavého čtenáře tı́mto odkazujeme na kurzy formálnı́ jazyky a automaty
a překladače. V následujı́cı́ ukázce je uvedeno použitı́ procedury expr->intern.

(expr->intern 1) Z=⇒ (number . 1)

(expr->intern '+) Z=⇒ (symbol . +)

(expr->intern '(1 . 2)) Z=⇒ (pair (number . 1) number . 2)

(expr->intern '(- 13)) Z=⇒ (pair (symbol . -) pair (number . 13) empty-list)
...

Printer se použı́vá předevšı́m k vypisovánı́ výsledků vyhodnocenı́. Při implementaci printeru musı́me oproti
readeru naprogramovat opačnou konverzi. Tedy konverzi elementu v internı́ reprezentaci na čitelnou
reprezentaci, jenž může být vytištěna na obrazovku, zapsána do souboru, a tak dále. Pochopitelně, že
naprogramovat printer je obecně vždy jednoduššı́ než naprogramovat reader (konverze řetězce znaků na
strukturovaná data je mnohem obtı́žnějšı́ než konverze strukturovaných dat na řetězec znaků18). Jedna
z možnostı́, jak naprogramovat printer je uvedena v programu 12.13. Printer bychom samozřejmě mohli
realizovat i mnohem jednodušeji, napřı́klad následujı́cı́ procedurou, která provede pouze vytištěnı́ internı́
reprezentace elementu na obrazovku:

(define scm-print

(lambda (elem)

(display elem)))

V tomto přı́padě by ale výpis některých elementů nebyl přehledný (uvidı́me dále).

12.5 Implementace vyhodnocovacı́ho procesu

V této sekci rozebereme implementaci vyhodnocovacı́ho procesu včetně aplikace procedur a speciálnı́ch
forem. Postup bude kopı́rovat teorii, kterou jsme probrali v prvnı́ch dvou lekcı́ch tohoto textu. Nejprve při-

18Toto pozorovánı́ by pro nás mělo být vlastně malým poučenı́m. Při programovánı́ čehokoliv se vždy vyplatı́ reprezentovat
data v co možná nejvı́c strukturované podobě. Nikdy tı́m nemůžeme nic ztratit (snad kromě většı́ pamět’ové náročnosti na jejich
uloženı́) a přidaná hodnota může být opravdu velká. Nenı́ náhodou, že metodám (automatického) strukturovánı́ velkých dat se
v informatice věnuje řada disciplı́n.

Program 12.13. Převod do externı́ reprezentace Implementace printeru.

(define intern->expr

(lambda (expr)

(cond ((scm-pair? expr) (cons (intern->expr (pair-car expr))

(intern->expr (pair-cdr expr))))

((scm-env? expr) "#<environment>")

((scm-primitive? expr) "#<primitive-procedure>")

((scm-user-procedure? expr) "#<user-defined-procedure>")

((scm-specform? expr) "#<special-form>")

((scm-undefined? expr) "#<undefined>")

(else (get-data expr)))))

(define scm-print

(lambda (elem)

(display (intern->expr elem))))

pomeňme, že aplikace primitivnı́ch procedur a speciálnı́ch forem bude řešena pomocı́ aplikace uživatelsky
definovaných metaprocedur.

Jak jsme již předeslali v předchozı́ sekci, v některých přı́padech je možné vytvořit primitivnı́ procedury
s využitı́m primitivnı́ch metaprocedur pomocı́ jejich „pouhého zabalenı́ “ do pomocné procedury. Pomocná
procedura sloužı́cı́ jako jakási obálka se stará o konverzi elementů na metaelementy (před aplikacı́ metapro-
cedury) a opačně o konverzi metaelementů na elementy (po aplikaci metaprocedury). Na provedenı́ tohoto
zabalenı́ „metaprocedury“ můžeme vytvořit překvapivě jednoduchou proceduru wrap-primitive, která
je zobrazena v programu 12.14. Použitı́ této procedury uvidı́me v dalšı́ch sekcı́ch. Princip wrap-primitive

Program 12.14. Reprezentace primitivnı́ch procedur.

(define wrap-primitive

(lambda (proc)

(make-primitive

(lambda arguments

(expr->intern (apply proc (map get-data arguments)))))))

si nejlépe uvědomı́me na následujı́cı́m přı́kladu, ve kterém nejprve na symbol p navážeme element jı́mž
je primitivnı́ procedura vzniklá z metaprocedury sčı́tánı́. Datovou složkou tohoto elementu je tedy meta-
procedura vzniklá vyhodnocenı́m vnitřnı́ho λ-výrazu uvedeného v těle procedury z programu 12.14. Viz
přı́klad:

(define p (wrap-primitive +))

p Z=⇒ (primitive . „metaprocedura realizujı́cı́ proceduru sčı́tánı́ čı́sel“)

Primitivnı́ proceduru bychom nynı́ mohli aplikovat následovně:

((cdr p) (make-number 10) (make-number 20)) Z=⇒ (number . 30)

Během předchozı́ aplikace byla provedena extrakce metačı́sel z elementů reprezentujı́cı́ch čı́sla 10 a 20.
Dále byla aplikována primitivnı́ metaprocedura sčı́tánı́ a jejı́m výsledkem je metačı́slo 30. To bylo nakonec
zkonvertováno na element pomocı́ procedury expr->intern, kterou jsme představili v programu 12.12 na
straně 302. Výše uvedenou metodou převezmeme v našem novém interpretu dalšı́ aritmetické procedury.

Jelikož bude během vyhodnocovánı́ občas potřeba konvertovat metaseznamy (seznamy složené z meta-
párů) na seznamy (seznamy složené z párů) a obráceně, vytvořı́me si pomocný konstruktor convert-list,

viz jeho kód v programu 12.15. Procedura convert-list je de facto obecný konstruktor seznamů a me-

Program 12.15. Obecný konvertor seznamu na seznam ve vnitřnı́ reprezentaci a obráceně.

(define convert-list

(lambda (a-null? a-car a-cdr b-cons b-nil f l)

(if (a-null? l)

b-nil

(b-cons (f (a-car l))

(convert-list a-null? a-car a-cdr b-cons b-nil f

(a-cdr l))))))

taseznamů. Pomocı́ něj můžeme vytvořit řadu konvertorů těchto datových struktur a metastruktur. Ty
nejdůležitějšı́ z nich jsou uvedeny v programu 12.16. Procedura scm-list->list konvertuje seznamy na

Program 12.16. Konverze seznamů na metaseznamy o obráceně.

(define scm-list->list

(lambda (scm-l)

(convert-list scm-null? pair-car pair-cdr cons '() (lambda (x) x) scm-l)))

(define list->scm-list

(lambda (l)

(convert-list null? car cdr make-pair the-empty-list (lambda (x) x) l)))

(define map-scm-list->list

(lambda (f scm-l)

(convert-list scm-null? pair-car pair-cdr cons '() f scm-l)))

metaseznamy. Naopak procedura list->scm-list konvertuje seznamy na metaseznamy. Pomocı́ pro-
cedury map-scm-list->list je rovněž možné převést seznam na metaseznam, ale během zpracovánı́
prvků výchozı́ho seznamu se ještě použı́vá procedura jednoho argumentu k modifikaci prvků (analogicky
jako u standardnı́ procedury map).

Nynı́ se již můžeme podı́vat na implementaci vyhodnocovacı́ho procesu. Nejprve se budeme zabývat im-
plementacı́ procedury provádějı́cı́ vyhodnocenı́ elementů. V nı́ použijeme několik procedur, které objasnı́me
dále. K vyhodnocovánı́ elementů bude sloužit procedura scm-eval, která je uvedena včetně jednoduchých
komentářů v programu 12.17 na straně 305. Všimněte si, že scm-eval má dva argumenty, prvnı́ z nich je
element, který vyhodnocujeme, a druhým je aktuálnı́ prostředı́, ve kterém tento element vyhodnocujeme.
To koresponduje s tı́m jak jsme zavedli Eval[E,P] v definici 2.7 na straně 47. V těle procedury scm-eval

je jeden cond-výraz, ve kterém se rozhoduje o způsobu vyhodnocenı́ elementu na základě jeho typu. Zde
uplatnı́me manifestaci typů a predikáty testujı́cı́ typ elementu, které jsme doposud zavedli.

V prvnı́m přı́padě je vyřešena situace, kdy je daný element symbol. V tomto přı́padě je hledá jeho vazba
(počı́naje předaným aktuálnı́m prostředı́m) pomocı́ procedury lookup-env z programu 12.10 na straně 301.

Druhá větev cond-výrazu ošetřuje přı́pad, kdy je daný element seznam nebo lépe řečeno, kdy je daný
element pár19. V tomto přı́padě, je nejprve v daném prostředı́ vyhodnocen prvnı́ prvek páru. Všimněte

19Uvědomte si, že test toho, zda-li daný element seznam nelze provést v konstantnı́m čase. Z důvodu efektivity by tedy bylo
nešt’astné definovat vyhodnocovánı́ seznamů, ale mnohem jednoduššı́ je pracovat přı́mo s páry, které daný seznam tvořı́. Tak je
tomu i v tomto přı́padě. Navı́c nám tento postup umožnı́ zapisovat program pomocı́ párů v tečkové notaci (i když to zřejmě nenı́
přı́liš užitečné a ani přehledné).

Program 12.17. Implementace vlastnı́ho vyhodnocovacı́ho procesu.

;; vyhodnot vyraz v danem prostredi

(define scm-eval

(lambda (elem env)

;; vyhodnocovani elementu podle jejich typu

(cond

;; symboly se vyhodnocuji na svou aktualni vazbu

((scm-symbol? elem)

(let* ((binding (lookup-env env elem #t #f)))

(if binding

(pair-cdr binding)

(error "EVAL: Symbol not bound"))))

;; vyhodnoceni seznamu

((scm-pair? elem)

;; nejprve vyhodnotime prvni prvek seznamu

(let* ((first (pair-car elem))

(args (pair-cdr elem))

(f (scm-eval first env)))

;; podle prvniho prvku rozhodni o co se jedna

(cond

;; pokud se jedna o proceduru, vyhodnot argumenty a aplikuj

((scm-procedure? f)

(scm-apply f (map-scm-list->list

(lambda (elem)

(scm-eval elem env))

args)))

;; pokud se jedna o formu, aplikuj s nevyhodnocenymi argumenty

((scm-specform? f)

(scm-specform-apply env f (scm-list->list args)))

;; na prvnim miste stoji nepripustny prvek

(error "EVAL: First element did not eval. to procedure"))))

;; vse ostatni (cisla, boolean, ... se vyhodnocuje na sebe sama)

(else elem))))

si, že zde docházı́ k rekurzivnı́ aplikaci scm-eval. Výsledek vyhodnocenı́ je navázán v lokálnı́m metapro-
středı́ na metasymbol f. Dále vyhodnocovánı́ postupuje dvěma směry podle toho jakého typu je element
navázaný na f. Pokud je to procedura (primitivnı́ nebo uživatelsky definovaná), provedeme aplikaci me-
taprocedury scm-apply, kterou si záhy popı́šeme. Tato metaprocedura má na starosti provedenı́ aplikace
procedury f s danými argumenty. Všimněte si, že argumenty jsou předány ve formě metaseznamu, jehož
prvku jsou vyhodnocené elementy ze seznamu argumentů (zde opět provádı́me rekurzivnı́ aplikaci scm-
eval). V přı́padě, kdy je na f navázána speciálnı́ forma je provedena jejı́ aplikace pomocı́ metaprocedury
scm-specform-apply. Zde si všimněme toho, že argumenty jsou scm-specform-apply předány bez vy-
hodnocenı́ a navı́c jako jeden z argumentů pro scm-specform-apply předáváme aktuálnı́ prostředı́. To je
nezbytné k tomu, aby mohla každá speciálnı́ forma provádět dalšı́ vyhodnocovánı́ a aby věděla, ve kterém
prostředı́ má vyhodnocovánı́ provádět.

Konečně v poslednı́ větvi cond-výrazu (else-větev) obsaženého v těle scm-eval je vyřešeno vyhodno-
covánı́ všech ostatnı́ch elementů (tedy čı́sel, pravdivostnı́ch hodnot, prázdného seznamu, nedefinované
hodnoty, procedur, speciálnı́ch forem a prostředı́): tyto elementy se vyhodnocujı́ na sebe sama.

Všimněte si, že procedura scm-eval de facto implementuje postup, který jsme uvedli v definici 2.7
na straně 47. Procedura scm-eval se od tohoto postupu lišı́ jen v technických drobnostech. Napřı́klad
implementace separátnı́ho bodu (A) z definice 2.7 nenı́ přı́tomna, protože je řešena již v rámci bodu
(D), viz else-větev v proceduře scm-eval. Nynı́ se tedy můžeme konečně „prakticky přesvědčit“, že
vyhodnocovacı́ proces skutečně funguje tak, jak jsme jej původně uvedli.

Abychom dokončili vyhodnocovacı́ proces, musı́me vytvořit metaprocedury provádějı́cı́ aplikaci procedur
a speciálnı́ch forem. Aplikace speciálnı́ch forem je elementárnı́. Jelikož každá forma řı́dı́ vyhodnocovánı́
svých argumentů jiným způsobem, necháváme při aplikaci speciálnı́ formy veškerý průběh vyhodnocovánı́
na metaproceduře, která je datovou složkou speciálnı́ formy. Proceduru scm-specform-apply použitou
ve scm-eval bychom tedy mohli naprogramovat takto:

(define scm-specform-apply

(lambda (env form args)

(cond ((scm-specform? form) (apply (get-data form) env args))

(else (error "APPLY: Expected special form")))))

Na předchozı́m kódu si opět všimněte, že metaproceduře realizujı́cı́ speciálnı́ formu je předáno prostředı́
jako prvnı́ z argumentů. Implementaci jednotlivých speciálnı́ch forem ukážeme v dalšı́ sekci.

V přı́padě aplikace procedur je situace složitějšı́. Musı́me jednak rozlišit mezi primitivnı́mi procedurami
a uživatelsky definovanými procedurami. Situace v přı́padě primitivnı́ch procedur bude podobně jedno-
duchá jako v přı́padě speciálnı́ch forem, jak záhy uvidı́me. V přı́padě aplikace uživatelsky definovaných
procedur však musı́me provést kroky popsané v definici 2.12 na straně 49. To jest, musı́me vytvořit nové
lokálnı́ prostředı́ s vazbami a v něm vyhodnotit tělo procedury.

K vytvořenı́ tabulky vazeb mezi formálnı́mi argumenty a předanými argumenty, která je nezbytnou sou-
částı́ nově vytvářeného lokálnı́ho prostředı́, bude sloužit procedura make-bindings z programu 12.18.
Procedura make-bindings akceptuje dva argumenty, prvnı́m s nich je seznam formálnı́ch argumentů
a druhým je metaseznam elementů (hodnot), které se majı́ na dané formálnı́ argumenty „navázat“. Výsled-
kem aplikace make-bindings je tabulka vazeb, která je posléze použita jako součást nového prostředı́.

Samotná aplikace procedur je prováděna pomocı́ scm-apply, viz program 12.19. Ještě před tı́m, že probe-
remescm-apply, se budeme zabývat obecnějšı́ pomocnou metaprocedurouscm-env-apply, která je rovněž
uvedena v programu 12.19. Metaprocedura scm-env-apply bere jako argumenty proceduru, prostředı́ a ar-
gumenty s nimiž má být procedura aplikována. Pokud jde o primitivnı́ proceduru, jejı́ aplikace je provedena
prostou aplikacı́ metaprocedury. V tomto přı́padě nehraje prostředı́ předané scm-env-apply žádnou roli.
V přı́padě, že je vyžadována aplikace uživatelsky definované procedury se vytvořı́ nová tabulka vazeb mezi
formálnı́mi argumenty této procedury a hodnotami, se kterými proceduru aplikujeme. Dále je pomocı́ této
tabulky a předaného prostředı́ (navázaného na env) vytvořeno nové prostředı́ a v něm je vyhodnoceno tělo

306



Program 12.18. Vytvořenı́ tabulky vazeb mezi formálnı́mi a skutečnými argumenty.

(define make-bindings

(lambda (formal-args args)

(cond ((scm-null? formal-args) the-empty-list)

((scm-symbol? formal-args)

(make-pair (make-pair formal-args (list->scm-list args))

the-empty-list))

(else (make-pair

(make-pair (pair-car formal-args) (car args))

(make-bindings (pair-cdr formal-args) (cdr args)))))))

Program 12.19. Implementace aplikace procedur.

(define scm-env-apply

(lambda (proc env args)

(cond ((scm-primitive? proc) (apply (get-data proc) args))

((scm-user-procedure? proc)

(scm-eval (procedure-body proc)

(make-env env

(make-bindings (procedure-arguments proc)

args))))

(else (error "APPLY: Expected procedure")))))

(define scm-apply

(lambda (proc args)

(cond ((scm-primitive? proc) (scm-env-apply proc #f args))

((scm-user-procedure? proc)

(scm-env-apply proc (procedure-environment proc) args))

(else (error "APPLY: Expected procedure")))))

procedury. Jinými slovy, scm-env-apply v přı́padě uživatelsky definovaných procedur provádı́ vyhodno-
cenı́ jejich těla v prostředı́, jehož předek je prostředı́ navázané na env. Tı́m je scm-env-apply obecnějšı́ než
tradičnı́ apply, kde je předek nově vzniklého prostředı́ dán prostředı́m vzniku procedury.

Naprogramovat scm-apply pomocı́ scm-env-apply je již velmi jednoduché. V programu 12.19 vidı́me, že
v přı́padě aplikace primitivnı́ch procedur je voláno scm-env-apply s prostřednı́m argumentem nastave-
ným na „nepravda“ (hodnota tohoto argumentu může být jakákoliv, protože jak jsme již zjistili, nebude
k ničemu použita). V přı́padě aplikace uživatelsky definovaných procedur je opět voláno scm-env-apply,
tentokrát je ale prostředı́ předka nastaveno na prostředı́ vzniku procedury (což je prostředı́ uložené v datové
složce elementu reprezentujı́cı́ho uživatelsky definovanou proceduru).

12.6 Počátečnı́ prostředı́ a cyklus REPL

Nynı́ nadefinujeme počátečnı́ prostředı́ našeho interpretu. Jelikož se snažı́me vytvářet čistě funkcionálnı́
interpret Scheme, tedy interpret, ve kterém žádná procedura ani speciálnı́ forma nemá vedlejšı́ efekt, náš inter-
pret nebude obsahovat speciálnı́ formudefine, jejı́mž vedlejšı́m efektem je modifikace aktuálnı́ho prostředı́.
V důsledku budeme muset v našem interpretu vytvářet rekurzivnı́ procedury pomocı́ y-kombinátorů.

Dalšı́m zajı́mavým rysem vašeho interpretu bude, že na počátku vyhodnocovánı́ nebudeme uvažovat
pouze jediné prostředı́, ale hned několik prostředı́, které mezi sebou budou mı́t určité vazby. Konkrétně

budeme rozlišovat dvě prostředı́:

• prostředı́ primitivnı́ch definic (anglický název toplevel environment) – prostředı́, které nemá předka (přı́sně
vzato je to tedy globálnı́ prostředı́), v němž jsou symboly navázány na primitivnı́ procedury, speciálnı́
formy a přı́padně dalšı́ elementy vyjma uživatelsky definovaných procedur,

• prostředı́ odvozených definic (anglický název midlevel environment) – prostředı́, jehož předkem je prostředı́
primitivnı́ch definic, v němž jsou symboly navázány na uživatelsky definované procedury.

Proč vůbec potřebujeme vytvořit dvě prostředı́? Přı́sně vzato bychom nemuseli, ale tı́m pásem bychom
v počátečnı́m prostředı́ nemohli mı́t na žádný symbol navázanou netriviálnı́ uživatelsky definovanou pro-
ceduru. Každá uživatelsky definovaná procedura má totiž v sobě obsaženu informaci o prostředı́ svého
vzniku. Jelikož nemáme k dispozici prostředky pro „změnu (mutaci) prostředı́ “, nemůžeme vytvořit pro-
ceduru, která by byla navázána v prostředı́ svého vlastnı́ho vzniku. Kromě prostředı́ primitivnı́ch definic
tedy musı́me mı́t k dispozici i nové prostředı́, v němž mohou být definice uživatelsky definovaných pro-
cedur jejichž prostředı́m vzniku je právě prostředı́ primitivnı́ch definic. Takovým novým prostředı́m je
právě prostředı́ odvozených definic20. Podotkněme, že uživatelsky definované procedury navázané v pro-
středı́ odvozených definic se „vzájemně nevidı́ “, protože prostředı́m vzniku všech procedur je prostředı́
primitivnı́ch definic. Kdybychom chtěli, aby některé z uživatelsky definovaných procedur mohly použı́vat
jiné, museli bychom vytvořit nové „prostředı́ odvozených definic II.“ jehož předkem by bylo existujı́cı́
prostředı́ odvozených definic. V tomto přı́padě by uživatelsky definované procedury navázané na symboly
v prostředı́ odvozených definic II. mohly použı́vat procedury navázané ve výchozı́m prostředı́ odvozených
definic. Dále bychom mohli dle potřeby vytvářet dalšı́ „prostředı́ odvozených definic III., IV., . . . “.

Nynı́ si tedy rozebereme obsah prostředı́ primitivnı́ch definic a prostředı́ odvozených definic. Začneme
prostředı́m primitivnı́ch definic. V našem interpretu bude prostředı́ zavedeno pomocı́ konstruktoru globálnı́ho
prostředı́ make-global-env z programu 12.9 na straně 300.

(define scheme-toplevel-env

(make-global-env

`(
...

)))

Výpustka uvedená v předchozı́m kódu značı́ mı́sto, kam budou uváděny definice vazeb symbolů, kterými
se budeme zabývat ve zbytku této sekce. Nejprve ukážeme definice několika základnı́ch speciálnı́ch forem.
V programu 12.20 je ukázána definice speciálnı́ formy if. Tato definice (a každá dalšı́ uvedená) je ve tvaru

Program 12.20. Definice speciálnı́ formy if v globálnı́m prostředı́.

(if . ,(make-specform

(lambda (env condition expr . alt-expr)

(let ((result (scm-eval condition env)))

(if (equal? result scm-false)

(if (null? alt-expr)

the-undefined-value

(scm-eval (car alt-expr) env))

(scm-eval expr env))))))

páru (〈symbol〉. 〈element〉), kde symbol je metasymbol určujı́cı́ „jméno symbolu“ a 〈element〉 je konkrétnı́
element navázaný na symbol. V našem přı́padě je metasymbolem if a element navázaný na přı́slušný
symbol v prostředı́ primitivnı́ch definic vznikne vyhodnocenı́m výrazu (make-specform · · · . Zde by nás
nemělo překvapit uvedenı́ „,“ před výrazem, protože si musı́me uvědomit, že celý pár(〈symbol〉. 〈element〉)

20Některé uživatelsky definované procedury bychom navázané mı́t mohli, třeba procedury vracejı́cı́ konstantnı́ čı́selnou hodnotu
nebo projekce. Tyto procedury totiž ve svém těle kromě formálnı́ch argumentů nepoužı́vajı́ žádné dalšı́ symboly. Tı́m pádem může
být prostředı́ jejich vzniku nastaveno na „nepravda“.

je vložen do předchozı́ho výrazu na mı́stě výpustky, která je v kvazikvotovaném seznamu (viz seznam
navázaný na scheme-toplevel-env).

Prohlédneme-li si výraz v programu 12.20 definujı́cı́ speciálnı́ formu if vidı́me, že odpovı́dá formálnı́mu
popisu if z definice 1.31 na straně 35. Všimněte si, že speciálnı́ forma je realizována metaprocedurou, jejı́ž
prvnı́ argument je prostředı́ a dalšı́ argumenty odpovı́dajı́ argumentům speciálnı́ formy if. Prostředı́ je
předáváno kvůli tomu, abychom mohli v těle metaprocedury realizujı́cı́ speciálnı́ formu provádět vyhod-
nocovánı́, viz aplikaci speciálnı́ch forem v programu 12.17. V těle metaprocedury je nejprve vyhodnocen
argument condition a výsledek je navázán na symbol result. Podle výsledné hodnoty je rozhodnuto,
zda-li se vyhodnotı́ expr nebo nepovinný poslednı́ argument alt-expr. Ošetřen je i přı́pad vrácenı́ nedefi-
nované hodnoty v přı́padě, že condition se vyhodnotilo na „nepravda“ a v if-výrazu chybı́ náhradnı́k, viz
definici 1.31 na straně 35. Všimněte si, že pro definici speciálnı́ formy if jsme, mimo jiné, použili speciálnı́
metaformu if.

Analogicky jako speciálnı́ formu if bychom mohli zavést speciálnı́ formu and jejı́ž definici nalezneme
v programu 12.21. Při naprogramovánı́ speciálnı́ formy jsme opět postupovali tak, že se chová jako and

Program 12.21. Definice speciálnı́ formy and v globálnı́m prostředı́.

(and . ,(make-specform

(lambda (env . exprs)

(let and-eval ((exprs exprs))

(cond ((null? exprs) scm-true)

((null? (cdr exprs)) (scm-eval (car exprs) env))

(else (let ((result (scm-eval (car exprs) env)))

(if (equal? result scm-false)

scm-false

(and-eval (cdr exprs))))))))))

představená v definici 2.22 na straně 64. Metaprocedura ve svém těle použı́vá iterativnı́ proceduru, která
postupně procházı́ a vyhodnocuje jednotlivé elementy předané speciálnı́ formě. Pokud se některý z ele-
mentů vyhodnotı́ na „nepravda“, je iterace ukončena a vrácena je pravdivostnı́ hodnota „nepravda“. Pokud
již zbývá jen poslednı́ element, je vrácena hodnota vzniklá jeho vyhodnocenı́m. Pokud byly zpracovány
všechny prvky, je výsledek aplikace speciálnı́ formy pravdivostnı́ hodnota „pravda“.

Kromě and můžeme vytvořit i primitivnı́ proceduru not následovně:

(not . ,(make-primitive

(lambda (elem)

(if (equal? elem scm-false)

scm-true

scm-false))))

Jelikož je not procedura a nikoliv speciálnı́ forma, nenı́ ji předáváno prostředı́ a předaný argument je již ve
své vyhodnocené podobě. Pomocı́ not a and můžeme vyjadřovat i disjunktivnı́ podmı́nky (viz komentář
v sekci 2.7). Samozřejmě, že z programátorského hlediska by bylo dobré naprogramovat rovněž i speciálnı́
formu or. Realizace speciálnı́ formy or je jednı́m z řešených přı́kladů na konci této sekce, proto ji nynı́
uvádět nebudeme.

V programu 12.22 jsou uvedeny definice speciálnı́ch forem lambda, the-environment a quote. Jak je vidět,
realizace těchto forem je velmi jednoduchá. Speciálnı́ forma lambda vytvořı́ nový element typu „uživatelsky
definovaná procedura“ na základě předaného prostředı́, seznamu argumentů a těla. Speciálnı́ forma the-

environment je realizována metaprocedurou, která pouze vracı́ aktuálnı́ prostředı́ (tato metaprocedura
je vlastně identita). Analogicky speciálnı́ forma quote vracı́ předaný element (bez jeho vyhodnocenı́),
metaprocedura realizujı́cı́ quote je tedy projekce.

Pro uživatelsky definované procedury můžeme vytvořit sadu selektorů, viz program 12.23. Se třemi

Program 12.22. Definice speciálnı́ch forem lambda, the-environment a quote v globálnı́m prostředı́.

(lambda . ,(make-specform

(lambda (env args body)

(make-procedure env args body))))

(the-environment . ,(make-specform (lambda (env) env)))

(quote . ,(make-specform (lambda (env elem) elem)))

Program 12.23. Definice selektorů uživatelsky definovaných procedur v globálnı́m prostředı́.

(procedure-environment . ,(make-primitive procedure-environment))

(procedure-arguments . ,(make-primitive procedure-arguments))

(procedure-body . ,(make-primitive procedure-body))

(environment-parent . ,(make-primitive get-pred))

(environment->list . ,(make-primitive

(lambda (elem)

(if (equal? elem scm-false)

scm-false

(get-table elem)))))

procedurami procedure-environment, environment-parent a environment->list jsme se už setkali
v lekci 6. Tyto procedury vracı́ prostředı́ vzniku procedury, nadřazené prostředı́ daného prostředı́ a po-
slednı́ procedura převádı́ tabulku na čitelný asociačnı́ seznam. V programu 12.23 máme navı́c selektor
procedure-arguments, který pro danou proceduru vracı́ seznam jejich argumentů. Selektor procedure-
body pro danou proceduru vracı́ výraz, který je tělem procedury.

Nynı́ můžeme popsat, jak do prostředı́ primitivnı́ch definic zabudujeme primitivnı́ procedury apply a eval.
Jelikož chceme, aby apply pracoval s libovolným počtem argumentů, to jest ve tvaru

(apply 〈procedura〉 〈arg1〉 〈arg2〉 · · · 〈argn〉 〈seznam〉),

zavedeme nejprve pomocnou metaproceduru apply-collect-arguments, která ze všech předaných ar-
gumentů ve tvaru 〈arg1〉 · · · 〈argn〉 〈seznam〉 sestavı́ (jediný) seznam všech argumentů. Kód této metaprocedury
je v programu 12.24. V programu 12.25 jsou pak uvedeny definice procedur eval (procedura dvou argu-

Program 12.24. Sestavenı́ seznamu argumentů pro obecný typ volánı́ procedury apply.

(define apply-collect-arguments

(lambda (args)

(cond ((null? args) (error "APPLY: argument missing"))

((and (not (null? args)) (null? (cdr args)))

(scm-list->list (car args)))

(else (cons (car args) (apply-collect-arguments (cdr args)))))))

mentů z nichž druhý – reprezentujı́cı́ prostředı́ – je vždy povinný), apply a env-apply (zobecněná verze
apply, které je jako druhý argument předáno prostředı́, viz program 12.19 na straně 307).

Program 12.25. Definice eval, apply and env-apply v globálnı́m prostředı́.

(eval . ,(make-primitive

(lambda (elem env)

(scm-eval elem env))))

(apply . ,(make-primitive

(lambda (proc . rest)

(scm-apply proc (apply-collect-arguments rest)))))

(env-apply . ,(make-primitive

(lambda (proc env . rest)

(scm-env-apply proc

env

(apply-collect-arguments rest)))))

Ostatnı́ procedury v prostředı́ primitivnı́ch vazeb mohou být vytvořeny rutinně, nebudeme je tedy všechny
vypisovat. Tı́mto čtenáře odkazujeme na zdrojové kódy interpretu jazyka Scheme, který je dodáván spolu
s tı́mto učebnı́m textem. Naznačme ale zhruba, jak definice vypadajı́. Budeme předpokládat, že v prostředı́
primitivnı́ch vazeb budeme mı́t na symbol pi navázánu čı́selnou hodnotu čı́sla π. Tuto vazbu bychom
provedli třeba takto:

(pi . ,(make-number (* 4 (atan 1))))

Aritmetické procedury (sčı́tánı́, odčı́tánı́, zaokrouhlovánı́ a podobně) můžeme vytvořit pomocı́ metapro-
cedury wrap-primitive, kterou jsme již popsali v programu 12.14 na straně 303. Při definici aritmetických
procedur tedy budeme postupovat následovně:

(* . ,(wrap-primitive *))

(+ . ,(wrap-primitive +))

(- . ,(wrap-primitive -))

(/ . ,(wrap-primitive /))

(< . ,(wrap-primitive <))

(<= . ,(wrap-primitive <=))

(= . ,(wrap-primitive =))
...

(tan . ,(wrap-primitive tan))

(truncate . ,(wrap-primitive truncate))

(zero? . ,(wrap-primitive zero?))

Rovněž definice konstruktorů a selektorů párů je jednoduchá. Pouze převedeme metaprocedury z pro-
gramu 12.7 na straně 297 na primitivnı́ procedury a uvedeme vazby na přı́slušné symboly.

(cons . ,(make-primitive make-pair))

(car . ,(make-primitive pair-car))

(cdr . ,(make-primitive pair-cdr))

Predikát testujı́cı́ prázdnost seznamu vytvořı́me pomocı́ porovnánı́ daného elementu s elementem repre-
zentujı́cı́m prázdný seznam:

(null? . ,(make-primitive

(lambda (l)

(expr->intern (equal? l the-empty-list)))))

Nynı́ se můžeme začı́t věnovat prostředı́ odvozených definic. Toto prostředı́ vytvořı́me analogicky jako
prostředı́ primitivnı́ch definice. Jediným rozdı́lem bude, že prostředı́ odvozených definic již bude mı́t

jako svého předka nastaveno jiné prostředı́, konkrétně právě prostředı́ primitivnı́ch definice. Následujı́cı́
fragment kódu ukazuje tvar, v jakém můžeme prostředı́ odvozených definic zavést.

(define scheme-midlevel-env

(make-env

scheme-toplevel-env

(assoc->env

`(
...

(sgn . ,(make-procedure

scheme-toplevel-env

(expr->intern '(x))

(expr->intern '(if (= x 0)

0

(if (> x 0)

1

-1)))))
...

))))

V předchozı́m kódu jsou opět uvedeny výpustky a je zde uveden přı́klad definice uživatelsky definované
procedury sgn. Element navázaný na symbol sgn v tomto prostředı́ je skutečně uživatelsky definovaná pro-
cedura, protože se jedná o element vytvořený aplikacı́ make-procedure, viz program 12.11 na straně 301.
Při vytvořenı́ této uživatelsky definované procedury jsme jako prostředı́ vzniku předali prostředı́ primi-
tivnı́ch vazeb (navázané na scheme-toplevel-env). Seznam argumentů procedury signum jsme vytvořili
převedenı́m metaseznamu (x) do jeho internı́ formy. Stejným způsobem jsme zapsali tělo procedury.

Ze vztahu obou prostředı́ je patrné, že ani v prostředı́ odvozených definic nemůžeme vytvářet rekurzivnı́
procedury bez použitı́ y-kombinátoru, protože procedura nenı́ navázána na symbol v prostředı́ svého
vzniku. Při definici rekurzivnı́ch procedur si tedy musı́me pomoci y-kombinátorem, viz sekci 9.2. Přı́klady
rekurzivnı́ch procedur definovaných v prostředı́ odvozených definic najdeme v programech 12.26 (výpočet
délky seznamu) a 12.27 (mapovánı́ přes jeden seznam).

Program 12.26. Procedura length v prostředı́ odvozených definic.

(length . ,(make-procedure

scheme-toplevel-env

(expr->intern '(l))

(expr->intern

'((lambda (y)

(y y l))

(lambda (length l)

(if (null? l)

0

(+ 1 (length length (cdr l)))))))))

Poslednı́ věcı́, kterou musı́me vyřešit, je implementace cyklu REPL, ve kterém poběžı́ samotné vyhodnoco-
vánı́. REPL bude realizován jednoduchou iterativnı́ metaprocedurou, která simuluje činnost vyhodnocova-
cı́ho cyklu tak, jak jsme jej popsali v sekci 1.5. Viz kód uvedený v programu 12.28. Nejprve je vytvořeno nové
prostředı́, jehož předkem je prostředı́ odvozených vazeb. Toto prostředı́ je navázáno na symbol init-env.
Dále se opakuje cyklus, ve kterém je vždy načten symbolický výraz, poté je převeden do internı́ repre-
zentace, vyhodnocen v prostředı́ navázaném na init-env, výsledek je vytištěn a celý cyklus se opakuje
dokud nenı́ vyčerpán vstup (nebo nedojde k chybě). Na poslednı́m řádku metaprogramu (interpretu) tedy

Program 12.27. Procedura map v prostředı́ odvozených definic.

(map . ,(make-procedure

scheme-toplevel-env

(expr->intern '(f l))

(expr->intern

'((lambda (y)

(y y l))

(lambda (map l)

(if (null? l)

()

(cons (f (car l)) (map map (cdr l)))))))))

Program 12.28. Implementace cyklu REPL.

(define scm-repl

(lambda ()

(let ((init-env (make-env scheme-midlevel-env the-empty-list)))

(let loop ()

(display "]=> ")

(let ((elem (scm-read)))

(if (not elem)

'bye-bye

(let ((result (scm-eval elem init-env)))

(newline)

(scm-print result)

(newline)

(newline)

(loop))))))))

spustı́me metaproceduru (scm-repl), která dále řı́dı́ průběh vyhodnocovánı́. Tı́m jsme završili vývoj prvnı́
verze našeho interpretu (dalšı́ vylepšenı́ ukážeme v dalšı́m dı́le tohoto učebnı́ho textu).

Zdrojový kód našeho interpretu (včetně komentářů) nepřesahuje 600 řádků, z pohledu velikosti se tedy
jedná o velmi malý program. Velké programy běžně přesahujı́ stovky tisı́c i miliony řádků. Napřı́klad
jádra operačnı́ch systémů mı́vajı́ kolem pěti milionů řádků, stejně tak kancelářské balı́ky. Mezi „největšı́
programy“ patřı́ bezpečnostnı́ software pro řı́zenı́ leteckého provozu a raketové systémy. I přes to, že
náš program je pozoruhodně malý, jedná se o implementaci interpretu Turingovsky úplného progra-
movacı́ho jazyka, tedy jazyka, který je z hlediska své vyjadřovacı́ sı́ly stejně silný jako běžně použı́vané
programovacı́ jazyky (napřı́klad C, C++, LISP, Pascal, . . . ).

12.7 Přı́klady použitı́ interpretu

V této sekci ukážeme přı́klady použitı́ nově vytvořeného interpretu. Přı́klady budeme komentovat pouze
stručně, protože všechny konstrukce jsou již čtenářům důvěrně známé. Výsledné hodnoty zobrazujeme
stejně jako je výstup našeho interpretu.

Nejprve ukážeme použitı́ a chovánı́ speciálnı́ formy quote:

quote Z=⇒ Specform: #<special-form>

(quote blah) Z=⇒ Symbol: blah

'blah Z=⇒ Symbol: blah

''blah Z=⇒ Pair: (quote blah)

Dalšı́ přı́klad ukazuje kvotovánı́ seznamu:

(+ 1 2 3) Z=⇒ Number: 6

'(+ 1 2 3) Z=⇒ Pair: (+ 1 2 3)

(+ 1 (* 2 3)) Z=⇒ Number: 7

'(+ 1 (* 2 3)) Z=⇒ (+ 1 (* 2 3))

Speciálnı́ elementy se vyhodnocujı́ na sebe sama, jako obvykle:

() Z=⇒ Empty-list: ()

'() Z=⇒ Empty-list: ()

#t Z=⇒ Boolean: #t

'#t Z=⇒ Boolean: #t

#f Z=⇒ Boolean: #f

'#f Z=⇒ Boolean: #f

Speciálnı́ formu if použı́váme obvyklým způsobem. Z poslednı́ho přı́kladu je vidět, že if se skutečně
chová jako speciálnı́ forma a nikoliv jako procedura:

if Z=⇒ Specform: #<special-form>

(if 1 2 3) Z=⇒ Number: 2

(if #f 1 2) Z=⇒ Number: 2

(if #f 1) Z=⇒ Undefined: #<undefined>

(if #t 1 blah-blah) Z=⇒ Number: 1

Následujı́cı́ přı́klad ukazuje použitı́ speciálnı́ formy and:

(and) Z=⇒ Boolean: #t

(and 10) Z=⇒ Number: 10

(and #f) Z=⇒ Boolean: #f

(and 0 2 4) Z=⇒ Number: 4

(and 3 (= 1 1)) Z=⇒ Boolean: #t

Pomocı́ speciálnı́ formy lambda vytvářı́me procedury:

lambda Z=⇒ Specform: #<special-form>

(lambda (x) (+ x 1)) Z=⇒ Procedure: #<user-defined-procedure>

((lambda (x) (+ x 1)) 10) Z=⇒ Number: 11

Náš interpret umožňuje práci se speciálnı́mi formami jako s elementy prvnı́ho řádu. V následujı́cı́m přı́kladu
je speciálnı́ forma předána jako argument proceduře. Podobnou konstrukci by nám drtivá většina interpretů
jazyka Scheme vůbec neumožnila (speciálnı́ formy nejsou v existujı́cı́ch interpretech jazyka Scheme chápány
jako elementy prvnı́ho řádu).

(((lambda (procedura)

(procedura (x) (+ x 1)))

lambda)

10) Z=⇒ Number: 11

Rekurzivnı́ procedury můžeme definovat pomocı́ y-kombinátoru, jako třeba:

((lambda (y)

(y y 6))

(lambda (fak n)

(if (= n 0)

1

(* n (fak fak (- n 1)))))) Z=⇒ Number: 720

Na symbol map je navázána uživatelsky definovaná procedura:

map Z=⇒ Procedure: #<user-defined-procedure>

(procedure-environment map) Z=⇒ Environment: #<environment>

(procedure-arguments map) Z=⇒ Pair: (f l)

(procedure-body map) Z=⇒ Pair: ((lambda (y) (y y l)) (lambda (map · · ·

Tato uživatelsky definovaná procedura funguje standardně:

(map - '(1 2 3 4)) Z=⇒ Pair: (-1 -2 -3 -4)

(map (lambda (x) (cons x ())) '(a b c d)) Z=⇒ Pair: ((a) (b) (c) (d))

(map even? '(0 1 2 3 4 5 6)) Z=⇒ Pair: (#t #f #t #f #t #f #t)

(map (lambda (x) (<= x 3)) '(0 1 2 3 4 5 6)) Z=⇒ Pair: (#t #t #t #t #f #f #f)

Následujı́cı́ přı́klad ukazuje použitı́ map a rekurzivnı́ procedury počı́tajı́cı́ faktoriál:

(map

(lambda (n)

((lambda (y)

(y y n))

(lambda (fak n)

(if (= n 0)

1

(* n (fak fak (- n 1)))))))

'(0 1 2 3 4 5 6 7 8 9)) Z=⇒ Pair: (1 1 2 6 24 120 720 5040 40320 362880)

Následujı́cı́ dvě ukázky demonstrujı́ použitı́ libovolných a nepovinných argumentů.

((lambda list (map - list)) 1 2 3 4 5 6) Z=⇒ Pair: (-1 -2 -3 -4 -5 -6)

((lambda (x y . list)

(cons x (cons y (map - list)))) 1 2 3 4 5 6) Z=⇒ Pair: (1 2 -3 -4 -5 -6)

Pomocı́ procedure-environment můžeme zı́skat prostředı́ vzniku procedury:

(procedure-environment

((lambda (x)

(lambda (y) (+ x y)))

10)) Z=⇒ Environment: #<environment>

Vazby v prostředı́ můžeme vypsat pomocı́ environment->list:

(environment->list

(procedure-environment

((lambda (x)

(lambda (y) (+ x y)))

10))) Z=⇒ Pair: ((x . 10))

V následujı́cı́m přı́kladu je zı́skáno a vypsáno počátečnı́ prostředı́ (jeho tabulka vazeb je prázdná). V druhém
přı́padě je pak vypsána tabulka vazeb v předchůdci aktuálnı́ho prostředı́, což je prostředı́ odvozených
definic.

(the-environment) Z=⇒ Environment: #<environment>

(environment->list (the-environment)) Z=⇒ Empty-list: ()

(environment->list

(environment-parent

(the-environment))) Z=⇒ Pair: ((sgn . #<user-defined-procedure>)

(length . #<user-defined-procedure>)

(map . #<user-defined-procedure>))

Procedura eval je možné použı́vat pouze se dvěma argumenty:

(eval '(+ 1 2) (the-environment)) Z=⇒ Number: 3

V následujı́cı́ ukázce je vyhodnocen seznam (length (quote (a b c))) ve třech různých prostředı́ch.
V poslednı́m přı́padě dojde při vyhodnocenı́ k chybě, protože v prostředı́ primitivnı́ch definic nenı́ pro-
cedura length definovaná.

(eval '(length '(a b c)) (the-environment)) Z=⇒ Number: 3

(eval '(length '(a b c)) (environment-parent

(the-environment))) Z=⇒ Number: 3

(eval '(length '(a b c)) (environment-parent

(environment-parent

(the-environment)))) Z=⇒ Error (symbol not bound)

Dalšı́ přı́klad ukazuje vyhodnocenı́ seznamu '(* x x) v prostředı́ vzniku procedury:

(eval '(* x x)

(procedure-environment

((lambda (x)

(lambda (y) (+ x y)))

10))) Z=⇒ Number: 100

Proceduru apply je možné použı́t i s vı́ce jak dvěma argumenty:

(apply + '(1 2 3 4)) Z=⇒ Number: 10

(apply + 1 2 '(3 4)) Z=⇒ Number: 10

(apply + 1 2 3 4 '()) Z=⇒ Number: 10

(apply map - '((1 2 3 4))) Z=⇒ Pair: (-1 -2 -3 -4)

(apply map - '(1 2 3 4) '()) Z=⇒ Pair: (-1 -2 -3 -4)

V dalšı́m přı́kladu využijeme faktu, že na symbol pi máme navázanou hodnotu čı́sla π:

pi Z=⇒ Number: 3.141592653589793

Při vyhodnocenı́ následujı́cı́ho výrazu nehraje globálnı́ vazba pi roli, protože interpret použı́vá lexikálnı́
rozsah platnosti:

(((lambda (pi)

(lambda (x)

(+ x pi)))

10)

20) Z=⇒ Number: 30

Totéž platı́ pro explicitnı́ aplikaci:

(apply ((lambda (pi)

(lambda (x)

(+ x pi)))

10)

20

'()) Z=⇒ Number: 30

V následujı́cı́ ukázce jsme provedli aplikaci uživatelsky definované procedury, přitom jsme „dočasně na-
stavili“ předka této procedury na lokálnı́ prostředı́ v němž je na y navázaná hodnota 100:

(env-apply (lambda (x) (+ x y))

((lambda (y) (the-environment)) 100)

20 '()) Z=⇒ Number: 120

Shrnutı́

Zabývali jsme se automatickým přetypovánı́m a generickými procedurami. Pro generické procedury jsme
zavedli jednoduchou metodu jejich aplikace prostřednictvı́m vyhledávánı́ metod pomocı́ vzorů uvedených
v tabulkách. Na konkrétnı́m přı́kladu generických procedur jsme ukázali jejich praktické použitı́. Dále

jsme se seznámili s konceptem manifestovaných typů. Pomocı́ manifestovaných typů jsme implementovali
reprezentaci elementů jazyka Scheme. S jejich využitı́m jsme dále naprogramovali vyhodnocovacı́ proces.
Nakonec jsme zkompletovali interpret čistě funkcionálnı́ podmnožiny jazyka Scheme.

Pojmy k zapamatovánı́

• přetypovánı́, implicitnı́/explicitnı́ přetypovánı́,
• automatické přetypovánı́, koerce,
• generická procedura, tabulka generických procedur,
• manifestované typy, visačky, tagy,
• metajazyk, metainterpret, metaelement, metaprocedura.

Nově představené prvky jazyka Scheme

• procedury number->string a string-append

Kontrolnı́ otázky

1. Co jsou to generické procedury?
2. Jaké jsou výhody a nevýhody automatického přetypovánı́?
3. Co jsou to manifestované typy?
4. Jak jsme v našı́ implementaci jazyka Scheme reprezentovali jednotlivé elementy jazyka?
5. Kolik jsme uvažovali počátečnı́ch prostředı́ a proč?
6. Jaký je rozdı́l mezi jazykem/interpretem a metajazykem/metainterpretem?

Cvičenı́

1. Bez použitı́ interpretu jazyka Scheme zjistěte, jak se vyhodnotı́ následujı́cı́ výrazy použı́vajı́cı́ generické
+ tak, jak jsme jej představili v sekci 12.1:
(+ 2/3 "." 0.5) Z=⇒
(+ (- 1) "x" (- 2)) Z=⇒
(+ 1 2 "+" 3 4) Z=⇒
(+ 2 (+ 2 "3") 10) Z=⇒
(+ 2 (log (exp 2))) Z=⇒
(apply + '("a" "b" (+ 1 2) 2)) Z=⇒
(apply + (map number->string '(1 2 3))) Z=⇒

2. Obohat’te vytvořený interpret jazyka Scheme o primitivnı́ proceduru foldr. To jest foldr by měla
být ve vytvořeném interpretu k dispozici přı́mo v počátečnı́m prostředı́ a mělo by se jednat o primi-
tivnı́ proceduru, tedy nikoliv o uživatelsky definovanou proceduru. Naprogramujte tuto primitivnı́
proceduru tak, aby mohla pracovat s libovolným množstvı́m seznamů (vždy však alespoň s jednı́m).

3. Analogicky proved’te obohacenı́ interpretu o primitivnı́ proceduru foldl.
4. Obohat’te vytvořený interpret jazyka Scheme o speciálnı́ formu or. Tuto speciálnı́ formu naprogramujte

tak, aby se chovala přesně jako speciálnı́ forma or v jazyku Scheme, viz [R5RS].
5. Obohat’te vytvořený interpret jazyka Scheme o speciálnı́ formu let (nepojmenovanou verzi).
6. Obohat’te vytvořený interpret jazyka Scheme o pojmenovaný let. Jelikož v jazyku nemáme k dispozici

define, musı́me si při programovánı́ pojmenovaného let pomocı́ y-kombinátoru. Speciálnı́ formu
ale vytvořte tak, abychom při rekurzivnı́ aplikaci nemuseli předávat proceduru prostřednictvı́m jejı́ho
prvnı́ho argumentu. Vytvořený pojmenovaný let by se tedy z uživatelského pohledu měl chovat jako
pojmenovaný let popsaný ve [R5RS].

7. Obohat’te interpret jazyka Scheme o speciálnı́ formu dyn-apply, která bude provádět aplikaci procedur
podobně jako apply, ale při aplikaci procedur pomocı́ dyn-apply se budou při vyhodnocovánı́ těla
procedury hledat vazby symbolů ve smyslu dynamického rozsahu platnosti, tedy v prostředı́ aplikace
procedury. Na následujı́cı́m přı́kladu je vidět rozdı́l použitı́ apply a dyn-apply:

((lambda (x)

(apply ((lambda (x)

(lambda (y)

(+ x y))) 100)

'(20)))

1000) Z=⇒ 120

((lambda (x)

(dyn-apply ((lambda (x)

(lambda (y)

(+ x y))) 100)

'(20)))

1000) Z=⇒ 1020

8. Obohat’te interpret o novou speciálnı́ formu delta, která je po syntaktické stránce shodná s formou
lambda. Vyhodnocenı́m δ-výrazu vznikne speciálnı́ typ uživatelsky definované procedury. Při aplikaci
procedury vzniklé vyhodnocenı́m δ-výrazů je uplatněn dynamický rozsah platnosti. Zavedenı́ δ-výrazů
do jazyka nám tedy umožnı́ vytvářet uživatelsky definované procedury, jejichž vyhodnocovánı́ probı́há
v souladu s dynamickým rozsahem platnosti. Viz přı́klad použitı́ a rozdı́l v použitı́ speciálnı́ch forem
lambda a delta.
((lambda (x)

(((lambda (x)

(lambda (y)

(+ x y))) 100) 20)) 1000) Z=⇒ 120

((lambda (x)

(((lambda (x)

(delta (y)

(+ x y))) 100) 20)) 1000) Z=⇒ 1020
Při implementaci využijte svých znalostı́ ze sekce 2.6.

9. Napište, jak interpret Scheme vyhodnotı́ následujı́cı́ symbolické výrazy:
(environment->list

((lambda (x) (the-environment)) 10)) Z=⇒

((lambda (x)

(eval '(- x) (the-environment)))

100) Z=⇒

(eval '(+ x 1)

((lambda (x) (the-environment)) 10)) Z=⇒

((lambda (x)

((lambda (x)

(eval '(cons x (quote x)) (procedure-environment x)))

(lambda (x) 'blah)))

1000) Z=⇒

(environment->list

(procedure-environment

((lambda (y)

(lambda (x)

(+ x 1)))

100))) Z=⇒

Úkoly k textu

1. Vytvořte systém generických procedur +, -, *, modulo, quotient a = sloužı́cı́ch pro práci s čı́sly
a polynomy. Polynomy (jedné proměnné) reprezentujte seznamy jejich koeficientů tak, jak jsme to
dělali v sekci 8.6. Základnı́ aritmetické procedury +, -, * sloužı́ k prováděnı́ aritmetických operacı́
s čı́sly a polynomy. Napřı́klad tedy pro * by to mělo pokrývat součin čı́sel, součin polynomů a součin
čı́sla s polynomem. Procedury modulo a quotient by měly vracet podı́l čı́sel nebo polynomů a zbytek
po dělenı́ čı́slem nebo polynomem. Predikát = by měl sloužit k porovnávánı́ čı́sel a polynomů.
Při implementaci si pomozte tı́m, že čı́sla jsou de facto konstantnı́ polynomy. Kde lze, provádějte
automatické přetypovánı́.

2. Po dokončenı́ předchozı́ho úkolu naprogramujte procedury pro sčı́tánı́ a násobenı́ matic. Pokud jsme
provedli správnou implementaci v předchozı́m bodě, měla by vaše implementace operacı́ s maticemi
umožňovat pracovat s maticemi jejichž prvky jsou čı́sla nebo polynomy. Důkladně vaši implementaci
otestujte.

3. Obohat’te interpret jazyka Scheme o speciálnı́ formy cond a let*. Obě speciálnı́ formy naprogramujte
tak, aby se chovaly stejně, jak jsme je představili v lekcı́ch 2 a 3. Implementace obou speciálnı́ch forem
důkladně otestujte a přesvědčte se o jejich správnosti.

4. Obohat’te interpret jazyka Scheme o speciálnı́ formu named-lambda, která je po syntaktické stránce
totožná se speciálnı́ formou lambda. Speciálnı́ forma named-lambda sloužı́ k vytvářenı́ procedur
stejně jako speciálnı́ forma lambda, navı́c však v těle procedury vytvořené pomocı́ named-lambda je
vždy na symbol self navázaná samotná procedura. Pomocı́ self je tedy možné provádět rekurzivnı́
volánı́, viz následujı́cı́ přı́klad použitı́.

(map (named-lambda (n)

(if (= n 0)

1

(* n (self (- n 1)))))

'(0 1 2 3 4 5 6 7 8)) Z=⇒ (1 1 2 6 24 120 720 5040 40320)

5. V předchozı́ch přı́kladech na procvičenı́ jsme viděli dva přı́stupy k aplikaci procedur řı́dı́cı́ se dy-
namickým rozsahem platnosti. Prvnı́ metodou bylo použitı́ explicitnı́ aplikace pomocı́ dyn-apply.
Druhým způsobem bylo zavedenı́ δ-výrazů, pomocı́ nichž vznikaly procedury aplikované vždy ve
smyslu dynamického rozsahu platnosti. Ani jedno z těchto řešenı́ nenı́ úplně ideálnı́, protože v ně-
kterých přı́padech by se mohlo hodit, abychom v těle jedné procedury uvažovali většinu symbolů ve
smyslu lexikálnı́ho rozsahu platnosti (zajı́mat se budeme o lexikálnı́ vazby symbolů) a některé sym-
boly ve smyslu dynamického rozsahu platnosti (zajı́mat se budeme o dynamické vazby symbolů).

Obohat’te interpret o speciálnı́ formu dynamic, jejı́mž jediným argumentem bude symbol. Výsledkem
vyhodnocenı́ (dynamic x) (v daném prostředı́) je dynamická vazba symbolu x, tedy vazba, která
je hledána v dynamicky nadřazených prostředı́ch (pokud nenı́ nalezena v lokálnı́m prostředı́), viz
sekci 2.6. Následujı́cı́ ukázka demonstruje použitı́ dynamic.

((lambda (x)

(((lambda (x)

(lambda (y)

(+ x y))) 100) 20)) 1000) Z=⇒ 120

((lambda (x)

(((lambda (x)

(lambda (y)

(+ (dynamic x) y))) 100) 20)) 1000) Z=⇒ 1020

Řešenı́ ke cvičenı́m

1. "2/3.0.5", "-1x-2", "12+7", "22310", 4.0, Error: No method for these types, "123"

2. Vytvořı́me pomocnou proceduru scm-foldr následovně:
(define scm-foldr

(lambda (f basis . lists)

(if (scm-null? (car lists))

basis

(scm-apply

f

(append (map pair-car lists)

(list (apply

scm-foldr

f basis (map pair-cdr lists))))))))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(foldr . ,(make-primitive scm-foldr))

3. Vytvořı́me pomocnou proceduru scm-foldl následovně:
(define scm-foldl

(lambda (f basis . lists)

(let iter ((lists lists)

(accum basis))

(if (scm-null? (car lists))

accum

(iter (map pair-cdr lists)

(scm-apply f (append (map pair-car lists)

(list accum))))))))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(foldl . ,(make-primitive scm-foldl))

4. Vytvořı́me pomocnou proceduru scm-or následovně:
(define scm-or

(lambda (env . exprs)

(let or-eval ((exprs exprs))

(cond ((null? exprs) scm-false)

((null? (cdr exprs)) (scm-eval (car exprs) env))

(else (let ((result (scm-eval (car exprs) env)))

(if (equal? result scm-false)

(or-eval (cdr exprs))

result)))))))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(or . ,(make-specform scm-or))

5. Vytvořı́me pomocnou proceduru map-scm-list pro mapovánı́ přes seznam ve vnitřnı́ reprezentaci:
(define map-scm-list

(lambda (f l)

(convert-list scm-null? pair-car pair-cdr

make-pair the-empty-list f l)))
Dále vytvořı́me pomocnou proceduru scm-let:
(define scm-let

(lambda (env bindings body)

(let ((proc (make-procedure

env

(map-scm-list pair-car bindings)

body)))

(scm-apply proc

(map-scm-list->list

(lambda (elem)

(scm-eval (pair-car (pair-cdr elem)) env))

bindings)))))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(let . ,(make-specform scm-let))

6. Vytvořı́me nejprve pomocný konstruktor seznamu v internı́ reprezentaci:
(define make-list

(lambda args

(if (null? args)

the-empty-list

(make-pair (car args)

(apply make-list

(cdr args))))))
Pro programovánı́ pomocné procedury, která bude v konstruovaném interpretu realizovat speciálnı́
formu „pojmenovaný let“ si musı́me uvědomit, že kód ve tvaru
(let 〈jméno〉 ((〈symbol1〉 〈výraz1〉)

(〈symbol2〉 〈výraz2〉)
...

(〈symboln〉 〈výrazn〉))
〈tělo〉)

stačı́ nahradit kódem ve tvaru
((lambda (y)

(y y 〈výraz1〉 〈výraz2〉 · · · 〈výrazn〉))
(lambda (〈jméno〉 〈symbol1〉 〈symbol2〉 · · · 〈symboln〉)
((lambda (〈jméno〉) 〈tělo〉)
(lambda (〈symbol1〉 〈symbol2〉 · · · 〈symboln〉)
(〈jméno〉 〈jméno〉 〈symbol1〉 〈symbol2〉 · · · 〈symboln〉)))))

Podle tohoto postupu naprogramujeme pomocnou proceduru scm-named-let:
(define scm-named-let

(lambda (env . args)

(if (not (scm-symbol? (car args)))

(apply scm-let env args)

(let* ((name (car args))

(bindings (cadr args))

(body (caddr args))

(y (make-symbol 'y))

(y-comb (make-procedure

env

(make-list y)

(make-pair y

(make-pair y

(map-scm-list

(lambda (elem)

(scm-eval (pair-car (pair-cdr elem)) env))

bindings)))))

(proc (make-procedure

env

(make-pair

name

(map-scm-list pair-car bindings))

(make-list

(make-list (make-symbol 'lambda)

(make-list name)

body)

(make-list (make-symbol 'lambda)

(map-scm-list pair-car bindings)

(make-pair name

(make-pair name

(map-scm-list pair-car bindings))))))))

(scm-apply y-comb (list proc))))))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(let . ,(make-specform scm-named-let))

7. Vytvořı́me pomocnou proceduru scm-dyn-apply následovně:
(define scm-dyn-apply

(lambda (env proc . rest)

(scm-eval

(make-pair (make-symbol 'env-apply)

(make-pair

proc

(make-pair

env

(let iter ((ar rest))

(if (null? (cdr ar))

(make-pair (car ar) the-empty-list)

(make-pair (car ar)

(iter (cdr ar))))))))

env)))
Dále je nutné do tabulky vazeb prostředı́ „toplevel“ přidat záznam:
(dyn-apply . ,(make-specform scm-dyn-apply))

8. Nejprve vytvořı́me datovou reprezentaci nového typu elementu:
(define make-dyn-procedure

(let ((make-physical-procedure (curry-make-elem 'dyn-procedure)))

(lambda (args body)

(make-physical-procedure (list args body)))))

(define dyn-procedure-arguments (lambda (proc) (car (get-data proc))))

(define dyn-procedure-body (lambda (proc) (cadr (get-data proc))))

(define scm-user-dyn-procedure? (curry-scm-type 'dyn-procedure))
Upravı́me scm-procedure? tak, aby brala ohled i na nový typ procedury:
(define scm-procedure?

(lambda (elem)

(or (scm-primitive? elem)

(scm-user-procedure? elem)

(scm-user-dyn-procedure? elem))))
Dále přidáme do tabulky vazeb prostředı́ „toplevel“ záznam:
(delta . ,(make-specform

(lambda (env args body)

(make-dyn-procedure args body))))
Upravı́me scm-eval tak, aby se při volánı́ scm-apply předávalo aktuálnı́ prostředı́ jako nový poslednı́
argument. Kritický fragment kódu by měl vypadat takto:
...

((scm-procedure? f)

(scm-apply f

(map-scm-list->list

(lambda (elem)

(scm-eval elem env))

args)

env))
...

Dále upravı́me scm-env-apply a scm-apply následovně tak, že do nich přidáme větve ošetřujı́cı́
průběh aplikaci v přı́padě procedur vzniklých vyhodnocenı́m δ-výrazů.
(define scm-env-apply

(lambda (proc env args)

(cond ((scm-primitive? proc) (apply (get-data proc) args))

((scm-user-procedure? proc)

(scm-eval (procedure-body proc)

(make-env env

(make-bindings (procedure-arguments proc)

args))))

((scm-user-dyn-procedure? proc)

(scm-eval (dyn-procedure-body proc)

(make-env env

(make-bindings (dyn-procedure-arguments proc)

args))))

(else (error "APPLY: Expected procedure")))))

(define scm-apply

(lambda (proc args . env)

(cond ((scm-primitive? proc) (scm-env-apply proc #f args))

((scm-user-procedure? proc)

(scm-env-apply proc (procedure-environment proc) args))

((scm-user-dyn-procedure? proc)

(scm-env-apply proc (car env) args))

(else (error "APPLY: Expected procedure")))))

9. ((x . 10)), -100, 11, (1000 . x), ((y . 100))

6.7 Součet druhých mocnin . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 150
6.8 Konstruktor seznamu (procedura list) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 150
7.1 Délka seznamu pomocı́ foldr (procedura length) . . . . . . . . . . . . . . . . . . . . . . . . . 174

7.8 Nahrazovánı́ prvků v seznamu pomocı́ foldr (procedura replace) . . . . . . . . . . . . . . . 178
7.9 Procedury genuine-foldl a foldl pomocı́ foldr a reverze seznamu . . . . . . . . . . . . . . 183

8.13 Délka seznamu pomocı́ rekurze (procedura length) . . . . . . . . . . . . . . . . . . . . . . . . 221
8.14 Délka seznamu pomocı́ iterace (procedura length) . . . . . . . . . . . . . . . . . . . . . . . . . 221

10.6 Výpočet permutace prvků množiny pomocı́ faktoradických čı́sel . . . . . . . . . . . . . . . . . 268

12.12 Převod do internı́ reprezentace a implementace readeru . . . . . . . . . . . . . . . . . . . . . . 302

8.6 Schématické zachycenı́ aplikace iterativnı́ verze fac. . . . . . . . . . . . . . . . . . . . . . . . . 208

8.11 Schématické zachycenı́ aplikace iterativnı́ verze fib. . . . . . . . . . . . . . . . . . . . . . . . . 217

11.2 Fyzická struktura seznamu (+ (* 2 x) (- (/ (+ x 2) z)) 5) . . . . . . . . . . . . . . . . 280

